
NOTICE
'
Two least squares macros of this MacroBundle, LS1 and 

'
WLS1, which gave no problems in Excel 2007, can mal-

'
function in Excel 2010 because it will misinterpret 

'
them as cell addresses, even though they are not used

'
in conjunction with an equal sign.

'


'
In order to avoid such problems, they have been re-

'
relabeled as LSI and WLSI, where the I signifies “with

'
Intercept”. For the sake of symmetry, their counter-

'
parts LS0 and WLS0 (which do not give such problems 

'
because Excel has no row number 0)have also been 

'
renamed LSO and WLSO respectively, with the O for

'
“through the Origin”. Moreover, to be consistent, 

'
Ortho0 and Ortho1, LSMulti0 and LSMulti1, as well as 

'
LSPoly0 and LSPoly1 have also been changed to OrthoO 

'
and OrthoI, LSMultiO and LSMultiI, and LSPolyO and 

'
LSPolyI. 

'
If you prefer to keep your old MacroBundle, and want to 

'
use Excel 2010, you can readily make these changes 

'
yourself using the Find function.

' LIST OF MACROS (including short descriptions)

'   Error analysis macro:

' Propagation computes the propagation of uncertainty for

'   a single function of one or more independent input

'   parameters with known standard deviations, or for

'   mutually dependent parameters with a known covariance

'   matrix.

'   Linear least squares macros:

' LS is a general least squares fitting routine for linear,

'   polynomial, and multivariate fitting, assuming one

'   dependent variable. LSO forces the fit through the

'   origin, LSI does not. The output provides the parameter

'   values, their standard deviations, the standard

'   deviation of the fit to the function, the covariance

'   matrix, and (optionally) the matrix of linear

'   correlation coefficients.

' ELS provides least squares smoothing and differentiation

'   for an equidistant (in the independent variable) but

'   otherwise arbitrary function using a 'Savitzky-Golay'

'   moving polynomial fit. ELSfixed uses a fixed-order

'   polynomial, ELSauto self-optimizes the order of the

'   fitting polynomial as it moves along the function.

' WLS is the equivalent of LS with the inclusion of user-

'   assignable weights.

' GradeBySf computes the standard deviations of the fit Sf

'   for all possible combinations and permutations of

'   unweighted least squares fits of a user-specified

'   multivariate expression of up to six terms.

' LSPoly applies LS to polynomial fitting to a polynomial

'   of gradually increasing order (up to 14).

' LSMulti applies LS to an increasing number of terms of a

'   multivariate least squares analysis.

' LSPermute computes the standard deviation of the fit for

'   all possible permutations of multivariate parameters

'   of up to six terms.

'   Non-linear least squares macros:

' SolverAid provides uncertainty estimates (standard

'   deviations and the covariance matrix) for Solver-

'   derived parameter values.

' ColumnSolver applies Solver to column-organized data,

'   and is especially useful for inverse interpolation

'   of algebraic functions.

' SolverScan lets Solver scan a two-dimensional array of

'   parameter values.

' Note that both ColumnSolver and SolverScan require that

'   Solver.xla be activated, as described in section 1.2.2.

'   Transform macros:

' FT is a general-purpose Fourier transform macro for

'   forward or inverse Fourier transformation of 2^n data

'   where n is an integer larger than 2.

' Gabor provides time-frequency analysis.

' Ortho yields a Gram-Schmidt orthogonalization

'   Convolution and deconvolution macros:

' Convolve provides general convolution.

' ConvolveFT yields convolution based on Fourier

'   transformation.

' Deconvolve provides deconvolution; this macro is not always

'   applicable.

' DeconvolveFT yields deconvolution based on Fourier

'   transformation.

' DeconvolveIt performs iterative (van Cittert)

'   deconvolution. DeconvolveIt0 has no constraints,

'   DeconvolveIt1 assumes that the function is everywhere

'   non-negative.

'   Calculus macros:

' Romberg performs Romberg integration of a function of a single

'   variable x defined either on the spreadsheet or in a special

'   function Equation(x). The calling macros are RombergAuto,

'   which evaluates the function on the spreadsheet, and

'   RombergSpecify, which requires a user-specified function.

' Trapez performs a trapezoidal integration of a function of a 

'   single variable x defined either on the spreadsheet or in a 

'   special function Equation(x). The calling macros are 

'   TrapezAuto, which evaluates the function on the spreadsheet, 

'   and TrapezSpecify, which requires a user-specified function.

' Deriv performs numerical differentiation using central

'   differencing.

' Deriv1 also performs numerical differentiation using central

'   differencing, but with a new, higher-accuracy algorithm.

' DerivScan applies Deriv to generate results for a range of 

'   Delta values.

' Semi-integrate & semi-differentiate are two small macros for

'   cyclic voltammetry assuming planar diffusion.

'   Mapper:

' Mapper generates colored (or gray-scale) two-dimensional maps.

'   Gradual mapper generates colored (or gray-scale) two-

'   dimensional maps with gradually changing colors. 

'   9-band mapper generates colored (or gray-scale) two-

'   dimensional maps with 9 distinct bands.

'   17-band mapper generates colored (or gray-scale) two-

'   dimensional maps with 17 distinct bands.

'   Miscellaneous macros:

' ScanF creates an array of values of thefunction F(x,y) and,

'   optionally, of an IsoListing for contour mapping with IsoL.

' RootFinder finds a single root of a function F(x) of x

'   by bisection.

' MovieDemos lists the macros used in section 1.6.

' InsertMBToolbar provides a toolbar for easy access to the

'   macros of the MacroBundle. Note: in Excel 2007 the

'   MBToolbar can only be displayed in the Add-Ins ribbon.

' RemoveMTToolbar
' This bundle also contains the necessary auxiliary subrou-

' tines and functions, as well as some freestanding 

' functions such as Lagrange for polynomial interpolation.

' Practical examples of the application of these macros or 

' functions in the most recent version of my book Advanced 
' Excel for scientific data analysis (3rd edition, Atlantic 

' Academic 2012) will often be mentioned as AE3 followed 

' by section and/or page numbers.

' PURPOSE
' The macros in this MacroBundle are primarily offered

' as examples of macro writing. They can also be used as

' such in scientific data analysis. Moreover, they can

' be modified by the user. All such uses and/or

' modifications are under the responsibility, and at the

' risk, of the user, and are subject to the conditions

' specified below.

' COPYLEFT & ABSENCE OF WARRANTY
' The material in this MacroBundle is free software: you

' can distribute it and/or modify it under the terms of

' the GNU General Public Licence as published by the Free

' Software Foundation, either version 3 of the License,

' or (at your option) any later version.

' This material is distributed in the hope that it may be

' useful, but WITHOUT ANY WARRANTY; without even the

' implied warranty of MERCHANTABILITY or FITNESS FOR A

' PARTICULAR PURPOSE. See the GNU General Public License

' for more details.

' Please download the GNU General Public License from the

' website http:/www.Bowdoin.edu/~rdelevie/excellaneous, 

' or see http:/www.gnu.org/licenses.

' ACKNOWLEDGEMENT
' When results obtained with these macros are communicated

' (orally, electronically, in print, or otherwise), please

' make reference to their source, so that others can learn 

' of their availability and usefulness. Therefore please 

' refer to the book Advanced Excel for scientific data 

' analysis, where these macros and functions are explained 

' and illustrated,and/or to the website of the author, 

' http://www.bowdoin.edu/~rdelevie/excellaneous, from where 

' they can be downloaded.

' INSTALLATION
' The macros in this MacroBundle are written in VBA (Visual

' BASIC for Applications) and will not work in versions of

' Excel preceding Excel 5, because those early versions did

' not use VBA as their macro language. Some may incorporate

' features that were not yet available in Excel versions 5

' and 95, and therefore may require some modifications to

' run under those versions. They were tested in Excel 97,

' 2000, and 2003, and appear to work equally well in Excel

' 2007 and 2010. Here we first describe the installation 

' procedure for Excel 97 and more recent versions, which 

' merely amounts to copying that text into an Excel VBEditor

' module. Thereafter we summarize the slightly different

' installation procedure for Excel 5 and 95.

' Here is a step-by-step account of how to do this, for

' those who have never done it. Note that almost all first

' spreadsheet commands were renamed in Excel 2007, and no 

' longer carry underlined letters to indicate their shortcut 

' key combinations. But most of these shortcut key combina-

' tions still work, including those shown here, which is why 

' we have not bothered to give the new 2007 command names.

' Select this text with Edit -> Select All or Alt+EL (Mac:

' Opt+EL) and copy it to the clipboard with Ctrl+C. Open

' Excel, then open the Visual Basic Editor with Tools ->

' Macro -> Visual Basic Editor or with Alt+F11 (Mac:

' Opt+F11) where F11 denotes the function key F11. In the

' VBEditor toolbar click on Insert -> Module. Then use

' Ctrl+V to paste the MacroBundle text into the module.

' Exit to the spreadsheet with Alt+F11 (Mac: Opt+F11).

' Save the macros with the spreadsheet (this happens

' automatically with earlier versions, but requires special

' care with Excel 2007), or in your Personal.xls file.

' For installation in Excel 5 or Excel 95, after Excel

' has been opened, use Insert -> Macro -> Module to open a

' module, then paste the MacroBundle text from the clip-

' board into the module. Note: most but not all of these

' programs have been tested to run properly in Excel 5 and

' Excel 95, but some of their recent embellishmnts may not

' work, since they use more recently introduced aspects of

' VBA.

' With the above you will have access to all macros of this

' MacroBundle via Alt+F8 (Mac: Opt+F8). For more convenient

' access, install the MacroBundle Toolbar with Alt+F8

' (Mac: Opt+F8), then double-click on InsertMBToolbar.

' The MacroBundle Toolbar can be saved with your spread-

' sheet, and can be removed with RemoveMBToolbar. You can

' customize the Toolbar by removing macros from it and/or

' adding your own custom macros to it.

' If you want these macros to be available every time you

' open a spreadsheet, you can incorporate them in the

' Personal.xls file, which will then be opened automatically 

' whenever Excel is opened. However, make sure only to place 

' well-tested macros in Personal.xls, and only modify them

' after you take them outside Personal.xls. A poor instruc- 

' tion during macro development, if tried in Personal.xls,

' may get it to 'hang up', in which case you may need expert

' help to extricate yourself.

' On any given line, all text to the right of an apostrophe

' (such as in this introduction) is considered a comment,

' and is therefore ignored by the VBEditor. Consequently

' it is not necessary to remove them. The VBEditor will

' also remove italics, boldfacing, and color.

' If you want to remove part or all of the MacroBundle, go

' to its VBEditor module, highlight the part(s) you want to

' remove, and delete them. If you want to update to a newer

' version of the MacroBundle, remove the old version (or,

' in case you don't have a back-up copy of it, move it to

' temporary storage until you have checked out its update),

' then insert the new version.

' THE MACROS
' Color code used: macros in blue,

' non-macro subroutines in purple,

' functions in brown.

'''''''''''''''''''''''''''''''''''''''''''''''''''''''''''

'''''''''''''''''''''''''''''''''''''''''''''''''''''''''''

'''''''''''''*********************************'''''''''''''

'''''''''''''**                             **'''''''''''''

'''''''''''''**         PROPAGATION         **'''''''''''''

'''''''''''''**                             **'''''''''''''

'''''''''''''*********************************'''''''''''''

''''''''''''''''''''''''''''''''''''''''''' (c) R. de Levie

''''''''''''''''''''''''''''''''''''''''''' v 13, Dec. 2013

' PURPOSE:

' This macro computes the propagated standard deviation in

' a single function F based on N input parameters, based on

' their standard deviations or on the corresponding covari-

' ance matrix. When only the standard deviations are given,

' the macro will assume that the input parameters are mutu-

' ally independent. No such assumption will be made when

' the covariance matrix is provided. For a single input

' parameter, there is no distinction between the two ap-

' proaches. For more than one input parameter, the macro

' recognizes which of the two computations to perform

' because the standard deviations will be provided as a

' vector, while the covariance matrix has the form of a

' square data array. The components of the N by N covari-

' ance matrix are assumed to be in the same order as those

' of the N input parameters.

' SUBROUTINES:

' This macro does not require any subroutines

' INPUT:

' The N independent input parameter values must be placed

' either in a contiguous row or in a contiguous column.

' They must be NUMBERS, i.e., they cannot be formulas,

' i.e., equations.

' The ORDER of the input parameters and of the standard

' deviations must be the same, because the (partial)

' derivative of the function and the corresponding standard

' deviation are combined strictly on the basis of their

' sequential order. The same applies to the covariance

' matrix if this is used instead.

' The N standard deviations or the corresponding covariance

' matrix can contain either values or formulas.

' The function must of course be a FUNCTION, because a 

' number cannot be differentiated. It is OK when it refers

' to the function, as long as that reference will respond

' to changes in the input parameters. 

' OUTPUT:

' The standard deviation of the single function F will be

' placed directly to the right of, or below, that function,

' in italics, provided that this cell is either unoccupied

' or its contents can be overwritten. Otherwise, the result

' will be displayed in a message box.

' PROCEDURE:

' In order to start this macro, call it. There is no need

' to highlight anything beforehand.

' You will see an input box in which to place (either

' by typing or by the 'highlight-and-click' method) the

' address(es) of the input parameter(s). After you have

' entered these, a second input box will request the

' addresses of either the standard deviations or the

' covariance matrix. These should have been arranged in

' the same order as the earlier-entered parameters.

' Finally, a third input box will ask for the address of

' the function. The output, the standard deviation of the 

' function, will be provided either on the spreadsheet, 

' or through message box(es).

' EXAMPLE:

' Use of this macro is illustrated starting in sections

' 2.8 and 2.9 of Advanced Excel.

' NOTATION:

'   v    value of

'   N:   the number of input parameters

'   X:   single input parameter (for N=1)

'   S:   the corresponding, single standard deviation of X

'   Xi:  multiple input parameters (for N>1) NOTE: THESE

'        MUST BE IN A SINGLE, CONTIGUOUS ROW OR COLUMN

'   Si:  standard deviations of the multiple input para-

'        meters. NOTE: THESE MUST BE IN A SINGLE, CONTI-

'        GUOUS ROW OR COLUMN

'   CM:  the covariance matrix

'   F:   the single function through which the error(s)

'        propagate(s)

'   VF:  the propagated variance of the function F

'   Sf:  the propagated standard deviation of function F

' We distinguish five cases:

' C = 1  one parameter P, one uncertainty U

' C = 2  parameters P and uncertainties U in column format

' C = 3  parameters P and uncertainties U in row format

' For C = 1 to 3, the uncertainty is the standard deviation

' C = 4  parameters P in column, uncertainty U in matrix

' C = 5  parameters P in row, uncertainty U in matrix

' For C = 4 or 5, the uncertainty must be in the form of

'   a covariance matrix, i.e., the uncertainties are in

'   terms of variances (the squares of the standard

'   deviations) and covariances (which may be zero).

Sub Propagation()
Dim C As Integer    'Case selector

Dim i As Integer, j As Integer, m As Integer

Dim number As Integer, LCCTest As Integer

Dim n As Integer    'Larger dimension of

                    'input parameter set

Dim NCF As Integer  'Number of Columns of output Function

Dim NCP As Integer  'Number of Columns of input Parameters

Dim NCU As Integer  'Number of Columns of the Uncertainty

                    'estimate

Dim NRF As Integer  'Number of Rows of output Function

Dim NRP As Integer  'Number of Rows of input Parameters

Dim NRU As Integer  'Number of Rows in the Uncertainty

                    'estimate

Dim vF As Double, vFF As Double

Dim vFiNew As Double, vFNew As Double

Dim vXNew As Double, vSF As Double, vS As Double

Dim vVFi As Double, vVF As Double, vX As Double

Dim vCM As Variant, vCMi As Variant

Dim vDel As Variant, vXiNew As Variant

Dim vSFi As Variant, vSi As Variant

Dim vXi As Variant

Dim myRange1 As Range, myRange2 As Range, myRange3 As Range

Dim Answer, FFormula, LocationEqualSign, Prompt, Title

Dim XFunction '?????

A:

' Select the input parameter values of the function

Prompt = "The the input parameter values" & Chr(13) & _

         "of the function are located in:"

Title = "Uncertainty Propagation InputBox 1: Input" _

  & " Parameters "

Set myRange1 = Application.InputBox(Prompt, Title, _

  Type:=8)

myRange1.Select

NRP = Selection.Rows.Count

NCP = Selection.Columns.Count

' Check the input format

LocationEqualSign = 0

If NRP = 0 Then End

If NRP <> 1 And NCP <> 1 Then

  MsgBox "The input parameter values should be" & Chr(13) _

    & "put in either a single contiguous row or" & _

    Chr(13) & "in a single contiguous column. Try again.", _

    , "Propagation of uncertainty"

  GoTo A

ElseIf NRP = 1 And NCP = 1 Then

  n = 1

  vX = Selection.Value          ' dimensioning the array

  FFunction = Selection.Formula     ' dimensioning the array

  LocationEqualSign = InStr(1, FFunction, "=")

  If LocationEqualSign = 1 Then GoTo B

ElseIf NRP > 1 And NCP = 1 Then

  n = NRP

  vXi = Selection.Value         ' dimensioning the array

  FFunction = Selection.Formula ' dimensioning the array

  vDel = Selection.Value        ' dimensioning the array

  vXiNew = Selection.Value      ' dimensioning the array

  vSFi = Selection.Value        ' dimensioning the array

  For j = 1 To n

    LocationEqualSign = InStr(1, FFunction(j, 1), "=")

    If LocationEqualSign = 1 Then GoTo B

  Next j

ElseIf NCP > 1 And NRP = 1 Then

  n = NCP

  vXi = Selection.Value

  FFunction = Selection.Formula ' dimensioning the array

  vDel = Selection.Value        ' dimensioning the array

  vXiNew = Selection.Value      ' dimensioning the array

  vSFi = Selection.Value        ' dimensioning the array

  For j = 1 To n

    LocationEqualSign = InStr(1, FFunction(1, j), "=")

    If LocationEqualSign = 1 Then GoTo B

  Next j

End If

GoTo C

B:

' Check that the "input parameter values"

' indeed only contain values, no formulas

MsgBox "The input parameter values can only " & _

  "contain NUMBERS" & Chr(13) & _

  "not functions (i.e., anything starting " & _

  "with an equal sign." & Chr(13) & _

  "Please re-enter the input parameter VALUES."

End

' Select the uncertainty estimates (standard deviations

' or covariance matrix) of the input parameters

C:

Prompt = " The standard deviations, or" _

  & Chr(13) & "the covariance matrix, of the" _

  & Chr(13) & " input parameters are located in:"

Title = "Uncertainty Propagation InputBox 2: " _

  & "Uncertainty estimates"

Set myRange2 = Application _

  .InputBox(Prompt, Title, Type:=8)

myRange2.Select

NRU = Selection.Rows.Count

NCU = Selection.Columns.Count

' Verify that the number of parameters

' matches that of the uncertainty estimates

number = NCP + NRP - 1

If NCU = NRU Then

  If (NCU + NRU) / 2 <> number Then

    MsgBox "The dimension of the covariance matrix does" _

      & Chr(13) & "not match the number of input " _

      & "parameters"

    End

  End If

End If

If NCU <> NRU Then

  If NCU + NRU - 1 <> number Then

    MsgBox "The number of uncertainty estimates does" _

      & Chr(13) & "not match the number of input " _

      & "parameters"

    End

  End If

End If

' Verify and categorize the inputs,

' then read the uncertainties

If NCU = 0 Then End

If n = 1 And NRU = 1 And NCU = 1 Then

  C = 1     'only one input parameter

  vS = Selection.Value

End If

If n > 1 Then   'multiple input parameters:

  If n = NRU And NCU = 1 Then

    C = 2   'the independent parameters and

            'their standard deviations are

            'located in single columns

    vSi = Selection.Value

  End If

  If n <> NRU And NCU = 1 Then

    MsgBox "The number of entries in the two" _

      & Chr(13) & "input columns must be the same.", _

      , "Propagation of uncertainty"

    End

  End If

  If n = NCU And NRU = 1 Then

    C = 3   'the independent parameters and

            'their standard deviations are

            'located in single rows

    vSi = Selection.Value

    If n <> NCU And NRU = 1 Then

      MsgBox "The number of entries in the two" _

        & Chr(13) & "input rows must be the same.", _

        , "Propagation of uncertainty"

      End

    End If

  End If

  If n = NCU And n = NRU Then

                'mutually dependent parameters:

    If NCP <> NCU And NRP <> NRU Then

      MsgBox "The number of entries for the " _

        & "uncertainties does" & Chr(13) & _

        "not match the number of input parameters." _

        , , "Propagation of uncertainty"

      End

    End If

    If n = NRP And NCP = 1 Then

      C = 4 'the dependent parameters are located

            'in a single column of N items, and

            'the covariance matrix is N by N

      vCM = Selection.Value

      vCMi = Selection.Value

    End If

    If n = NCP And NRP = 1 Then

      C = 5 'the dependent parameters are located

            'in a single row of N items, and

            'the covariance matrix is N by N

      vCM = Selection.Value

      vCMi = Selection.Value

    End If

  End If

End If

If NCU > 1 And NRU > 1 And NCU <> NRU Then

  MsgBox "The covariance matrix has an" & _

    Chr(13) & " incorrect dimension. Try again.", _

    , "Propagation of uncertainty"

  GoTo C

End If

If NCU > 1 And NRU > 1 Then

  LCCTest = 0

  For i = 1 To n

    If vCM(i, i) > 0.99999 And _

      vCM(i, i) < 1.00001 Then LCCTest = LCCTest + 1

  Next i

  If LCCTest = n Then

    Answer = MsgBox("You appear to have entered the" _

      & Chr(13) & "array of linear correlation " _

      & "coefficients" & Chr(13) & "rather than the " _

      & "covariance matrix." & Chr(13) _

      & "Do you want to correct the input?", _

      vbYesNo, "Propagation of uncertainty")

    If Answer = vbYes Then GoTo A

  End If

End If

' Select the function

Set myRange3 = Application.InputBox(Prompt:= _

  "The FUNCTION (not just its value) is located in:", Title:= _

  "Uncertainty Propagation InputBox 3: Function", Type:=8)

myRange3.Select

NRF = Selection.Rows.Count

NCF = Selection.Columns.Count

If NRF = 0 Then End

If NRF = 1 And NCF = 1 Then

  vF = Selection.Value

End If

If NRF <> 1 Or NCF <> 1 Then

  MsgBox "The propagation is computed only for a SINGLE " _

    & "function.", , "Propagation of uncertainty"

  End

End If

' Test and prepare the default output cell for the

' standard deviation of the function F

' Place the standard deviation to the right of the cell

' containing F for a single input parameter, or when the

' input parameters are in a single column

If n = 1 Or NCP = 1 Then

  Selection.Offset(0, 1).Select

  m = 0

  vFF = Selection.Value

  If (IsEmpty(vFF) Or vFF = 0) Then

    m = m

  Else

    m = m + 1

  End If

  Selection.Offset(0, -1).Select

End If

' Place the standard deviation below F when the

' input parameters are organized in a single row.

If n > 1 And NRP = 1 Then

  Selection.Offset(1, 0).Select

  m = 0

  vFF = Selection.Value

  If (IsEmpty(vFF) Or vFF = 0) Then

    m = m

  Else

    m = m + 1

  End If

  Selection.Offset(-1, 0).Select

End If

Application.ScreenUpdating = False

vVF = 0

myRange1.Select

' Compute and display the sought standard deviation vSF

Select Case C

Case 1

  vXNew = vX * 1.000001

  Selection.Replace vX, vXNew

  myRange3.Select

  vFNew = Selection.Value

  myRange1.Select

  Selection.Replace vXNew, vX

  vSF = Abs(1000000 * (vFNew - vF) * vS _

    / vX)

  myRange3.Select

  If m > 0 Then

    Answer = MsgBox _

      ("The cell to the right of the function is " _

      & "not empty." & Chr(13) & "Can its contents " _

      & "be overwritten?", vbYesNo, "Overwrite?")

    If Answer = vbYes Then m = 0

  End If

  If m = 0 Then

    Selection.Offset(0, 1).Select

    Selection.Font.Italic = True

    Selection.Value = vSF

  Else

    MsgBox "The standard deviation of the function is ", _

      , "Propagation of uncertainty" & vSF

  End If

Case 2

  For i = 1 To n

    vXiNew(i, 1) = vXi(i, 1) * 1.000001

    Selection.Item(i, 1).Replace vXi(i, 1), vXiNew(i, 1)

    myRange3.Select

    vFiNew = Selection.Value

    myRange1.Select

    Selection.Item(i, 1).Replace vXiNew(i, 1), vXi(i, 1)

    vVFi = 1000000 * (vFiNew - vF) _

      * vSi(i, 1) / vXi(i, 1)

    vVF = vVF + vVFi ^ 2

  Next i

  vSF = Sqr(vVF)

  myRange3.Select

  If m > 0 Then

    Answer = MsgBox _

      ("The cell to the right of the function is " _

      & "not empty." & Chr(13) & "Can its contents " _

      & "be overwritten?", vbYesNo, "Overwrite?")

    If Answer = vbYes Then m = 0

  End If

  If m = 0 Then

    Selection.Offset(0, 1).Select

    Selection.Font.Italic = True

    Selection.Value = vSF

  Else

    MsgBox "The standard deviation of the function is ", _

      , "Propagation of uncertainty" & vSF

  End If

Case 3

  For j = 1 To n

    vXiNew(1, j) = vXi(1, j) * 1.000001

    Selection.Item(1, j).Replace vXi(1, j), vXiNew(1, j)

    myRange3.Select

    vFiNew = Selection.Value

    myRange1.Select

    Selection.Item(1, j).Replace vXiNew(1, j), vXi(1, j)

    vVFi = 1000000 * (vFiNew - vF) _

      * vSi(1, j) / vXi(1, j)

    vVF = vVF + vVFi ^ 2

  Next j

  vSF = Sqr(vVF)

  myRange3.Select

  If m > 0 Then

    Answer = MsgBox("The cell below the function is " _

    & "not empty." & Chr(13) & "Can its contents be " _

    & "overwritten?", vbYesNo, "Overwrite?")

    If Answer = vbYes Then m = 0

  End If

  If m = 0 Then

    Selection.Offset(1, 0).Select

    Selection.Font.Italic = True

    Selection.Value = vSF

  Else

  End If

Case 4

  For i = 1 To n

    vXiNew(i, 1) = vXi(i, 1) * 1.000001

    Selection.Item(i, 1).Replace vXi(i, 1), vXiNew(i, 1)

    myRange3.Select

    vFiNew = Selection.Value

    myRange1.Select

    Selection.Item(i, 1).Replace vXiNew(i, 1), vXi(i, 1)

    vDel(i, 1) = 1000000 * (vFiNew - vF) _

      / vXi(i, 1)

  Next i

  For j = 1 To n

    For i = 1 To n

      vVF = vVF + vDel(i, 1) _

        * vDel(j, 1) * vCM(i, j)

    Next i

  Next j

  vSF = Sqr(vVF)

  myRange3.Select

  If m > 0 Then

    Answer = MsgBox _

    ("The cell to the right of the function is " _

    & "not empty." & Chr(13) & "Can its contents " _

    & "be overwritten?", vbYesNo, "Overwrite?")

    If Answer = vbYes Then m = 0

  End If

  If m = 0 Then

    Selection.Offset(0, 1).Select

    Selection.Font.Italic = True

    Selection.Value = vSF

  Else

    MsgBox "The standard deviation of the function is ", _

      , "Propagation of uncertainty" & vSF

  End If

Case 5

  For i = 1 To n

    vXiNew(1, i) = vXi(1, i) * 1.000001

    Selection.Item(1, i).Replace vXi(1, i), vXiNew(1, i)

    myRange3.Select

    vFiNew = Selection.Value

    myRange1.Select

    Selection.Item(1, i).Replace vXiNew(1, i), vXi(1, i)

    vDel(1, i) = 1000000 * (vFiNew - vF) _

      / vXi(1, i)

  Next i

  For i = 1 To n

    For j = 1 To n

      vVF = vVF + vDel(1, i) _

        * vDel(1, j) * vCM(j, i)

    Next j

  Next i

  vSF = Sqr(vVF)

  myRange3.Select

  If m > 0 Then

    Answer = MsgBox("The cell below the function is " _

      & "not empty." & Chr(13) & "Can its contents be " _

      & "overwritten?", vbYesNo, "Overwrite?")

    If Answer = vbYes Then m = 0

  End If

  If m = 0 Then

    Selection.Offset(1, 0).Select

    Selection.Font.Italic = True

    Selection.Value = vSF

  Else

    MsgBox "The standard deviation of the function is ", _

      , "Propagation of uncertainty" & vSF

  End If

End Select

End Sub
'''''''''''''''''''''''''''''''''''''''''''''''''''''''''''

'''''''''''''''''''''''''''''''''''''''''''''''''''''''''''

'''''''''''''********************************''''''''''''''

'''''''''''''*                              *''''''''''''''

'''''''''''''*     Linear Least Squares     *''''''''''''''

'''''''''''''*                              *''''''''''''''

'''''''''''''********************************''''''''''''''

''''''''''''''''''''''''''''''''''''''''''' (c) R. de Levie

'''''''''''''''''''''''''''''''''''''''''''  v 12, May 2012

' PURPOSE:

' The macros LSI and LSO set the input parameter p for the

' subroutine LeastSquares, which in turn computes the indi-

' vidual coefficients or parameters with their standard

' deviations, plus the corresponding covariance matrix CM

' and the array of linear correlation coefficients CC for

' an unweighted least squares fit to data in 2 or more

' columns. Moreover, it provides the standard deviation Sf

' of the over-all fit of the function to the data.

' Color coding is used to alert the user when the indivi-

' dual standard deviations are of the same order of magni-

' tude as the corresponding coefficients, and when the

' independent variables exhibit significant collinearity.

' This macro uses the traditional least squares algorithm.

' When the LC matrix indicates high collinearity, and you

' use Excel 2003 or a more recent version, you may instead

' want to use Excel's own LinEst or Regression, which are

' now based on QRDecomposition. Otherwise use the Singular

' Value Decomposition macros in Volpi's Matrix.xla. But

' neither of these provides the covariance matrix, which

' you may therefore have to construct yourself using

' Solver and SolverAid.

' The input parameter p = 1 starts a general unweighted

' least squares fit to the data while p = 0 forces the fit

' to pass through the origin, i.e., it assumes that y = 0

' for x = 0.

' SUBROUTINES:

' This macro requires the subroutines Multiply, Invert, and

' Transpose

' INPUT:

' The input data must be organized in columns, arranged

' as follows. The first column must contain the dependent

' variable y. The second (and any subsequent) column(s)

' must contain the independent variable(s) x.

' OUTPUT:

' The macro labels the coefficients and standard devi-

' ations, except when there is no space available for

' labels because the first data column is at the left

' edge of the spreadsheet.

' Comment boxes are used:

' (1) to identify the version number, the data input range,

' the number N of data points, the number P of adjustable

' fitting parameters, and date and time of the analysis;

' (2) to explain the color code used to warn the user about

' the reliability of the coefficient ai, by displaying a

' categorical measure of the absolute value of the ratio of

' ai to its standard deviation si; and

' (3) to explain the color code used to alert the user of

' high collinearity, again using a crude categorization.

' PROCEDURE:

' Before calling the macro, make sure that the output area

' (at minimum three lines below the input data block) does

' not contain valuable data.

' In order to start the process, highlight the entire input

' data block, and call LSI or LSO.

' EXAMPLES:

' Use of this macro is illustrated starting in sections

' 2.5 and 2.9 of Advanced Excel.

'''''''''''''''''''''''''''''''''''''''''''''''''''''''''''

' The function of the following two drivers is merely to

' set the value of one parameter, p, equal to either one or

' zero, in order to choose between a general least squares

' fitting (p = 1) or one that forces the curve through the

' origin (p = 0).

Sub LSO()           ' for an unweighted least squares

                    ' fit through the origin

Dim P As Double

P = 0

Call LeastSquares(P)

End Sub

'''''''''''''''''''''''''''''''''''''''''''''''''''''''''''

Sub LSI()           ' for a general unweighted

                    ' least squares fit

Dim P As Double

P = 1

Call LeastSquares(P)

End Sub

'''''''''''''''''''''''''''''''''''''''''''''''''''''''''''

Sub LeastSquares(P)

Dim CMax As Long, i As Long, j As Long, jj As Long

Dim lccC As Integer, lccR As Integer, m As Integer

Dim N As Integer, NN As Integer, RMax As Long

Dim a As Double, Q As Double, Root As Double

Dim Sf As Double, SSR As Double, varY As Double

Dim DataArray As Variant, OutputArray As Variant

Dim lccArray As Variant, vArray As Variant

Dim v0Array As Variant, Cell As Variant

Dim myRange As Range

Dim aA, AC, aD, Answer, hAnswer, jAnswer

Dim BoxAddress, myAddress1, myAddress2

Dim bArray, btArray

Dim pArray, piArray, qArray

Dim XArray, XBArray, xtArray

Dim YArray, ytArray

' Determination of the array size

Begin:

RMax = Selection.Rows.Count

CMax = Selection.Columns.Count

' If area is not highlighted before the macro is called

If RMax = 1 And CMax = 1 Then

  hAnswer = MsgBox("You forgot to highlight" _

    & Chr(13) & "the block of input data." _

    & Chr(13) & "Do you want to do so now?" _

    , vbYesNo, "Least Squares Fit")

  If hAnswer = vbNo Then End

  If hAnswer = vbYes Then

    Set myRange = Application.InputBox(Prompt:= _

      "The input data are located in:", Type:=8)

    myRange.Select

  End If

  GoTo Begin

End If

' Check that the number of columns is at least 2

If CMax < 2 Then

  MsgBox "There must be at least two columns," & _

    Chr(13) & "one for Y, and one or more for X.", _

    , "Least Squares Fit"

  End

End If

' Check that there are more data than coefficients

If RMax - CMax - P + 1 <= 0 Then

  MsgBox "With " & RMax & " data, LS" & P & _

    " can only deter-" & Chr(13) & "mine " & RMax - 1 & _

    " least squares coefficients." & Chr(13) & Chr(13) _

    & "Add more data, or reduce the re-" & Chr(13) & _

    "quested number of coefficients."

  End

End If

' Dimension the arrays:

ReDim bArray(1 To CMax, 1 To 1) As Double

ReDim btArray(1 To 1, 1 To CMax) As Double

ReDim lccArray(1 To CMax, 1 To CMax) As Double

ReDim pArray(1 To CMax, 1 To CMax) As Double

ReDim piArray(1 To CMax, 1 To CMax) As Double

ReDim qArray(1 To CMax, 1 To 1) As Double

ReDim vArray(1 To CMax, 1 To CMax) As Double

ReDim v0Array(1 To CMax - 1 + P, 1 To CMax - 1 + P) _

  As Double

ReDim XArray(1 To RMax, 1 To CMax) As Double

ReDim XBArray(1 To RMax, 1 To 1) As Double

ReDim xtArray(1 To CMax, 1 To RMax) As Double

ReDim YArray(1 To RMax, 1 To 1) As Double

ReDim ytArray(1 To 1, 1 To RMax) As Double

' Read the dataArray, then fill yArray and xArray.

DataArray = Selection.Value

aD = Selection.Address

For i = 1 To RMax

  YArray(i, 1) = DataArray(i, 1)

Next i

For i = 1 To RMax

  If IsEmpty(DataArray(i, 1)) Then

    MsgBox "Y-value(s) missing", , "Least Squares Fit"

    End

  End If

Next i

For j = 2 To CMax

  For i = 1 To RMax

    If IsEmpty(DataArray(i, j)) Then

      MsgBox "X-value(s) missing", , "Least Squares Fit"

      End

    End If

  Next i

Next j

' Fill the first column of xArray with zeroes (for p = 0)

' or ones (for p = 1), the rest with the data in the

' x-column(s)

For i = 1 To RMax

  XArray(i, 1) = CDbl(P)

Next i

For j = 2 To CMax

  For i = 1 To RMax

    XArray(i, j) = DataArray(i, j)

  Next i

Next j

' Compute B = (X' X)" X' Y , where ' or t denote

' transposition, and " or i indicate inversion

' The various arrays, their names, and

' their dimensions (rows, columns) are:

'   Y          = YArray        (  rMax,     1)

'   X          = XArray        (  rMax,  cMax)

'   X'         = XtArray       (  cMax,  rMax)

'   X' X       = pArray        (  cmax,  cMax)

'  (X' X)"     = piArray       (  cMax,  cMax)

'   X' Y       = qArray        (  cMax,     1)

'   B          = BArray        (  cMax,     1)

Call Transpose(XArray, RMax, CMax, xtArray)

Call Multiply(xtArray, CMax, RMax, XArray, CMax, pArray)

Call Invert(pArray, CMax, piArray)

Call Multiply(xtArray, CMax, RMax, YArray, 1, qArray)

Call Multiply(piArray, CMax, CMax, qArray, 1, bArray)

' Check against overwriting spreadsheet data

m = 0

If (P = 0 And CMax = 2) Then

  For i = 1 To 3

    Selection.Offset(1, 0).Select

    OutputArray = Selection.Value

    For j = 1 To CMax

      If IsEmpty(OutputArray(RMax, j)) Then

        m = m

      Else

        m = m + 1

      End If

    Next j

  Next i

  Selection.Offset(-3, 0).Select

  If m > 0 Then Answer = MsgBox("There are data in the " _

    & "three lines below the" & Chr(13) & _

    "input data array. " & "Can they be overwritten?", _

    vbYesNo, "Overwrite?")

  If Answer = vbNo Then End

Else

  For i = 1 To 1 + 2 * (P + CMax)

    Selection.Offset(1, 0).Select

    OutputArray = Selection.Value

    For j = 1 To CMax

      If IsEmpty(OutputArray(RMax, j)) Then

        m = m

      Else

        m = m + 1

      End If

    Next j

  Next i

  Selection.Offset(-1 - 2 * (P + CMax), 0).Select

  If m > 0 Then Answer = MsgBox("There are data in the " _

    & 1 + 2 * (P + CMax) & " lines below the" & Chr(13) & _

    "input data array. " & "Can they be overwritten?", _

    vbYesNo, "Overwrite?")

  If Answer = vbNo Then End

End If

' The additional arrays, names, and

' dimensions (rows, columns) are:

'   Y'         = YtArray       (    1, rMax)

'   B'         = BtArray       (    1, cMax)

'   X B        = XBArray       ( rMax,    1)

'   v          = vArray        ( cMax, cMax)

Call Transpose(YArray, RMax, 1, ytArray)

Call Transpose(bArray, CMax, 1, btArray)

Call Multiply(XArray, RMax, CMax, bArray, 1, XBArray)

' Calculate SSR = Sum(Y - Xb) ^ 2; then compute the

' variance of y as varY = SSR/(rMax-cMax-p+1); and the

' covariance matrix vArray as V = (X'X)" times varY.

SSR = 0

For j = 1 To RMax

  SSR = SSR + (YArray(j, 1) - XBArray(j, 1)) ^ 2

Next j

varY = SSR / (RMax - CMax - P + 1)

Sf = Sqr(Abs(varY))

For i = 1 To CMax

  For j = 1 To CMax

    vArray(i, j) = varY * piArray(i, j)

  Next j

Next i

Application.ScreenUpdating = False

ActiveCell.Offset(RMax, 0).Select

' Prepare the output format

For j = 1 To CMax

  ActiveCell.Font.Bold = True

  ActiveCell.Font.Italic = True

  ActiveCell.HorizontalAlignment = xlCenter

  ActiveCell.Offset(0, 1).Select

Next j

ActiveCell.Offset(1, -CMax).Select

If (P = 0 And CMax = 2) Then

  For i = 1 To 2

    For j = 1 To CMax

      ActiveCell.Font.Bold = False

      ActiveCell.Font.Italic = True

      ActiveCell.HorizontalAlignment = xlCenter

      ActiveCell.Offset(0, 1).Select

    Next j

    ActiveCell.Offset(1, -CMax).Select

  Next i

  ActiveCell.Offset(-3, 0).Select

Else

  For i = 1 To 1 + P + CMax

    For j = 1 To CMax

      ActiveCell.Font.Bold = False

      ActiveCell.Font.Italic = True

      ActiveCell.HorizontalAlignment = xlCenter

      ActiveCell.Offset(0, 1).Select

    Next j

    ActiveCell.Offset(1, -CMax).Select

  Next i

  ActiveCell.Offset(-2 - P - CMax, 0).Select

End If

' Prepare the output labels, suppressing them when space

' for them is unavailable or data will be overwritten

aA = ActiveCell.Address

AC = Mid(aA, 1, 3)

If (AC = "$A$" And P = 1) Then GoTo NoLabel

ActiveCell.Offset(0, -P).Select

If P = 1 Then

  If (IsEmpty(ActiveCell) Or ActiveCell.Value = "Coeff:") _

  Then

    GoTo Step1

  Else

    ActiveCell.Offset(0, P).Select

    GoTo NoLabel

  End If

End If

Step1:                      ' Make label for Coeff

With ActiveCell

  .Value = "Coeff:"

  .Font.Bold = True

  .Font.Italic = True

  .Font.ColorIndex = 1

  .Interior.ColorIndex = xlNone

  .ClearComments

  .HorizontalAlignment = xlRight

End With

ActiveCell.Offset(1, 0).Select

If P = 1 Then

  If (IsEmpty(ActiveCell) Or ActiveCell.Value = "StDev:") _

  Then

    GoTo Step2

  Else

    ActiveCell.Offset(-1, P).Select

    GoTo NoLabel

  End If

End If

Step2:                      ' Make label for StDev

With ActiveCell

  .Value = "StDev:"

  .Font.Bold = True

  .Font.Italic = True

  .Font.ColorIndex = 1

  .Interior.ColorIndex = xlNone

  .HorizontalAlignment = xlRight

  .ClearComments

  .AddComment

  .Comment.Visible = False

  .Comment.Text Text:= _

    "Q = abs(P/S);  Colors:" & Chr(10) & _

    "gray:       3 < Q <= 5" & Chr(10) & _

    "orange:   2 < Q <= 3" & Chr(10) & _

    "red:         1 < Q <= 2" & Chr(10) & _

    "bold red:        Q <= 1"

End With

ActiveCell.Offset(1, 0).Select

If P = 1 Then

  If (IsEmpty(ActiveCell) Or _

    ActiveCell.Value = "Sf:") Then

    GoTo Step3

  Else

    ActiveCell.Offset(-1, P).Select

    GoTo NoLabel

  End If

End If

Step3:                      ' Make label for Sf

With ActiveCell

  .Value = "Sf:"

  .Font.Bold = True

  .Font.Italic = True

  .Font.ColorIndex = 1

  .Interior.ColorIndex = xlNone

  .ClearComments

  .HorizontalAlignment = xlRight

End With

ActiveCell.Offset(-1, P).Select

If P = 0 And CMax = 2 Then

  ActiveCell.Offset(-1, P).Select

  GoTo NoLabel

End If

ActiveCell.Offset(3, -P).Select

If P = 1 Then

  If (IsEmpty(ActiveCell) Or _

    ActiveCell.Value = "CM:") Then GoTo Step4

Else

  ActiveCell.Offset(-3, P).Select

  GoTo NoLabel

End If

Step4:                      ' Make label for CM

ActiveCell.Offset(-1, 0).Select

With ActiveCell

  .Value = "CM:"

  .Font.Bold = True

  .Font.Italic = True

  .Font.ColorIndex = 5

  .Interior.ColorIndex = xlNone

  .ClearComments

  .HorizontalAlignment = xlRight

End With

ActiveCell.Offset(-3, P).Select

NoLabel:

ActiveCell.Offset(0, 1 - P).Select

For j = 2 - P To CMax

With ActiveCell

  .Value = bArray(j, 1)

  .NumberFormat = "General"

  .Font.Italic = True

  .Font.ColorIndex = 1

  .Interior.ColorIndex = xlNone

  .ClearComments

  .Offset(0, 1).Select

End With

Next j

ActiveCell.Offset(1, 1 - P - CMax).Select

NN = 0

For j = 2 - P To CMax

  ActiveCell.Font.ColorIndex = 3

  ActiveCell.Interior.ColorIndex = xlNone

  ActiveCell.ClearComments

  ActiveCell.Font.Bold = True

' Warning if variance is zero or negative

  If vArray(j, j) <= 0 Then

    With ActiveCell

      .Value = "var <= 0"

      .Font.Bold = True

      .Font.Italic = True

      .HorizontalAlignment = xlCenter

      .Interior.ColorIndex = xlNone

      .ClearComments

    End With

    NN = NN + 1

  Else

' Displaying the standard deviation in color code:

'   bold red when Q <= 1

'   red      when 1 < Q <= 2

'   orange   when 2 < Q <= 3

'   gray     when 3 < Q <= 5

'   black    when Q > 5

    ActiveCell.Value = Sqr(vArray(j, j))

    Q = Abs(bArray(j, 1)) / Sqr(vArray(j, j))

    With ActiveCell.Font

    If Q <= 1 Then

      .ColorIndex = 3    'bold red

      .Bold = True

    End If

    If Q > 1 Then

      .ColorIndex = 3    'red

      .Bold = False

    End If

    If Q > 2 Then _

      .ColorIndex = 46   'orange

    If Q > 3 Then _

      .ColorIndex = 16   'gray

    If Q > 5 Then _

      .ColorIndex = 1    'black

    End With

  End If

  ActiveCell.Offset(0, 1).Select

Next j

If NN > 0 Then MsgBox _

  "This macro has difficulties with the " & _

  "least squares analysis." & Chr(13) & _

  "Distrust its numerical answers. " & _

  "You may be able to solve" & Chr(13) & _

  "this problem with multivariate or " & _

  "polynomial centering. Or," & Chr(13) & _

  "if you use Excel 2003 or a " & _

  "more recent version, and you" & Chr(13) & _

  "do not need the covariance, " & _

  "use LinEst or Regression." & Chr(13) & _

  "Otherwise, use xLS if you " & _

  "have Xnumbers.dll installed," & Chr(13) & _

  "or svd plus SolverAid if you have " & _

  "Matrix.xla available.", 48, "Warning!"

With ActiveCell

  .Offset(1, 1 - P - CMax).Select

  .Interior.ColorIndex = xlNone

  .NumberFormat = "General"

  .ClearComments

End With

With ActiveCell

  .Value = Sf

  .Font.Italic = True

  .HorizontalAlignment = xlCenter

  .Font.ColorIndex = 1

  .Interior.ColorIndex = xlNone

  .ClearComments

End With

ActiveCell.Offset(0, 1).Select

ActiveCell.Offset(0, 1).Select

With ActiveCell

  .Value = ""

  .Interior.ColorIndex = xlNone

  .ClearComments

End With

ActiveCell.Offset(0, -1).Select

If P = 0 And CMax = 2 Then ActiveCell.Offset(1, -1).Select

With ActiveCell

  .Value = "LS" & P

  .Font.Bold = True

  .Font.Italic = False

  .Font.ColorIndex = 5

  .HorizontalAlignment = xlCenter

  .Interior.ColorIndex = 20

  .ClearComments

' In a comment box, identify the macro used, its

' optional parameters, and its date & time of use

  .AddComment

  .Comment.Visible = False

  .Comment.Text Text:= _

    "LS" & P & " version 8" & Chr(10) & _

    "Input: " & aD & Chr(10) & _

    "N = " & RMax & ",   P = " & CMax + P - 1 & Chr(10) & _

    "date = " & Date & Chr(10) & "time = " & Time

End With

If P = 0 And CMax = 2 Then ActiveCell.Offset(1, 0).Select

ActiveCell.Offset(0, -1).Select

' Write the covariance matrix

If P = 0 And CMax = 2 Then GoTo LastLine

ActiveCell.Offset(1, 0).Select

For i = 2 - P To CMax

  For j = 2 - P To CMax

    With ActiveCell

      .Value = vArray(i, j)

      .Font.ColorIndex = 5

      .Font.Italic = True

      .HorizontalAlignment = xlCenter

      .NumberFormat = "General"

      .Interior.ColorIndex = xlNone

      .ClearComments

     End With

     ActiveCell.Offset(0, 1).Select

  Next j

  ActiveCell.Offset(1, 1 - P - CMax).Select

Next i

Application.ScreenUpdating = True

' Provide as optional output the array of linear

' correlation coefficients. The user specifies

' the cell block in which to write this array

If P = 0 And CMax = 2 Then GoTo LastLine

jAnswer = MsgBox("Do you want to see the " _

  & "matrix of linear correlation" _

  & Chr(13) & "coefficients? It will need a " _

  & "block of " & CMax + P - 1 _

  & " by " & CMax + P - 1 & " cells.", vbYesNo, _

  "Least Squares Fit")

' Read location of active cell

ActiveCell.Select

myAddress1 = Selection.Address

ActiveCell.Offset(CMax + P - 2, CMax + P - 2).Select

myAddress2 = Selection.Address

BoxAddress = myAddress1 & ":" & myAddress2

ActiveCell.Offset(2 - CMax - P, 2 - CMax - P).Select

OutlineMatrix:

If jAnswer = vbYes Then

  Set myRange = Application.InputBox(Prompt:= _

    "Select the default array location with OK," _

    & vbCr & "otherwise type in your own coordinates:", _

    Default:=BoxAddress, Type:=8)

  myRange.Select

' Make sure that the selected block has the correct size

  If Selection.Rows.Count <> CMax + P - 1 Then

    MsgBox "The selected range does not have " _

      & CMax + P - 1 & " rows. Please correct.", _

      , "Least Squares Fit"

    GoTo OutlineMatrix

  End If

  If Selection.Columns.Count <> CMax + P - 1 Then

    MsgBox "The selected range does not have " _

      & CMax + P - 1 & " columns. Please correct", _

      , "Least Squares Fit"

    GoTo OutlineMatrix

  End If

  Application.ScreenUpdating = False

' Draw a box around the reserved area

  Selection.Borders(xlEdgeTop).LineStyle = xlDashDot

  Selection.Borders(xlEdgeTop).Weight = xlThin

  Selection.Borders(xlEdgeRight).LineStyle = xlDashDot

  Selection.Borders(xlEdgeRight).Weight = xlThin

  Selection.Borders(xlEdgeBottom).LineStyle = xlDashDot

  Selection.Borders(xlEdgeBottom).Weight = xlThin

  Selection.Borders(xlEdgeLeft).LineStyle = xlDashDot

  Selection.Borders(xlEdgeLeft).Weight = xlThin

' Format cells in array

  For Each Cell In Selection.Cells

    Cell.HorizontalAlignment = xlCenter

    Cell.Font.ColorIndex = 1

    Cell.Font.Bold = False

    Cell.Font.Italic = False

    Cell.Interior.ColorIndex = xlNone

    Cell.ClearComments

  Next Cell

' Write the array of linear correlation coefficients

  If P = 0 Then

    For i = 1 To CMax - 1

      For j = 1 To CMax - 1

        v0Array(i, j) = vArray(i + 1, j + 1)

      Next j

    Next i

    For i = 1 To CMax - 1

      For j = 1 To CMax - 1

        Root = Sqr(v0Array(i, i) * v0Array(j, j))

        lccArray(i, j) = v0Array(i, j) / Root

      Next j

    Next i

    Selection.Value = lccArray

  End If

  If P = 1 Then

    For i = 1 To CMax

      For j = 1 To CMax

        v0Array(i, j) = vArray(i, j)

      Next j

    Next i

    For i = 1 To CMax

      For j = 1 To CMax

        Root = Sqr(Abs(vArray(i, i) * vArray(j, j)))

        lccArray(i, j) = vArray(i, j) / Root

      Next j

    Next i

    Selection.Value = lccArray

  End If

End If

' Format and color-code the cells in the CC array

' Color alerts indicating collinearity:

'   black     for         abs(CoVar) < 0.90

'   gray      for 0.90 <= abs(CoVar) < 0.95

'   orange    for 0.95 <= abs(CoVar) < 0.98

'   red       for 0.98 <= abs(CoVar) < 0.99

'   bold red  for 0.99 <= abs(CoVar) < 1

For Each Cell In Selection.Cells

  Cell.HorizontalAlignment = xlCenter

  Cell.Font.Italic = True

  Cell.Font.Bold = False

  a = Abs(Cell.Value)

  If a < 0.9 Then

    Cell.Font.ColorIndex = 1        ' black

  ElseIf (a >= 0.9 And a < 0.95) Then

    Cell.Font.ColorIndex = 16       ' gray

  ElseIf (a >= 0.95 And a < 0.98) Then

    Cell.Font.ColorIndex = 46       ' orange

  ElseIf (a >= 0.98 And a < 0.99) Then

    Cell.Font.ColorIndex = 3        ' red

  ElseIf a >= 0.99 Then

    Cell.Font.Bold = True           ' bold

    Cell.Font.ColorIndex = 3        ' red

  End If

  If Cell.Value = 1 Then Cell.Font.Italic = False

Next Cell

' Do not color-code the main diagonal

lccR = Selection.Rows.Count

lccC = Selection.Columns.Count

For i = 1 To lccR

  For j = 1 To lccC

    If i = j Then

      Selection(i, j).Font.Italic = False

      Selection(i, j).Font.Bold = False

      Selection(i, j).Font.ColorIndex = 1

    End If

  Next j

Next i

' Label the linear correlation coefficients

ActiveCell.Select

On Error Resume Next

ActiveCell.Offset(0, -1).Select

If (IsEmpty(ActiveCell) Or _

  ActiveCell.Value = "CC:") Then

  With ActiveCell

    .Value = "CC:"

    .Font.Bold = True

    .Font.Italic = True

    .Font.ColorIndex = 1

    .Interior.ColorIndex = xlNone

    .HorizontalAlignment = xlRight

    .ClearComments

    .AddComment

    .Comment.Visible = False

    .Comment.Text Text:= _

      "Collinearity alerts:" & Chr(10) & _

      "gray:       absCC >= 0.90" & Chr(10) & _

      "orange:   absCC >= 0.95" & Chr(10) & _

      "red:         absCC >= 0.98" & Chr(10) & _

      "bold red: absCC >= 0.99"

  End With

  ActiveCell.Offset(0, 1).Select

Else

  ActiveCell.Offset(0, 1).Select

  With ActiveCell

    .ClearComments

    .AddComment

    .Comment.Visible = False

    .Comment.Text Text:= _

      "Collinearity alerts:" & Chr(10) & _

      "gray:       absCC >= 0.90" & Chr(10) & _

      "orange:   absCC >= 0.95" & Chr(10) & _

      "red:         absCC >= 0.98" & Chr(10) & _

      "bold red: absCC >= 0.99"

  End With

End If

LastLine:

End Sub

'''''''''''''''''''''''''''''''''''''''''''''''''''''''''''

'''''''''''''''''''''''''''''''''''''''''''''''''''''''''''

'''''''''''*************************************'''''''''''

'''''''''''*                                   *'''''''''''

'''''''''''*     Equidistant least squares     *'''''''''''

'''''''''''*                                   *'''''''''''

'''''''''''*************************************'''''''''''

''''''''''''''''''''' © P. Barak, pwbarak@facstaff.wisc.edu

'''''''''''''''''''''''''''''''''''''' v 5.3, March 1, 2005

' PURPOSE:

' This program for a least-squares fit to EQUIDISTANT data

' y,x with a moving polynomial uses the approach pioneered

' by Gram, further developed by Sheppard [Proc. London

' Math. Soc. (2) 13 (1914) 81] and Sherriff [Proc. Royal

' Soc. Edinburgh 40 (1920) 112], and subsequently advocated

' by Whittaker & Robinson [The calculus of observations,

' Blackie & Son, 1924] and by Savitzky & Golay [Anal. Chem.

' 36 (1964) 1627]. It computes smoothed values of the data

' set, or (if so desired) its first or second derivative.

' There are two options: in ELSfixed() the user also

' selects the order of the polynomial, whereas in ELSauto()

' the program optimizes the order of the polynomial (bet-

' ween 1 and an upper limit set by the user) each time the

' moving polynomial slides one data point along the data

' set, using an algorithm described by Barak in Anal. Chem.

' 67 (1995) 2758.

' The program compares the ratio of the variances for a

' given order and that for the next-lower order with the

' corresponding F test as its first criterion. Since symme-

' trical functions often contain mostly even powers, a

' single, final comparison is made between the variances of

' the next-higher and the next-lower order. If the latter

' is not desired, simply comment out the section following

' the comment line "Second test for optimum Order".

' FUNCTIONS & SUBROUTINES:

' There are two drivers, ELSfixed and ELSauto, that call

' the main subroutine, ELS(iOrder). The latter uses the

' functions GenFact() and Smooth(). It also calls the sub-

' routine ConvolutionFactors, which calculates the Gram

' polynomials and the corresponding convolution weights.

' INPUT:

' The user must select the moving polynomial length; the

' corresponding value is restricted to odd integers between

' 3 and 31. This choice involves a compromise: the longer

' the polynomial, the more noise is removed, and the more

' the underlying signal is distorted. A useful rule of

' thumb is to make the length such that it encompasses no

' more than the half-width of the most narrow feature in

' the signal that should be preserved.

' Moreover, the length of the moving polynomial, PL, cannot

' exceed the length of the data set, NPts. The selected

' (maximum) order of the polynomial, MaxOrder, must be a

' positive integer, MaxOrder>0. Moreover, for ELSfixed,

' MaxOrder<PL, while for ELSauto we have MaxOrder<(PL-1).

' OUTPUT:

' The output is written in one or two columns to the right

' of the input data. The first output column contains the

' smoothed or differentiated data. The second column, which

' appears only with ELSauto(), displays the order selected

' by the program. Make sure that the output space is free,

' or can be overwritten.

' EXAMPLES:

' Use of this macro is illustrated in section 3.16 of

' AE3.

' Some of the abbreviations and indices used:

'     DerOrder: the derivative order, selected by the

'       answer to ELS InputBox 4: Derivative Order:

'         DerOrder = 0 for smoothing,

'         DerOrder = 1 for the 1st derivative,

'         DerOrder = 2 for the 2nd derivative.

'     FValueTable: the Fisher criterion, obtained

'       from the Excel function FInv.

'     GP: the Gram Polynomial

'     j: index for the order of the polynomial.

'     k: index for the position of the center of the

'       moving polynomial in the data set;

'       k ranges from m+1 to NPts-m.

'     m: number of points on each side of the central

'       point in the moving polynomial, calculated as

'       m = (PL-1)/2.

'     MaxOrder: the maximum order of the polynomial

'       (from ELS InputBox 3: Polynomial Order) or its

'       maximum value (from ELS InputBox 3: Maximum Poly-

'       nomial Order). MaxOrder must be an integer, > 0,

'       and < PL for ELSfixed, or < (PL-1) for ELSauto.

'     NPts: number of points in the data set, computed by

'       the macro from the data range provided in ELS

'       InputBox 1: Input data.

'     OptOrder: array of optimized Order values.

'     Order: working value of the polynomial order.

'       In ELSfixed, Order = MaxOrder; in ELSauto, Order

'       starts at 1, and has a maximum value of MaxOrder.

'     OutputData: the final (smoothed or derivative) result

'     OutputOrder:  the values of OptOrder used in ELSauto.

'     PL: length (in number of points) of the moving poly-

'       nomial, selected in ELS InputBox 2: Length of Mo-

'       ving Polynomial.

'     s: index for the order of the derivative; s ranges

'       from 0 to DerivOrder.

'     t: index for the individual points in the moving

'       polynomial; t ranges from -m to +m.

'     tries:  number of attempts to enter data in input box

'     Y() and YData(): the input data containing the entire

'       data set.

'''''''''''''''''''''''''''''''''''''''''''''''''''''''''''

Sub ELSfixed()

' Selects a fixed polynomial order

' by setting iOrder equal to 1

Dim iOrder As Integer

iOrder = 1

Call ELS(iOrder)

End Sub

'''''''''''''''''''''''''''''''''''''''''''''''''''''''''''

Sub ELSauto()

' Selects a variable polynomial order (between 1 and a

' user-selectable maximum value,MaxOrder) by setting iOrder

' equal to -1. Note that the high-order polynomials may be

' undesirable for use with, e.g., differentiation or inter-

' polation.

Dim iOrder As Integer

iOrder = -1

Call ELS(iOrder)

End Sub

'''''''''''''''''''''''''''''''''''''''''''''''''''''''''''

Sub ELS(iOrder)

Dim a As Integer, B As Integer, DerOrder As Integer

Dim i As Integer, ii As Integer, j As Integer

Dim jj As Integer, k As Integer, m As Integer

Dim MaxOrder As Integer, N As Integer, NPts As Integer

Dim Order As Integer, PL As Integer, Q As Integer

Dim s As Integer, t As Integer, Tries2 As Integer

Dim Tries3 As Integer, Tries4 As Integer

Dim GenFact As Double, Percentage As Long

Dim aA As Double, bb As Double, DeltaX As Double

Dim FTest1 As Double, FTest2 As Double, Length As Double

Dim Perc As Double, Sum As Double, SumSq As Double

Dim SumXY As Double, SumY As Double, SumY2 As Double

Dim GP() As Double, SumSquares() As Double

Dim SumX2() As Double, w() As Double, Y() As Double

Dim OptOrder As Variant, outputData As Variant

Dim OutputOrder As Variant, TestContents As Variant

Dim YData As Variant

Dim myRange As Range

Dim Ans As String

Dim Z

' Preliminaries and data input

Set myRange = _

  Application.InputBox(Prompt:="The input data are " _

  & "located in column:", _

  Title:="ELS InputBox 1: Input data", Type:=8)

myRange.Select

If Selection.Columns.Count <> 1 Then

  MsgBox "Only enter a single column of input data."

  End

End If

NPts = Selection.Rows.Count

If NPts = 0 Then End

YData = myRange.Value

outputData = myRange.Value          'defines the array size

OutputOrder = myRange.Value         'defines the array size

OptOrder = myRange.Value            'defines the array size

' Test and prepare the default output range

N = 0

Selection.Offset(0, 1).Select

TestContents = Selection.Value

For i = 1 To NPts

  Z = TestContents(i, 1)

  If IsEmpty(Z) Then

    N = N

  Else

    N = N + 1

  End If

Next i

If iOrder = 1 Then

  If N > 0 Then

    Ans = MsgBox(" There are data in the " _

      & "column where the output " & Chr(13) & _

      "will be written. Proceed anyway and overwrite " _

      & "those data?", vbYesNo, _

      "Equidistant Least Squares Fit")

    If Ans = vbNo Then

      MsgBox ("Safeguard the data in the highlighted " _

        & "area by" & Chr(13) & "moving them to another " _

        & "place, then try again."), _

        , "Equidistant Least Squares Fit"

    End

    End If

  End If

  Selection.Offset(0, -1).Select

End If

If iOrder = -1 Then

  Selection.Offset(0, 1).Select

  TestContents = Selection.Value

  For i = 1 To NPts

    Z = TestContents(i, 1)

    If IsEmpty(Z) Then

      N = N

    Else

      N = N + 1

    End If

  Next i

  If N > 0 Then

    Ans = MsgBox("There are data in the TWO columns " _

      & "where the output " & Chr(13) & "will be " _

      & "written. Proceed anyway and overwrite " _

      & Chr(13) & "those data?", vbYesNo, "Equidistant " _

      & "Least Squares Fit")

    If Ans = vbNo Then

      MsgBox ("Safeguard the data in the highlighted " _

      & "area by" & Chr(13) & "moving them to another " _

      & "place, then try again."), _

      , "Equidistant Least Squares Fit"

    End

    End If

  End If

  Selection.Offset(0, -2).Select

End If

' Select the length of the moving polynomial, PL

Tries2 = 0

Line2:

Length = InputBox(Prompt:="The length of the moving " _

  & "polynomial is:", _

  Title:="ELS InputBox 2: Polynomial Length")

PL = CInt(Length)

' Make sure that PL is an odd integer

' larger than 0 and smaller than NPts

If (Length <= 0 Or Length >= NPts Or PL - Length <> 0 _

  Or CInt((PL - 1) / 2) - ((Length - 1) / 2) <> 0) Then

  MsgBox "The length of the moving polynomial must" & _

    Chr(13) & "    be an odd integer larger than zero " _

    & "and " & Chr(13) & "smaller than the length of " & _

    "the input column.", , "Equidistant Least Squares Fit"

  Tries2 = Tries2 + 1

  If Tries2 = 2 Then End

  GoTo Line2

End If

' Select the order of the moving polynomial, MaxOrder

Tries3 = 0

Line3:

If iOrder = 1 Then

  MaxOrder = InputBox(Prompt:="The order of the moving" _

    & "polynomial is:", _

    Title:="ELS InputBox 3: Polynomial Order")

Else

  MaxOrder = InputBox(Prompt:="The maximum order of the " _

    & "moving polynomial is:", _

    Title:="ELS InputBox 3: Polynomial Order")

End If

' Make sure that MaxOrder > 0 and that either MaxOrder < PL

' (for ELSfixed) or MaxOrder < PL - 1 (for ELSauto).

If iOrder = 1 Then

  If (MaxOrder <= 0 Or MaxOrder >= PL) Then

    MsgBox "The order of the moving polynomial" & Chr(13) _

      & "must be larger than zero, and smaller" & Chr(13) _

      & "than the length of the moving polynomial.", , _

      "Equidistant Least Squares Fit"

    Tries3 = Tries3 + 1

    If Tries3 = 2 Then End

    GoTo Line3

  End If

Else

  If (MaxOrder <= 0 Or MaxOrder >= PL - 1) Then

    MsgBox "The maximum order of the moving polynomial" & _

      Chr(13) & "must be larger than zero, and smaller " _

      & "than" & Chr(13) & "the length of the moving " _

      & "polynomial minus 1.", , _

      "Equidistant Least Squares Fit"

    Tries3 = Tries3 + 1

    If Tries3 = 2 Then End

    GoTo Line3

  End If

End If

' Select smoothing, first derivative, or second derivative

Tries4 = 0

Line4:

DerOrder = InputBox(Prompt:="Select the order of the " _

  & "derivative" & Chr(13) & "(either 1 or 2); for " _

  & "smoothing, select 0." & Chr(13) & Chr(13) _

  & "The order of the derivative is:", _

  Title:="ELS InputBox 4:" & "Derivative Order")

' Make sure that DerOrder has the value 0, 1, or 2

If DerOrder = 0 Then

  GoTo Line6

ElseIf DerOrder = 1 Then

  GoTo Line5

ElseIf DerOrder = 2 Then

  GoTo Line5

Else

  MsgBox " The order of the moving polynomial must be" & _

  Chr(13) & "either 0 (for smoothing), 1 (for the first" _

  & Chr(13) & "derivative), or 2 (for the second " & _

  "derivative).", , "Equidistant Least Squares Fit"

  Tries4 = Tries4 + 1

  If Tries4 = 2 Then End

  GoTo Line4

End If

Line5:

DeltaX = InputBox(Prompt:="The data spacing in x is:", _

  Title:="ELS InputBox 5: X Increment")

Line6:

m = (PL - 1) / 2

ReDim Y(1 To NPts), OptOrder(1 To NPts, 1)

For i = 1 To NPts

  Y(i) = YData(i, 1)

  OptOrder(i, 1) = MaxOrder

Next i

Call ConvolutionFactors(PL, MaxOrder, DerOrder, GP, w, Y)

' THE FOLLOWING SECTION IS USED ONLY BY ELSauto

If iOrder = -1 Then

  ReDim SumX2(1 To MaxOrder)

  For j = 1 To MaxOrder

    Sum = 0

    For i = -m To m

      Sum = Sum + GP(i, j, 0) ^ 2

    Next i

    SumX2(j) = Sum

  Next j

' Calculate FValueTable(MaxOrder,PL)

  ReDim FValueTable(1 To MaxOrder, 1 To PL)

  jj = 0

  Do

    jj = jj + 1

    For ii = 1 To MaxOrder

      FValueTable(ii, jj) = Application.FInv(0.05, ii, jj)

    Next ii

  Loop Until jj = PL

  For k = m + 1 To NPts - m

    ReDim SumSquares(0 To MaxOrder)

' Calculate SumSquares for Order = 0

    Order = 0

    SumY = 0

    SumY2 = 0

    For t = -m To m

      SumY = SumY + Y(k + t)

      SumY2 = SumY2 + Y(k + t) ^ 2

    Next t

    SumSquares(0) = SumY2 - SumY ^ 2 / (2 * t + 1)

' Calculate SumSquares for Order = 1

    Order = 1

    SumSq = 0

    For t = -m To m

      SumSq = SumSq + (Smooth(PL, k, Order, t, 0, w, Y) _

        - Y(k + t)) ^ 2

    Next t

    SumSquares(1) = SumSq

' Test whether one-higher order satisfies the criterion

    Do

      Order = Order + 1

      If Order > MaxOrder Then GoTo line10

' Calculate SumSquares for Order > 1

      SumXY = 0

      For t = -m To m

        SumXY = SumXY + Y(k + t) * GP(t, Order, 0)

      Next t

      SumSquares(Order) = SumSquares(Order - 1) _

        - SumXY ^ 2 / SumX2(Order)

' First test for optimum Order

      FTest1 = _

        (SumSquares(Order - 1) - SumSquares(Order)) * _

        (PL - Order) / (SumSquares(Order))

    Loop Until _

      (FTest1 / FValueTable(1, PL - Order - 1)) < 1

' Second test for optimum Order

    If Order < MaxOrder Then

      Order = Order + 1

      SumXY = 0

      For t = -m To m

        SumXY = SumXY + Y(k + t) * GP(t, Order, 0)

      Next t

      SumSquares(Order) = SumSquares(Order - 1) _

        - SumXY ^ 2 / SumX2(Order)

      FTest2 = _

        (SumSquares(Order - 2) - SumSquares(Order)) * _

        (PL - Order) / (SumSquares(Order))

      If (FTest2 / FValueTable(2, PL - Order - 1)) < 1 _

        Then Order = Order - 1

    End If

line10:

    OptOrder(k, 1) = Order - 1

    Perc = 100 * (k / NPts)

    Percentage = Int(Perc)

    Application.StatusBar = _

      "Calculation " & Percentage & "% done."

  Next k

End If

' THIS ENDS THE PART USED ONLY BY ELSauto

' Prepare the output files

For k = m + 1 To NPts - m

  If k = m + 1 Then

    For t = -m To -1

      outputData(k + t, 1) = _

        Smooth(PL, k, OptOrder(k + t, 1), _

        t, DerOrder, w, Y) / (DeltaX ^ DerOrder)

      OutputOrder(k + t, 1) = OptOrder(k, 1)

    Next t

  End If

  outputData(k, 1) = _

    Smooth(PL, k, OptOrder(k, 1), 0, DerOrder, w, Y) _

    / (DeltaX ^ DerOrder)

  OutputOrder(k, 1) = OptOrder(k, 1)

  If k = NPts - m Then

    For t = 1 To m

      outputData(k + t, 1) = _

        Smooth(PL, k, OptOrder(k + t, 1), _

        t, DerOrder, w, Y) / (DeltaX ^ DerOrder)

      OutputOrder(k + t, 1) = OptOrder(k, 1)

    Next t

  End If

Next k

' Write the output files

Selection.Offset(0, 1).Select

Selection.Value = outputData

If iOrder = -1 Then

  Selection.Offset(0, 1).Select

  Selection.Value = OutputOrder

  Selection.Offset(0, -1).Select

End If

Application.StatusBar = False

End Sub

'''''''''''''''''''''''''''''''''''''''''''''''''''''''''''

Function GenFact(a, B)

' Computes the generalized factorial

Dim gf As Double, j As Integer

gf = 1

For j = (a - B + 1) To a

  gf = gf * j

Next j

GenFact = gf

End Function

'''''''''''''''''''''''''''''''''''''''''''''''''''''''''''

Public Function Smooth(PL, k, j, t, s, w, Y)

' Computes the appropriately weighted sum of the Y-values

Dim i As Integer, m As Integer, Sum As Double

m = (PL - 1) / 2

Sum = 0

For i = -m To m

  Sum = Sum + w(i, t, j, s) * Y(k + i)

Next i

Smooth = Sum

End Function

'

'''''''''''''''''''''''''''''''''''''''''''''''''''''''''''

'

Sub ConvolutionFactors(PL, MaxOrder, DerOrder, GP, w, Y)

' Calculates tables of GP(i = -m to m, k = -1 to MaxOrder,

' s = 1 to DerOrder), and of W(i = -m to m, t = -m to m,

' k = -1 to MaxOrder, s = -1 to DerOrder)

' Abbreviations used:

'   DerOrder: Derivative Order

'   GP:       Gram Polynomial

'   MaxOrder: Maximum Order,

'   PL:       Polynomial Length

'   W:        Weight

Dim i As Integer, k As Integer, m As Integer

Dim s As Integer, t As Integer

Dim aA As Double, bb As Double

m = (PL - 1) / 2

ReDim GP(-m To m, -1 To MaxOrder, -1 To DerOrder)

'Evaluate the Gram polynomials for DerOrder=0

For i = -m To m

  GP(i, 0, 0) = 1

  GP(i, 1, 0) = 0

Next i

For i = -m To -1

  GP(i, 1, 0) = i / m

Next i

For i = 1 To m

  GP(i, 1, 0) = i / m

Next i

For k = 2 To MaxOrder

  aA = 2 * (2 * k - 1) / (k * (2 * m - k + 1))

  bb = ((k - 1) * (2 * m + k)) / (k * (2 * m - k + 1))

  For i = 0 To m

    GP(i, k, 0) = aA * i * GP(i, k - 1, 0) - _

      bb * GP(i, k - 2, 0)

  Next i

  For i = -m To -1

    If k Mod 2 = 0 Then

      GP(i, k, 0) = GP(-i, k, 0)

    Else

      GP(i, k, 0) = -GP(-i, k, 0)

    End If

  Next i

Next k

'Evaluate the Gram polynomials for DerOrder>0

If DerOrder > 0 Then

  For s = 1 To DerOrder

    For i = -m To m

      GP(i, -1, s) = 0

      GP(i, 0, s) = 0

    Next i

    For k = 1 To MaxOrder

      aA = 2 * (2 * k - 1) / (k * (2 * m - k + 1))

      bb = ((k - 1) * (2 * m + k)) / (k * (2 * m - k + 1))

      For i = -m To m

        GP(i, k, s) = aA * (i * GP(i, k - 1, s) + _

        s * GP(i, k - 1, s - 1)) - bb * GP(i, k - 2, s)

      Next i

    Next k

  Next s

End If

'Calculate the convolution weights

ReDim w(-m To m, -m To m, -1 To MaxOrder, -1 To DerOrder)

For k = 0 To MaxOrder

  aA = (2 * k + 1) * GenFact(2 * m, k) / _

    GenFact(2 * m + k + 1, k + 1)

  For s = 0 To DerOrder

    For i = -m To m

      For t = -m To m

        w(i, t, k, s) = w(i, t, k - 1, s) _

        + aA * GP(i, k, 0) * GP(t, k, s)

      Next t

    Next i

  Next s

Next k

End Sub

'''''''''''''''''''''''''''''''''''''''''''''''''''''''''''

'''''''''''''''''''''''''''''''''''''''''''''''''''''''''''

''''''''''''*********************************''''''''''''''

''''''''''''*                               *''''''''''''''

''''''''''''*    Weighted least squares     *''''''''''''''

''''''''''''*                               *''''''''''''''

''''''''''''*********************************''''''''''''''

''''''''''''''''''''''''''''''''''''''''''' (c) R. de Levie

'''''''''''''''''''''''''''''''''''''''''''  v 12, May 2012

' PURPOSE:

' The macros LSI and LSO set the input parameter p for the

' subroutine LeastSquares, which in turn computes the indi-

' vidual coefficients or parameters with their standard

' deviations, plus the corresponding covariance matrix CM

' and the array of linear correlation coefficients CC for

' an unweighted least squares fit to data in 2 or more

' columns. Moreover, it provides the standard deviation Sf

' of the over-all fit of the function to the data.

' Color coding is used to alert the user when the indivi-

' dual standard deviations are of the same order of magni-

' tude as the corresponding coefficients, and when the

' independent variables exhibit significant collinearity.

' This macro uses the traditional least squares algorithm.

' When the LC matrix indicates high collinearity, and you

' use Excel 2003 or a more recent version, you may instead

' want to use Excel's own LinEst or Regression, which are

' now based on QRDecomposition. Otherwise use the Singular

' Value Decomposition macros in Volpi's Matrix.xla. But

' neither of these provides the covariance matrix, which

' you may therefore have to construct yourself using

' Solver and SolverAid.

' The input parameter p = 1 starts a general unweighted

' least squares fit to the data, while p = 0 forces the fit

' to pass through the origin, i.e., it assumes that y = 0

' for x = 0.

' SUBROUTINES:

' This macro requires the subroutines Multiply, VMultiply,

' Invert, and Transpose

' NOTE REGARDING VMultiply:

' Logically, WLS should involve an (rMax by rMax) matrix

' containing, on its main diagonal, the weights assigned

' to the various individual measurements, and with zeros

' everywhere else. For large values of rMax this matrix

' can readily become so large as to slow down the macro.

' Therefore, the weights have instead been placed in a

' vector, WVector, and the multiplication procedure has

' been adjusted accordingly, in the subroutine VMultiply.

' INPUT:

' The input data must be organized in columns, arranged as

' follows. The first column must contain the dependent va-

' riable y. The second column should contain the weights w

' of the individual points. The third (and any subsequent)

' column(s) must contain the independent variable(s) x.

' If an unweighted least squares is desired, leave the

' weights blank, or enter the same number (say, 1, or 13)

' for all individual weights.

' Do NOT delete the second column, as in that case the

' macro will misinterpret the input.

' OUTPUT:

' The macro labels the coefficients and standard devia-

' tions, except when there is no space available for them

' because the first data column is at the left edge of the

' spreadsheet.

' If the data to be analyzed are the result of a trans-

' formation (such as taking logarithms of an exponential

' decay), the standard deviation of the overall fit will

' not yield the correct answer with respect to those

' original, untransformed data, because the macro does not

' read those data. If that standard deviation is desired,

' it is best computed separately on the spreadsheet.

' Comment boxes are used:

' (1) to identify the version number, the data input range,

' the number N of data points, the number P of adjustable

' fitting parameters, and date and time of the analysis;

' (2) to explain the color code used to warn the user about

' the reliability of the coefficient ai, by displaying a

' categorical measure of the absolute value of the ratio of

' ai to its standard deviation si; and

' (3) to explain the color code used to alert the user of

' high collinearity, again using a crude categorization.

' PROCEDURE:

' Before calling the macro, make sure that the output area

' is free of valuable data. To find out the reach of the

' output, try out the macro but, in the "Overwrite?"

' message box, note the number of rows required, answer

' "No" to whether the data can be overwritten, move the

' sensitive data, then try the macro again.

' In order to start the process, highlight the entire input

' data block, then call WLSI or WLSO.

' EXAMPLES:

' Use of this macro is illustrated starting in sections

' 3.17 through 3.20 of AE3.

'''''''''''''''''''''''''''''''''''''''''''''''''''''''''''

' NOTATION:

' This subroutine computes the parameters and their

' standard deviations for a weighted or unweighted least

' squares fit to data in 3 or more columns. The columns

' must be arranged as follows:

' The first column must contain the dependent variable y.

' The second column must contain the weights w for y. For

' an unweighted least squares fit, leave this column blank,

' or fill it with ones, but do not omit it. The third (and

' possibly subsequent) column(s) must contain the indepen-

' dent variable(s) x.

' The subroutine requires an input parameter p: p = 1

' causes a general weighted least squares fit to the data,

' while p = 0 forces the fit to pass through the origin,

' i.e., it assumes that Y = 0 for X = 0. The subroutine is

' therefore called by a driver (WLSO or WLSI) that sets the

' value of p.

Sub WLSO()          ' for least squares fit

                    ' through the origin

Dim P As Double

P = 0

Call WeightedLeastSquares(P)

End Sub

'''''''''''''''''''''''''''''''''''''''''''''''''''''''''''

Sub WLSI()          ' for general least squares fit

Dim P As Double

P = 1

Call WeightedLeastSquares(P)

End Sub

'''''''''''''''''''''''''''''''''''''''''''''''''''''''''''

Sub WeightedLeastSquares(P)

Dim ccMax As Integer, CMax As Integer

Dim j As Integer, m As Integer, N As Integer

Dim i As Long, RMax As Long

Dim Root As Double, SSR As Double, StDevF As Double

Dim sumW As Double, u As Double, varY As Double

Dim DataArray As Variant, OutputArray As Variant

Dim lccArray As Variant

Dim vArray As Variant, v0Array As Variant

Dim myRange As Range

Dim Answer, hAnswer, iAnswer, jAnswer, kAnswer

Dim BoxAddress, myAddress1, myAddress2

Dim bArray, btArray, btqArray

Dim pArray, piArray, qArray, WVector

Dim XArray, xtArray, YArray, ytArray, ytyArray

' Determination of the array size:

Begin:

RMax = Selection.Rows.Count

CMax = Selection.Columns.Count

ccMax = CMax - 1

u = 1

' If area was not highlighted

If RMax = 1 And CMax = 1 Then

  hAnswer = MsgBox("You forgot to highlight" _

    & Chr(13) & "the block of input data." _

    & Chr(13) & "Do you want to do so now?" _

    , vbYesNo, "Weighted Least Squares")

  If hAnswer = vbNo Then End

  If hAnswer = vbYes Then

    Set myRange = Application.InputBox(Prompt:= _

      "The input data are located in:", Type:=8)

    myRange.Select

  End If

  GoTo Begin

End If

' Check that the number of columns is at least 3:

If CMax < 3 Then

  MsgBox "There must be at least three columns, one" & _

  Chr(13) & "for Y, one for W, and one or more for X.", _

  , "Weighted Least Squares Fit"

  End

End If

' Check that there are more data than coefficients :

If RMax - CMax - P + 2 <= 0 Then

  MsgBox "With " & RMax & " data, WLS" & P & " can only deter-" & _

    Chr(13) & "mine " & RMax - 1 & " least squares coefficients." & _

    Chr(13) & Chr(13) & "Add more data, or reduce the re-" & _

    Chr(13) & "quested number of coefficients."

  End

End If

' Dimension the arrays:

ReDim YArray(1 To RMax, 1 To 1) As Double

ReDim WVector(1 To RMax) As Double

ReDim XArray(1 To RMax, 1 To ccMax) As Double

ReDim ytArray(1 To 1, 1 To RMax) As Double

ReDim ytwArray(1 To 1, 1 To RMax) As Double

ReDim ytwyArray(1 To 1, 1 To 1) As Double

ReDim xtArray(1 To ccMax, 1 To RMax) As Double

ReDim xtwArray(1 To ccMax, 1 To RMax) As Double

ReDim pArray(1 To ccMax, 1 To ccMax) As Double

ReDim piArray(1 To ccMax, 1 To ccMax) As Double

ReDim qArray(1 To ccMax, 1 To 1) As Double

ReDim bArray(1 To ccMax, 1 To 1) As Double

ReDim btArray(1 To 1, 1 To ccMax) As Double

ReDim btqArray(1 To 1, 1 To 1) As Double

ReDim vArray(1 To ccMax, 1 To ccMax) As Double

ReDim v0Array(1 To ccMax + P - 1, 1 To ccMax + P - 1) _

  As Double

ReDim lccArray(1 To ccMax + P - 1, 1 To ccMax + P - 1) _

  As Double

' Read the dataArray

DataArray = Selection.Value

' Fill YArray

For i = 1 To RMax

  YArray(i, 1) = DataArray(i, 1)

Next i

For i = 1 To RMax

  If IsEmpty(DataArray(i, 1)) Then

    MsgBox "Y-value(s) missing", _

      , "Weighted Least Squares"

    End

  End If

Next i

For i = 1 To RMax

  If IsEmpty(DataArray(i, 2)) Then DataArray(i, 2) = u

Next i

For j = 3 To CMax

  For i = 1 To RMax

    If IsEmpty(DataArray(i, j)) Then

      MsgBox "X-value(s) missing", _

        , "Weighted Least Squares"

      End

    End If

  Next i

Next j

' Fill wVector with the individual, normalized weights.

sumW = 0

For i = 1 To RMax

  sumW = sumW + DataArray(i, 2)

Next i

For i = 1 To RMax

  WVector(i) = DataArray(i, 2) * RMax / sumW

Next i

' Fill the first column of XArray with zeroes (for p = 0)

' or ones (for p = 1), the rest with the data in the

' x-column(s)

For i = 1 To RMax

  XArray(i, 1) = P

Next i

For j = 3 To CMax

  For i = 1 To RMax

    XArray(i, (j - 1)) = DataArray(i, j)

  Next i

Next j

' Compute b = (X' W X)" X' W Y , where ' or t denote

' transposition, and " or i indicate inversion

' The various arrays and their dimensions (rows, columns)

' are:

'   Y          = yArray        (  rMax,     1)

'   W          = wVector       (  rMax)

'   X          = xArray        (  rMax, ccMax)

'   X'         = xtArray       ( ccMax,  rMax)

'   X' W       = xtwArray      ( ccMax,  rMax)

'   X' W X     = pArray        ( ccMax, ccMax)

'  (X' W X)"   = piArray       ( ccMax, ccMax)

'   X' W Y     = qArray        ( ccMax,     1)

'   b          = bArray        ( ccMax,     1)

Call Transpose(XArray, RMax, ccMax, xtArray)

Call VMultiply(xtArray, ccMax, RMax, WVector, xtwArray)

Call Multiply(xtwArray, ccMax, RMax, XArray, ccMax, pArray)

Call Invert(pArray, ccMax, piArray)

Call Multiply(xtwArray, ccMax, RMax, YArray, 1, qArray)

Call Multiply(piArray, ccMax, ccMax, qArray, 1, bArray)

' Check against overwriting spreadsheet data

Selection.Offset(2, 0).Select

OutputArray = Selection.Value

Selection.Offset(-2, 0).Select

m = 0

If (P = 0 And CMax = 3) Then

  For i = 1 To 3

    Selection.Offset(1, 0).Select

    OutputArray = Selection.Value

    For j = 1 To CMax

      If IsEmpty(OutputArray(RMax, j)) Then

        m = m

      Else

        m = m + 1

      End If

    Next j

  Next i

  Selection.Offset(-3, 0).Select

  If m > 0 Then Answer = MsgBox("There are data in the " _

    & "three lines below the" & Chr(13) & _

    "input data array. " & "Can they be overwritten?", _

    vbYesNo, "Overwrite?")

  If Answer = vbNo Then End

Else

  For i = 1 To 3 + P + CMax

    Selection.Offset(1, 0).Select

    OutputArray = Selection.Value

    For j = 1 To CMax

      If IsEmpty(OutputArray(RMax, j)) Then

        m = m

      Else

        m = m + 1

      End If

    Next j

  Next i

  Selection.Offset(-3 - P - CMax, 0).Select

  If m > 0 Then Answer = MsgBox("There are data in the " _

    & 1 + P + CMax & " lines below the" & Chr(13) & _

    "input data array. " & "Can they be overwritten?", _

    vbYesNo, "Overwrite?")

  If Answer = vbNo Then End

End If

' The additional arrays and their dimensions (rows,

' columns) are:

'   Y'         = ytArray       (     1,  rMax)

'   Y' W       = ytwArray      (     1,  rMax)

'   Y' W Y     = ytwyArray     (     1,     1)

'   b'         = btArray       (     1, ccMax)

'   b' X' W Y  = btqArray      (     1,     1)

Call Transpose(YArray, RMax, 1, ytArray)

Call Transpose(bArray, ccMax, 1, btArray)

Call VMultiply(ytArray, 1, RMax, WVector, ytwArray)

Call Multiply(ytwArray, 1, RMax, YArray, 1, ytwyArray)

Call Multiply(btArray, 1, ccMax, qArray, 1, btqArray)

' Calculate SSR = Y'WY - b'X'WY, and then varY, the vari-

' ance of y, as varY = SSR/(rmax-ccmax); and vArray, the

' covariance matrix, as V = (X'WX)" times varY, of which

' we here only use the diagonal elements, i.e., the vari-

' ances.

SSR = ytwyArray(1, 1) - btqArray(1, 1)

varY = SSR / (RMax - ccMax - P + 1)

If SSR >= 0 Then StDevF = Sqr(varY)

If SSR < 0 Then

  StDevF = Sqr(-varY)

  MsgBox "SSR is " & SSR & ", which is negative!"

End If

For i = 1 To ccMax

  For j = 1 To ccMax

    vArray(i, j) = varY * piArray(i, j)

  Next j

Next i

Application.ScreenUpdating = False

ActiveCell.Offset(RMax, 0).Select

For j = 1 To CMax

  ActiveCell.Font.Bold = True

  ActiveCell.Font.Italic = True

  ActiveCell.Value = ""

  ActiveCell.Offset(0, 1).Select

Next j

ActiveCell.Offset(0, -CMax).Select

ActiveCell.Value = "Coeff:"

ActiveCell.HorizontalAlignment = xlRight

ActiveCell.Offset(0, 1).Select

For j = 2 - P To ccMax

  ActiveCell.Value = bArray(j, 1)

  ActiveCell.HorizontalAlignment = xlCenter

  ActiveCell.Offset(0, 1).Select

Next j

ActiveCell.Offset(1, 1 - P - CMax).Select

For j = 1 To CMax

  ActiveCell.Font.Bold = False

  ActiveCell.Font.Italic = True

  ActiveCell.Value = ""

  ActiveCell.HorizontalAlignment = xlCenter

  ActiveCell.Offset(0, 1).Select

Next j

ActiveCell.Offset(0, -CMax).Select

ActiveCell.Value = "StDev:"

ActiveCell.HorizontalAlignment = xlRight

ActiveCell.Offset(0, 1).Select

For j = 2 - P To ccMax

  If vArray(j, j) < 1E-40 Then

    ActiveCell.Value = "<1E-20"

    ActiveCell.HorizontalAlignment = xlCenter

  Else

    ActiveCell.Value = Sqr(vArray(j, j))

  End If

  ActiveCell.Offset(0, 1).Select

Next j

ActiveCell.Offset(1, 1 - P - CMax).Select

ActiveCell.Font.Bold = False

ActiveCell.Font.Italic = True

ActiveCell.Value = "Sf:"

ActiveCell.HorizontalAlignment = xlRight

ActiveCell.Offset(0, 1).Select

ActiveCell.Value = StDevF

ActiveCell.HorizontalAlignment = xlCenter

ActiveCell.Offset(1, -1).Select

' Compute and display the covariance matrix

If (P = 0 And CMax = 3) Then GoTo LastLine

If P = 0 Then

  For i = 1 To CMax - 2

    For j = 1 To CMax

      ActiveCell.Font.Bold = False

      ActiveCell.Font.Italic = True

      ActiveCell.Font.ColorIndex = 1

      ActiveCell.Value = ""

      ActiveCell.HorizontalAlignment = xlCenter

      ActiveCell.Offset(0, 1).Select

    Next j

    ActiveCell.Offset(1, -CMax).Select

  Next i

  ActiveCell.Offset(2 - CMax, 0).Select

  ActiveCell.Font.Bold = True

  ActiveCell.Font.ColorIndex = 11

  ActiveCell.Value = "CM:"

  ActiveCell.HorizontalAlignment = xlRight

  ActiveCell.Offset(0, 1).Select

  For i = 1 To CMax - 2

    For j = 1 To CMax - 2

      ActiveCell.Font.ColorIndex = 11

      ActiveCell.Value = vArray(i, j)

      ActiveCell.Offset(0, 1).Select

    Next j

    ActiveCell.Offset(1, 2 - CMax).Select

  Next i

Else

  For i = 1 To CMax - 1

    For j = 1 To CMax

      ActiveCell.Font.Bold = False

      ActiveCell.Font.Italic = True

      ActiveCell.Font.ColorIndex = 1

      ActiveCell.Value = ""

      ActiveCell.Offset(0, 1).Select

    Next j

    ActiveCell.Offset(1, -CMax).Select

  Next i

  ActiveCell.Offset(1 - CMax, 0).Select

  ActiveCell.Font.Bold = True

  ActiveCell.Font.ColorIndex = 11

  ActiveCell.Value = "CM:"

  ActiveCell.HorizontalAlignment = xlRight

  ActiveCell.Offset(0, 1).Select

  For i = 1 To CMax - 1

    For j = 1 To CMax - 1

      ActiveCell.Font.ColorIndex = 11

      ActiveCell.Value = vArray(i, j)

      ActiveCell.HorizontalAlignment = xlCenter

      ActiveCell.Offset(0, 1).Select

    Next j

    ActiveCell.Offset(1, 1 - CMax).Select

  Next i

End If

Application.ScreenUpdating = True

' Provide as optional output the array of linear

' correlation coefficients. The user specifies

' the cell block in which to write this array

If P = 0 And CMax = 3 Then GoTo LastLine

jAnswer = MsgBox("Do you want to see the " _

  & "matrix of linear correlation" _

  & Chr(13) & "coefficients? It will need a " _

  & "block of " & CMax + P - 2 _

  & " by " & CMax + P - 2 & " cells.", vbYesNo, _

  "Least Squares Fit")

' Read location of active cell

ActiveCell.Select

myAddress1 = Selection.Address

ActiveCell.Offset(CMax + P - 3, CMax + P - 3).Select

myAddress2 = Selection.Address

ActiveCell.Offset(3 - CMax - P, 3 - CMax - P).Select

BoxAddress = myAddress1 & ":" & myAddress2

OutlineMatrix:

If jAnswer = vbYes Then

  Set myRange = Application.InputBox(Prompt:= _

    "Select the default array location with OK," _

    & vbCr & "otherwise type in your own coordinates:", _

    Default:=BoxAddress, Type:=8)

  myRange.Select

' Make sure that the selected block has the correct size

  If Selection.Rows.Count <> CMax + P - 2 Then

    MsgBox "The selected range does not have " _

      & CMax + P - 2 & " rows. Please correct.", _

      , "Least Squares Fit"

    GoTo OutlineMatrix

  End If

  If Selection.Columns.Count <> CMax + P - 2 Then

    MsgBox "The selected range does not have " _

      & CMax + P - 2 & " columns. Please correct", _

      , "Least Squares Fit"

    GoTo OutlineMatrix

  End If

' Draw a box around the reserved area

  Selection.Borders(xlEdgeTop).LineStyle = xlDashDot

  Selection.Borders(xlEdgeTop).Weight = xlThin

  Selection.Borders(xlEdgeRight).LineStyle = xlDashDot

  Selection.Borders(xlEdgeRight).Weight = xlThin

  Selection.Borders(xlEdgeBottom).LineStyle = xlDashDot

  Selection.Borders(xlEdgeBottom).Weight = xlThin

  Selection.Borders(xlEdgeLeft).LineStyle = xlDashDot

  Selection.Borders(xlEdgeLeft).Weight = xlThin

' Format cells in array

  For Each Cell In Selection.Cells

    Cell.HorizontalAlignment = xlCenter

  Next Cell

' Write the array of linear correlation coefficients

  If P = 0 Then

    For i = 1 To CMax - 2

      For j = 1 To CMax - 2

        v0Array(i, j) = vArray(i + 1, j + 1)

      Next j

    Next i

    For i = 1 To CMax - 2

      For j = 1 To CMax - 2

        Root = Sqr(v0Array(i, i) * v0Array(j, j))

        lccArray(i, j) = v0Array(i, j) / Root

      Next j

    Next i

    Selection.Value = lccArray

  End If

  If P = 1 Then

    For i = 1 To CMax - 1

      For j = 1 To CMax - 1

        v0Array(i, j) = vArray(i, j)

      Next j

    Next i

    For i = 1 To CMax - 1

      For j = 1 To CMax - 1

        Root = Sqr(vArray(i, i) * vArray(j, j))

        lccArray(i, j) = vArray(i, j) / Root

      Next j

    Next i

    Selection.Value = lccArray

  End If

End If

LastLine:

End Sub

'''''''''''''''''''''''''''''''''''''''''''''''''''''''''''

'''''''''''''''''''''''''''''''''''''''''''''''''''''''''''

''''''''''''''*******************************''''''''''''''

''''''''''''''*                             *''''''''''''''

''''''''''''''*          GradeBySf          *''''''''''''''

''''''''''''''*                             *''''''''''''''

''''''''''''''*******************************''''''''''''''

''''''''''''''''''''''''''''''''''''''''''' (c) R. de Levie

'''''''''''''''''''''''''''''''''''''''''''  v 12, May 2012

' PURPOSE:

' The macro GradeBySf computes the standard deviations of

' the fit, Sf, for all possible combinations and permuta-

' tions of unweighted least squares fits of a dependent

' variable y to a user-specified multivariate expression

' of the form y = a1x1 + a2x2 + … (for "Thru 0", i.e.,

' for a0 = 0) or y = a0 + a1x1 + a2x2 + … (for the

' "General" case where the value of a0 is not constrained),

' for up to six linear terms xi. The approach used can

' be extended readily to more than six variables.

' SUBROUTINES:

' The macro GradeBySf calls the subroutine LLSS which,

' in turn, calls the subroutines Invert, Multiply, and

' Transpose.

' INPUT:

' The input data must be arranged in a block of at least 3

' CONTIGUOUS COLUMNS, with the dependent variable y in the

' left-most column, and the independent variables xi to

' its right.

' OUTPUT:

' In order to keep the output compact, the macro only

' provides three columns of output information:

' (1) in its first column it lists the indices i of the

'     parameters xi considered;

' (2) its second column shows the values of the standard

'     deviation of the fit, Sf, assuming a0 = 0; and

' (3) its third column the values of the standard deviation

'     of the fit Sf for the general case.

' However, the output will sweep clear an area as wide as

' the input array, and extending by rMax lines below the

' bottom line of the output

' WARNING:

' Make sure that there are no valuable data below the

' input array, because these will be overwritten. This

' macro does NOT check whether that space is empty.

' PROCEDURE:

' Before calling the macro, make sure that the space below

' the input data array is clear of valuable data (see above

' under WARNING). The output requires a space as wide as

' the input array, even though that may not be obvious from

' the results displayed.

' In order to start the process, highlight the input data

' array, which should occupy at least two adjacent columns,

' then call GradeBySf.

' EXAMPLE:

' AE3 section 3.15 gives an example. 

Sub GradeBySf()

Dim C As Integer, c2 As Integer, c3 As Integer

Dim c4 As Integer, c5 As Integer, c6 As Integer

Dim cc As Integer, CMax As Integer, ccMax As Integer

Dim Down As Integer

Dim i As Long, R As Long, RMax As Long

Dim Sf As Double

Dim InputArray As Variant, OutputArray As Variant

Dim aD, hAnswer, XArray, YArray

Down = 0

' Read input array dataArray

Begin:

InputArray = Selection.Value

CMax = Selection.Columns.Count

RMax = Selection.Rows.Count

aD = Selection.Address

' If area was not highlighted

If RMax = 1 And CMax = 1 Then

  hAnswer = MsgBox("You forgot to highlight" _

    & Chr(13) & "the block of input data." _

    & Chr(13) & "Do you want to do so now?" _

    , vbYesNo, "GradeBySf")

  If hAnswer = vbNo Then End

  If hAnswer = vbYes Then

    Set myRange = Application.InputBox(Prompt:= _

      "The input data are located in:", Type:=8)

    myRange.Select

  End If

  GoTo Begin

End If

' Check the size of the input array

If CMax < 3 Then

  MsgBox "There should be at least three input columns." _

  , , "GradeBySf"

  End

End If

If CMax > 7 Then

  MsgBox "This macro can only handle" & Chr(13) & _

    "six independent parameters.", , "GradeBySf"

  End

End If

If RMax < CMax + 1 Then

  MsgBox "There are too few rows for the num-" & _

  Chr(13) & "ber of independent parameters used.", _

  , "GradeBySf"

  End

End If

' Check for missing input data

For i = 1 To RMax

  If IsEmpty(InputArray(i, 1)) Then

    MsgBox "Y-value(s) missing", , "GradeBySf"

    End

  End If

Next i

For j = 2 To CMax

  For i = 1 To RMax

    If IsEmpty(InputArray(i, j)) Then

      MsgBox "X-value(s) missing", , "GradeBySf"

      End

    End If

  Next i

Next j

' Dimension the data array and the output array

ReDim OutputArray(1 To RMax, 1 To CMax)

ReDim YArray(1 To RMax, 1 To 1) As Double

ReDim XArray(1 To RMax, 1 To 1) As Double

' Fill yArray

For R = 1 To RMax

  YArray(R, 1) = InputArray(R, 1)

Next R

' Initialize xArray

For R = 1 To RMax

  For C = 1 To CMax

    OutputArray(R, C) = ""

  Next C

Next R

Selection.Offset(RMax + 1, 0).Select

Down = Down + RMax + 1

' Write column headings

OutputArray(1, 1) = "Standard deviation of fit"

Selection.Value = OutputArray

Selection.Offset(1, 0).Select

OutputArray(1, 1) = "Indices:"

OutputArray(1, 2) = "Thru 0"

OutputArray(1, 3) = "General"

Selection.Value = OutputArray

Selection.Offset(1, 1).Select

With ActiveCell

  .Value = "GradeBySf"

  .Font.Bold = True

  .Font.ColorIndex = 5

  .Font.Italic = False

  .HorizontalAlignment = xlCenter

  .Interior.ColorIndex = 20

  .ClearComments

  .AddComment

  .Comment.Visible = False

  .Comment.Text Text:= _

    "GradeBySf version 7" & Chr(10) & _

    "Input: " & aD & Chr(10) & Chr(10) & _

    "date: " & Date & Chr(10) & "time: " & Time

End With

Selection.Offset(1, -1).Select

Down = Down + 3

' Compute the output for 1 variable:

ccMax = 4

ReDim XArray(1 To RMax, 1 To ccMax) As Double

For C = 2 To CMax

  For R = 1 To RMax

    XArray(R, 1) = 0

    XArray(R, 2) = InputArray(R, C)

  Next R

  Call LLSS(ccMax, RMax, 0, YArray, XArray, Sf)

  OutputArray(1, 1) = C - 1

  OutputArray(1, 2) = Sf

  For R = 1 To RMax

    XArray(R, 1) = 1

  Next R

  Call LLSS(ccMax, RMax, 1, YArray, XArray, Sf)

  OutputArray(1, 3) = Sf

  Selection.Value = OutputArray

  Selection.Offset(1, 0).Select

  Down = Down + 1

Next C

Selection.Offset(1, 0).Select

Down = Down + 1

' Compute the output for 2 variables:

ccMax = 5

ReDim XArray(1 To RMax, 1 To ccMax) As Double

For C = 2 To CMax

  For c2 = C + 1 To CMax

    For R = 1 To RMax

      XArray(R, 1) = 0

      XArray(R, 2) = InputArray(R, C)

      XArray(R, 3) = InputArray(R, c2)

    Next R

    Call LLSS(ccMax, RMax, 0, YArray, XArray, Sf)

    OutputArray(1, 1) = C - 1 & "," & c2 - 1

    OutputArray(1, 2) = Sf

    For R = 1 To RMax

      XArray(R, 1) = 1

    Next R

    Call LLSS(ccMax, RMax, 1, YArray, XArray, Sf)

    OutputArray(1, 3) = Sf

    Selection.Value = OutputArray

    Selection.Offset(1, 0).Select

    Down = Down + 1

  Next c2

Next C

Selection.Offset(1, 0).Select

Down = Down + 1

' Compute the output for 3 variables:

ccMax = 6

ReDim XArray(1 To RMax, 1 To ccMax) As Double

For C = 2 To CMax

 For c2 = C + 1 To CMax

  For c3 = c2 + 1 To CMax

   For R = 1 To RMax

     XArray(R, 1) = 0

     XArray(R, 2) = InputArray(R, C)

     XArray(R, 3) = InputArray(R, c2)

     XArray(R, 4) = InputArray(R, c3)

   Next R

   Call LLSS(ccMax, RMax, 0, YArray, XArray, Sf)

   OutputArray(1, 1) = C - 1 & "," & c2 - 1 & "," & c3 - 1

   OutputArray(1, 2) = Sf

   For R = 1 To RMax

     XArray(R, 1) = 1

   Next R

   Call LLSS(ccMax, RMax, 1, YArray, XArray, Sf)

   OutputArray(1, 3) = Sf

   Selection.Value = OutputArray

   Selection.Offset(1, 0).Select

   Down = Down + 1

  Next c3

 Next c2

Next C

Selection.Offset(1, 0).Select

Down = Down + 1

' Compute the output for 4 variables:

ccMax = 7

ReDim XArray(1 To RMax, 1 To ccMax) As Double

For C = 2 To CMax

 For c2 = C + 1 To CMax

  For c3 = c2 + 1 To CMax

   For c4 = c3 + 1 To CMax

    For R = 1 To RMax

      XArray(R, 1) = 0

      XArray(R, 2) = InputArray(R, C)

      XArray(R, 3) = InputArray(R, c2)

      XArray(R, 4) = InputArray(R, c3)

      XArray(R, 5) = InputArray(R, c4)

    Next R

    Call LLSS(ccMax, RMax, 0, YArray, XArray, Sf)

    OutputArray(1, 1) = C - 1 & "," & c2 - 1 & _

      "," & c3 - 1 & "," & c4 - 1

    OutputArray(1, 2) = Sf

    For R = 1 To RMax

      XArray(R, 1) = 1

    Next R

    Call LLSS(ccMax, RMax, 1, YArray, XArray, Sf)

    OutputArray(1, 3) = Sf

    Selection.Value = OutputArray

    Selection.Offset(1, 0).Select

    Down = Down + 1

   Next c4

  Next c3

 Next c2

Next C

Selection.Offset(1, 0).Select

Down = Down + 1

' Compute the output for 5 variables:

ccMax = 8

ReDim XArray(1 To RMax, 1 To ccMax) As Double

For C = 2 To CMax

 For c2 = C + 1 To CMax

  For c3 = c2 + 1 To CMax

   For c4 = c3 + 1 To CMax

    For c5 = c4 + 1 To CMax

      For R = 1 To RMax

        XArray(R, 1) = 0

        XArray(R, 2) = InputArray(R, C)

        XArray(R, 3) = InputArray(R, c2)

        XArray(R, 4) = InputArray(R, c3)

        XArray(R, 5) = InputArray(R, c4)

        XArray(R, 6) = InputArray(R, c5)

      Next R

      Call LLSS(ccMax, RMax, 0, YArray, XArray, Sf)

      OutputArray(1, 1) = C - 1 & "," & c2 - 1 & _

        "," & c3 - 1 & "," & c4 - 1 & "," & c5 - 1

      OutputArray(1, 2) = Sf

      For R = 1 To RMax

        XArray(R, 1) = 1

      Next R

      Call LLSS(ccMax, RMax, 1, YArray, XArray, Sf)

      OutputArray(1, 3) = Sf

      Selection.Value = OutputArray

      Selection.Offset(1, 0).Select

      Down = Down + 1

    Next c5

   Next c4

  Next c3

 Next c2

Next C

Selection.Offset(1, 0).Select

Down = Down + 1

' Compute the output for 6 variables:

ccMax = 9

ReDim XArray(1 To RMax, 1 To ccMax) As Double

For C = 2 To CMax

 For c2 = C + 1 To CMax

  For c3 = c2 + 1 To CMax

   For c4 = c3 + 1 To CMax

    For c5 = c4 + 1 To CMax

     For c6 = c5 + 1 To CMax

      For R = 1 To RMax

        XArray(R, 1) = 0

        XArray(R, C) = InputArray(R, 2)

        XArray(R, c2) = InputArray(R, 3)

        XArray(R, c3) = InputArray(R, 4)

        XArray(R, c4) = InputArray(R, 5)

        XArray(R, c5) = InputArray(R, 6)

        XArray(R, c6) = InputArray(R, 7)

      Next R

      Call LLSS(ccMax, RMax, 0, YArray, XArray, Sf)

      OutputArray(1, 1) = C - 1 & "," & _

        "," & c2 - 1 & "," & c3 - 1 & "," & _

        c4 - 1 & c5 - 1 & "," & c6 - 1

      OutputArray(1, 2) = Sf

      For R = 1 To RMax

        XArray(R, 1) = 1

      Next R

      Call LLSS(ccMax, RMax, 1, YArray, XArray, Sf)

      OutputArray(1, 3) = Sf

      Selection.Value = OutputArray

      Selection.Offset(1, 0).Select

      Down = Down + 1

     Next c6

    Next c5

   Next c4

  Next c3

 Next c2

Next C

Selection.Offset(-Down, 0).Select

End Sub

'''''''''''''''''''''''''''''''''''''''''''''''''''''''''''

'''''''''''''''''''''''''''''''''''''''''''''''''''''''''''

''''''''''''''''*****************************''''''''''''''

''''''''''''''''*                           *''''''''''''''

''''''''''''''''*    Least squares poly     *''''''''''''''

''''''''''''''''*                           *''''''''''''''

''''''''''''''''*****************************''''''''''''''

''''''''''''''''''''''''''''''''''''''''''' (c) R. de Levie

'''''''''''''''''''''''''''''''''''''''''''  v 12, May 2012

' PURPOSE:

' The macros LSPoly0 and LSPoly1 compute the parameters and

' their standard deviations for an unweighted least squares

' fit of a dependent variable y to a power series in x of

' increasing (positive integer) order m, up to 14. The user

' specifies the highest order to be displayed.

'

' Use the custom macros Ortho1 or Ortho0 in order to decide

' what polynomial order might be optimal.

' SUBROUTINES:

' The macros LSPolyO and LSPolyI call the subroutine

' LeastSquaresPoly which, in turn, calls the subroutines

' Invert, Multiply, and Transpose.

' INPUT:

' The input data must be arranged in a block of two adja-

' cent columns, with the dependent variable y in the left-

' most column, and the independent variable x to its right.

' Highlight the data in these two columns, and call the

' macro LSPolyO or LSPolyI. Then specify the maximum poly-

' nomial order for which you want the results displayed.

' OUTPUT:

' The macro provides:

' (1) the coefficients of the least squares fit of y to the

'     power series a(1)x + a(2)x^2 + ... = Sum[a(i)x^i] for

'     i = 1(1)m for LSPolyO, or to a(0) + a(1)x + a(2)x^2

'     + ... = Sum[a(i)x^i] for i = 0(1)m for LSPoly1;

' (2) the corresponding standard deviations s

'     for these coefficients;

' (3) the standard deviation of the fit

'     for each value of m;

' (4) the F-ratios for alpha = 0.05 and 0.01

'     for each value of m.

'

' For each order the output consists of three rows. The top

' row lists the order of the polynomial, and the standard

' deviation of the fit. The second row displays the coeffi-

' cients of the fitted polynomial, and the third row the

' corresponding standard deviations. The second and third

' rows also list the values for the F-ratios for 5% and 1%

' respectively. For easy reading the rows are color-coded

' by order.

' If you want the output to display more digits, change

' the instruction ActiveCell.NumberFormat = "0.00E+00" to

' ActiveCell.NumberFormat = "0.000E+00", "0.0000E+00", etc.

' WARNINGS:

' The output is displayed below the highlighted input

' array, and therefore might escape your notice. If

' the macro doesn't seem to respond, look below it.

'

' If you find cells filled with ###### the macro needs

' wider columns to display its results. Click on the

' column name(s) at the top of the column(s), right-click,

' and insert a larger number in Column Width.

' The size of the macro output will be determined by your

' choice of the maximum polynomial order. The output will

' be displayed under the input data, and will gradually

' expand to the right as results for higher orders occupy

' increasingly more space. The first-order output will

' occupy a block of 3 rows of 5 columns for LSPoly0 (6

' for LSPoly1), and each additional order will use a

' block that is one column wider than its predecessor.

' Please move valuable data from the output region before

' running this macro, lest those data be overwritten.

' PROCEDURE:

' Before calling the macro, make sure that the output area

' is clear of valuable data (see above under WARNING)

'

' In order to start the process, highlight the (two columns

' wide, and at least 3 rows high) input data block, then

' call LSPoly0 or LSPoly1.

' EXAMPLE:

' Use of this macro is illustrated in sections 3.10, 3.13 

' and 3.14 of AE3.

'''''''''''''''''''''''''''''''''''''''''''''''''''''''''''

' The function of the following two drivers is merely to

' set the value of one parameter, p, equal to either one or

' zero, in order to choose between a general least squares

' fitting (p = 1) or one that forces the curve through the

' origin (p = 0).

Sub LSPolyO()           ' for unweighted least squares

                        ' fit through the origin

Dim P As Double

P = 0

Call LeastSquaresPoly(P)

End Sub

'''''''''''''''''''''''''''''''''''''''''''''''''''''''''''

Sub LSPolyI()           ' for general unweighted

                        ' least squares fit

Dim P As Double

P = 1

Call LeastSquaresPoly(P)

End Sub

'''''''''''''''''''''''''''''''''''''''''''''''''''''''''''

Sub LeastSquaresPoly(P)

Dim CMax As Integer, Color(1 To 7) As Integer

Dim j As Integer, m As Integer, N As Integer

Dim MaxOrder As Integer, MaxOrderMax As Integer

Dim i As Long, RMax As Long

Dim FRatio1 As Double, FRatio5 As Double

Dim Root As Double, StDevF As Double

Dim u As Double, Z As Double

Dim DataArray As Variant, F As Variant, lccArray As Variant

Dim OutputArray As Variant, SSR As Variant

Dim varY As Variant, vArray As Variant, v0Array As Variant

Dim myRange As Range

Dim bArray, btArray, btqArray, pArray, piArray

Dim qArray, XArray, xtArray, YArray, ytArray, ytyArray

Dim Answer, iAnswer, hAnswer, kAnswer

' Determination of the array size:

Begin:

RMax = Selection.Rows.Count

CMax = Selection.Columns.Count

ReDim F(1 To RMax, 1 To 1) As Double

ReDim SSR(1 To RMax, 1 To 1) As Double

u = 1

Z = 0

' If area was not highlighted

If RMax = 1 And CMax = 1 Then

  hAnswer = MsgBox("You forgot to highlight" _

    & Chr(13) & "the block of input data." _

    & Chr(13) & "Do you want to do so now?" _

    , vbYesNo, "LSPoly")

  If hAnswer = vbNo Then End

  If hAnswer = vbYes Then

    Set myRange = Application.InputBox(Prompt:= _

      "The input data are located in:", Type:=8)

    myRange.Select

  End If

  GoTo Begin

End If

' Check that the number of columns is 2:

If CMax <> 2 Then

  MsgBox "There must be two columns," & _

  Chr(13) & "one for Y, and one for X.", _

    , "Least Squares Polynomial Fit"

  End

End If

' Check that rmax > 3, so that the number of data

' points is sufficient to define the problem:

If RMax < 4 Then

  MsgBox "There must be at least 4 input" & _

  Chr(13) & "data pairs to define the problem.", _

    , "Least Squares Polynomial Fit"

  End

End If

MaxOrderMax = 14

If RMax < 16 Then MaxOrderMax = RMax - 2

DataArray = Selection.Value

Selection.Item(1, 1).Select

' Select the maximum order:

MaxOrder = InputBox _

  ("Select as maximum order an integer" & Chr(13) & _

  "between 3 and " & MaxOrderMax & "." & Chr(13) & _

  Chr(13) & "Enter the maximum order: ", "MaxOrder")

If MaxOrder < 3 Then MaxOrder = 3

If MaxOrder > MaxOrderMax Then MaxOrder = MaxOrderMax

RedimensionArrays:

' Dimension the arrays:

ReDim YArray(1 To RMax, 1 To 1) As Double

ReDim XArray(1 To RMax, 1 To CMax) As Double

ReDim ytArray(1 To 1, 1 To RMax) As Double

ReDim ytyArray(1 To 1, 1 To 1) As Double

ReDim xtArray(1 To CMax, 1 To RMax) As Double

ReDim pArray(1 To CMax, 1 To CMax) As Double

ReDim piArray(1 To CMax, 1 To CMax) As Double

ReDim qArray(1 To CMax, 1 To 1) As Double

ReDim bArray(1 To CMax, 1 To 1) As Double

ReDim btArray(1 To 1, 1 To CMax) As Double

ReDim btqArray(1 To 1, 1 To 1) As Double

ReDim vArray(1 To CMax, 1 To CMax) As Double

ReDim v0Array(1 To CMax - 1 + P, 1 To CMax - 1 + P) _

  As Double

ReDim lccArray(1 To CMax, 1 To CMax) As Double

' Read the dataArray, then fill yArray and xArray.

For i = 1 To RMax

  YArray(i, 1) = DataArray(i, 1)

Next i

If CMax = 2 Then

  For i = 1 To RMax

    If IsEmpty(DataArray(i, 1)) Then

      MsgBox "Y-value(s) missing", _

        , "Least Squares Polynomial Fit"

      End

    End If

  Next i

  For i = 1 To RMax

    If IsEmpty(DataArray(i, 2)) Then

      MsgBox "X-value(s) missing", _

        , "Least Squares Polynomial Fit"

      End

    End If

  Next i

End If

' Fill the first column of xArray with zeroes

' (for p = 0) or ones (for p = 1), the rest

' with the data in the x-column(s)

For i = 1 To RMax

  XArray(i, 1) = CDbl(P)

Next i

For i = 1 To RMax

  XArray(i, 2) = DataArray(i, 2)

Next i

If CMax > 2 Then

  For j = 3 To CMax

    For i = 1 To RMax

      XArray(i, j) = XArray(i, 2) * XArray(i, j - 1)

    Next i

  Next j

End If

' Compute b = (X' X)" X' Y , where ' or t denote

' transposition, and " or i indicate inversion

' The various arrays and their dimensions (rows,

' columns) are:

'   Y          = yArray        (  rmax,     1)

'   X          = xArray        (  rmax,  cmax)

'   X'         = xtArray       (  cmax,  rmax)

'   X' X       = pArray        (  cmax,  cmax)

'  (X' X)"     = piArray       (  cmax,  cmax)

'   X' Y       = qArray        (  cmax,     1)

'   b          = bArray        (  cmax,     1)

'   Y'         = ytArray       (     1,  rmax)

'   Y' Y       = ytyArray      (     1,     1)

'   b'         = btArray       (     1,  cmax)

'   b' X' Y    = btqArray      (     1,     1)

Call Transpose(XArray, RMax, CMax, xtArray)

Call Multiply(xtArray, CMax, RMax, XArray, CMax, pArray)

Call Invert(pArray, CMax, piArray)

Call Multiply(xtArray, CMax, RMax, YArray, 1, qArray)

Call Multiply(piArray, CMax, CMax, qArray, 1, bArray)

Call Transpose(YArray, RMax, 1, ytArray)

Call Transpose(bArray, CMax, 1, btArray)

Call Multiply(ytArray, 1, RMax, YArray, 1, ytyArray)

Call Multiply(btArray, 1, CMax, qArray, 1, btqArray)

' Calculate SSR = Y'Y - b'X'Y, and then varY, the variance

' of y, as varY = SSR/(rmax-cmax-p+1); and vArray, the co-

' variance matrix, as V = (X'X)" times varY, of which we

' here only use the diagonal elements, i.e., the variances.

SSR(CMax, 1) = ytyArray(1, 1) - btqArray(1, 1)

varY = SSR(CMax, 1) / (RMax - CMax + 1 - P)

StDevF = Sqr(Abs(varY))

If CMax > 2 Then

  F(CMax, 1) = ((SSR(CMax - 1, 1) / SSR(CMax, 1)) - 1) _

    * (RMax - CMax + 1 - P)

  FRatio5 = (F(CMax, 1)) / _

    (Application.FInv(0.05, 1, RMax - CMax + 1 - P))

  FRatio1 = (F(CMax, 1)) / _

    (Application.FInv(0.01, 1, RMax - CMax + 1 - P))

End If

For i = 1 To CMax

  For j = 1 To CMax

    vArray(i, j) = varY * piArray(i, j)

  Next j

Next i

Application.ScreenUpdating = False

ActiveCell.Offset(RMax + 1, 0).Select

' Paint color bands for up to 14 orders, and set the

' numerical format to scientific with 3 decimal places

Color(1) = 38

Color(2) = 40

Color(3) = 36

Color(4) = 35

Color(5) = 34

Color(6) = 37

Color(7) = 39

For i = 1 To 3

  For j = 1 To CMax + P + 2

    ActiveCell.ClearContents

    If CMax < 9 Then

      ActiveCell.Interior.ColorIndex = Color(CMax - 1)

    Else

      ActiveCell.Interior.ColorIndex = Color(CMax - 8)

    End If

    ActiveCell.Offset(0, 1).Select

    If CMax < 9 Then

      ActiveCell.Interior.ColorIndex = Color(CMax - 1)

    Else

      ActiveCell.Interior.ColorIndex = Color(CMax - 8)

    End If

    ActiveCell.Font.Bold = False

    ActiveCell.Font.Italic = False

    ActiveCell.NumberFormat = "0.00E+00"

  Next j

  ActiveCell.Offset(1, -CMax - P - 2).Select

Next i

ActiveCell.Offset(-3, 0).Select

' Display the top line of the output

With ActiveCell

  .Font.Bold = True

  .Font.Italic = False

  .HorizontalAlignment = xlLeft

  .Value = "Order " & CMax - 1

End With

For j = 1 To CMax + P - 1

  Selection.Offset(0, 1).Select

  With ActiveCell

    .Font.Bold = False

    .Font.Italic = True

    .HorizontalAlignment = xlCenter

    .Value = "term " & j - P

  End With

Next j

Selection.Offset(0, 2).Select

With ActiveCell

  .Font.Bold = False

  .Font.Italic = True

  .HorizontalAlignment = xlRight

  .Value = "Sf:"

End With

Selection.Offset(0, 1).Select

Selection.HorizontalAlignment = xlLeft

Selection.Value = StDevF

Selection.Offset(1, -CMax - P - 2).Select

' Display the center line of the output

With ActiveCell

  .Font.Bold = False

  .Font.Italic = True

  .HorizontalAlignment = xlRight

  .Value = "Coeff:"

End With

For j = 1 To CMax + P - 1

  Selection.Offset(0, 1).Select

  With ActiveCell

    .Font.Bold = False

    .Font.Italic = False

    .HorizontalAlignment = xlCenter

    .Value = bArray(j + 1 - P, 1)

  End With

Next j

Selection.Offset(0, 2).Select

With ActiveCell

  .Font.Bold = False

  .Font.Italic = True

  .HorizontalAlignment = xlRight

  .Value = "FR5:"

End With

Selection.Offset(0, 1).Select

Selection.HorizontalAlignment = xlLeft

If CMax = 2 Then Selection.Value = "N/A"

If CMax > 2 Then Selection.Value = FRatio5

Selection.Offset(1, -CMax - P - 2).Select

' Display the bottom line of the output

With ActiveCell

  .Font.Bold = False

  .Font.Italic = True

  .HorizontalAlignment = xlRight

  .Value = "StDev:"

End With

For j = 1 To CMax + P - 1

  Selection.Offset(0, 1).Select

  With ActiveCell

    .Font.Bold = False

    .Font.Italic = False

    .HorizontalAlignment = xlCenter

  End With

  If vArray(j - P + 1, j - P + 1) < 1E-40 Then

    ActiveCell.Value = "<1E-20"

  Else

    ActiveCell.Value = _

      Sqr(Abs(vArray(j - P + 1, j - P + 1)))

  End If

Next j

Selection.Offset(0, 2).Select

With ActiveCell

  .Font.Bold = False

  .Font.Italic = True

  .HorizontalAlignment = xlRight

  .Value = "FR1:"

End With

Selection.Offset(0, 1).Select

Selection.HorizontalAlignment = xlLeft

If CMax = 2 Then Selection.Value = "N/A"

If CMax > 2 Then Selection.Value = FRatio1

Selection.Offset(1, -CMax - P - 2).Select

If CMax < MaxOrder + 2 Then CMax = CMax + 1

ActiveCell.Offset(-RMax - 1, 0).Select

If CMax = MaxOrder + 2 Then End

GoTo RedimensionArrays

End Sub

'''''''''''''''''''''''''''''''''''''''''''''''''''''''''''

'''''''''''''''''''''''''''''''''''''''''''''''''''''''''''

''''''''''''''''****************************'''''''''''''''

''''''''''''''''*                           *'''''''''''''''

''''''''''''''''*    Least squares multi    *''''''''''''''

''''''''''''''''*                           *'''''''''''''''

''''''''''''''''****************************'''''''''''''''

''''''''''''''''''''''''''''''''''''''''''' (c) R. de Levie

'''''''''''''''''''''''''''''''''''''''''''  v 12, May 2012

' PURPOSE:

' The macros LSMultiO and LSMultiI compute the parameters

' and their standard deviations for an unweighted least

' squares fit of a dependent variable y to a user-specified

' multivariate expression of the form y = a1x1 + a2x2 + ..

' (for LSMultiO) or y = a0 + a1x1 + a2x2 + ...(for

' LSMultiI) by gradually (one term at a time) increasing

' the number of terms included in the least squares

' analysis from left to right

' Note that the macro does NOT rearrange the columns, and

' therefore does not provide all possible permutations.

' If those possible permutations are needed, instead use

' the macros LSPermute0 or LSPermute1.

' The macro yields: the fitting coefficients a(i), their

' standard deviations s(i), the standard deviation of the

' over-all fit, s(y), and the results of two F-tests, FR5

' and FR1, for 5% and 1% probability respectively.

' SUBROUTINES:

' The macros LSMulti0 and LSMulti1 call the subroutine

' LSMulti which, in turn, calls the subroutines Invert,

' Multiply, and Transpose.

' INPUT:

' The input data must be arranged in a block of at least 3

' (for LSMulti0) or 4 (for LSMulti1) adjacent columns, with

' the dependent variable y in the left-most column, and the

' independent variables x to its right. Highlight the data

' in these columns, and call the macro LSMulti0 or LSMulti1

' OUTPUT:

' The macro provides:

' (1) the coefficients of the least squares fit of y to

'     the series a(1)x(1) + a(2)x^2 + ... + a(i)x^i

'     = Sum[a(i)x(i)] for i = 1(1)m for LSMulti0, or

'     a(0) + a(1)x(1) + a(1)x(2) + ... = Sum[a(i)x(i)]

'     for i = 0(1)m for LSMulti1;

' (2) the corresponding standard deviations s for these

'     coefficients;

' (3) the corresponding standard deviation of the fit; and

' (4) the corresponding F-ratios for alpha = 0.05 and 0.01.

'

' For each order the output consists of three rows. The top

' row lists the standard deviation of the fit. The second

' row displays the coefficients of the fitted polynomial,

' and the third row the corresponding standard deviations.

' The second and third rows also list the values for the

' F-ratios for 5% and 1% respectively. For easy reading

' the rows are color-coded.

' If you want the output to display more digits, change

' the instruction ActiveCell.NumberFormat = "0.00E+00" to

' ActiveCell.NumberFormat = "0.000E+00", "0.0000E+00", etc.

' WARNING:

' The macro output will take up much space below and to the

' right of the input data. Please move valuable data from

' the output region before running this macro, lest those

' data be overwritten.

'

' PROCEDURE:

'

' Before calling the macro, make sure that the output area

' is clear of valuable data (see above under WARNING). The

' macro does NOT check whether that area is free of data.

' In order to start the process, highlight the input data,

' which should occupy at least three adjacent columns, then

' call LSMulti0 or LSMulti1.

'''''''''''''''''''''''''''''''''''''''''''''''''''''''''''

' The function of the following two drivers is merely to

' set the value of one parameter, p, equal to either one or

' zero, in order to choose between a general least squares

' fitting (p = 1) or one that forces the curve through the

' origin (p = 0).

Sub LSMultiO()      ' for unweighted least squares

                    ' fit through the origin

Dim P As Double

P = 0

Call LSMulti(P)

End Sub

'''''''''''''''''''''''''''''''''''''''''''''''''''''''''''

Sub LSMultiI()      ' for general unweighted

                    ' least squares fit

Dim P As Double

P = 1

Call LSMulti(P)

End Sub

'''''''''''''''''''''''''''''''''''''''''''''''''''''''''''

Sub LSMulti(P)

Dim CMax As Integer, Color(1 To 7) As Integer

Dim j As Integer, m As Integer, MM As Integer

Dim MaxOrder As Integer, N As Integer

Dim i As Long, RMax As Long

Dim FRatio1 As Double, FRatio5 As Double

Dim Root As Double, StDevF As Double

Dim u As Double, Z As Double

Dim DataArray As Variant, F As Variant

Dim lccArray As Variant, OutputArray As Variant

Dim SSR As Variant, varY As Variant

Dim vArray As Variant, v0Array As Variant

Dim myRange As Range

Dim bArray, btArray, btqArray, M1, M2, pArray, piArray

Dim qArray, XArray, xtArray, YArray, ytArray, ytyArray

Dim Answer, iAnswer, hAnswer, kAnswer

' Determination of the array size:

Begin:

RMax = Selection.Rows.Count

CMax = Selection.Columns.Count

ReDim F(1 To RMax, 1 To 1) As Double

ReDim SSR(1 To RMax, 1 To 1) As Double

u = 1

Z = 0

' If area was not highlighted

If RMax = 1 And CMax = 1 Then

  hAnswer = MsgBox("You forgot to highlight" _

    & Chr(13) & "the block of input data." _

    & Chr(13) & "Do you want to do so now?" _

    , vbYesNo, "LSMulti")

  If hAnswer = vbNo Then End

  If hAnswer = vbYes Then

    Set myRange = Application.InputBox(Prompt:= _

      "The input data are located in:", Type:=8)

    myRange.Select

  End If

  GoTo Begin

End If

' Check that the number of columns is at least (3+p):

If CMax < 3 + P Then

  MsgBox "There must be at least " & 3 + P & " columns.", _

    , "Least Squares Multivariate Fit"

  End

End If

' Check that rmax > cmax+1, so that the number of data

' points is sufficient to define the problem:

If RMax < CMax + 1 Then

  MsgBox "     There must be at least " & CMax + 1 & _

    Chr(13) & "input data pairs to define the problem.", _

    , "Least Squares Multivariate Fit"

  End

End If

DataArray = Selection.Value

Selection.Item(1, 1).Select

MM = 2

RedimensionArrays:

' Dimension the arrays:

ReDim YArray(1 To RMax, 1 To 1) As Double

ReDim XArray(1 To RMax, 1 To MM) As Double

ReDim ytArray(1 To 1, 1 To RMax) As Double

ReDim ytyArray(1 To 1, 1 To 1) As Double

ReDim xtArray(1 To MM, 1 To RMax) As Double

ReDim pArray(1 To MM, 1 To MM) As Double

ReDim piArray(1 To MM, 1 To CMax) As Double

ReDim qArray(1 To MM, 1 To 1) As Double

ReDim bArray(1 To MM, 1 To 1) As Double

ReDim btArray(1 To 1, 1 To MM) As Double

ReDim btqArray(1 To 1, 1 To 1) As Double

ReDim vArray(1 To MM, 1 To MM) As Double

ReDim v0Array(1 To MM - 1 + P, 1 To MM - 1 + P) As Double

ReDim lccArray(1 To MM, 1 To MM) As Double

' Read the dataArray, then fill yArray and xArray.

For i = 1 To RMax

  YArray(i, 1) = DataArray(i, 1)

Next i

' Check the input data for contiguity.

If MM = 2 Then

  For i = 1 To RMax

    If IsEmpty(DataArray(i, 1)) Then

      MsgBox "Y-value(s) missing", _

        , "Least Squares Multivariate Fit"

      End

    End If

  Next i

  For i = 1 To RMax

    For j = 2 To CMax

      If IsEmpty(DataArray(i, j)) Then

        MsgBox "X-value(s) missing", _

          , "Least Squares Multivariate Fit"

        End

      End If

    Next j

  Next i

End If

' Fill the first column of xArray with either

' zeros (for p = 0) or ones (for p = 1),

' the rest with the data in the x-column(s)

For i = 1 To RMax

  XArray(i, 1) = CDbl(P)

Next i

For i = 1 To RMax

  XArray(i, 2) = DataArray(i, 2)

Next i

If MM > 2 Then

  For i = 1 To RMax

    For j = 3 To MM

      XArray(i, j) = DataArray(i, j)

    Next j

  Next i

End If

' Compute b = (X' X)" X' Y , where ' or t denote

' transposition, and " or i indicate inversion

' The various arrays and their dimensions (rows,

' columns) are:

'   Y          = yArray        (  rmax,     1)

'   X          = xArray        (  rmax,    mm)

'   X'         = xtArray       (  mm,    rmax)

'   X' X       = pArray        (  mm,      mm)

'  (X' X)"     = piArray       (  mm,      mm)

'   X' Y       = qArray        (  mm,       1)

'   b          = bArray        (  mm,       1)

'   Y'         = ytArray       (     1,  rmax)

'   Y' Y       = ytyArray      (     1,     1)

'   b'         = btArray       (     1,    mm)

'   b' X' Y    = btqArray      (     1,     1)

Call Transpose(XArray, RMax, MM, xtArray)

Call Multiply(xtArray, MM, RMax, XArray, MM, pArray)

Call Invert(pArray, MM, piArray)

Call Multiply(xtArray, MM, RMax, YArray, 1, qArray)

Call Multiply(piArray, MM, MM, qArray, 1, bArray)

Call Transpose(YArray, RMax, 1, ytArray)

Call Transpose(bArray, MM, 1, btArray)

Call Multiply(ytArray, 1, RMax, YArray, 1, ytyArray)

Call Multiply(btArray, 1, MM, qArray, 1, btqArray)

' Calculate SSR = Y'Y - b'X'Y, and then varY, the variance

' of y, as varY = SSR/(rmax-cmax-p+1); and vArray, the

' covariance matrix, as V = (X'X)" times varY, of which we

' here only use the diagonal elements, i.e., the variances

For i = 1 To MM

  For j = 1 To MM

    vArray(i, j) = varY * piArray(i, j)

  Next j

Next i

Application.ScreenUpdating = False

ActiveCell.Offset(RMax + 1, 0).Select

' Paint color bands for up to 14 orders, and set the

' numerical format to scientific with 3 decimal places

Color(1) = 38

Color(2) = 40

Color(3) = 36

Color(4) = 35

Color(5) = 34

Color(6) = 37

Color(7) = 39

For i = 1 To 3

  For j = 1 To MM + P + 2

    ActiveCell.ClearContents

    If MM < 9 Then

      ActiveCell.Interior.ColorIndex = Color(MM - 1)

    Else

      ActiveCell.Interior.ColorIndex = Color(MM - 8)

    End If

    ActiveCell.Offset(0, 1).Select

    If MM < 9 Then

      ActiveCell.Interior.ColorIndex = Color(MM - 1)

    Else

      ActiveCell.Interior.ColorIndex = Color(MM - 8)

    End If

    ActiveCell.Font.Bold = False

    ActiveCell.Font.Italic = False

    ActiveCell.NumberFormat = "0.00E+00"

  Next j

  ActiveCell.Offset(1, -MM - P - 2).Select

Next i

ActiveCell.Offset(-3, 0).Select

' Display the top line of the output

With ActiveCell

  .Font.Bold = True

  .Font.Italic = False

  .HorizontalAlignment = xlLeft

  .Value = "Set # " & MM - 1

End With

For j = 1 To MM + P - 1

  Selection.Offset(0, 1).Select

  With ActiveCell

    .Font.Bold = False

    .Font.Italic = True

    .HorizontalAlignment = xlCenter

    .Value = "term " & j - P

  End With

Next j

Selection.Offset(0, 2).Select

With ActiveCell

  .Font.Bold = False

  .Font.Italic = True

  .HorizontalAlignment = xlRight

  .Value = "Sf:"

End With

Selection.Offset(0, 1).Select

Selection.HorizontalAlignment = xlLeft

Selection.Value = StDevF

Selection.Offset(1, -MM - P - 2).Select

' Display the center line of the output

With ActiveCell

  .Font.Bold = False

  .Font.Italic = True

  .HorizontalAlignment = xlRight

  .Value = "Coeff:"

End With

For j = 1 To MM + P - 1

  Selection.Offset(0, 1).Select

  With ActiveCell

    .Font.Bold = False

    .Font.Italic = False

    .HorizontalAlignment = xlCenter

    .Value = bArray(j + 1 - P, 1)

  End With

Next j

Selection.Offset(0, 2).Select

With ActiveCell

  .Font.Bold = False

  .Font.Italic = True

  .HorizontalAlignment = xlRight

  .Value = "FR5:"

End With

Selection.Offset(0, 1).Select

Selection.HorizontalAlignment = xlLeft

If MM = 2 Then Selection.Value = "N/A"

If MM > 2 Then Selection.Value = FRatio5

Selection.Offset(1, -MM - P - 2).Select

' Display the bottom line of the output

With ActiveCell

  .Font.Bold = False

  .Font.Italic = True

  .HorizontalAlignment = xlRight

  .Value = "StDev:"

End With

For j = 1 To MM + P - 1

  Selection.Offset(0, 1).Select

  With ActiveCell

    .Font.Bold = False

    .Font.Italic = False

    .HorizontalAlignment = xlCenter

  End With

  If vArray(j - P + 1, j - P + 1) < 1E-40 Then

    ActiveCell.Value = "<1E-20"

  Else

    ActiveCell.Value = _

      Sqr(Abs(vArray(j - P + 1, j - P + 1)))

  End If

Next j

Selection.Offset(0, 2).Select

With ActiveCell

  .Font.Bold = False

  .Font.Italic = True

  .HorizontalAlignment = xlRight

  .Value = "FR1:"

End With

Selection.Offset(0, 1).Select

Selection.HorizontalAlignment = xlLeft

If MM = 2 Then Selection.Value = "N/A"

If MM > 2 Then Selection.Value = FRatio1

Selection.Offset(1, -MM - P - 2).Select

If MM < CMax + 1 Then MM = MM + 1

ActiveCell.Offset(-RMax - 1, 0).Select

If MM = CMax + 1 Then End

GoTo RedimensionArrays

End Sub

'''''''''''''''''''''''''''''''''''''''''''''''''''''''''''

''''''''''''''*******************************''''''''''''''

''''''''''''''*                             *''''''''''''''

''''''''''''''*    Least squares permute    *''''''''''''''

''''''''''''''*                             *''''''''''''''

''''''''''''''*******************************''''''''''''''

''''''''''''''''''''''''''''''''''''''''''' (c) R. de Levie

'''''''''''''''''''''''''''''''''''''''''''  v 12, May 2012

' PURPOSE:

' The macro LSPermute computes the standard deviations of

' the fit, sf, for all possible combinations and permuta-

' tions of unweighted least squares fits of a dependent

' variable y to a user-specified multivariate expression of

' the form y = a1x1 + a2x2 + .. (for "Thru 0", i.e., for a0

' = 0) or y = a0 + a1x1 + a2x2 + .. (for the "General" case

' where the value of a0 is not constrained), for up to six

' linear terms xi.

' SUBROUTINES:

' The macro LSPermute calls the subroutine LLSS which,

' in turn, calls the subroutines Invert, Multiply, and

' Transpose.

' INPUT:

' The input data must be arranged in a block of at least 3

' contiguous columns, with the dependent variable y in the

' left-most column, and the independent variables xi to

' its right.

' OUTPUT:

' In order to keep the output compact, the macro only

' provides three columns of output information:

' (1) in its first column the indices i of the parameters

'     xi considered;

' (2) in its second column the values of the standard

'     deviation of the fit, Sf, assuming a0 = 0; and

' (3) in its third column the values of the standard

'     deviation of the fit Sf for the general case.

' However, the output will sweep clear an area as wide as

' the input array, and extending by rMax lines below the

' bottom line of the output

' WARNING:

' Make sure that there are no valuable data below the

' input array, because these will be overwritten. This

' macro does NOT check whether that space is empty.

' PROCEDURE:

' Before calling the macro, make sure that the space below

' the input data array is clear of valuable data (see above

' under WARNING). The output requires a space as wide as

' the input array, even though that may not be obvious from

' the results displayed.

' In order to start the process, highlight the input data

' array, which should occupy at least two adjacent columns,

' then call LSPermute.

Sub LSPermute()

Dim C As Integer, c2 As Integer, c3 As Integer

Dim c4 As Integer, c5 As Integer, c6 As Integer

Dim cc As Integer, CMax As Integer, ccMax As Integer

Dim Down As Integer

Dim i As Long, R As Long, RMax As Long

Dim StDevF As Double

Dim InputArray As Variant, OutputArray As Variant

Dim hAnswer, XArray, YArray

Down = 0

' Read input array dataArray

Begin:

InputArray = Selection.Value

CMax = Selection.Columns.Count

RMax = Selection.Rows.Count

' If area was not highlighted

If RMax = 1 And CMax = 1 Then

  hAnswer = MsgBox("You forgot to highlight" _

    & Chr(13) & "the block of input data." _

    & Chr(13) & "Do you want to do so now?" _

    , vbYesNo, "LSPermute")

  If hAnswer = vbNo Then End

  If hAnswer = vbYes Then

    Set myRange = Application.InputBox(Prompt:= _

      "The input data are located in:", Type:=8)

    myRange.Select

  End If

  GoTo Begin

End If

' Check the size of the input array

If CMax < 3 Then

  MsgBox "There should be at least three input columns." _

  , , "LSPermute"

  End

End If

If CMax > 7 Then

  MsgBox "This macro can only handle" & Chr(13) & _

    "six independent parameters.", , "LSPermute"

  End

End If

If RMax < CMax + 1 Then

  MsgBox "There are too few rows for the num-" & _

  Chr(13) & "ber of independent parameters used.", _

  , "LSPermute"

  End

End If

' Check for missing input data

For i = 1 To RMax

  If IsEmpty(InputArray(i, 1)) Then

    MsgBox "Y-value(s) missing", , "LSPermute"

    End

  End If

Next i

For j = 2 To CMax

  For i = 1 To RMax

    If IsEmpty(InputArray(i, j)) Then

      MsgBox "X-value(s) missing", , "LSPermute"

      End

    End If

  Next i

Next j

' Dimension the data array and the output array

ReDim OutputArray(1 To RMax, 1 To CMax)

ReDim YArray(1 To RMax, 1 To 1) As Double

ReDim XArray(1 To RMax, 1 To 1) As Double

' Fill yArray

For R = 1 To RMax

  YArray(R, 1) = InputArray(R, 1)

Next R

' Initialize xArray

For R = 1 To RMax

  For C = 1 To CMax

    OutputArray(R, C) = ""

  Next C

Next R

Selection.Offset(RMax + 1, 0).Select

Down = Down + RMax + 1

' Write column headings

OutputArray(1, 1) = "Standard deviation of fit"

Selection.Value = OutputArray

Selection.Offset(1, 0).Select

OutputArray(1, 1) = "Indices:"

OutputArray(1, 2) = "Thru 0"

OutputArray(1, 3) = "General"

Selection.Value = OutputArray

Selection.Offset(2, 0).Select

Down = Down + 3

' Compute the output for 1 variable:

ccMax = 4

ReDim XArray(1 To RMax, 1 To ccMax) As Double

For C = 2 To CMax

  For R = 1 To RMax

    XArray(R, 1) = 0

    XArray(R, 2) = InputArray(R, C)

  Next R

  Call LLSS(ccMax, RMax, 0, YArray, XArray, StDevF)

  OutputArray(1, 1) = C - 1

  OutputArray(1, 2) = StDevF

  For R = 1 To RMax

    XArray(R, 1) = 1

  Next R

  Call LLSS(ccMax, RMax, 1, YArray, XArray, StDevF)

  OutputArray(1, 3) = StDevF

  Selection.Value = OutputArray

  Selection.Offset(1, 0).Select

  Down = Down + 1

Next C

Selection.Offset(1, 0).Select

Down = Down + 1

' Compute the output for 2 variables:

ccMax = 5

ReDim XArray(1 To RMax, 1 To ccMax) As Double

For C = 2 To CMax

  For c2 = C + 1 To CMax

    For R = 1 To RMax

      XArray(R, 1) = 0

      XArray(R, 2) = InputArray(R, C)

      XArray(R, 3) = InputArray(R, c2)

    Next R

    Call LLSS(ccMax, RMax, 0, YArray, XArray, StDevF)

    OutputArray(1, 1) = C - 1 & "," & c2 - 1

    OutputArray(1, 2) = StDevF

    For R = 1 To RMax

      XArray(R, 1) = 1

    Next R

    Call LLSS(ccMax, RMax, 1, YArray, XArray, StDevF)

    OutputArray(1, 3) = StDevF

    Selection.Value = OutputArray

    Selection.Offset(1, 0).Select

    Down = Down + 1

  Next c2

Next C

Selection.Offset(1, 0).Select

Down = Down + 1

' Compute the output for 3 variables:

ccMax = 6

ReDim XArray(1 To RMax, 1 To ccMax) As Double

For C = 2 To CMax

 For c2 = C + 1 To CMax

  For c3 = c2 + 1 To CMax

   For R = 1 To RMax

     XArray(R, 1) = 0

     XArray(R, 2) = InputArray(R, C)

     XArray(R, 3) = InputArray(R, c2)

     XArray(R, 4) = InputArray(R, c3)

   Next R

   Call LLSS(ccMax, RMax, 0, YArray, XArray, StDevF)

   OutputArray(1, 1) = C - 1 & "," & c2 - 1 & "," & c3 - 1

   OutputArray(1, 2) = StDevF

   For R = 1 To RMax

     XArray(R, 1) = 1

   Next R

   Call LLSS(ccMax, RMax, 1, YArray, XArray, StDevF)

   OutputArray(1, 3) = StDevF

   Selection.Value = OutputArray

   Selection.Offset(1, 0).Select

   Down = Down + 1

  Next c3

 Next c2

Next C

Selection.Offset(1, 0).Select

Down = Down + 1

' Compute the output for 4 variables:

ccMax = 7

ReDim XArray(1 To RMax, 1 To ccMax) As Double

For C = 2 To CMax

 For c2 = C + 1 To CMax

  For c3 = c2 + 1 To CMax

   For c4 = c3 + 1 To CMax

    For R = 1 To RMax

      XArray(R, 1) = 0

      XArray(R, 2) = InputArray(R, C)

      XArray(R, 3) = InputArray(R, c2)

      XArray(R, 4) = InputArray(R, c3)

      XArray(R, 5) = InputArray(R, c4)

    Next R

    Call LLSS(ccMax, RMax, 0, YArray, XArray, StDevF)

    OutputArray(1, 1) = C - 1 & "," & c2 - 1 & _

      "," & c3 - 1 & "," & c4 - 1

    OutputArray(1, 2) = StDevF

    For R = 1 To RMax

      XArray(R, 1) = 1

    Next R

    Call LLSS(ccMax, RMax, 1, YArray, XArray, StDevF)

    OutputArray(1, 3) = StDevF

    Selection.Value = OutputArray

    Selection.Offset(1, 0).Select

    Down = Down + 1

   Next c4

  Next c3

 Next c2

Next C

Selection.Offset(1, 0).Select

Down = Down + 1

' Compute the output for 5 variables:

ccMax = 8

ReDim XArray(1 To RMax, 1 To ccMax) As Double

For C = 2 To CMax

 For c2 = C + 1 To CMax

  For c3 = c2 + 1 To CMax

   For c4 = c3 + 1 To CMax

    For c5 = c4 + 1 To CMax

      For R = 1 To RMax

        XArray(R, 1) = 0

        XArray(R, 2) = InputArray(R, C)

        XArray(R, 3) = InputArray(R, c2)

        XArray(R, 4) = InputArray(R, c3)

        XArray(R, 5) = InputArray(R, c4)

        XArray(R, 6) = InputArray(R, c5)

      Next R

      Call LLSS(ccMax, RMax, 0, YArray, XArray, StDevF)

      OutputArray(1, 1) = C - 1 & "," & c2 - 1 & _

        "," & c3 - 1 & "," & c4 - 1 & "," & c5 - 1

      OutputArray(1, 2) = StDevF

      For R = 1 To RMax

        XArray(R, 1) = 1

      Next R

      Call LLSS(ccMax, RMax, 1, YArray, XArray, StDevF)

      OutputArray(1, 3) = StDevF

      Selection.Value = OutputArray

      Selection.Offset(1, 0).Select

      Down = Down + 1

    Next c5

   Next c4

  Next c3

 Next c2

Next C

Selection.Offset(1, 0).Select

Down = Down + 1

' Compute the output for 6 variables:

ccMax = 9

ReDim XArray(1 To RMax, 1 To ccMax) As Double

For C = 2 To CMax

 For c2 = C + 1 To CMax

  For c3 = c2 + 1 To CMax

   For c4 = c3 + 1 To CMax

    For c5 = c4 + 1 To CMax

     For c6 = c5 + 1 To CMax

      For R = 1 To RMax

        XArray(R, 1) = 0

        XArray(R, C) = InputArray(R, 2)

        XArray(R, c2) = InputArray(R, 3)

        XArray(R, c3) = InputArray(R, 4)

        XArray(R, c4) = InputArray(R, 5)

        XArray(R, c5) = InputArray(R, 6)

        XArray(R, c6) = InputArray(R, 7)

      Next R

      Call LLSS(ccMax, RMax, 0, YArray, XArray, StDevF)

      OutputArray(1, 1) = C - 1 & "," & _

        "," & c2 - 1 & "," & c3 - 1 & "," & _

        c4 - 1 & c5 - 1 & "," & c6 - 1

      OutputArray(1, 2) = StDevF

      For R = 1 To RMax

        XArray(R, 1) = 1

      Next R

      Call LLSS(ccMax, RMax, 1, YArray, XArray, StDevF)

      OutputArray(1, 3) = StDevF

      Selection.Value = OutputArray

      Selection.Offset(1, 0).Select

      Down = Down + 1

     Next c6

    Next c5

   Next c4

  Next c3

 Next c2

Next C

Selection.Offset(-Down, 0).Select

End Sub

'''''''''''''''''''''''''''''''''''''''''''''''''''''''''''

'''''''''''''''''''''****************''''''''''''''''''''''

'''''''''''''''''''''*              *''''''''''''''''''''''

'''''''''''''''''''''*     LLSS     *''''''''''''''''''''''

'''''''''''''''''''''*              *''''''''''''''''''''''

'''''''''''''''''''''****************''''''''''''''''''''''

'''''''''''''''''''''''''''''''''''''''''''''''''''''''''''

'''''''''''''''''''''''''''''''''''''''''''''''''''''''''''

Sub LLSS(ccMax, RMax, P, YArray, XArray, StDevF)

' This subroutine provides the standard deviation StDevF

' of the least squares fit. It calls the subroutines

' Invert, Multiply, and Transpose.

Dim btArray, btqArray

Dim SSR As Double, varY As Double

Dim pArray, piArray, qArray

Dim xtArray, ytArray, ytyArray

ReDim ytArray(1 To 1, 1 To RMax) As Double

ReDim ytyArray(1 To 1, 1 To 1) As Double

ReDim xtArray(1 To ccMax, 1 To RMax) As Double

ReDim pArray(1 To ccMax, 1 To ccMax) As Double

ReDim piArray(1 To ccMax, 1 To ccMax) As Double

ReDim qArray(1 To ccMax, 1 To 1) As Double

ReDim bArray(1 To ccMax, 1 To 1) As Double

ReDim btArray(1 To 1, 1 To ccMax) As Double

ReDim btqArray(1 To 1, 1 To 1) As Double

' Compute the least squares fit

Call Transpose(XArray, RMax, ccMax, xtArray)

Call Multiply(xtArray, ccMax, RMax, XArray, ccMax, pArray)

Call Invert(pArray, ccMax, piArray)

Call Multiply(xtArray, ccMax, RMax, YArray, 1, qArray)

Call Multiply(piArray, ccMax, ccMax, qArray, 1, bArray)

Call Transpose(YArray, RMax, 1, ytArray)

Call Transpose(bArray, ccMax, 1, btArray)

Call Multiply(ytArray, 1, RMax, YArray, 1, ytyArray)

Call Multiply(btArray, 1, ccMax, qArray, 1, btqArray)

' Compute the standard deviation of the fit

SSR = ytyArray(1, 1) - btqArray(1, 1)

varY = SSR / (RMax - ccMax - P + 3)

StDevF = Sqr(Abs(varY))

End Sub

'''''''''''''''''''''''''''''''''''''''''''''''''''''''''''

'''''''''''''''''''''''''''''''''''''''''''''''''''''''''''

'''''''''''''*********************************'''''''''''''

'''''''''''''*                               *'''''''''''''

'''''''''''''*           SolverAid           *'''''''''''''

'''''''''''''*                               *'''''''''''''

'''''''''''''*********************************'''''''''''''

''''''''''''''''''''''''''''''''''''''''''' (c) R. de Levie

'''''''''''''''''''''''''''''''''''''''''''  v 12, May 2012

' PURPOSE:

' SolverAid takes the results of Solver and computes the

' corresponding standard deviations, including color coding

' in case the absolute value of ai/si is close to 1.

' When there are multiple parameters, the macro displays

' the covariance matrix, CM, in dark blue italics, and

' provides the option to show the corresponding array of

' linear correlation coefficients, CC, color-coded to alert

' the user in case of collinearity.

' All necessary derivatives are computed with second-order

' central differencing.

' REQUIREMENTS:

' This macro requires the subroutine Invert

' INPUT:

' The input information must be provided in response to

' three input boxes, which ask for (1) the location of

' the SINGLE CELL or CONTIGUOUS COLUMN containing the

' parameter(s) determined by Solver; (2) the location of

' the cell containing the target to be minimized, such

' as SSR; and (3) the CONTIGUOUS COLUMN containing the y-

' values calculated with the Solver-determined parameters.

' Note that these data must contain the formulas referring

' to the Solver-determined parameters; a column merely

' containing their numerical values will NOT do.

' For convenient output, place the target (e.g., SSR) such

' that there is an empty cell directly above it for the

' macro label, and an empty cell to its immediate right for

' the standard deviation of the fit of the function, Sf.

' Likewise leave empty cells to the right of the Solver

' parameters for their individual standard deviations. If

' this is not done, message boxes will display the output.

' OUTPUT:

' Provided that these spaces are unoccupied, the standard

' deviations of the fit, and the standard deviation(s) of

' the parameter(s), will be placed directly, in italics, to

' the right of the corresponding parameters (when these

' were entered as a column), or below them (for row input).

' Otherwise, the result(s) will be displayed in message

' boxes. It is therefore most convenient to leave blank

' the spreadsheet cells to the right of (or below) the

' parameters and of SSR.

' Warning: If space to be used for output contains text or

' other non-numerical information (such as time or date),

' the macro will crash.

' The covariance matrix shows the variances on its main

' (top-left to bottom-right) diagonal, and the correspon-

' ding covariances in the off-diagonal positions. The user

' must specify where on the spreadsheet it should be dis-

' played. If you do not wish to see the covariance matrix

' and the corresponding linear correlation coefficients,

' select "Cancel" when asked to specify the location of

' the covariance matrix.

' The user also specifies the location on the spreadsheet

' of the optional array of the correlation coefficients,

' for which all elements on the main diagonal are 1, and

' all off-diagonal elements are calculated as r(i,j) =

' v(i,j)/sqr[(v(i,i)*v(j,j)].

' PROCEDURE:

' To start the process, call the macro, and enter or

' drag the requested addresses into the input boxes.

' COLOR CODES USED:

' A color code is used with the standard deviation S of

' a parameter P, in order to alert the user when that

' standard deviation is not much smaller than the absolute

' value of that parameter. The color code used is:

'   black     for Abs(P/S) > 5

'   gray      for 3 < Abs(P/SP) <= 5

'   orange    for 2 < Abs(P/SP) <= 3

'   red       for 1 < Abs(P/SP) <= 2

'   bold red  for Abs(P/SP) <= 1

' A color code is also used with the array of linear

' correlation coefficients CC, to alert the user to

' collinearity. In this case the color code used is:

'   black     for abs(CC) < 0.90

'   gray      for 0.90 <= absCC < 0.95

'   orange    for 0.95 <= absCC < 0.98

'   red       for 0.98 <= absCC < 0.99

'   bold red  for 0.99 <= absCC < 1

' In both cases, a cell comment summarizing the color

' code used is attached to the display.

' EXAMPLES:

' Use of this macro is illustrated throughout AE3, starting

' with chapter 4.

Sub SolverAid()

' NOTATION USED:

' P:     single Parameter determined by Solver,

' PP:    multiple Parameters determined by Solver

'        which must be in a single, contiguous column.

' SF:    the Standard deviation on the Function

' SP:    the Standard deviations of a single Parameter

' SPP:   the Standard deviations of multiple Parameters

' T:     the Solver Target (e.g., the sum of squares

'        of the residuals) used in optimizing Solver

' Ycalc: the Y-values computed with the parameters P,

'        which must be in a contiguous single column

' prefixes used:

' a:     the address of

' c:     the number of columns of

' o:     occupancy index (0: empty, 1: occupied)

' r:     the number of rows of

' rg:    the range of

' suffixes used:

' a:     Above

' m#:    Minus # times Delta

' p#:    Plus # times Delta

' r:     to the Right

' Dimension the parameters

Dim cP As Integer, cT As Integer, cY As Integer

Dim i As Integer, ii As Integer, N As Integer

Dim oPr As Integer

Dim oTa As Integer, oTr As Integer

Dim rP As Integer, rT As Integer

Dim j As Long, rY As Long, rrY As Long

Dim Delta As Double

Dim P As Double, Pm1 As Double, Pm2 As Double

Dim Pp1 As Double, Pp2 As Double, Pr As Double

Dim Q As Double, Root As Double, Sf As Double, SDD1 As Double

Dim SP As Double, t As Double, Z As Double

Dim cc As Variant, CoVar As Variant

Dim PC As Variant, PCs As Variant

Dim PP As Variant, PPm1 As Variant, PPm2 As Variant

Dim PPp1 As Variant, PPp2 As Variant, PPr As Variant

Dim QQ As Variant, SPP As Variant

Dim Y As Variant, Ym1 As Variant, Ym2 As Variant

Dim Yp1 As Variant, Yp2 As Variant

Dim rgP As Range, rgT As Range

Dim rgY As Range, rgCM As Range

Dim Answer, iAnswer, iiAnswer, aLeftTop, aRightBottom

Dim s, TryAgainA, TryAgainB, TryAgainC, TryAgainD

Dim LastLine

Delta = 0.00003

' Select the computed Solver parameter P1 or parameters

Set rgP = Application.InputBox(Prompt:= _

  "The location of the cell or column containing" & _

    Chr(13) & "the parameter(s) determined by Solver is:", _

  Title:="SolverAid input: Solver parameter(s)", Type:=8)

rgP.Select

cP = Selection.Columns.Count

rP = Selection.Rows.Count

If cP = 0 Then End

' Multiple parameters must be in a single column

If cP > 1 Then

  MsgBox "The Solver-determined parameter" & Chr(13) & _

    "must be in a contiguous SINGLE COLUMN."

  End

End If

' Read the multiple parameters

' and dimension the related variables

If rP > 1 Then

  vpP = Selection.Value             ' reading

  fPP = Selection.Formula           ' reading

  PPm2 = Selection.Value            ' dimensioning

  PPm1 = Selection.Value            ' dimensioning

  PPp1 = Selection.Value            ' dimensioning

  PPp2 = Selection.Value            ' dimensioning

End If

' The column of multiple parameters must be contiguous

If rP > 1 Then

  N = 0

  For i = 1 To rP

    If IsEmpty(fPP(i, 1)) Then N = N + 1

  Next i

  If N > 0 Then

    MsgBox "The Solver-determined parameters" & Chr(13) _

       & "must be in a CONTIGUOUS single column."

    End

  End If

End If

ReDim SPP(1 To rP, 1 To cP)

ReDim QQ(1 To rP, 1 To cP)

' Replace any zeros in the input parameters by

' small non-zero numbers, in order to prevent

' subsequent divide-by-zero errors

If rP = 1 Then

  P = Selection.Value

  If P = 0 Then P = 1E-300

ElseIf rP > 1 Then

  For i = 1 To rP

    If vpP(i, 1) = 0 Then vpP(i, 1) = 1E-300

  Next i

End If

' Test the default output range for the standard

' deviations SP to the right of the parameter(s) P.

oPr = 0

Selection.Offset(0, 1).Select

If rP = 1 Then

  Pr = Selection.Value

  If (IsEmpty(Pr) Or Pr = 0) Then

    oPr = oPr

  Else

    oPr = 1

  End If

ElseIf rP > 1 Then

  PPr = Selection.Value

  For i = 1 To rP

    If IsEmpty(PPr(i, 1)) Then

      oPr = oPr

    Else

      oPr = 1

    End If

  Next i

End If

Selection.Offset(0, -1).Select

' Select the Solver target, such as

' the sum of squares of the residuals

Set rgT = Application.InputBox(Prompt:= _

  "The Solver Target (such as the sum of" & _

  Chr(13) & "squares of the residuals) is located in: ", _

  Title:="SolverAid input: Target", Type:=8)

rgT.Select

cT = Selection.Columns.Count

rT = Selection.Rows.Count

If cT <> 1 Then End

If rT <> 1 Then End

t = Selection.Value

' Test the default output range for space for the

' standard deviation of the fit to the function,

' Sf, as well as for the macro label

oTa = 1      ' tests cell Occupancy Above T

oTr = 1      ' tests cell Occupancy to the Right of T

Selection.Offset(0, 1).Select

s = Selection.Value

If (IsEmpty(s) Or s = 0) Then oTr = 0

Selection.Offset(0, -1).Select

On Error Resume Next     ' in case T is on row 1

Selection.Offset(-1, 0).Select

If Err = 0 Then

  s = Selection.Value

  If (IsEmpty(s) Or s = 0 Or s = "SolverAid") Then oTa = 0

  Selection.Offset(1, 0).Select

End If

' Select the computed Y-values, Ycalc

Set rgY = Application.InputBox(Prompt:= _

  "The column containing Ycalc is:", _

  Title:="SolverAid input: Ycalc", Type:=8)

rgY.Select

rY = Selection.Rows.Count

cY = Selection.Columns.Count

If cY <> 1 Then

  MsgBox "The Ycalc values should be in" & _

    Chr(13) & "a single, contiguous COLUMN"

  End

End If

If rY <= rP Then

  MsgBox " The number N of data pairs must be at least" & _

    Chr(13) & "larger by one than the number of " _

    & "parameters P.", , "SolverAid"

  End

End If

rgY.Select

Y = Selection.Value                 ' reading

Ym2 = Selection.Value               ' dimensioning

Ym1 = Selection.Value               ' dimensioning

Yp1 = Selection.Value               ' dimensioning

Yp2 = Selection.Value               ' dimensioning

Application.ScreenUpdating = False

' Place the macro label above, below, or on T

rgT.Select

If oTa = 0 Then

  Selection.Offset(-1, 0).Select

  With Selection

    .Value = "SolverAid"

    .Font.Bold = True

    .Font.Italic = False

    .Font.ColorIndex = 5

    .HorizontalAlignment = xlCenter

    .Interior.ColorIndex = 20

  End With

End If

With Selection

  .ClearComments

  .AddComment

  .Comment.Visible = False

  .Comment.Text Text:= _

    "SolverAid vs. 7" & Chr(10) & _

    "Param: " & rgP.Address & Chr(10) & _

    "Target: " & rgT.Address & Chr(10) & _

    "Ycalc: " & rgY.Address & Chr(10) & _

    Date & ",   " & Time

End With

If (oTa = 0 Or (oTa = 1 And Selection.Value = _

  "SolverAid")) Then Selection.Offset(1, 0).Select

' Compute and display the value of Sf

rgT.Select

Sf = Sqr(t / (rY - rP))

If oTr = 1 Then

  Answer = MsgBox("The cell to the right of SSR is" _

    & Chr(13) & "not empty. Can it be overwritten?", vbYesNo, "Overwrite?")

  If Answer = vbYes Then oTr = 0

End If

If oTr = 0 Then

  Selection.Offset(0, 1).Select

  With Selection

    .Value = Sf

    .Font.Bold = False

    .Font.Italic = True

    .Font.ColorIndex = 1

    .HorizontalAlignment = xlCenter

    .ClearComments

  End With

  Selection.Offset(0, -1).Select

End If

If oTr = 1 Then MsgBox "The standard deviation of " _

  & "the fit of the function is " & Sf, , "SolverAid"

' The ONE-PARAMETER case

' Compute the partial differentials dY/dP

' and the standard deviations

If rP = 1 Then

  rgY.Select

  Y = Selection.Value

  Ym2 = Selection.Value

  Ym1 = Selection.Value

  Yp1 = Selection.Value

  Yp2 = Selection.Value

  rgP.Select

  P = rgP.Value

  Pm2 = P * (1 - 2 * Delta)

  Pm1 = P * (1 - Delta)

  Pp1 = P * (1 + Delta)

  Pp2 = P * (1 + 2 * Delta)

  rgP.Select

  Selection.Replace P, Pm2

  rgY.Select

  Ym2 = rgY.Value

  rgP.Select

  Selection.Replace Pm2, Pm1

  rgY.Select

  Ym1 = rgY.Value

  rgP.Select

  Selection.Replace Pm1, Pp1

  rgY.Select

  Yp1 = rgY.Value

  rgP.Select

  Selection.Replace Pp1, Pp2

  rgY.Select

  Yp2 = rgY.Value

  rgP.Select

  Selection.Replace Pp2, P

  ReDim D1(1 To rY) As Double, DD1(1 To rY) As Double

  SDD1 = 0

  For j = 1 To rY

    D1(j) = (Ym2(j, 1) - Yp2(j, 1) + 8 * _

      (Yp1(j, 1) - Ym1(j, 1))) / (12 * Delta * P)

    DD1(j) = D1(j) * D1(j)

    SDD1 = SDD1 + DD1(j)

  Next j

  rgP.Select

  Selection.Value = P

  SP = Sf / Sqr(SDD1)

' Write the standard deviation of the parameter

  If oPr > 0 Then

    Answer = MsgBox("The cells to the right of the" _

      & Chr(13) & "Solver parameter is not empty." _

      & "Can it be overwritten?", vbYesNo, "Overwrite?")

    If Answer = vbNo Then

      For i = 1 To rP

    MsgBox "The standard deviation of the" & _

      Chr(13) & "parameter is " & SP, , "SolverAid"

      Next i

    End If

    If Answer = vbYes Then oPr = 0

  End If

  If oPr = 0 Then

    rgP.Select

    Selection.Offset(0, 1).Select

    Selection.Value = SP

    Q = Abs(P / SP)

' Apply color code

' Color code used:

'   black     for Q > 5

'   gray      for 3 < Q <= 5

'   orange    for 2 < Q <= 3

'   red       for 1 < Q <= 2

'   bold red  for Q <= 1

    With Selection.Font

      .Italic = True

      .Bold = False

      .ColorIndex = 1

      If Q <= 5 Then .ColorIndex = 16

      If Q <= 3 Then .ColorIndex = 46

      If Q <= 3 Then .ColorIndex = 3

      If Q <= 1 Then .Bold = True

    End With

' Add explanatory cell comment

    ActiveCell.Select

    With ActiveCell

      .ClearComments

      .AddComment

      .Comment.Visible = False

      .Comment.Text Text:= _

        "Q = abs(P/S);  Colors:" & Chr(10) & _

        "gray:       3 < Q <= 5" & Chr(10) & _

        "orange:   2 < Q <= 3" & Chr(10) & _

        "red:         1 < Q <= 2" & Chr(10) & _

        "bold red:        Q <= 1"

    End With

  End If

  GoTo LastLine

End If

' The MULTIPLE-PARAMETER case

' Compute the partial differentials

If rP > 1 Then

  ReDim D(1 To rP, 1 To rY) As Double

  For i = 1 To rP

    rgP.Select

    Selection.Formula = fPP

    PPm2(i, 1) = vpP(i, 1) * (1 - 2 * Delta)

    Selection.Value = PPm2

    rgY.Select

    Ym2 = Selection.Value

    rgP.Select

    PPm2(i, 1) = vpP(i, 1)

    Selection.Formula = fPP

    PPm1 = Selection.Value

    PPm1(i, 1) = vpP(i, 1) * (1 - Delta)

    Selection.Value = PPm1

    rgY.Select

    Ym1 = Selection.Value

    rgP.Select

    Selection.Formula = fPP

    PPp1 = Selection.Value

    PPp1(i, 1) = vpP(i, 1) * (1 + Delta)

    Selection.Value = PPp1

    rgY.Select

    Yp1 = Selection.Value

    rgP.Select

    Selection.Formula = fPP

    PPp2 = Selection.Value

    PPp2(i, 1) = vpP(i, 1) * (1 + 2 * Delta)

    Selection.Value = PPp2

    rgY.Select

    Yp2 = Selection.Value

    rgP.Select

    Selection.Formula = fPP

    For j = 1 To rY

    D(i, j) = (Ym2(j, 1) - Yp2(j, 1) + 8 * _

      (Yp1(j, 1) - Ym1(j, 1))) / (12 * Delta * vpP(i, 1))

    Next j

  Next i

  rgP.Select

  Selection.Value = vpP

  ReDim dd(1 To rP, 1 To rP, 1 To rY) As Double

  ReDim SDD(1 To rP, 1 To rP) As Double

  ReDim SDDInv(1 To rP, 1 To rP) As Double

  For i = 1 To rP

    For ii = 1 To rP

      SDD(i, ii) = 0

      For j = 1 To rY

        dd(i, ii, j) = D(i, j) * D(ii, j)

        SDD(i, ii) = SDD(i, ii) + dd(i, ii, j)

      Next j

    Next ii

  Next i

  Call Invert(SDD, rP, SDDInv)

' Display the standard deviations of the

' Solver parameters

  Application.ScreenUpdating = True

  For i = 1 To rP

    SPP(i, 1) = Sf * Sqr(SDDInv(i, i))

    QQ(i, 1) = Abs(vpP(i, 1) / SPP(i, 1))

  Next i

  If oPr > 0 Then

    Answer = MsgBox("There are data in the cells to the " _

      & "right of the" & Chr(13) & "Solver parameters. " _

      & "Can they be overwritten?", vbYesNo, "Overwrite?")

    If Answer = vbNo Then

      For i = 1 To rP

        MsgBox "The standard deviation #" & i & _

          Chr(13) & "is " & SPP(i, 1), , "SolverAid"

      Next i

    End If

    If Answer = vbYes Then oPr = 0

  End If

  If oPr = 0 Then

    rgP.Select

    Selection.Offset(0, 1).Select

' Apply the color code based on QQ,

' i.e., on the value of abs[P(i)/SP(i)], where

' the P(i) are the Solver-determined parameters,

' and the SP(i) their standard deviations.

' Color code used:

'   black     for QQ > 5

'   gray      for 3 < QQ <= 5

'   orange    for 2 < QQ <= 3

'   red       for 1 < QQ <= 2

'   bold red  for QQ <= 1

    Selection.Value = QQ

    For Each Cell In Selection.Cells

      Cell.HorizontalAlignment = xlCenter

      Cell.Font.Italic = True

      If Cell.Value > 5 Then

        Cell.Font.Bold = False

        Cell.Font.ColorIndex = 1   ' black

      ElseIf (Cell.Value <= 5 And Cell.Value > 3) Then

        Cell.Font.Bold = False

        Cell.Font.ColorIndex = 16  ' gray

      ElseIf (Cell.Value <= 3 And Cell.Value > 2) Then

        Cell.Font.Bold = False

        Cell.Font.ColorIndex = 46   ' orange

      ElseIf (Cell.Value <= 2 And Cell.Value > 1) Then

        Cell.Font.Bold = False

        Cell.Font.ColorIndex = 3   ' red

      ElseIf Cell.Value <= 1 Then

        Cell.Font.Bold = True      ' bold

        Cell.Font.ColorIndex = 3   ' red

      End If

    Next Cell

' Overwrite QQ with SSP but leave color code intact

    Selection.Value = SPP

' Add an explanatory cell comment

    ActiveCell.Select

    With ActiveCell

      .ClearComments

      .AddComment

      .Comment.Visible = False

      .Comment.Text Text:= _

        "Q = abs(P/S);  Colors:" & Chr(10) & _

        "gray:       3 < Q <= 5" & Chr(10) & _

        "orange:   2 < Q <= 3" & Chr(10) & _

        "red:         1 < Q <= 2" & Chr(10) & _

        "bold red:        Q <= 1"

    End With

  End If

' Select the option, and the block

' in which to write the matrix

TryAgainA:

  If t = 0 Then GoTo LastLine

  Application.ScreenUpdating = True

  Err = 0

  On Error Resume Next      ' Abolish CM with "Cancel"

  Set rgCM = Application.InputBox(Prompt:= _

    "The covariance matrix needs a block" & Chr(13) & _

    "of " & rP & " by " & rP & " cells." & _

    " It should be located in:", Type:=8)

  If Err <> 0 Then GoTo LastLine

  rgCM.Select

  CoVar = Selection.Value

  cc = Selection.Value

' Make sure that the selected block has the correct size

  If Selection.Rows.Count <> rP Then

    MsgBox "The selected range does not have " & _

      rP & " rows." & Chr(13) & _

      "Push OK and then select a proper block."

    GoTo TryAgainA

  End If

  If Selection.Columns.Count <> rP Then

    MsgBox "The selected range does not have " & rP & _

      " columns." & Chr(13) & _

      "Push OK and then select a proper block."

    GoTo TryAgainA

  End If

' Draw a box around the selected block

  Selection.Borders(xlEdgeTop).LineStyle = xlDashDotDot

  Selection.Borders(xlEdgeTop).Weight = xlThin

  Selection.Borders(xlEdgeRight).LineStyle = xlDashDotDot

  Selection.Borders(xlEdgeRight).Weight = xlThin

  Selection.Borders(xlEdgeBottom).LineStyle = _

    xlDashDotDot

  Selection.Borders(xlEdgeBottom).Weight = xlThin

  Selection.Borders(xlEdgeLeft).LineStyle = xlDashDotDot

  Selection.Borders(xlEdgeLeft).Weight = xlThin

' Format cells in array

  For Each Cell In Selection.Cells

    Cell.HorizontalAlignment = xlCenter

    Cell.Font.Italic = True

    Cell.Font.ColorIndex = 5

  Next Cell

' Write the covariance matrix in the selected block

  For i = 1 To rP

    For j = 1 To rP

      CoVar(i, j) = Sf * Sf * SDDInv(i, j)

    Next j

  Next i

  For i = 1 To rP

    For j = 1 To rP

      Root = Sqr(CoVar(i, i) * CoVar(j, j))

      cc(i, j) = CoVar(i, j) / Root

    Next j

  Next i

  Selection.Value = CoVar

' Label the Covariance Matrix

  ActiveCell.Select

  ActiveCell.Offset(0, -1).Select

  If IsEmpty(ActiveCell) Then

    With ActiveCell

      .Value = "CM:"

      .Font.Bold = True

      .Font.Italic = True

      .Font.ColorIndex = 5

      .Interior.ColorIndex = xlNone

      .HorizontalAlignment = xlRight

    End With

  End If

  ActiveCell.Offset(0, 1).Select

' Read the default location, directly below CM,

' for the Correlation Coefficients

  Selection.Offset(rP, 0).Select

  ActiveCell.Select

  aLeftTop = Selection.Address

  ActiveCell.Offset(rP - 1, rP - 1).Select

  aRightBottom = Selection.Address

  BoxAddress = aLeftTop & ":" & aRightBottom

  rgCM.Select

' Provide the optional display of the

' linear correlation coefficients

  iAnswer = MsgBox("Do you also want to see the" _

    & Chr(13) & "linear correlation coefficients?", _

    vbYesNo, "SolverAid")

TryAgainB:

  If iAnswer = vbYes Then

    Application.ScreenUpdating = True

    Set rgCM = Application.InputBox(Prompt:= _

      "The array of linear correlation coefficients" & _

      "also needs" & Chr(13) & "a block of " & rP & _

      " by " & rP & " cells." & _

      " It should be located in:", Default:=BoxAddress, _

      Type:=8)

    rgCM.Select

' Make sure that the selected block has the correct size

    If Selection.Rows.Count <> rP Then

      MsgBox "The selected range does not have " & _

        rP & " rows." & Chr(13) & _

        "Push OK and then select a proper block."

      GoTo TryAgainB

    End If

    If Selection.Columns.Count <> rP Then

      MsgBox "The selected range does not have " & rP & _

        " columns." & Chr(13) & _

        "Push OK and then select a proper block."

      GoTo TryAgainB

    End If

' Draw a box around the selected block

    Selection.Borders(xlEdgeTop).LineStyle = xlDashDotDot

    Selection.Borders(xlEdgeTop).Weight = xlThin

    Selection.Borders(xlEdgeRight).LineStyle = xlDashDotDot

    Selection.Borders(xlEdgeRight).Weight = xlThin

    Selection.Borders(xlEdgeBottom).LineStyle = _

      xlDashDotDot

    Selection.Borders(xlEdgeBottom).Weight = xlThin

    Selection.Borders(xlEdgeLeft).LineStyle = xlDashDotDot

    Selection.Borders(xlEdgeLeft).Weight = xlThin

' Write the linear correlation coefficients

' in the selected block

    Selection.Value = cc

' Format and color-code the cells in the CC array

' Color alerts indicating collinearity:

'   black     for         abs(CoVar) < 0.90

'   gray      for 0.90 <= abs(CoVar) < 0.95

'   orange    for 0.95 <= abs(CoVar) < 0.98

'   red       for 0.98 <= abs(CoVar) < 0.99

'   bold red  for 0.99 <= abs(CoVar) < 1

    For Each Cell In Selection.Cells

      Cell.HorizontalAlignment = xlCenter

      Cell.Font.Italic = True

      Cell.Font.Bold = False

      a = Abs(Cell.Value)

      If a < 0.9 Then

        Cell.Font.ColorIndex = 1        ' black

      ElseIf (a >= 0.9 And a < 0.95) Then

        Cell.Font.ColorIndex = 16       ' gray

      ElseIf (a >= 0.95 And a < 0.98) Then

        Cell.Font.ColorIndex = 46       ' orange

      ElseIf (a >= 0.98 And a < 0.99) Then

        Cell.Font.ColorIndex = 3        ' red

      ElseIf (a <> 1 And a >= 0.99) Then

        Cell.Font.Bold = True           ' bold

        Cell.Font.ColorIndex = 3        ' red

      End If

      If Cell.Value = 1 Then Cell.Font.Italic = False

    Next Cell

' Label the linear correlation coefficients

    ActiveCell.Select

    ActiveCell.Offset(0, -1).Select

    If (IsEmpty(ActiveCell) Or _

      ActiveCell.Value = "CC:") Then

      With ActiveCell

        .Value = "CC:"

        .Font.Bold = True

        .Font.Italic = True

        .Font.ColorIndex = 1

        .Interior.ColorIndex = xlNone

        .HorizontalAlignment = xlRight

        .ClearComments

        .AddComment

        .Comment.Visible = False

        .Comment.Text Text:= _

          "Collinearity alerts:" & Chr(10) & _

          "gray:       absCC >= 0.90" & Chr(10) & _

          "orange:   absCC >= 0.95" & Chr(10) & _

          "red:         absCC >= 0.98" & Chr(10) & _

          "bold red: absCC >= 0.99"

      End With

      ActiveCell.Offset(0, 1).Select

    Else

      ActiveCell.Offset(0, 1).Select

      With ActiveCell

        .ClearComments

        .AddComment

        .Comment.Visible = False

        .Comment.Text Text:= _

          "Collinearity alerts:" & Chr(10) & _

          "gray:       absCC >= 0.90" & Chr(10) & _

          "orange:   absCC >= 0.95" & Chr(10) & _

          "red:         absCC >= 0.98" & Chr(10) & _

          "bold red: absCC >= 0.99"

      End With

    End If

  End If

End If

LastLine:

End Sub

'''''''''''''''''''''''''''''''''''''''''''''''''''''''''''

'''''''''''''''''''''''''''''''''''''''''''''''''''''''''''

'''''''''''''''''************************''''''''''''''''''

'''''''''''''''''*                      *''''''''''''''''''

'''''''''''''''''*     ColumnSolver     *''''''''''''''''''

'''''''''''''''''*                      *''''''''''''''''''

'''''''''''''''''************************''''''''''''''''''

''''''''''''''''''''''''''''''''''''''''''' (c) R. de Levie

'''''''''''''''''''''''''''''''''''''''''''  v 12, May 2012

' PURPOSE:

'

' ColumnSolver allows the user to apply Solver sequentially

' to data located within in single row, whereupon it moves

' to the next row and repeats the procedure.

' The user specifies the location of the target cells, and

' the location of the adjustable parameter(s). Moreover,

' for ColumnSolverValue, the location of the wanted target

' values must be specified. All input information must be

' in columns and blocks of the same length, with all input

' information for any individual Solver application

' confined to a single row. When there are several

' adjustable parameters, these must be in a single

' contiguous block.

'

' ColumnSolverMax maximizes the value in the target cell,

' ColumnSolverMin minimizes the value in the target cell,

' while ColumnSolverValue makes the values of the target

' cells equal to the values specified in the WantedValue

' column. The column or block of parameter values should

' contain appropriate initial values for Solver, which

' ColumnSolver then changes into the final parameter values

' computed by Solver.

'

' The level of precision, tolerance, and convergence of

' Solver, and all other preferences and constraints, should

' be set in the Solver Options before calling ColumnSolver.

'

' INPUT:

'

' Set up Solver as if for a single operation, including all

' necessary specifications of constraints and options, if

' any. Then call ColumnSolver and answer the questions in

' its input boxes. ColumnSolver will start once the needed

' input information has been entered.

'

' OUTPUT:

'

' ColumnSolver adjusts the parameters so that the target cells

' yield the specified criterion, Min, Max, or EqualToValue.

'

' REQUIREMENTS:

'

' ColumnSolver requires that both Solver and Solver.xla have

' been installed.

'

' ONE-TIME PROCEDURE FOR INSTALLING SOLVER.XLA

'

' ColumnSolver can only call Solver as a subroutine when

' Solver.xla is properly referenced in the VBAProject.

' To see whether this is the case, go to the spreadsheet,

' select its VBA module, and look under Tools -> References

' to see whether SOLVER has been activated. It should be

' part of the contiguous, check-marked top items in the

' listed References - VBAProject.

'

' If it is not, first prepare your computer to make Solver

' callable. The procedure to do so is outlined below, and

' is needed only once until the operating system and/or

' Excel is re-installed or updated.

'

' Click on the Start button, and in the resulting menu, on

' Search, under For Files or Folders...(or Find > Files or

' Folders), search for Solver.xla. Note down where

' Solver.xla is located. If you have Microsoft Office, it

' most likely is in C:\Program Files\MicrosoftOffice\Office

' \Library\Solver.

'

' Exit the Search or Find dialog box, which will bring you

' back to the spreadsheet, select the VBA editor with

' Alt+F11, then go to Tools -> References, which will

' display the References - VBAProject dialog box. Click on

' Browse, select Filers of type: Microsoft Excel Files

' (*.xls, *.xla), and now that you know where to find it,

' navigate your way to Solver, and Open it. This will

' return you to the References - VBAProject dialog box,

' where you now use the Priority Up button to move it up,

' so that it is listed contiguously with the other, already

' activated add-ins. Click OK; from then on (i.e., until

' you reload or upgrade Excel) VBA will know how to find

' Solver when it is called from a macro.

'

' EXAMPLE:

'

' Use of this macro is illustrated in section 4.18 of AE3.

Sub ColumnSolverMax()

Dim D As Integer

D = 1

Call ColumnSolver(D)

End Sub

'''''''''''''''''''''''''''''''''''''''''''''''''''''''''''

Sub ColumnSolverMin()

Dim D As Integer

D = 2

Call ColumnSolver(D)

End Sub

'''''''''''''''''''''''''''''''''''''''''''''''''''''''''''

Sub ColumnSolverValue()

Dim D As Integer

D = 3

Call ColumnSolver(D)

End Sub

'''''''''''''''''''''''''''''''''''''''''''''''''''''''''''

Sub ColumnSolver(D)

Dim i As Integer, nc As Integer, nr As Integer

Dim PositionOf2ndDollar As Integer

Dim ParameterAddress As Variant, RowNumber As Variant

Dim TargetAddress As Variant, WantedValue As Variant

Dim myRow As Range, ParameterRange As Range

Dim TargetRange As Range, WantedValues As Range

Dim aAA, Cell, EqualTo, Prompt, Title

' Read the location of the target column

a:

Prompt = "The target cells are located in:" & Chr(13) _

  & "(Highlight a column or enter its address)"

Title = "ColumnSolver: Address of target column"

Set TargetRange = Application.InputBox(Prompt, Title, _

  Type:=8)

TargetRange.Select

nr = Selection.Rows.Count

i = 0

ReDim TargetAddress(1 To nr)

For Each Cell In TargetRange

  i = i + 1

  TargetAddress(i) = Cell.Address

Next Cell

ReDim OutputArray(1 To nr)

' Verify that the target range is a single column

nc = Selection.Columns.Count

If nc <> 1 Then

  MsgBox "There can be only one single Target column"

  GoTo a

End If

' Read the column containing the target values

B:

If D = 3 Then

  Prompt = "The wanted target values are located in:" & Chr(13) _

    & "(Highlight a column or enter its address)"

  Title = "ColumnSolver: Wanted target values"

  Set WantedValues = Application.InputBox(Prompt, Title, _

    Type:=8)

  WantedValues.Select

  ReDim WantedValue(1 To nr) As Double

  i = 0

  For Each Cell In WantedValues

    i = i + 1

    WantedValue(i) = Cell.Value

  Next Cell

' Verify that the target range is a single column

  nc = Selection.Columns.Count

  If nc <> 1 Then

    MsgBox "There can be only one single" & _

      Chr(13) & " column with target values"

    GoTo B

  End If

End If

' Verify that the target range has the correct length

If nr - Selection.Rows.Count <> 0 Then

  MsgBox "The column of target values must have the" _

    & Chr(13) & "same number of rows as the target column."

  GoTo B

End If

' Read the location of the adjustable parameter column

C:

Prompt = "The location of the adjustable parameter(s) is:" & _

  Chr(13) & "(Highlight a column or block, or enter its address)"

Title = "ColumnSolver: Address of parameter range"

Set ParameterRange = Application.InputBox(Prompt, Title, _

  Type:=8)

ParameterRange.Select

nc = Selection.Columns.Count

If nc < 1 Then

  MsgBox "There must be at least one" _

    & Chr(13) & "column of adjustable parameters."

  GoTo C

End If

' Verify that the parameter range has the correct length

If nr - Selection.Rows.Count <> 0 Then

  MsgBox "The parameter block must have the same" _

    & Chr(13) & "number of rows as the target column."

  GoTo C

End If

ReDim ParameterAddress(1 To nr)

Application.ScreenUpdating = False

' For one parameter per row

If nc = 1 Then

  i = 0

  For Each Cell In ParameterRange

    i = i + 1

    ParameterAddress(i) = Cell.Address

  Next Cell

' Run Solver

  For i = 1 To nr

    If D < 3 Then EqualTo = " "

    If D = 3 Then EqualTo = WantedValue(i)

    SolverOK SetCell:=TargetAddress(i), MaxMinVal:=D, _

      valueof:=EqualTo, ByChange:=ParameterAddress(i)

    SolverSolve UserFinish:=True

  Next i

  GoTo E

' For multiple parameters per row

Else

' Find the parameter range for the proper row

' at the intersection of ParameterRange and myRow

  ReDim ParameterAddress(1 To nr)

  ReDim RowNumber(1 To nr)

  For i = 1 To nr

    PositionOf2ndDollar = InStr(2, TargetAddress(i), "$")

    RowNumber(i) = Mid(TargetAddress(i), PositionOf2ndDollar)

    aAA = "$A" & RowNumber(i) & ":" & "$IV" & RowNumber(i)

    Set myRow = Range(aAA)

    Intersect(ParameterRange, myRow).Select

    ParameterAddress = Selection.Address

' Run Solver

    If D < 3 Then EqualTo = " "

    If D = 3 Then EqualTo = WantedValue(i)

    SolverOK SetCell:=TargetAddress(i), MaxMinVal:=D, _

      valueof:=EqualTo, ByChange:=ParameterAddress

    SolverSolve UserFinish:=True

  Next i

End If

E:

End Sub

'''''''''''''''''''''''''''''''''''''''''''''''''''''''''''

'''''''''''''''''''''''''''''''''''''''''''''''''''''''''''

'''''''''''''''''************************''''''''''''''''''

'''''''''''''''''*                      *''''''''''''''''''

'''''''''''''''''*      SolverScan      *''''''''''''''''''

'''''''''''''''''*                      *''''''''''''''''''

'''''''''''''''''************************''''''''''''''''''

''''''''''''''''''''''''''''''''''''''''''' (c) R. de Levie

'''''''''''''''''''''''''''''''''''''''''''  v 12, May 2012

' PURPOSE:

'

' SolverScan uses Solver to scan possible combinations of

' two adjustable parameters, P and Q, by changing the

' initial guess values that Solver uses as its points of

' departure. Its output displays the value of the criterion

' (such as SSR) used to optimize P and Q.

'

' The user specifies the initial values, the step sizes to

' be used to vary those initial values, and the numbers of

' such steps. The output is presented as an array suitable

' for display with Mapper or as a 3-D plot.

'

' Selecting a fine search grid (i.e., many small steps)

' increases the execution time of the macro. The macro has

' built-in limits of 251 steps per parameter, but using far

' fewer steps is usually sufficient, and certainly makes

' the macro complete its task much faster.

'

' INPUT:

'

' Set up Solver as if for a single operation, including

' all necessary specifications of constraints and options,

' if any. Then call SolverScan and answer the questions in

' the input boxes. SolverScan will start once the output

' area has been specified.

'

' OUTPUT:

'

' The user selects the location where the output array will

' appear, complete with top and left-hand axis labels for

' subsequent visual display with Mapper or as a 3-D plot

' of SSR as a function of P and Q, where P and Q are the

' two Solver-adjusted parameters, and where P is plotted

' in the X-direction, and Q in the Y-direction.

'

' SUBROUTINES:

'

' SolverScan requires that both Solver and Solver.xla

' have been installed.

'

' ONE-TIME PROCEDURE FOR INSTALLING SOLVER.XLA

'

' SolverScan can only call Solver as a subroutine, when

' Solver.xla is properly referenced in the VBAProject.

' To see whether this is the case, go to the spreadsheet,

' select its VBA module, and look under Tools ->

' References to see whether SOLVER has been activated.

' It should be part of the contiguous, check-marked top

' items in the listed References - VBAProject. If it is

' not, first prepare your computer to make Solver callable.

' The procedure to do so is outlined below, and is needed

' only once until the operating system and/or Excel is

' re-installed or updated.

'

' Click on the Start button, and in the resulting menu, on

' Search, For Files or Folders...(or Find > Files or

' Folders). In the Find: All Files dialog box, under the

' tab Name & Location, in the Name: window type Solver.xla,

' and, if necessary, enter Systemdisk[C:] in the Look in:

' window, activate Include subfolders, and click on Find

' now. Note down where Solver.xla is located. (If you have

' Microsoft Office, it most likely is in Systemdisk[C:],

' Program Files\Microsoft Office\Office\Library\Solver.)

' Exit the Search or Find dialog box, which will bring you

' back to the spreadsheet, select the VBEditor with

' Alt+F11, then go to Tools -> References, which will

' display the References - VBAProject dialog box. Click on

' Browse, select Files of type: Microsoft Excel Files

' (*.xls, *.xla), and now that you know where to find it,

' navigate your way to Solver.xla, and Open it. This

' will return you to the References - VBAProject dialog

' box, where you now use the Priority up button to move

' it up, so that it is listed contiguously with the other,

' already activated add-ins. Click OK; from then on (i.e.,

' until you reload or upgrade Excel) VBA will know how to

' find Solver when it is called from a macro.

'

' EXAMPLE:

'

' Use of this macro is illustrated in section 4.17 of EA3.

Sub SolverScan()

Dim nc As Integer, ns As Integer, nz As Integer

Dim P As Integer, PassOut As Integer, PassP As Integer

Dim PassQ As Integer, PassTarget As Integer, Q As Integer

Dim StepNumberP As Integer, StepNumberQ As Integer

Dim InitialValueP As Double, InitialValueQ As Double

Dim StepSizeP As Double, StepSizeQ As Double

Dim SSR As Variant, Z As Variant

Dim RangeOut As Range, RangeP As Range

Dim RangeQ As Range, RangeTarget As Range

Dim AddressOut, AddressP, AddressQ, AddressTarget

Dim Answer, Default, Prompt, Title

' Specify the location of the Target criterion

Prompt = "Enter a CELL ADDRESS:" & Chr(13) & Chr(13) & _

  "The target cell (containing the opti-" _

  & Chr(13) & "mization criterion) is located in:"

Title = "SolverScan: Address of target cell"

Set RangeTarget = Application.InputBox(Prompt, Title, Type:=8)

RangeTarget.Select

AddressTarget = RangeTarget.Address

SSR = RangeTarget.Value

PassP = 1

AP:

' Read the address of the first adjustable parameter, P

Prompt = "Enter a CELL ADDRESS:" & Chr(13) & Chr(13) & _

  "The first adjustable parameter, P, is located in:"

Title = "SolverScan: Address of P"

If PassP = 1 Then Default = " "

If PassP > 1 Then Default = RangeP.Address

Set RangeP = Application.InputBox(Prompt, Title, Default, Type:=8)

RangeP.Select

AddressP = RangeP.Address

' Specify the initial guess value for P

Prompt = "Enter a value:" & Chr(13) & Chr(13) & _

  "The initial value of P is:"

Title = "SolverScan: Initial value of P"

If PassP = 1 Then Default = " "

If PassP > 1 Then Default = InitialValueP

InitialValueP = InputBox(Prompt, Title, Default)

' Specify the step size for P

Prompt = "Enter a value:" & Chr(13) & Chr(13) & _

  "The step size for P is:"

Title = "SolverScan: Step size for P"

If PassP = 1 Then Default = " "

If PassP > 1 Then Default = StepSizeP

StepSizeP = Application.InputBox(Prompt, Title, Default)

' Specify the number of steps for P

Prompt = "Enter a value:" & Chr(13) & Chr(13) & _

  "The number of steps for P is:"

Title = "SolverScan: Step number for P"

If PassP = 1 Then Default = " "

If PassP > 1 Then Default = StepNumberP

StepNumberP = Application.InputBox(Prompt, Title, Default)

PassP = PassP + 1

' Check that StepNumberP is smaller than 252

If StepNumberP > 251 Then

  MsgBox "The number of steps cannot be larger than 251."

  GoTo AP

End If

PassQ = 1

AQ:

' Read the address of the first adjustable parameter, Q

Prompt = "Enter a CELL ADDRESS:" & Chr(13) & Chr(13) & _

  "The first adjustable parameter, Q, is located in:"

Title = "SolverScan: Address of Q"

If PassQ = 1 Then Default = ""

If PassQ > 1 Then Default = RangeQ.Address

Set RangeQ = Application.InputBox(Prompt, Title, _

  Default, Type:=8)

RangeQ.Select

AddressQ = RangeQ.Address

' Specify the initial guess value for Q

Prompt = "Enter a value:" & Chr(13) & Chr(13) & _

  "The initial value of Q is:"

Title = "SolverScan: Initial value of Q"

If PassQ = 1 Then Default = ""

If PassQ > 1 Then Default = InitialValueQ

InitialValueQ = InputBox(Prompt, Title, Default)

' Specify the step size for Q

Prompt = "Enter a value:" & Chr(13) & Chr(13) & _

  "The step size for Q is:"

Title = "SolverScan: Step size for Q"

If PassQ = 1 Then Default = " "

If PassQ > 1 Then Default = StepSizeQ

StepSizeQ = Application.InputBox(Prompt, Title, Default)

' Specify the number of steps for Q

Prompt = "Enter a value:" & Chr(13) & Chr(13) & _

  "The number of steps for Q is:"

Title = "SolverScan: Step number for Q"

If PassQ = 1 Then Default = " "

If PassQ > 1 Then Default = StepNumberQ

StepNumberQ = Application.InputBox(Prompt, Title, Default)

PassQ = PassQ + 1

' Check that StepNumberQ is smaller than 252

If StepNumberQ > 251 Then

  MsgBox "The number of steps cannot be larger than 251."

  GoTo AQ

End If

' Specify the top left-hand corner of the Output array

PassOut = 1

AOut:

Prompt = "The output array, including its axes, " _

  & "will be" & Chr(13) & StepNumberP + 1 _

  & " columns wide, and " & StepNumberQ + 1 _

  & " rows high." & Chr(13) & _

  "                     Enter a CELL ADDRESS:" & Chr(13) _

  & "The top left-hand corner of the output" & Chr(13) _

  & "array should be located at:"

Title = "SolverScan: Top left-hand corner of output array"

If PassOut = 1 Then Default = " "

If PassOut > 1 Then Default = RangeOut.Address

Set RangeOut = Application.InputBox(Prompt, Title, Type:=8)

RangeOut.Select

AddressOut = RangeOut.Address

PassOut = PassOut + 1

' Count the number of columns nc to the left of RangeOut

nc = 0

AEdge:

On Error GoTo OnEdge

Selection.Offset(0, -1).Select

nc = nc + 1

GoTo AEdge

OnEdge:

RangeOut.Select

' Make sure that there is enough space for the output array

If nc + StepNumberP > 254 Then

  MsgBox "Move the left-hand corner of the" & _

    Chr(13) & "output array to the left, in order" & _

    Chr(13) & "to accommodate its projected size."

  GoTo AOut

End If

' Verify that the output will not overwrite valuable data

Application.ScreenUpdating = False

nz = 0

RangeOut.Select

For P = 0 To StepNumberP + 1

  Selection.Offset(0, 1).Select

  For Q = 0 To StepNumberQ + 1

    Selection.Offset(1, 0).Select

    Z = Selection.Value

    nz = nz + 1

    If (IsEmpty(Z) Or Z = Null) Then nz = nz - 1

  Next Q

Next P

Application.ScreenUpdating = True

If nz > 0 Then

  Answer = MsgBox("There are data in the array in which" _

    & Chr(13) & "the output data will be written. " _

    & "Proceed" & Chr(13) _

    & "anyway and overwrite those data?", vbYesNo)

  If Answer = vbNo Then GoTo AOut

End If

Application.ScreenUpdating = False

ReDim SSR(StepNumberQ + 1, StepNumberP + 1) As Double

' Enter the initial guess values,

' run Solver, and write output file

ns = 0

For P = 0 To StepNumberP

  RangeP.Select

  RangeP.Value = InitialValueP + P * StepSizeP

  For Q = 0 To StepNumberQ

    RangeQ.Select

    RangeQ.Value = InitialValueQ + Q * StepSizeQ

    Application.DisplayStatusBar = False

    SolverOK SetCell:="addressTarget", MaxMinVal:=2, _

      ByChange:="addressP, addressQ"

    SolverSolve UserFinish:=True

    RangeTarget.Select

    SSR(Q, P) = RangeTarget.Value

    ns = ns + 1

    Application.DisplayStatusBar = True

    Application.StatusBar = "SolverScan has completed " & _

      ns & " out of " & (StepNumberP + 1) * _

      (StepNumberQ + 1) & " calculations."

  Next Q

Next P

' Write the axis scale labels

RangeOut.Select

For P = 1 To StepNumberP + 1

  Selection.Offset(0, P).Select

  Selection.Value = InitialValueP + (P - 1) * StepSizeP

  ActiveCell.Font.Italic = True

  ActiveCell.Font.Bold = True

  Selection.Offset(0, -P).Select

Next P

For Q = 1 To StepNumberQ + 1

  Selection.Offset(Q, 0).Select

  ActiveCell.Font.Italic = True

  ActiveCell.Font.Bold = True

  Selection.Value = InitialValueQ + (Q - 1) * StepSizeQ

  Selection.Offset(-Q, 0).Select

Next Q

' Output the SSR data

RangeOut.Select

For P = 0 To StepNumberP

  For Q = 0 To StepNumberQ

    Selection.Offset(Q + 1, P + 1).Select

    Selection.Value = SSR(Q, P)

    Selection.Offset(-Q - 1, -P - 1).Select

  Next Q

Next P

End Sub

'''''''''''''''''''''''''''''''''''''''''''''''''''''''''''

''''''''''''''''''''''******************'''''''''''''''''''

''''''''''''''''''''''*                *'''''''''''''''''''

''''''''''''''''''''''*     Invert     *'''''''''''''''''''

''''''''''''''''''''''*                *'''''''''''''''''''

''''''''''''''''''''''******************'''''''''''''''''''

'''''''''''''''''''''''''''''''''''''''''''''''''''''''''''

'''''''''''''''''''''''''''''''''''''''''''''''''''''''''''

Sub Invert(m, nrc, MInv)

' The square input and output matrices are M and MInv

' respectively; nrc is the number of rows and columns in

' M and MInv

Dim i As Integer, icol As Integer, irow As Integer

Dim j As Integer, k As Integer, L As Integer, LL As Integer

Dim big As Double, dummy As Double

Dim N As Integer, pivinv As Double

Dim u As Double

N = nrc + 1

ReDim bb(1 To N, 1 To N) As Double

ReDim ipivot(1 To N) As Double

ReDim Index(1 To N) As Double

ReDim indexr(1 To N) As Double

ReDim indexc(1 To N) As Double

u = 1

' Copy the input matrix in order to retain it

For i = 1 To nrc

  For j = 1 To nrc

    MInv(i, j) = m(i, j)

  Next j

Next i

' The following is the Gauss-Jordan elimination routine

' GAUSSJ from J. C. Sprott, "Numerical Recipes: Routines

' and Examples in BASIC", Cambridge University Press,

' Copyright (C)1991 by Numerical Recipes Software. Used by

' permission. Use of this routine other than as an integral

' part of the present book requires an additional license

' from Numerical Recipes Software. Further distribution is

' prohibited. The routine has been modified to yield

' double-precision results.

For j = 1 To nrc

  ipivot(j) = 0

Next j

For i = 1 To nrc

  big = 0

  For j = 1 To nrc

    If ipivot(j) <> u Then

      For k = 1 To nrc

        If ipivot(k) = 0 Then

          If Abs(MInv(j, k)) >= big Then

            big = Abs(MInv(j, k))

            irow = j

            icol = k

          End If

          ElseIf ipivot(k) > 1 Then Exit Sub

        End If

      Next k

    End If

  Next j

  ipivot(icol) = ipivot(icol) + 1

  If irow <> icol Then

    For L = 1 To nrc

      dummy = MInv(irow, L)

      MInv(irow, L) = MInv(icol, L)

      MInv(icol, L) = dummy

    Next L

    For L = 1 To nrc

      dummy = bb(irow, L)

      bb(irow, L) = bb(icol, L)

      bb(icol, L) = dummy

    Next L

  End If

  indexr(i) = irow

  indexc(i) = icol

  If MInv(icol, icol) = 0 Then Exit Sub

  pivinv = u / MInv(icol, icol)

  MInv(icol, icol) = u

  For L = 1 To nrc

    MInv(icol, L) = MInv(icol, L) * pivinv

    bb(icol, L) = bb(icol, L) * pivinv

  Next L

  For LL = 1 To nrc

    If LL <> icol Then

      dummy = MInv(LL, icol)

      MInv(LL, icol) = 0

      For L = 1 To nrc

        MInv(LL, L) = MInv(LL, L) - MInv(icol, L) * dummy

        bb(LL, L) = bb(LL, L) - bb(icol, L) * dummy

      Next L

    End If

  Next LL

Next i

For L = nrc To 1 Step -1

  If indexr(L) <> indexc(L) Then

    For k = 1 To nrc

      dummy = MInv(k, indexr(L))

      MInv(k, indexr(L)) = MInv(k, indexc(L))

      MInv(k, indexc(L)) = dummy

    Next k

  End If

Next L

Erase indexc, indexr, ipivot

End Sub

''''''''''''''''''''********************'''''''''''''''''''

''''''''''''''''''''*                  *'''''''''''''''''''

''''''''''''''''''''*     Multiply     *'''''''''''''''''''

''''''''''''''''''''*                  *'''''''''''''''''''

''''''''''''''''''''********************'''''''''''''''''''

Sub Multiply(M1, r1, C1, M2, c2, MOut)

' Computes the product of two matrices: MOut = M1 times M2

'   r1: number of rows in M1

'   c1: number of columns in M1

'   c2: number of columns in M2

' M2 must have c1 rows

' MOut will have r1 rows and c2 columns

Dim j As Integer

Dim i As Long, k As Long

For i = 1 To r1

  For j = 1 To c2

    MOut(i, j) = 0

    For k = 1 To C1

      MOut(i, j) = MOut(i, j) + M1(i, k) * M2(k, j)

    Next k

  Next j

Next i

End Sub

''''''''''''''''''''********************'''''''''''''''''''

''''''''''''''''''''*                  *'''''''''''''''''''

''''''''''''''''''''*     VMultiply    *'''''''''''''''''''

''''''''''''''''''''*                  *'''''''''''''''''''

''''''''''''''''''''********************'''''''''''''''''''

Sub VMultiply(m, R, C, V, MOut)

' This function computes the product of the matrix M

' and the diagonal matrix DM, where the latter has been

' condensed into a vector V

' where: MOut = M times DM

'   r: number of rows in M

'   c: number of columns in M

' V must have c elements

' MOut will have r rows and c columns

Dim i As Long, j As Integer, k As Integer

For i = 1 To R

  For j = 1 To C

    MOut(i, j) = m(i, j) * V(j)

  Next j

Next i

End Sub

''''''''''''''''''''*********************''''''''''''''''''

''''''''''''''''''''*                   *''''''''''''''''''

''''''''''''''''''''*     Transpose     *''''''''''''''''''

''''''''''''''''''''*                   *''''''''''''''''''

''''''''''''''''''''*********************''''''''''''''''''

Sub Transpose(m, R, C, MT)

' Computes the transpose MT of matrix M

'   r: number of rows in M

'   c: number of columns in M

' MT will have c rows and r columns

Dim i As Long, j As Long

For i = 1 To C

  For j = 1 To R

    MT(i, j) = m(j, i)

  Next j

Next i

End Sub

'''''''''''''''''''''''''''''''''''''''''''''''''''''''''''

'''''''''''''''''''''''''''''''''''''''''''''''''''''''''''

'''''''''''''*********************************'''''''''''''

'''''''''''''*                               *'''''''''''''

'''''''''''''*     Fourier Transformation    *'''''''''''''

'''''''''''''*                               *'''''''''''''

'''''''''''''*********************************'''''''''''''

''''''''''''''''''''''''''''''''''''''''''' (c) R. de Levie

'''''''''''''''''''''''''''''''''''''''''''  v 12, May 2012

' PURPOSE:

' The macros ForwardFT and InverseFT perform fast Fourier

' transformations on an array of complex data, arranged in

' a single block of three adjacent columns.

' SUBROUTINES:

' The macros are merely drivers that set iSign to +1 (for

' the forward transform) or to -1 (for inverse transforma-

' tion). The main subroutine Fourier then reads the input

' information, makes a number of checks, and computes the

' output. For the actual Fourier transformation it then

' calls the subroutine FT. This latter part is placed in a

' separate subroutine in order to make it readily available

' to other macros, such as ConvolveFT, DeconvolveFT,

' DeconvolveIt, and Gabor.

' INPUT:

' The first column must contain the variable (e.g., time,

' or frequency), the second column the real components of

' the input data, the third column their imaginary com-

' ponents. For purely real input data, either leave the

' third column blank, or fill it with zeros. Likewise, for

' purely imaginary input data, the second column should

' only contain zeros, blanks, or combinations thereof.

' The macro accepts two input formats: (1) data with time

' or frequency values centered around zero (jSign = -1),

' and (2) data with time or frequency values starting at

' zero (jSign = 1). The output format is commensurate with

' the input format used.

' OUTPUT:

' The output is written in the three columns to the right

' of the input data block, thereby overwriting any prior

' data in that region. The output columns contain (from

' left to right) frequency or time, the real components

' of the output data, and their imaginary components.

' PROCEDURE:

' To initiate the transform, highlight the three columns

' of the input array, and call one of the two macros,

' ForwardFT() or InverseFT().

' EXAMPLES:

' Uses of this macro are illustrated in chapters 5 and 6

' of Advanced Excel, starting with section 5.1.

'''''''''''''''''''''''''''''''''''''''''''''''''''''''''''

Sub ForwardFT()

Dim iSign As Integer

iSign = 1

Call Fourier(iSign)

End Sub

'''''''''''''''''''''''''''''''''''''''''''''''''''''''''''

Sub InverseFT()

Dim iSign As Integer

iSign = -1

Call Fourier(iSign)

End Sub

'''''''''''''''''''''''''''''''''''''''''''''''''''''''''''

Sub Fourier(iSign)

' Check the array length n, which must be a power of 2,

' and be at least 2

Dim cN As Integer, cnMax As Integer, jSign As Integer

Dim N As Long, NN As Long

Dim rN As Long, rnMax As Long

Dim Length As Single

Dim Check1 As Double, Check2 As Double

Dim Denom As Double, Interval As Double, Z As Double

Dim DataArray As Variant, OutputArray As Variant

Dim myRange As Range

Dim Ans, hAnswer

Begin:

N = 0

rnMax = Selection.Rows.Count

Length = CSng(rnMax)

cnMax = Selection.Columns.Count

' If area was not highlighted

If rnMax = 1 And cnMax = 1 Then

  hAnswer = MsgBox("You forgot to highlight" _

    & Chr(13) & "the block of input data." _

    & Chr(13) & "Do you want to do so now?" _

    , vbYesNo, "Fourier transformation")

  If hAnswer = vbNo Then End

  If hAnswer = vbYes Then

    Set myRange = Application.InputBox(Prompt:= _

      "The input data are located in:", Type:=8)

    myRange.Select

  End If

  GoTo Begin

End If

If Length < 2 Then

  MsgBox "There must be at least two rows."

  End

End If

Do While Length > 1

  Length = Length / 2

Loop

If Length <> 1 Then

  MsgBox "The current number of rows is " & rnMax _

    & "which is not an integer power of two."

  End

End If

' Check that there are three input columns

If cnMax <> 3 Then

  MsgBox "There must be three input columns," _

    & Chr(13) & "one for the variable (time, frequen-," _

    & Chr(13) & "cy, etc.), the next two for the real" _

    & Chr(13) & "and imaginary parts of the input data."

  End

End If

' Read the input data

DataArray = Selection.Value

' Check that the first column has its first two elements

Check1 = VarType(DataArray(1, 1))

If Check1 = 0 Then

  MsgBox "Enter the top left value."

  End

End If

Check2 = VarType(DataArray(2, 1))

If Check2 = 0 Then

  MsgBox "Enter a value in row 2 of the first column."

  End

End If

' Determine what input convention is used:

' jSign = -1 for input data centered around zero,

' jSign = 1 for input data starting at zero

jSign = 0

Interval = (DataArray(2, 1) - DataArray(1, 1)) * rnMax

If DataArray(1, 1) > (-0.5 * Interval / rnMax) And _

  DataArray(1, 1) < (0.5 * Interval / rnMax) Then jSign = 1

If DataArray(1, 1) < (-0.5 * Interval / rnMax) And _

  DataArray(rnMax / 2 + 1, 1) > (-0.5 * Interval / rnMax) _

  And DataArray(rnMax / 2 + 1, 1) < (0.5 * Interval / _

  rnMax) Then jSign = -1

If jSign = 0 Then

  MsgBox "The input format is incorrect." _

    & Chr(13) & "It should either be centered" _

    & Chr(13) & "around zero, or start at zero."

  End

End If

' Read and rearrange the input data

NN = 2 * rnMax

ReDim Term(NN) As Double

If jSign = 1 Then

  For rN = 1 To rnMax

    Term(2 * rN - 1) = DataArray(rN, 2)

    Term(2 * rN) = DataArray(rN, 3)

  Next rN

End If

If jSign = -1 Then

  For rN = 1 To rnMax / 2

    Term(2 * rN - 1) = DataArray(rnMax / 2 + rN, 2)

    Term(2 * rN) = DataArray(rnMax / 2 + rN, 3)

    Term(rnMax + 2 * rN - 1) = DataArray(rN, 2)

    Term(rnMax + 2 * rN) = DataArray(rN, 3)

  Next rN

End If

' Check that the output does not overwrite valuable data

Selection.Offset(0, 3).Select

OutputArray = Selection.Value

For rN = 1 To rnMax

  For cN = 1 To cnMax

    Z = OutputArray(rN, cN)

    If (IsEmpty(Z) Or Z = 0) Then

      N = N

    Else

      N = N + 1

    End If

  Next cN

Next rN

If N > 0 Then

  Ans = MsgBox("  There are data in the space where" _

    & Chr(13) & "the output will be written. Proceed " _

    & Chr(13) & "  anyway and overwrite those data?", _

    vbYesNo)

  If Ans = vbNo Then

    Selection.Offset(0, -3).Select

    End

  End If

End If

' Calculate and write the frequency or time scale

If jSign = 1 Then

  For rN = 1 To rnMax / 2

    DataArray(rN, 1) = (rN - 1) / Interval

  Next rN

  For rN = (rnMax / 2 + 1) To rnMax

    If iSign > 0 Then

      DataArray(rN, 1) = (rN - rnMax - 1) / Interval

    Else

      DataArray(rN, 1) = (rN - 1) / Interval

    End If

  Next rN

End If

If jSign = -1 Then

  For rN = 1 To rnMax

    DataArray(rN, 1) = (-(rnMax / 2) + rN - 1) / Interval

  Next rN

End If

' Calculate the Fourier transform

Call FT(Term, NN, iSign)

' Arrange and write the output data

Denom = (rnMax + 1 + iSign * (rnMax - 1)) / 2

If jSign = 1 Then

  For rN = 1 To rnMax

    DataArray(rN, 2) = Term(2 * rN - 1) / Denom

    DataArray(rN, 3) = Term(2 * rN) / Denom

  Next rN

End If

If jSign = -1 Then

  For rN = 1 To rnMax / 2

    DataArray(rN, 2) = Term(rnMax + 2 * rN - 1) / Denom

    DataArray(rN, 3) = Term(rnMax + 2 * rN) / Denom

    DataArray((rnMax / 2) + rN, 2) = _

      Term(2 * rN - 1) / Denom

    DataArray((rnMax / 2) + rN, 3) = Term(2 * rN) / Denom

  Next rN

End If

Application.ScreenUpdating = False

Selection.Value = DataArray

End Sub

'''''''''''''''''''''''''''''''''''''''''''''''''''''''''''

Sub FT(Term, NN, iSign)

' The following is the Fourier transform routine FOUR1 from

' J. C. Sprott, "Numerical Recipes: Routines and Examples

' in BASIC", Cambridge University Press, Copyright (C)1991

' by Numerical Recipes Software. Used here by permission.

' Use of this routine other than as an integral part of the

' present book requires an additional license from Numeri-

' cal Recipes Software. Further distribution is prohibited.

' The routine has been modified to yield double-precision

' results, and to conform to the standard mathematical sign

' convention for Fourier transformation.

Dim i As Long, iStep As Long, j As Long

Dim m As Long, mMax As Long

Dim tr As Double, ti As Double, theta As Double

Dim wtemp As Double, wi As Double, wr As Double

Dim wpi As Double, wpr As Double

j = 1

For i = 1 To NN Step 2

  If j > i Then

    tr = Term(j)

    ti = Term(j + 1)

    Term(j) = Term(i)

    Term(j + 1) = Term(i + 1)

    Term(i) = tr

    Term(i + 1) = ti

  End If

  m = Int(NN / 2)

  While m >= 2 And j > m

    j = j - m

    m = Int(m / 2)

  Wend

  j = j + m

Next i

mMax = 2

While NN > mMax

  iStep = 2 * mMax

  theta = 2 * Application.Pi() / (-iSign * mMax)

  wpr = -2 * Sin(0.5 * theta) ^ 2

  wpi = Sin(theta)

  wr = 1

  wi = 0

  For m = 1 To mMax Step 2

    For i = m To NN Step iStep

      j = i + mMax

      tr = wr * Term(j) - wi * Term(j + 1)

      ti = wr * Term(j + 1) + wi * Term(j)

      Term(j) = Term(i) - tr

      Term(j + 1) = Term(i + 1) - ti

      Term(i) = Term(i) + tr

      Term(i + 1) = Term(i + 1) + ti

    Next i

    wtemp = wr

    wr = wr * wpr - wi * wpi + wr

    wi = wi * wpr + wtemp * wpi + wi

  Next m

  mMax = iStep

Wend

End Sub

'''''''''''''''''''''''''''''''''''''''''''''''''''''''''''

'''''''''''''''''''''''''''''''''''''''''''''''''''''''''''

''''''''''''''*******************************''''''''''''''

''''''''''''''*                             *''''''''''''''

''''''''''''''*     Gabor Transformation    *''''''''''''''

''''''''''''''*                             *''''''''''''''

''''''''''''''*******************************''''''''''''''

'''''''''''''''''''''''''''''''''''''''''''''''''''''''''''

''''''''''''''''''''''''''''''''''''''''''' (c) R. de Levie

'''''''''''''''''''''''''''''''''''''''''''  v 12, May 2012

' PURPOSE:

' Gabor transformation analyzes a time-dependent signal in

' terms of its frequency content as a function of time. It

' does this by multiplying the time-dependent signal by a

' moving Gaussian window, and by then Fourier-transforming

' the resulting product. The result is stored in two

' arrays, ReOut() and ImOut(), which contain the real and

' imaginary parts of the Fourier transform as a function of

' time and frequency. The contents of one of these arrays,

' or of the corresponding magnitude or phase angle, can

' then be displayed in a color map (using Mapper) or a 3-D

' plot.

' SUBROUTINES:

' This macro requires the subroutine FT.

' INPUT:

' In input information should be placed in two adjacent

' columns, with the left-most column containing time, the

' right-most the corresponding signal. The macro is re-

' stricted to equidistant data, and the time increments

' must therefore be equal. The width of the moving window

' must be an odd power of 2 (here set to be at least a

' factor 2 smaller than the number of data in the set),

' and is selected with an input box.

' OUTPUT:

' The result is displayed with a time axis on top, and a

' frequency axis to the left, ready for immediate visuali-

' zation with Mapper.

' PROCEDURE:

' In order to start the Gabor transform, call the macro

' and enter the requested information in the input boxes.

' WARNING:

' This macro produces a large output array to the right of

' the input data. It does NOT check whether that array

' contains any information that should not be overwritten.

' Therefore make sure that there is no valuable information

' on the spreadsheet to the right of the input columns.

' EXAMPLE:

' Use of this macro is illustrated in sections 6.9 and 6.10

' of AE3.

' dataArray   input data set, two columns wide, the left-

'               most for time, the rightmost for signal

' M           the number of data in the set, i.e., the

'               number of rows in dataArray

' N           the number of data used for the window,

'               smaller than about half of M

' Gaussian    the window function used, containing N points

' ReOut       the real component of the output signal,

'               as a function of time and frequency

' ImOut       the imaginary component of the output signal,

'               as a function of time and frequency

' MagOut      the magnitude (absolute value) of the output

'               signal, as a function of time and frequency

' LogOut      the 10-based logarithm of the magnitude of

'               the output signal, as a function of time

'               and frequency

' PhaseOut    the phase angle (in radians) of the output

'               signal, as a function of time and frequency

Sub Gabor()

Dim i As Integer, j As Integer, jj As Integer

Dim m As Integer, MM As Integer, N As Integer

Dim NN As Integer, MinSpacing As Integer

Dim P As Integer, Spacing As Integer

Dim Gaussian() As Double, ImOut() As Double

Dim Interval As Double, LogOut() As Double

Dim MagOut() As Double, Norm As Double

Dim PhaseOut() As Double, Pi As Double, ReOut() As Double

Dim s As Double, Term() As Double

Dim DataArray As Variant, freqArray As Variant

Dim OutputArray As Variant

Dim myRange As Range

Dim Message1, Message3, Title1, Title3, Default3

' Read the time interval

Message1 = "The time interval between successive data is:"

Title1 = "Gabor InputBox 1: Time interval"

Interval = InputBox(Message1, Title1)

' Read the input data

Set myRange = Application.InputBox(Prompt:= _

  "The input data are located in column:", Title:= _

  "Gabor InputBox 2: Input data", Type:=8)

myRange.Select

' Check that there is only one data input column

If Selection.Columns.Count <> 1 Then

  MsgBox "There can only be ONE data input column."

  End

End If

DataArray = Selection.Value

freqArray = Selection.Value

OutputArray = Selection.Value

m = Selection.Rows.Count

' Determine the size of the window width N

MM = m / 4

N = 2

Do

N = 2 * N

Loop Until N > MM

' Set the minimum sample spacing

MinSpacing = Int((m - N + 1) / 125)

jj = MinSpacing

' Select a sample spacing

Message3 = "The minimum sample spacing is:"

Title3 = "Gabor InputBox 3: Sample spacing"

Default3 = MinSpacing

Spacing = InputBox(Message3, Title3, Default3)

If Spacing >= MinSpacing Then jj = Spacing

' Compute the frequency scale

For i = 1 To N

  freqArray(i, 1) = " "

Next i

For i = 1 + N / 2 To N

  freqArray(i, 1) = (-N + 2 * i - 2) / (2 * N * Interval)

Next i

For i = N + 1 To N + 1

  freqArray(i, 1) = (-N + 2 * i - 2) / (2 * N * Interval)

Next i

' Display the frequency scale

Selection.Offset(0, 1).Select

Selection.Value = freqArray

Application.ScreenUpdating = False

' Compute the window function Gaussian()

ReDim Gaussian(1 To N)

Pi = [Pi()]

s = N / 6

Norm = 12 / Sqr(2 * Pi)

For i = 1 To N

  Gaussian(i) = _

    Norm * Exp(-(i - (N + 1) / 2) ^ 2 / (2 * s * s))

Next i

' The main loop, repeating the analysis M-N+1 times

For j = jj To m - N + 1 Step jj

' Window the input data fragment and rearrange it for FFT

  NN = 2 * N

  ReDim Term(NN) As Double

  For i = 1 To N

    Term(2 * i - 1) = DataArray(i + j - 1, 1) * Gaussian(i)

    Term(2 * i) = 0

  Next i

'  Call the forward Fourier transform subroutine

  Call FT(Term, NN, 1)

' Compute the output data. The following section presents

' several output options, of which only one combination can

' be selected at any one time. For general visualization of

' the result, MagOut is usually the preferred output format

  ReDim MagOut(N, m - N + 1)

'  ReDim ReOut(N, M - N + 1)

'  ReDim ImOut(N, M - N + 1)

'  ReDim LogMagOut(N, M - N + 1)

'  ReDim PhaseOut(N, M - N + 1)

  For i = 1 To N

    MagOut(i, j) = _

      (Sqr(Term(2 * i - 1) ^ 2 + Term(2 * i) ^ 2)) / N

'    ReOut(i, j) = Term(2 * i - 1) / N

'    ImOut(i, j) = Term(2 * i) / N

'    LogMagOut(i, j) = 0.5 * Log((Term(2 * i - 1) ^ 2 _

       + Term(2 * i) ^ 2) / N)

'    PhaseOut(i, j) = Atn(Term(2 * i) / Term(2 * i - 1))

  Next i

  P = Int(100 * (j / (m - N + 1)))

  Application.StatusBar = "Calculation " & P & "% done"

' Data output

' (shown here for MagOut; modify by replacing MagOut

' by alternative output file name XXX(i, j) if so desired)

  For i = 1 To N / 2 - 1

    OutputArray(i, 1) = " "

  Next i

  For i = N / 2 To N / 2

    OutputArray(i, 1) = j

  Next i

  For i = 1 + N / 2 To N

    OutputArray(i, 1) = MagOut(i, j)

  Next i

  For i = N + 1 To N + 2

    OutputArray(i, 1) = MagOut(1, j)

  Next i

  For i = N + 2 To m

    OutputArray(i, 1) = " "

  Next i

  Selection.Offset(0, 1).Select

  Selection.Value = OutputArray

Next j

Application.StatusBar = False

End Sub

'''''''''''''''''''''''''''''''''''''''''''''''''''''''''''

'''''''''''''''''''''''''''''''''''''''''''''''''''''''''''

'''''''''''''*********************************'''''''''''''

'''''''''''''*                               *'''''''''''''

'''''''''''''*         Gram-Schmidt          *'''''''''''''

'''''''''''''*       orthogonalization       *'''''''''''''

'''''''''''''*                               *'''''''''''''

'''''''''''''*********************************'''''''''''''

''''''''''''''''''''''''''''''''''''''''''' (c) R. de Levie

'''''''''''''''''''''''''''''''''''''''''''  v 12, May 2012

' PURPOSE:

' This subroutine uses Gram-Schmidt orthogonalization to

' compute and display the orthogonal polynomials correspon-

' ding to one or more independent variables x, which need

' not be equidistant. It also provides the fitting para-

' meters, their standard deviations, and their ratios.

' The latter are convenient for determining which terms

' are statistically significant and should be retained.

' SUBROUTINES:

' This macro requires the subroutines Multiply, Invert,

' and Transpose

' INPUT:

' The input data must be arranged as follows. The first

' column must contain the dependent variable y. The second

' (and subsequent) column(s) must contain the independent

' variable(s) x.

' The subroutine requires an input parameter p: p = 1

' causes a general unweighted least squares fit to the

' data, while p = 0 forces the fit to pass through the

' origin, i.e., it assumes that Y = 0 for X = 0. The sub-

' routine Ortho is therefore called by a macro, OrthoI or

' OrthoO, that sets the value of p.

' OUTPUT:

' The macro produces its output to the right of the input,

' separated by one empty column. It therefore repeats the

' y column. It then computes the fitting parameters and

' their standard deviations. Since these are uncorrelated,

' a covariance matrix is neither needed nor provided. The

' standard deviation of the fit is shown directly below the

' label "Coeff:", as in LSI and LSO. A third row lists the

' ratio of the absolute values of the coefficient and the

' corresponding standard deviation.

' In general you will NOT want to use the listed coeffi-

' cients and standard deviations, which apply to the ortho-

' gonalized independent parameters. The most useful part of

' the output is the row listing their ratios, which indi-

' cates which orders should be considered (green if ratio >

' 5) and which should not be (red if ratio < 1).

' PROCEDURE:

' Before calling the macro, make sure that the output area

' (a block to the right of the input data block, but exten-

' ding one more column and three more rows) contains no

' valuable data, since these will be overwritten.

' In order to start the process, highlight the entire input

' data array, i.e., the rectangular block containing the

' column for y and the adjacent column(s) for the corres-

' ponding values of x, then call either Ortho(1) or

' Ortho(0)

' The function of the following two drivers is merely to

' set the value of one parameter, p, equal to either one or

' zero, in order to choose between a general least squares

' fitting (p = 1) or one that forces the curve through the

' origin (p = 0).

' EXAMPLES:

' Use of this macro is illustrated starting in sections

' 3.11 through 3.14 of AE3.

Sub OrthoO()    ' for unweighted LS fit through the origin

Dim P As Double

P = 0

Call Ortho(P)

End Sub

'''''''''''''''''''''''''''''''''''''''''''''''''''''''''''

Sub OrthoI()    ' for general unweighted least squares fit

Dim P As Double

P = 1

Call Ortho(P)

End Sub

'''''''''''''''''''''''''''''''''''''''''''''''''''''''''''

Sub Ortho(P)

Dim CMax As Integer, cN As Integer, j As Integer

Dim MM As Integer, N As Integer

Dim i As Long, m As Long, RMax As Long

Dim Num As Double, Resid As Double

Dim Root As Double, SSR As Double, StDevF As Double

Dim Sf As Double, SYZ As Double, SZZ As Double

Dim u As Double, varY As Double

Dim bArray As Variant, coeffArray As Variant

Dim DataArray As Variant, Denom As Variant

Dim lccArray As Variant, orthoArray As Variant

Dim OutputArray As Variant, stdevArray As Variant

Dim vArray As Variant, v0Array As Variant

Dim XArray As Variant, YArray As Variant

Dim Z As Variant

Dim myRange As Range

Dim Answer, hAnswer, iAnswer, kAnswer

' Determination of the array size:

Begin:

RMax = Selection.Rows.Count

CMax = Selection.Columns.Count

ReDim a(0 To CMax) As Double, s(0 To CMax) As Double

u = 1

' If area was not highlighted

If RMax = 1 And CMax = 1 Then

  hAnswer = MsgBox("You forgot to highlight" _

    & Chr(13) & "the block of input data." _

    & Chr(13) & "Do you want to do so now?" _

    , vbYesNo, "Ortho")

  If hAnswer = vbNo Then End

  If hAnswer = vbYes Then

    Set myRange = Application.InputBox(Prompt:= _

      "The input data are located in:", Type:=8)

    myRange.Select

  End If

  GoTo Begin

End If

' Check that the number of columns is at least 2:

If CMax < 2 Then

  MsgBox "There must be at least two columns," & _

    Chr(13) & "one for Y, and one or more for X.", _

    , "Orthogonalization"

  End

End If

' Check that rmax > cmax, so that the number of data

' points is sufficient to define the problem:

If RMax < CMax Then

  MsgBox "There must be at least " & CMax & " input" & _

    Chr(13) & "  data to define the problem.", _

    , "Orthogonalization"

  End

End If

' Dimension the arrays:

ReDim bArray(1 To CMax) As Double

ReDim Denom(1 To CMax)

ReDim vArray(1 To CMax, 1 To CMax) As Double

ReDim XArray(1 To RMax, 1 To CMax) As Double

ReDim YArray(1 To RMax) As Double

ReDim Z(1 To RMax, 1 To CMax) As Double

' Read the dataArray, then fill yArray and xArray.

DataArray = Selection.Value

For i = 1 To RMax

  YArray(i) = DataArray(i, 1)

Next i

For i = 1 To RMax

  If IsEmpty(DataArray(i, 1)) Then

    MsgBox "Y-value(s) missing", , "Orthogonalization"

    End

  End If

Next i

For j = 2 To CMax

  For i = 1 To RMax

    If IsEmpty(DataArray(i, j)) Then

      MsgBox "X-value(s) missing", , "Orthogonalization"

      End

    End If

  Next i

Next j

' Fill the first column of xArray with zeroes

' (for p = 0) or ones (for p = 1), the rest with

' the data in the x-column(s)

For i = 1 To RMax

  XArray(i, 1) = CDbl(P)

Next i

For j = 2 To CMax

  For i = 1 To RMax

    XArray(i, j) = DataArray(i, j)

  Next i

Next j

cN = 1

' Dimension and fill Z, the matrix of column vectors

' already transformed, initially Z is the 1st column of

' xArray if p = 1, the 2nd if p = 0

ReDim Z(1 To RMax, 1 To cN) As Double

' Check against overwriting valuable spreadsheet

' data by the output of orthogonal polynomials

Selection.Offset(0, CMax + 1).Select

orthoArray = Selection.Value

OutputArray = Selection.Value

m = 0

For i = 1 To RMax

  For j = 1 To CMax

    If IsEmpty(OutputArray(i, j)) Then

      m = m

    Else

      m = m + 1

    End If

  Next j

Next i

If m > 0 Then

  Answer = MsgBox("There are data in the highlighted " _

    & "output block to the" & Chr(13) & _

    "right of the input data array. Can they be " _

    & "overwritten? ", vbYesNo, "Overwrite?")

  If Answer = vbNo Then

    End

  End If

End If

' Check against overwriting valuable spreadsheet data

' by the output of coefficients and standard deviations

Selection.Offset(4, 0).Select

OutputArray = Selection.Value

MM = 0

For i = RMax - 1 To RMax

  For j = 1 To CMax

    If IsEmpty(OutputArray(i, j)) Then

      MM = MM

    Else

      MM = MM + 1

    End If

  Next j

Next i

If MM > 0 Then

  Answer = MsgBox("There are data in the two bottom " _

    & "lines of the " & Chr(13) & "highlighted array. " _

    & "Can they be overwritten? ", vbYesNo, "Overwrite?")

  If Answer = vbNo Then

    End

  End If

End If

Selection.Offset(-4, 0).Select

For i = 1 To RMax

  orthoArray(i, 1) = DataArray(i, 1)

Next i

Application.ScreenUpdating = False

Do

  For i = 1 To RMax

    For j = 1 To cN

      Z(i, cN) = XArray(i, j)

    Next j

  Next i

' Dimension and fill Zi, the next column vector to be

' transformed

  ReDim Zi(1 To RMax, 1 To 1) As Double

  For i = 1 To RMax

    Zi(i, 1) = XArray(i, cN + 1)

  Next i

' Dimension the other vectors and matrices

  ReDim Ztp(1 To cN, 1 To RMax) As Double     ' = Z'

  ReDim ZtpZ(1 To cN, 1 To cN) As Double      ' = Z' Z

  ReDim Z1(1 To cN, 1 To cN) As Double        ' = (Z' Z)"

  ReDim ZtpZi(1 To cN, 1 To 1) As Double      ' = Z' Zi

  ReDim Z2(1 To cN, 1 To 1) As Double

                              ' = Z1 ZtpZi = (Z' Z)" Z' Zi

  ReDim Z3(1 To RMax, 1 To 1) As Double

                              ' = Z Z2 = Z (Z' Z)" Z' Zi

' Compute Z1 = (Z' Z)"

  Call Transpose(Z, RMax, cN, Ztp)

  Call Multiply(Ztp, cN, RMax, Z, cN, ZtpZ)

  Call Invert(ZtpZ, cN, Z1)

  Call Multiply(Ztp, cN, RMax, Zi, 1, ZtpZi)

  Call Multiply(Z1, cN, cN, ZtpZi, 1, Z2)

  Call Multiply(Z, RMax, cN, Z2, 1, Z3)

' Dimension and compute Ztf = Zi - Z3, the transformed

' vector, orthogonal to the vector(s) in Z

  ReDim Ztf(1 To RMax, 1 To 1) As Double

                                ' = Zi - Z (Z' Z)" (Z' Zi)

  For i = 1 To RMax

    Ztf(i, 1) = Zi(i, 1) - Z3(i, 1)

  Next i

  SYZ = 0

  SZZ = 0

  For i = 1 To RMax

    SYZ = SYZ + YArray(i) * Ztf(i, 1)

    SZZ = SZZ + Ztf(i, 1) * Ztf(i, 1)

    orthoArray(i, cN + 1) = Ztf(i, 1)

  Next i

  If cN < CMax Then cN = cN + 1

  ReDim Preserve Z(1 To RMax, 1 To cN) As Double

Loop Until cN = CMax

' Display the orthogonal array

Selection.Value = orthoArray

' Update yArray and xArray.

For i = 1 To RMax

  YArray(i) = orthoArray(i, 1)

Next i

For i = 1 To RMax

  XArray(i, 1) = CDbl(P)

Next i

For j = 2 To CMax

  For i = 1 To RMax

    XArray(i, j) = orthoArray(i, j)

  Next i

Next j

' Compute the coefficients

For j = 2 - P To CMax

  Num = 0

  Denom(j) = 0

  For i = 1 To RMax

    Num = Num + XArray(i, j) * YArray(i)

    Denom(j) = Denom(j) + XArray(i, j) * XArray(i, j)

  Next i

  bArray(j) = Num / Denom(j)

Next j

' Compute the variance of y

SSR = 0

For i = 1 To RMax

  Resid = YArray(i)

  For j = 2 - P To CMax

    Resid = Resid - bArray(j) * XArray(i, j)

  Next j

  SSR = SSR + Resid * Resid

Next i

varY = SSR / (RMax - CMax - P + 1)

StDevF = Sqr(Abs(varY))

' Compute the variances of the coefficients

For j = 2 - P To CMax

  vArray(j, j) = varY / Denom(j)

Next j

ActiveCell.Offset(RMax, 0).Select

' Arrange for the data output

For j = 1 To CMax

  With ActiveCell.Font

    .Bold = True

    .Italic = True

    .ColorIndex = 1

  End With

  ActiveCell.Offset(0, 1).Select

Next j

ActiveCell.Offset(1, -CMax).Select

For j = 1 To CMax

  ActiveCell.Font.Italic = True

  ActiveCell.Font.ColorIndex = 1

  ActiveCell.Offset(0, 1).Select

Next j

ActiveCell.Offset(1, -CMax).Select

For j = 1 To CMax

  ActiveCell.Font.Italic = True

  ActiveCell.Font.ColorIndex = 1

  ActiveCell.Offset(0, 1).Select

Next j

ActiveCell.Offset(-2, -CMax - P).Select

With ActiveCell

  .Font.Bold = True

  .Font.Italic = True

  .Font.ColorIndex = 1

  .HorizontalAlignment = xlRight

  .Value = "Coeff:"

End With

ActiveCell.Offset(0, 1).Select

For j = 2 - P To CMax

  With ActiveCell.Font

    .Bold = True

    .Italic = True

    .ColorIndex = 1

  End With

  ActiveCell.Value = bArray(j)

  ActiveCell.Offset(0, 1).Select

Next j

ActiveCell.Offset(1, -CMax - P).Select

With ActiveCell.Font

  .Bold = False

  .Italic = True

  .ColorIndex = 1

End With

ActiveCell.HorizontalAlignment = xlRight

ActiveCell.Value = "StDev:"

ActiveCell.Offset(0, 1).Select

ActiveCell.Font.ColorIndex = 1

For j = 2 - P To CMax

  If vArray(j, j) < 1E-40 Then

    ActiveCell.Value = "<1E-20"

  Else

    ActiveCell.Value = Sqr(Abs(vArray(j, j)))

  End If

  ActiveCell.Offset(0, 1).Select

Next j

ActiveCell.Offset(1, -CMax - P).Select

With ActiveCell

  .Font.Bold = False

  .Font.Italic = True

  .Font.ColorIndex = 1

  .HorizontalAlignment = xlRight

  .Value = "Ratio:"

End With

ActiveCell.Offset(0, 1).Select

For j = 2 - P To CMax

  If vArray(j, j) <= 0 Then

    ActiveCell.Value = "N/A"

    ActiveCell.Font.ColorIndex = 1

  End If

  If vArray(j, j) > 0 Then

    ActiveCell.Font.Bold = False

    ActiveCell.Font.Italic = True

    ActiveCell.Font.ColorIndex = 16

    ActiveCell.Value = Abs(bArray(j) _

      / Sqr(vArray(j, j)))

    If ActiveCell.Value > 5 Then _

      ActiveCell.Font.ColorIndex = 10

    If ActiveCell.Value < 1 Then _

      ActiveCell.Font.ColorIndex = 3

  End If

  ActiveCell.Offset(0, 1).Select

Next j

ActiveCell.Offset(1, -CMax).Select

' Provide the standard deviation of the fit

ActiveCell.Offset(0, 1 - P).Select

With ActiveCell.Font

  .Bold = False

  .Italic = True

  .ColorIndex = 1

End With

Selection.Value = StDevF

End Sub

'''''''''''''''''''''''''''''''''''''''''''''''''''''''''''

'''''''''''''''''''''''''''''''''''''''''''''''''''''''''''

''''''''''''***********************************''''''''''''

''''''''''''*                                 *''''''''''''

''''''''''''*      Direct (De)Convolution     *''''''''''''

''''''''''''*                                 *''''''''''''

''''''''''''***********************************''''''''''''

''''''''''''''''''''''''''''''''''''''''''' (c) R. de Levie

'''''''''''''''''''''''''''''''''''''''''''  v 12, May 2012

' PURPOSE:

' These macros perform a straightforward (i.e., non-Fourier

' transform) convolution or deconvolution of 2 columns of

' real data, arranged in a single block of 3 adjacent

' columns, the first column containing the variable (time),

' the second column the signals to be convolved, and the

' third column the transfer function. The signal in the

' third column is normalized to have an average of 1.

' The number of points in the data set is arbitrary, but

' the data must be equidistant, since the signal and trans-

' fer functions will be sliding past each other.

' Convolution tends to spread the function. Make sure that

' there is enough space for it to spread into, e.g., by ad-

' ding zeros before and/or after the signal for expansion.

' Likewise, for a symmetrical transfer function, signal

' shifting can be prevented by centering the transfer

' function at t = 0.

' The deconvolution routine works well when the deconvol-

' ving function (in the third column) is a monotonically

' decreasing function of time, but otherwise may lead to

' oscillations.

' SUBROUTINES:

' The macros Convolve and Deconvolve set the value of d

' to +1 (for convolution) or -1 (for deconvolution), then

' call the subroutine Convolution. The latter requires no

' further subroutines.

' INPUT:

' The macro accepts any number of input data, but requires

' that they be equidistant. The output is independent of

' the starting value of the time scale, but does depend on

' the phase relationship between the two input signals.

' OUTPUT:

' The output is written in one column, immediately to the

' right of the input data block, thereby overwriting any

' prior data in that region. After writing the output, the

' activated area returns to its original position and

' contents.

' PROCEDURE:

' Before starting the macro, make sure that the column to

' the right of the input data block contains no valuable

' data.

' To initiate the transform, highlight the three columns of

' the input array, and call the macro.

' EXAMPLE:

' Use of this macro is illustrated in sections 6.2 and 6.3

' of Advanced Excel.

'''''''''''''''''''''''''''''''''''''''''''''''''''''''''''

Sub Convolve()

Dim D As Integer

D = 1

Call Convolution(D)

End Sub

'''''''''''''''''''''''''''''''''''''''''''''''''''''''''''

Sub Deconvolve()

Dim D As Integer

D = -1

Call Convolution(D)

End Sub

'''''''''''''''''''''''''''''''''''''''''''''''''''''''''''

Sub Convolution(D)

Dim CMax As Integer, rM As Integer, RMax As Integer

Dim i As Integer, j As Integer, N As Integer

Dim Incr As Double, mySum As Double

Dim Sum As Double, Z As Double

Dim Inp As Variant, myInput As Variant

Dim myInput2 As Variant, myOutput As Variant

Dim Outp As Variant, Trans As Variant

Dim Ans, hAnswer

' Read the input

Begin:

CMax = Selection.Columns.Count

RMax = Selection.Rows.Count

' If area was not highlighted

If RMax = 1 And CMax = 1 Then

  hAnswer = MsgBox("You forgot to highlight" _

    & Chr(13) & "the block of input data." _

    & Chr(13) & "Do you want to do so now?" _

    , vbYesNo, "(De)Convolution")

  If hAnswer = vbNo Then End

  If hAnswer = vbYes Then

    Set myRange = Application.InputBox(Prompt:= _

      "The input data are located in:", Type:=8)

    myRange.Select

  End If

  GoTo Begin

End If

' Check that there are three input columns

If CMax <> 3 Then

  MsgBox "There must be three input columns," _

    & Chr(13) & "one each for time, signal s, and" _

    & Chr(13) & "transfer function t, in that order."

  End

End If

' Read the input data

myInput = Selection.Value

myInput2 = Selection.Formula

Incr = myInput(2, 1) - myInput(1, 1)

rM = RMax - 1

ReDim Inp(-rM To rM)

ReDim Outp(-rM To rM)

ReDim Trans(-rM To rM)

' Check whether the fourth column can be overwritten

N = 0

Selection.Offset(0, 1).Select

myOutput = Selection.Value

For i = 1 To RMax

  Z = myOutput(i, 3)

  If (IsEmpty(Z) Or Z = 0) Then

    N = N

  Else

    N = N + 1

  End If

Next i

If N > 0 Then

  Ans = MsgBox("There are data in the space where" _

  & Chr(13) & "the output will be written. Proceed" _

  & Chr(13) & "anyway and overwrite those data?", vbYesNo)

  If Ans = vbNo Then

    Selection.Offset(0, -1).Select

    End

  End If

End If

Selection.Offset(0, -1).Select

' Normalize and invert the transfer function

Sum = 0

For i = 1 To RMax

  Sum = Sum + myInput(i, 3)

Next i

For i = 1 To RMax

  Inp(1 - i) = 0

  Inp(i - 1) = myInput(i, 2)

  Trans(i - 1) = 0

  Trans(1 - i) = myInput(i, 3)

  Outp(1 - i) = 0

  Outp(i - 1) = 0

Next i

' Compute the convolution (for d = 1)

If D = 1 Then

  For j = 0 To rM

    mySum = 0

    For i = 0 To rM

      mySum = mySum + Inp(j - i) * Trans(-i)

    Next i

  Outp(j) = mySum / Sum

  myOutput(j + 1, 3) = Outp(j)

  Next j

End If

' Compute the deconvolution (for d = -1)

If D = -1 Then

  For j = 0 To rM

    mySum = Inp(j)

    For i = 1 To rM

      mySum = mySum - Outp(j - i) * Trans(-i)

    Next i

  Outp(j) = mySum

  Next j

  For j = 1 To RMax

    myOutput(j, 3) = Outp(j - 1) * Sum

  Next j

End If

' Output the result

Selection.Offset(0, 1).Select

Selection.Value = myOutput

Selection.Offset(0, -1).Select

Selection.Formula = myInput2

End Sub

'''''''''''''''''''''''''''''''''''''''''''''''''''''''''''

'''''''''''''''''''''''''''''''''''''''''''''''''''''''''''

'''''''*********************************************'''''''

'''''''*                                           *'''''''

'''''''*      Fourier Transform (De)Convolution    *'''''''

'''''''*                                           *'''''''

'''''''*********************************************'''''''

''''''''''''''''''''''''''''''''''''''''''' (c) R. de Levie

'''''''''''''''''''''''''''''''''''''''''''  v 12, May 2012

' PURPOSE:

' These macros perform either a convolution or a deconvolu-

' tion of two sets of real data, arranged in a single block

' of three adjacent columns, the first column containing

' their common variable (e.g., time), and the next columns

' the data sets to be convolved or deconvolved.

' For convolution the order of the second and third column

' does not matter, and interchanging the data sets in the

' second and third column will yield the same result.

' For deconvolution, the second column should contain the

' data to be deconvolved, and the third column the data set

' with which the data in the second column are to be decon-

' volved. In this case, exchanging the second and third

' column will in general yield quite different results.

' For deconvolution the macro incorporates an adjustable

' von Hann window with a single, continually adjustable

' positive parameter, w. The default, w = 0, yields a

' rectangular window (i.e., the absence of filtering), but

' allows the user to select a sharp high-frequency cut-off.

' Setting w = 1 selects the traditional von Hann window,

' while other w-values >0 produce filtering that is less

' severe (0 < w < 1) or more severe (w > 1) than the stan-

' dard von Hann filter. Note that the filters have not been

' normalized.

' SUBROUTINES:

' The macros Convolve and Deconvolve require the subroutine

' Convolution which, in turn, calls the subroutine FT.

' INPUT:

' The macro accepts any input format in which the time

' scale contains 2^n equidistant values (where n is a posi-

' tive integer larger than 1), regardless of its starting

' value. While the output is independent of the starting

' value of the time scale, it does depend on the phase

' relationship between the two input signals.

' Provided that the function in the third column is symme-

' trical, such phase shifts can be avoided by starting that

' function at its point of symmetry (typically: at its

' maximum), and placing the part before that point of

' symmetry at the very end of the column. What makes this

' 'wrap-around' method work is the assumption of the

' Fourier transform method that the functions are self-

' repeating.

' OUTPUT:

' The output is written in one column, immediately to the

' right of the input data block, thereby overwriting any

' prior data in that region. After writing the output,

' the activated area returns to its original position and

' contents.

' PROCEDURE:

' Before starting the macro, make sure that the column to

' the right of the input data block contains no valuable

' data.

' To initiate the transform, highlight the three columns

' of the input array, and call the macro.

' EXAMPLE:

' Use of this macro is illustrated in section 6.5 of AE3.

'''''''''''''''''''''''''''''''''''''''''''''''''''''''''''

' The following are the two drivers for the (de)convolution

' subroutine. They set kSign to either +1 (for convolution)

' or -1 (for deconvolution).

Sub ConvolveFT()

Dim kSign As Integer

kSign = 1

Call ConvolutionFT(kSign)

End Sub

'''''''''''''''''''''''''''''''''''''''''''''''''''''''''''

Sub DeconvolveFT()

Dim kSign As Integer

kSign = -1

Call ConvolutionFT(kSign)

End Sub

'''''''''''''''''''''''''''''''''''''''''''''''''''''''''''

Sub ConvolutionFT(kSign)

Dim cN As Integer, cnMax As Integer

Dim N As Integer, NN As Integer, nfZ As Integer

Dim rN As Integer, rnMax As Integer, rrn As Integer

Dim nfZero As Single, Q As Single

Dim Average3 As Double, Check1 As Double, Check2 As Double

Dim D As Double, dataSpacing As Double, Length As Double

Dim Min As Double, Norm As Double, Pi As Double

Dim Ratio As Double, Rnm2 As Double, s As Double

Dim Sum As Double, Sum3 As Double

Dim WindowFunction As Double, Z As Double

Dim DataArray As Variant, InputArray As Variant

Dim OutputArray As Variant

Dim Answer, hAnswer, vAnswer, Default1, Default2

Dim Message1, Message2, Title1, Title2

' Read the input

Begin:

N = 0

Z = 0

rnMax = Selection.Rows.Count

NN = 2 * rnMax

cnMax = Selection.Columns.Count

Length = CDbl(rnMax)

' If area was not highlighted

If RMax = 1 And CMax = 1 Then

  hAnswer = MsgBox("You forgot to highlight" _

    & Chr(13) & "the block of input data." _

    & Chr(13) & "Do you want to do so now?" _

    , vbYesNo, "Fourier transform (de)convolution")

  If hAnswer = vbNo Then End

  If hAnswer = vbYes Then

    Set myRange = Application.InputBox(Prompt:= _

      "The input data are located in:", Type:=8)

    myRange.Select

  End If

  GoTo Begin

End If

' Check the array length n, which must be a power of 2,

' and be at least 2

If Length < 2 Then

  MsgBox "There must be at least two rows."

  End

End If

Do While Length > 1

  Length = Length / 2

Loop

If Length <> 1 Then

  MsgBox "The number of rows must be a power of two."

  End

End If

' Check that there are three input columns

cnMax = Selection.Columns.Count

If cnMax <> 3 Then

  If kSign = 1 Then

    MsgBox " There must be 3 input columns," _

      & Chr(13) & "one for the independent variable" _

      & Chr(13) & " (e.g., time), the next for the" _

      & Chr(13) & " two functions to be convolved."

    End

  End If

  If kSign = -1 Then

    MsgBox " There must be 3 input columns," _

      & Chr(13) & "one for the independent variable" _

      & Chr(13) & " (e.g., time), the next for the" _

      & Chr(13) & " two functions to be deconvolved."

    End

  End If

End If

' Read the input data

DataArray = Selection.Value

InputArray = Selection.Formula

' Check that the first column has its first two elements

Check1 = VarType(DataArray(1, 1))

If Check1 = 0 Then

  MsgBox "Enter the top left value."

  End

End If

Check2 = VarType(DataArray(2, 1))

If Check2 = 0 Then

  MsgBox "Enter a value in row 2 of the first column."

  End

End If

dataSpacing = DataArray(2, 1) - DataArray(1, 1)

' Read and rearrange the input data

' from the second and third columns

ReDim Term2(NN) As Double, Term3(NN) As Double

ReDim Term4(NN) As Double

For rN = 1 To rnMax

  Term2(2 * rN - 1) = DataArray(rN, 2)

  Term2(2 * rN) = Z

  Term3(2 * rN - 1) = DataArray(rN, 3)

  Term3(2 * rN) = Z

Next rN

' Normalize the data in the third column

Sum = 0

For rN = 1 To rnMax

  Sum = Sum + Term3(2 * rN - 1)

Next rN

For rN = 1 To rnMax

' If normalization is NOT desired,

' comment out the next line:

  Term3(2 * rN - 1) = Term3(2 * rN - 1) * rnMax / Sum

Next rN

' Check that the output does not overwrite valuable data.

Selection.Offset(0, 1).Select

OutputArray = Selection.Value

For rN = 1 To rnMax

  Q = OutputArray(rN, 3)

  If (IsEmpty(Q) Or Q = 0) Then

    N = N

  Else

    N = N + 1

  End If

Next rN

Selection.Offset(0, -1).Select

If N > 0 Then

  Answer = MsgBox(" There are data in the column where" _

    & Chr(13) & "the output will be written. Proceed" & _

    Chr(13) & " anyway and overwrite those data?", vbYesNo)

  If Answer = vbNo Then End

End If

' Transform the data from the second and third columns

Call FT(Term2, NN, 1)

Call FT(Term3, NN, 1)

' Multiply in the frequency domain for convolution, or

' divide in the frequency domain for deconvolution. For

' deconvolution, a minimum value "min" (arbitrarily set

' here to 1E-6) prevents division by zero. Moreover, an

' input box is provided for optional noise filtering.

Rnm2 = CDbl(rnMax) * CDbl(rnMax)

' For convolution, multiply in the frequency domain:

' (a + jb) (c + jd) = (ac - bd) + j(ad + bc)

If kSign = 1 Then

  For rN = 1 To rnMax

    Term4(2 * rN - 1) = _

      (Term2(2 * rN - 1) * Term3(2 * rN - 1) _

      - Term2(2 * rN) * Term3(2 * rN)) / rnMax

    Term4(2 * rN) = _

      (Term2(2 * rN - 1) * Term3(2 * rN) _

      + Term2(2 * rN) * Term3(2 * rN - 1)) / rnMax

  Next rN

End If

' For deconvolution, select the window parameter s

If kSign = -1 Then

 Min = 0.000001

 Pi = [Pi()]

 Message1 = "Enter the window parameter, a " & _

   "non-negative number." & Chr(13) & "The default " & _

   "is 0, which corresponds to no filtering," & Chr(13) & _

   "but allows you to insert a sharp frequency cut-off." _

   & Chr(13) & Chr(13) & "For the standard von Hann " _

   & "filter select 1." & Chr(13) & Chr(13) _

   & "For an adjustable von Hann / Tukey window," _

   & Chr(13) & " select a value other than 0 or 1."

  Title1 = "Adjustable von Hann / Tukey window"

  Default1 = "0"

  s = InputBox(Message1, Title1, Default1)

  If s < Min Then s = 0

' Provide the option of removing

' high-frequency data for M1 = 0

  If s = 0 Then

    Message2 = "Enter the number of frequencies" _

      & Chr(13) & "(an integer smaller than " _

      & rnMax & ") for which" & Chr(13) _

      & "you want the signal to be set to zero." & _

      Chr(13) & Chr(13) & "Press ENTER or OK for the " _

      & "default, 0," & Chr(13) & "in which case no " _

      & "data will be set to zero."

    Title2 = "Zeroing out high frequencies"

    Default2 = "0"

    nfZero = InputBox(Message2, Title2, Default2)

    If nfZero < 0 Then nfZ = 0

    If nfZero > 0 And nfZero < rnMax / 2 Then _

      nfZ = CInt(nfZero)

    If nfZero > rnMax / 2 Then nfZ = rnMax / 2

    If nfZ > 0 Then

      For rN = (rnMax / 2 + 2 - nfZ) To (rnMax / 2 + nfZ)

        Term4(2 * rN - 1) = 0

        Term4(2 * rN - 2) = 0

      Next rN

    End If

  End If

' For deconvolution, divide in the frequency domain:

' (a+jb)/(c+jd) = (a+jb)(c-jd)/(cc+dd)

'   = [(ac+bd)+j(bc-ad)]/(cc+dd)

  For rN = 1 To rnMax

    D = (Term3(2 * rN - 1) * Term3(2 * rN - 1) + _

      Term3(2 * rN) * Term3(2 * rN)) * dataSpacing / rnMax

    If D < Min Then D = Min

    Term4(2 * rN - 1) = _

      (Term2(2 * rN - 1) * Term3(2 * rN - 1) _

      + Term2(2 * rN) * Term3(2 * rN)) / D

    Term4(2 * rN) = (Term2(2 * rN) * Term3(2 * rN - 1) _

      - Term2(2 * rN - 1) * Term3(2 * rN)) / D

  ' Compute and apply the window function

    Ratio = 2 * CDbl(rN - 1) / CDbl(rnMax)

    If s < 1 Then WindowFunction = 1

    If s > 0 And s < 1 Then

      If Abs(Ratio - 1) < s Then WindowFunction = _

        0.5 + 0.5 * Cos(Pi * (Abs(Ratio) - 1 + s) / s)

    End If

    If s >= 1 Then

      WindowFunction = (0.5 + 0.5 * Cos(Pi * Ratio)) ^ s

    End If

    Term4(2 * rN - 1) = Term4(2 * rN - 1) * WindowFunction

    Term4(2 * rN) = Term4(2 * rN) * WindowFunction

  Next rN

End If

' Calculate output data

Call FT(Term4, NN, -1)

' Arrange and write the output data

For rN = 1 To rnMax

  OutputArray(rN, 3) = Term4(2 * rN - 1) / rnMax

Next rN

Application.ScreenUpdating = False

Selection.Offset(0, 1).Select

Selection.Value = OutputArray

Selection.Offset(0, -1).Select

Selection.Value = InputArray

End Sub

'''''''''''''''''''''''''''''''''''''''''''''''''''''''''''

'''''''''''''''''''''''''''''''''''''''''''''''''''''''''''

'''''''''''************************************''''''''''''

'''''''''''*                                  *''''''''''''

'''''''''''*      Iterative Deconvolution     *''''''''''''

'''''''''''*                                  *''''''''''''

'''''''''''************************************''''''''''''

''''''''''''''''''''''''''''''''''''''''''' (c) R. de Levie

'''''''''''''''''''''''''''''''''''''''''''  v 12, May 2012

' PURPOSE:

' This macro performs an iterative deconvolution along the

' lines of the van Cittert approach. It convolves (rather

' than deconvolves) the data in the second (signal) column

' with the filter function in the third column, then

' corrects the signal by the difference between it and its

' convolution. The macro provides the option of smoothing

' intermediate results with a moving quadratic of selec-

' table length, using the recursive expressions given by

' Gorry in Anal. Chem. 62 (1990) 570. The iteration stops

' when the sum of the squares of the residuals has reached

' a minimum, or the number of iterations has reached a pre-

' set limit.

' There are two choices: DeconvolveIt0 involves no

' constraints, while DeconvolveIt1 assumes one constraint,

' namely that the function is always positive. It therefore

' replaces all negative intermediate results by zero.

' INPUT:

' The input requires three adjacent columns of equal

' length. The first column should contain the independent

' parameter, such as time, wavelength, etc. The second

' column should contain the (real) input data that need to

' be deconvolved with the (real) filter data in the third

' column.

' The macro accepts any number of input data, but requires

' that they be equidistant. The output is independent of

' the starting value of the time scale, but does depend on

' the phase relationship between the two input signals.

' An optional parabolic noise filter is provided to reduce

' the effect of noise. Use of such a filter introduces its

' own distortion, the more so the longer the length of the

' moving parabola. If used at all, the parabola length

' should therefore be kept to a minimum.

' INTERRUPT:

' Because the iteration can take more time than you may

' have available or care to spend, the process can be

' interrupted, by pressing either the Escape key or the

' Control+Break key combination. A message box then

' provides the option of ending the computation and

' displaying the result obtained so far, or of resuming

' the calculation where it was interrupted.

' OUTPUT:

' The output is written in one column, immediately to the

' right of the input data block, thereby overwriting any

' prior data in that region. After writing the output,

' the activated area returns to its original position and

' contents. Moreover, the number of iterations used, and

' the value of the termination criterion are shown at the

' bottom of the data set.

' SUBROUTINES AND FUNCTIONS:

' The macros DeconvIt0 and DeconvolveIt1 require the

' subroutine DeconvolveIteratively which, in turn, calls

' the subroutine FT. Moreover, when filtering is used,

' the functions ConvolutionWeight, GramPolynomial, and

' GenFact are also required.

' PROCEDURE:

' Before starting the macro, make sure that the column to

' the right of the input data block contains no valuable

' data.

' To initiate the macro, highlight the three columns of the

' input array, and call the macro. The macro will then ask

' you to specify the maximum number of iterations; its

' default value is 10, and its maximum value has been set

' at 100,000. (If that takes you too long, interrupt the

' iterations with Esc.) The iterations will stop whenever

' the sum of the squares of the residuals has reached a

' minimum, or the set maximum number of iterations has been

' performed, whichever comes first.

' EXAMPLE:

' Use of this macro is illustrated in section 6.6 of AE3.

'''''''''''''''''''''''''''''''''''''''''''''''''''''''''''

' For iterative deconvolution without constraints:

Sub DeconvolveIt0()

Dim zz As Integer

zz = 0

Call DeconvolveIteratively(zz)

End Sub

'''''''''''''''''''''''''''''''''''''''''''''''''''''''''''

' For iterative deconvolution constrained at x = 0:

Sub DeconvolveIt1()

Dim zz As Integer

zz = 1

Call DeconvolveIteratively(zz)

End Sub

'''''''''''''''''''''''''''''''''''''''''''''''''''''''''''

Sub DeconvolveIteratively(zz)

Dim C As Integer, cN As Integer, cnMax As Integer

Dim ItNo As Integer, N As Integer

Dim NN As Integer, nfZ As Integer

Dim rN As Integer, rnMax As Integer, rrn As Integer

Dim nfZero As Single, Q As Single

Dim Average3 As Double, Check1 As Double, Check2 As Double

Dim D As Double, dataSpacing As Double, Delta As Double

Dim Factor As Double, Length As Double, MaxItNo As Double

Dim Min As Double, myScale As Double, Norm As Double

Dim P As Double, Pi As Double, previousSSR As Double

Dim Ratio As Double, Rnm2 As Double, SSR As Double

Dim s As Double, Sum As Double

Dim Sum3 As Double, WindowFunction As Double, Z As Double

Dim DataArray As Variant, InputArray As Variant

Dim OutputArray As Variant

Dim aA, bb, cc, Answer, hAnswer, vAnswer, ContinueHere

Dim Default, InterruptHandler, Message, myFilter

Dim myFunction, Reply, Title, TryAgain

' Read the input

Begin:

N = 0

Z = 0

rnMax = Selection.Rows.Count

NN = 2 * rnMax

cnMax = Selection.Columns.Count

Length = CDbl(rnMax)

Pi = [Pi()]

' If area was not highlighted

If rnMax = 1 And cnMax = 1 Then

  hAnswer = MsgBox("You forgot to highlight" _

    & Chr(13) & "the block of input data." _

    & Chr(13) & "Do you want to do so now?" _

    , vbYesNo, "Iterative (de)convolution")

  If hAnswer = vbNo Then End

  If hAnswer = vbYes Then

    Set myRange = Application.InputBox(Prompt:= _

      "The input data are located in:", Type:=8)

    myRange.Select

  End If

  GoTo Begin

End If

On Error GoTo InterruptHandler

' Check the array length n, which must

' be a power of 2, and be at least 2

If Length < 2 Then

  MsgBox "There must be at least two rows."

  End

End If

Do While Length > 1

  Length = Length / 2

Loop

If Length <> 1 Then

  MsgBox "The number of rows must be a power of two."

  End

End If

' Check that there are three input columns

If cnMax <> 3 Then

  MsgBox "There must be three input columns, one for " _

    & "the" & Chr(13) & "   independent variable (e.g., " _

    & "time), the next" & Chr(13) _

    & "   two for the two functions to be deconvolved."

  End

End If

' Read the input data

DataArray = Selection.Value

InputArray = Selection.Formula

' Check that the first column has its first two elements

Check1 = VarType(DataArray(1, 1))

If Check1 = 0 Then

  MsgBox "Enter the top left value."

  End

End If

Check2 = VarType(DataArray(2, 1))

If Check2 = 0 Then

  MsgBox "Enter a value in row 2 of the first column."

  End

End If

dataSpacing = DataArray(2, 1) - DataArray(1, 1)

' Check that the output does not overwrite valuable data.

Selection.Offset(0, 1).Select

OutputArray = Selection.Value

For rN = 1 To rnMax

  Q = OutputArray(rN, 3)

  If (IsEmpty(Q) Or Q = 0) Then

    N = N

  Else

    N = N + 1

  End If

Next rN

Selection.Offset(0, -1).Select

If N > 0 Then

  Answer = MsgBox("There are data in the column where" & _

    Chr(13) & "the output will be written. Proceed" & _

    Chr(13) & " anyway and overwrite those data?", vbYesNo)

  If Answer = vbNo Then End

End If

' Select a filter

TryAgain:

Message = "You can select a moving parabola to filter" & _

Chr(13) & "the data. Specify a moving parabola with an" & _

Chr(13) & "odd integer number of data points, at least" & _

Chr(13) & "5 but always much smaller than the number" & _

Chr(13) & "of points in your data set, or select the" & _

Chr(13) & "default, 0, if you do NOT want any filtering."

Title = "Deconvolve Iteratively: Select moving polynomial length"

Default = "0"

myFilter = InputBox(Message, Title, Default)

If myFilter = 0 Then GoTo aA

m = CInt((myFilter - 1) / 2)

If (m <> 0 And m < 2) Then

  MsgBox "Select either 0 (for no filtering)," & _

  Chr(13) & "or an odd number larger than 3 for" & _

  Chr(13) & "the length of the moving parabola."

  GoTo TryAgain

End If

If rnMax < 2 * m + 3 Then

  MsgBox "Select a shorter moving parabola length."

  GoTo TryAgain

End If

aA:

' Save the input signal

ReDim signal(1 To rnMax) As Double

ReDim newSignal(1 To rnMax) As Double

For rN = 1 To rnMax

  signal(rN) = DataArray(rN, 2)

Next rN

' Read and rearrange the input data from the third column

ReDim Term3(1 To NN) As Double

For rN = 1 To rnMax

  Term3(2 * rN - 1) = DataArray(rN, 3)

  Term3(2 * rN) = Z

Next rN

' Normalize the data in the third column

Sum = 0

For rN = 1 To rnMax

  Sum = Sum + DataArray(rN, 3)

Next rN

myScale = rnMax / Sum

' Select the maximum number of iterations; the

' default value is 10, and its maximum value is

' limited at 10000 to prevent an endless loop.

Message = "Select the maximum number of iterations"

Title = _

  "Deconvolve Iteratively: Select max. # of iterations"

Default = "10"

MaxItNo = InputBox(Message, Title, Default)

If MaxItNo > 10000 Then MaxItNo = 10000

' Make code uninterruptible for next section

Application.EnableCancelKey = xlDisabled

' Set the iteration number and start the iteration loop

ItNo = 1

bb:

previousSSR = SSR

' Read and rearrange the input data from the second column

ReDim Term2(NN) As Double, Term4(NN) As Double

For rN = 1 To rnMax

  Term2(2 * rN - 1) = DataArray(rN, 2)

  Term2(2 * rN) = Z

  Term3(2 * rN - 1) = DataArray(rN, 3) * myScale

  Term3(2 * rN) = Z

Next rN

' Transform the data from the second and third columns

Call FT(Term2, NN, 1)

Call FT(Term3, NN, 1)

' Multiply in the frequency domain for convolution

Rnm2 = CDbl(rnMax) * CDbl(rnMax)

' Multiply in the frequency domain, using

' (a + jb) * (c + jd) = (ac - bd) + j(ad + bc)

' where j is the square root of -1

For rN = 1 To rnMax

  Term4(2 * rN - 1) = _

    (Term2(2 * rN - 1) * Term3(2 * rN - 1) _

    - Term2(2 * rN) * Term3(2 * rN)) / rnMax

  Term4(2 * rN) = (Term2(2 * rN - 1) * Term3(2 * rN) _

    + Term2(2 * rN) * Term3(2 * rN - 1)) / rnMax

Next rN

' Inverse transform to obtain the convolution

Call FT(Term4, NN, -1)

For rN = 1 To rnMax

  OutputArray(rN, 3) = Term4(2 * rN - 1) / rnMax

Next rN

' Compute the corrected signal value

If zz < 2 Then

  For rN = 1 To rnMax

    P = 1

    If zz = 1 And signal(rN) < 0 Then P = 0

    newSignal(rN) = P * signal(rN) + _

      DataArray(rN, 2) - OutputArray(rN, 3)

  Next rN

ElseIf zz = 2 Then

  For rN = 1 To rnMax

    newSignal(rN) = signal(rN) + (1 - 2 * _

      Abs(signal(rN) - 0.5)) * (DataArray(rN, 2) - _

      OutputArray(rN, 3))

  Next rN

End If

' Compute SSR

SSR = 0

For rN = 1 To rnMax

  SSR = SSR + (OutputArray(rN, 3) - signal(rN)) ^ 2

Next rN

If myFilter = 0 Then GoTo cc

' Apply the filter

ReDim SmoothedArray(rnMax) As Double

For rN = 1 To rnMax

  SmoothedArray(rN) = DataArray(rN, 2)

Next rN

k = 2           ' for a parabola

s = 0           ' for smoothing

For rN = 1 To m

  SmoothedArray(rN) = 0

  For i = -m To m

    SmoothedArray(rN) = SmoothedArray(rN) _

    + ConvolutionWeight(i, rN - m - 1, m, k, s) _

    * DataArray(i + m + 1, 2)

  Next i

Next rN

For rN = m + 1 To rnMax - m

  SmoothedArray(rN) = 0

  For i = -m To m

    SmoothedArray(rN) = SmoothedArray(rN) _

    + ConvolutionWeight(i, 0, m, k, s) _

    * DataArray(rN + i, 2)

  Next i

Next rN

For rN = rnMax - m + 1 To rnMax

  SmoothedArray(rN) = 0

  For i = -m To m

    SmoothedArray(rN) = SmoothedArray(rN) _

    + ConvolutionWeight(i, rN - rnMax + m, m, k, s) _

    * DataArray(rnMax - m + i, 2)

  Next i

Next rN

For rN = 1 To rnMax

  DataArray(rN, 2) = SmoothedArray(rN)

Next rN

cc:

' Set negative values equal to zero for DeconvolveIt1

If zz = 1 Then

  For rN = 1 To rnMax

    If DataArray(rN, 2) < 0 Then DataArray(rN, 2) _

      = -DataArray(rN, 2)

  Next rN

End If

ContinueHere:

' Make code briefly interruptible

Application.EnableCancelKey = xlErrorHandler

' Update the second (signal) column of dataArray

For rN = 1 To rnMax

  DataArray(rN, 2) = signal(rN) _

    + DataArray(rN, 2) - OutputArray(rN, 3)

Next rN

' increment ItNo and loop back

If ItNo = 1 Then previousSSR = SSR

ItNo = ItNo + 1

Application.StatusBar = "Iteration # " & _

  ItNo & ", SSR = " & SSR

If (ItNo < MaxItNo And previousSSR >= SSR) Then GoTo bb

Output:

' Prepare for landing

For rN = 1 To rnMax

  OutputArray(rN, 3) = signal(rN) _

    + DataArray(rN, 2) - OutputArray(rN, 3)

Next rN

If zz > 0 Then

  For rN = 1 To rnMax

    If OutputArray(rN, 3) < 0 Then _

      OutputArray(rN, 3) = 0

  Next rN

End If

' Summarize the parameters used

' If you do not want a reminder of the parameters chosen,

' and a display of the final score on the status bar, place

' the apostrophe before the word MsgBox

MsgBox "The program DeconvolveIt" & zz & "," & _

  Chr(13) & "using a " & myFilter & _

  "-point parabolic filter," & Chr(13) & _

  "found an answer after " & ItNo & " iterations" & _

  Chr(13) & "with an SSR value of " & SSR & "."

Application.StatusBar = False

Application.ScreenUpdating = False

Selection.Offset(0, 1).Select

Selection.Value = OutputArray

Selection.Offset(0, -1).Select

Selection.Value = InputArray

End

InterruptHandler:

If Err.Number = 18 Then

  Reply = MsgBox("Do you want to terminate the" & _

    vbCr & "computation? If so, press Yes", vbYesNo, _

    "DeconvolveIt:   Stop iteration?")

  If Reply = vbYes Then GoTo Output

  If Reply = vbNo Then

    Resume ContinueHere

  End If

End If

End Sub

'''''''''''''''''''''''''''''''''''''''''''''''''''''''''''

Function ConvolutionWeight(i, t, m, k, s) As Double

Dim H As Integer

Dim Sum As Double

Sum = 0

For H = 0 To k

  Sum = Sum + (2 * H + 1) * (GenFact(2 * m, H) _

  / GenFact(2 * m + H + 1, H + 1)) _

  * GramPolynomial(i, m, H, 0) _

  * GramPolynomial(t, m, H, s)

Next H

ConvolutionWeight = Sum

End Function

'''''''''''''''''''''''''''''''''''''''''''''''''''''''''''

Function GramPolynomial(i, m, k, s) As Double

If k > 0 Then

  GramPolynomial = ((4 * k - 2) / (k * (2 * m - k + 1))) _

  * (i * GramPolynomial(i, m, k - 1, s) _

  + s * GramPolynomial(i, m, k - 1, s - 1)) _

  - ((k - 1) / k) * ((2 * m + k) / (2 * m - k + 1)) _

  * GramPolynomial(i, m, k - 2, s)

ElseIf k = 0 And s = 0 Then

  GramPolynomial = 1

Else

  GramPolynomial = 0

End If

End Function

'''''''''''''''''''''''''''''''''''''''''''''''''''''''''''

'''''''''''''''''''''''''''''''''''''''''''''''''''''''''''

'''''''''''*************************************'''''''''''

'''''''''''*                                   *'''''''''''

'''''''''''*             Integrate             *'''''''''''

'''''''''''*                                   *'''''''''''

'''''''''''*************************************'''''''''''

''''''''''''''''''''''''''''''''''''''''''' (c) R. de Levie

'''''''''''''''''''''''''''''''''''''''''''  v 12, May 2012

' PURPOSE:

' The four macros under this title use trapezoidal integra-

' tion, or its Romberg refinement, to compute the definite

' integral of a function F(x) of a single variable x. You

' can let the macro read the function AUTOmatically from

' the spreadsheet (as long  as F(x) contains x in no more

' than six different places), or you can SPECIFY it in the

' function Equation(x), in which case x can occur any num-

' ber of times in the mathematical expression. This leaves

' the choice between convenience and execution speed to the

' user: having to specify the equation is less convenient

' but yields much faster execution, a consideration that

' may become important when complicated equations are inte-

' grated,and/or when high accuracy is required.

' The macro starts with 2 data points, finds a third point

' half-way between them, then doubles the number of

' intervals by halving their widths, thereby increasing

' the number of data points to 2^k+1, where k counts

' successive iterations. Just counting the number of data

' points limits the number ofiterations to 30, because

' 2^30+1 exceeds the range of integers defined As Long.

' The macro allows the user to specify the required number

' of significant (i.e., accurate) decimals, as well as the

' maximum number kMax of iterations allowed. Note: the

' macro will display its results in the standard Excel

' format, regardless of the number of significant digits in

' that result. It also displayes the accumulated time taken

' by the calculation at that stage.

' The integration will stop when successive estimates of

' the integral agree to within the specified accuracy

' (which typically yields an accuracy in slight excess

' over the specified value), or when the specified number

' of iterations has been reached.

' A cell comment will alert the user when  the desired

' accuracy is not reached within the specified  number of

' iterations. In that case the user may want to repeat the

' integration after increasing the number of iterations

' and/or decreasing the specified number of significant

' digits.

' SUBROUTINES:

' This macro does not require any subroutines

' INPUT:

' Place x in the first, left-most column, and F(x) in

' the second, right-most column of a highlighted, two-

' columns wide rectangular block of input data. The top

' row of the x-column should contain the lower integration

' limit, x = a, the bottom row its upper integration limit,

' x = b. Likewise, the top item in the F(x) column should

' be F(a), its bottom item F(b). (If the positions of a

' and b are interchanged, the negative of the integral

' is obtained.)

' Make sure that the two columns of the input data block,

' for up to 33 rows below that data block, do not contain

' valuable material, which otherwise may be overwritten by

' the output.

' Apart from the first few data points, which are obtained

' by invasive computing, the AUTO version of this macro uses

' reconstruction of the spreadsheet equation to compute the

' values of the function F(x) for successive x-values. The

' typically much faster SPECIFY version instead uses the

' expression for F(x) specified in the function Equation().

'       When you use TrapezSPECIFY or RombergSPECIFY,

'        please MAKE SURE THAT THE EQUATION FUNCTION

'          INDEED CONTAINS THE CORRECT EXPRESSION

'            rather than an expression left in

'              that function from earlier use

' The macro includes a timer, which displays the total

' elapsed time (in seconds, with a resolution of 0.1 s)

' for all iterations made so far. Beyond the first few

' iterations, the integration time roughly doubles for

' every unit increase in the number of iterations used.

' The timer illustrates the trade-off between accuracy

' and iteration time.

' For the computation, the input data block should have at

' least three rows, but can have many more, as long as the

' first and last row contain the lower and higher integra-

' tion limit respectively. However, unless you know the

' properties of your function F(x), you should first plot

' the function (using many more rows) to see whether it

' contains any indeterminate points, discontinuities,

' and/orsingularities. For such a plot, three points

' obviously would not suffice.

' OUTPUT:

' The macro will write its output below the input data

' block, in two labeled columns, using up to 33 rows.

' The macro does NOT check whether the output might over-

' write valuable data, so please make sure to remove such

' data before highlighting the input data block and

' calling the macro.

' CHOICES:

' The following choices are provided:

'   TrapezAuto      trapezoidal integration using the

'               spreadsheet expression (slow)

'   TrapezSpecify   trapezoidal integration using the

'               function Equation(x) (fast)

'   RombergAuto Romberg integration using the

'               spreadsheet expression (slow)

'   RombergSpecify  Romberg integration using the

'               function Equation(x) (fast)

' Romberg integration is usually considerably faster than

' trapezoidal integration, except when the various

' derivatives at both endpoints are the same.

' EXAMPLE:

' Use of this macro is illustrated in section 9.4 of AE3.

''''''''''''''''''''''''''''''''''''''''''''''''''''''''''''

Function Equation(X)

' For use with integration routines that require a

' function to SPECIFY the equation to be integrated.

' The equation can have only one variable, x, which

' can be invoked any number of times in the equation.

' Enter the formula in VBA format, i.e., square root

' as Sqr rather than Sqrt, ten-based logarithm as

' Log(x)/Log(10) rather than as Log(x), etc.

' Here are a few example expressions:

' Equation = Sqr(1 - x ^ 2)

' Equation = Exp(x ^ 2) * Erfc(x)

' Equation = 4 * x ^ 3 - 3 * x ^ 2 + 2 * x - 1

Equation = Exp(-1 * X ^ 2)

End Function

''''''''''''''''''''''''''''''''''''''''''''''''''''''''''''

Sub TrapezAuto()

Dim P As Integer, R As Integer

P = 0

R = 0

Call Integrate(P, R)

End Sub

''''''''''''''''''''''''''''''''''''''''''''''''''''''''''''

Sub TrapezSpecify()

Dim P As Integer, R As Integer

P = 1

R = 0

Call Integrate(P, R)

End Sub

''''''''''''''''''''''''''''''''''''''''''''''''''''''''''''

Sub RombergAuto()

Dim P As Integer, R As Integer

P = 0

R = 1

Call Integrate(P, R)

End Sub

''''''''''''''''''''''''''''''''''''''''''''''''''''''''''''

Sub RombergSpecify()

Dim P As Integer, R As Integer

P = 1

R = 1

Call Integrate(P, R)

End Sub

''''''''''''''''''''''''''''''''''''''''''''''''''''''''''''

Sub Integrate(P, R)

Dim j As Integer, jj As Integer, k As Integer, kk As Integer

Dim kMax As Integer, LenF As Integer, LenXX As Integer

Dim m As Integer, mMax As Integer

Dim PosDollar1 As Integer, PosDollar2 As Integer

Dim PosXinF As Integer, RMax As Integer

Dim CMax As Long, i As Long, ii As Long, iMax As Long

Dim N As Long

Dim H As Double, HH As Double, pErel As Double

Dim Sum As Double, Term As Double, vA As Double

Dim vFA As Double, vB As Double, vFB As Double

Dim fF As Variant, fIn As Variant, Integral As Variant

Dim Interval As Variant, OutputArray As Variant

Dim Rom As Variant

Dim vF As Variant, vIn As Variant, vX As Variant

Dim ColumnStringX As String

Dim LeftF(1 To 20) As String, LeftFF As String

Dim RightF(1 To 20) As String, RowStringX As String

Dim X As String, XX As String

Dim rgIn As Range, rgX As Range, rgF As Range

Dim aF, aIn, Answer, aX, D, fA, fB

Dim Time1, Time2

D = 8

k = 1

kk = k

kMax = 15

m = 1

RMax = Selection.Rows.Count

CMax = Selection.Columns.Count

aIn = Selection.Address

' Make sure that the input area is highlighted

If RMax = 1 And CMax = 1 Then

  Answer = MsgBox("You forgot to highlight" _

    & Chr(13) & "the block of input data." _

    & Chr(13) & "Do you want to do so now?" _

    , vbYesNo, "Trapezoidal integration")

  If Answer = vbNo Then End

  If Answer = vbYes Then

    Set rgIn = Application.InputBox(Prompt:= _

      "The input data block is located in:", Type:=8)

    rgIn.Select

    aIn = Selection.Address

  End If

RMax = Selection.Rows.Count

CMax = Selection.Columns.Count

End If

' Check the size of the input range

If CMax <> 2 Then

  MsgBox "The data must be in two columns," & Chr(13) _

    & "the first for x, the second for F(x).", _

    , "Integration"

  End

End If

If RMax < 3 Then

  MsgBox "The input block must have" & Chr(13) & _

    "at least be three rows.", , "Integration"

  End

End If

D = InputBox _

  (Prompt:="The desired number of significant" _

  & Chr(13) & "decimals in the integral is ", _

  Title:="Accuracy", Default:=6)

If D > 14 Then D = 14

kMax = InputBox _

  (Prompt:="The maximum number of iterations is ", _

  Title:="Iterations", Default:=15)

If kMax > 30 Then kMax = 30 ' kMax = 31 yields integer

  ' overflow even though n is dimensioned As Long

Application.ScreenUpdating = False

Time1 = Timer ' this will measure the time taken by the

              ' algorithm, excluding user data entry time

ReDim OutputArray(1 To kMax, 1 To 2)

ReDim Integral(1 To kMax)

ReDim Interval(1 To kMax)

ReDim Rom(1 To kMax, 1 To kMax)

If P = 0 Then

' Read the input

  aX = Selection(2, 1).Address

  fF = Selection(2, 2).Formula

' Reconstruct the equation F(x) of fF. Use the address and

' formula in the second row of the input data block, to

' allow the right-hand cells in the top and bottom rows of

' that input data block to be modified in the presence of

' indeterminate values at one or both of these limits.

  PosDollar1 = InStr(1, aX, "$")

  PosDollar2 = InStr(2, aX, "$")

  ColumnStringX = Mid(aX, 2, _

    PosDollar2 - PosDollar1 - 1)

  RowStringX = Mid(aX, PosDollar2 + 1, 8)

  XX = ColumnStringX & RowStringX

  LenXX = Len(XX)

  LenF = Len(fF)

  PosXinF = InStr(1, fF, XX)

  LeftF(m) = Left(fF, PosXinF - 1)

  RightF(m) = Right(fF, LenF - PosXinF - LenXX + 1)

' Check whether RightF(1) contains more terms XX

  PosXinF = InStr(1, RightF(1), XX)

' Continue parsing if RightF(m) still contains XX

  Do

    If PosXinF = 0 Then Exit Do

    LenF = Len(RightF(m))

    PosXinF = InStr(1, RightF(m), XX)

    If PosXinF <= 0 Then Exit Do

    LeftF(m + 1) = Left(RightF(m), PosXinF - 1)

    RightF(m + 1) = Right(RightF(m), _

      LenF - PosXinF - LenXX + 1)

    m = m + 1

    PosXinF = InStr(1, RightF(m), XX)

  Loop Until PosXinF <= 0

  mMax = m

  If mMax > 6 Then

    MsgBox "The variable x appears more than" & _

      Chr(10) & "six times in the expression for F(x)." & _

      Chr(10) & "Specify the expression in the function" _

      & Chr(10) & "Equation(), and then use TrapezSpecify."

    End

  End If

End If

' Read some input data from the input range rgIn. For

' convenience, the first few data are obtained by

' invasive computing, while causing negligible delay.

fIn = Selection.Formula   ' Save for restoration

fA = fIn(1, 2)

fB = fIn(RMax, 2)

vIn = Selection.Value

vA = vIn(1, 1)       ' A: lower integration limit

vFA = vIn(1, 2)

vB = vIn(RMax, 1)    ' B: higher integration limit

vFB = vIn(RMax, 2)

HH = vB - vA         ' HH: x-range of integration

' Select the sub-ranges rgX and rgF

vIn = Selection.Value

aX = Selection(2, 1).Address

aF = Selection(2, 2).Address

Set rgX = Range(aX)

rgX.Select

vX = Selection.Value        ' The value x

Set rgF = Range(aF)

rgF.Select

vF = Selection.Value        ' The value of F(x)

fF = Selection.Formula      ' The formula F(x)

' Determine the integration limits

Sum = vFA + vFB

j = k

N = 2 ^ j - 1

H = HH / (2 ^ j)

' First cycle

For i = 1 To N Step kk

  vX = vA + H * i

  rgX.Select

  Selection.Value = vX

  rgF.Select

  vF = Selection.Value

  Sum = Sum + 2 * vF

  Integral(1) = Sum * H / 2

Next i

OutputArray(1, 1) = Integral(1)

Time2 = Timer

Interval(1) = Round(Time2 - Time1, 1)

OutputArray(1, 2) = Interval(1)

' Subsequent cycles

For j = k + 1 To kMax

  N = 2 ^ j - 1

  H = HH / (2 ^ j)

  For i = 1 To N Step 2 * kk

    vX = vA + H * i

' Interrupt the summation to reconstruct or read F(x)

    If P = 0 Then

      If mMax = 1 Then vF = Application.Evaluate _

        (LeftF(1) & Str(vX) & RightF(1))

      If mMax = 2 Then vF = Application.Evaluate _

        (LeftF(1) & Str(vX) & LeftF(2) & Str(vX) & RightF(2))

      If mMax = 3 Then vF = Application.Evaluate _

        (LeftF(1) & Str(vX) & LeftF(2) & Str(vX) & _

        LeftF(3) & Str(vX) & RightF(3))

      If mMax = 4 Then vF = Application.Evaluate _

        (LeftF(1) & Str(vX) & LeftF(2) & Str(vX) & _

        LeftF(3) & Str(vX) & LeftF(4) & Str(vX) & RightF(4))

      If mMax = 5 Then vF = Application.Evaluate _

        (LeftF(1) & Str(vX) & LeftF(2) & Str(vX) & _

        LeftF(3) & Str(vX) & LeftF(4) & Str(vX) & _

        LeftF(5) & Str(vX) & RightF(5))

      If mMax = 6 Then vF = Application.Evaluate _

        (LeftF(1) & Str(vX) & LeftF(2) & Str(vX) & _

        LeftF(3) & Str(vX) & LeftF(4) & Str(vX) & _

        LeftF(5) & Str(vX) & LeftF(6) & Str(vX) & RightF(6))

    End If

    If P = 1 Then vF = Equation(vX)

' In case the SPECIFY option is used, make sure that the

' function Equation() specifies the correct expression.

' Continue the summation

    Sum = Sum + 2 * vF

    Integral(j) = Sum * H / 2

  Next i

  If R = 0 Then

    OutputArray(j, 1) = Integral(j)

    If j > k + 1 Then

      If Integral(j) = 0 Then pErel = _

        -Log(Abs(Integral(j) - Integral(j - 1))) / Log(10)

      If Integral(j) = Integral(j - 1) Then pErel = 14

      If Integral(j) <> 0 And Integral(j) <> _

        Integral(j - 1) Then pErel = -Log(Abs((Integral(j) _

        - Integral(j - 1)) / Integral(j))) / Log(10)

      If pErel > D Then

        kMax = j

        Exit For

      End If

    End If

  End If

  If R = 1 Then

    Rom(j, 1) = Integral(j)

    For jj = 2 To j

      Rom(j, jj) = (2 ^ (2 * (jj - 1)) * Rom(j, jj - 1) - _

        Rom(j - 1, jj - 1)) / (2 ^ (2 * (jj - 1)) - 1)

    Next jj

    If j > k + 1 Then

      If Rom(j, j) = 0 Then pErel = _

        -Log(Abs(Rom(j, j) - Rom(j - 1, j - 1))) / Log(10)

      If Rom(j, j) = Rom(j - 1, j - 1) Then pErel = 14

      If Rom(j, j) <> 0 And Rom(j, j) <> Rom(j - 1, j - 1) _

        Then pErel = -Log(Abs((Rom(j, j) - Rom(j - 1, j - 1)) _

          / Rom(j, j))) / Log(10)

      End If

    OutputArray(j, 1) = Rom(j, j)

    If pErel > D Then

      kMax = j

      Exit For

    End If

  End If

  Time2 = Timer

  Interval(j) = Round(Time2 - Time1, 1)

  OutputArray(j, 2) = Interval(j)

  Application.StatusBar = True

  Application.StatusBar = "Iteration # " & j & _

    ". Each successive iteration takes about " & _

    "twice as long as its immediate predecessor."

Next j

Time2 = Timer

Interval(j) = Round(Time2 - Time1, 1)

OutputArray(j, 2) = Interval(j)

Debug.Print "Integral(" & j & ") = " & OutputArray(j, 1)

Debug.Print "Interval(" & j & ") = " & OutputArray(j, 2)

' Select the output range

Range(aIn).Select

Selection.Formula = fIn

Selection.Offset(RMax + 1, 0).Select

Selection.Resize(kMax, 2).Select

' Output the results

Selection.Value = OutputArray

' Label the macro

ActiveCell.Select

ActiveCell.ClearComments

ActiveCell.AddComment

ActiveCell.Comment.Visible = False

ActiveCell.Comment.Shape.Width = 170

If P = 0 And R = 0 Then _

  ActiveCell.Comment.Text Text:="TrapezAuto" & _

    Chr(10) & "Trapezoidal integration of " & fF & _

    Chr(10) & "from x = " & vA & " to " & vB & _

      ", in block " & aIn & "," & _

    Chr(10) & "to " & D & " significant digits," & _

      " in max. " & kMax & " iterations." & _

    Chr(10) & "Date: " & Date & ",  Time: " & Time

If P = 1 And R = 0 Then _

  ActiveCell.Comment.Text Text:="TrapezSpecify" & _

    Chr(10) & "Trapezoidal integration of " & fF & _

    Chr(10) & "from x = " & vA & " to " & vB & _

      ", in block " & aIn & "," & _

    Chr(10) & "to " & D & " significant digits," & _

      " in max. " & kMax & " iterations." & _

    Chr(10) & "Date: " & Date & ",  Time: " & Time

If P = 0 And R = 1 Then _

  ActiveCell.Comment.Text Text:="RombergAuto" & _

    Chr(10) & "Romberg integration of " & fF & _

    Chr(10) & "from x = " & vA & " to " & vB & _

      ", in block " & aIn & "," & _

    Chr(10) & "to " & D & " significant digits," & _

      " in max. " & kMax & " iterations." & _

    Chr(10) & "Date: " & Date & ",  Time: " & Time

If P = 1 And R = 1 Then _

  ActiveCell.Comment.Text Text:="RombergSpecify" & _

    Chr(10) & "Romberg integration of " & fF & _

    Chr(10) & "from x = " & vA & " to " & vB & _

      ", in block " & aIn & "," & _

    Chr(10) & "to " & D & " significant digits," & _

      " in max. " & kMax & " iterations." & _

    Chr(10) & "Date: " & Date & ",  Time: " & Time

' Alert the user when goal is not reached

ActiveCell.Select

Selection.Offset(kMax - 1, 0).Select

ActiveCell.ClearComments

If pErel < D Then

  ActiveCell.AddComment

  ActiveCell.Comment.Visible = True

  ActiveCell.Comment.Shape.Width = 105

  ActiveCell.Comment.Text Text:= _

    "This numerical integration" & _

    Chr(10) & "did NOT reach " & D & " significant" & _

    Chr(10) & "digits in " & kMax & " iterations." & _

    Chr(10) & "Increase kMax" & _

    Chr(10) & "and/or decrease D."

End If

' Restore the input screen

Range(aIn).Select

Selection.Formula = fIn

Application.StatusBar = False

End Sub

'''''''''''''''''''''''''''''''''''''''''''''''''''''''''''

'''''''''''''''''''''''''''''''''''''''''''''''''''''''''''

'''''''''''''''''''*********************'''''''''''''''''''

'''''''''''''''''''*                   *'''''''''''''''''''

'''''''''''''''''''*       Deriv       *'''''''''''''''''''

'''''''''''''''''''*                   *'''''''''''''''''''

'''''''''''''''''''*********************'''''''''''''''''''

''''''''''''''''''''''''''''''''''''''''''' (c) R. de Levie

'''''''''''''''''''''''''''''''''''''''''''  v 12, May 2012

Sub Deriv()

' ALERT: This function and/or its reference derivative, if

' used for computing pE, can become undefined (as with

' log(x) for x <= 0, and for sqrt(x) for x < 0). This may

' occur during the computation even though X0 is near but

' outside that region. This risk becomes much greater at

' larger j-values because then the footprint of the

' algorithm then becomes much wider. In such a case, set

' OptionB = 0 and start from the lowest j-value, j = 3.

' If the problem persists, use forward or backward 

' derivatives instead, see AE3 sections 9.2.2 and 9.2 6, 

' possibly with extended precision when a smaller foot-

' print is needed. 

' This macro computes the first derivative dF(x)/dx at

' x = x(0) of a function F(x) using compact central

' differencing, using the algorithm described in section

' 9.2 of my book "Advanced Excel for scientific data

' analysis", Oxford University Press, 2nd ed. 2008.

' The macro samples (j+1)/2 equidistant x-values on

' either side of x(0), where j is a positive odd integer:

' j = 3, 5, 7, 9, 11, 13, or 15.

' The macro assumes that the function F(x) is smooth

' and differentiable, and has no discontinuities or

' singularities, within the sampling range of x-values

' probed by the macro. If that is not the case, a more

' appropriate, asymmetric differencing formula should

' be used instead.

' EXAMPLES:

' Section 9.2 in EA3 discusses computing the derivative in 

' great detail.

' USER-SELECTABLE OPTIONS

' The macro provides several options that are selected

' by setting several switches IN THE VBA CODE,

' immediately following the dimension statements. The

' chosen options are specified in the cell comment. The

' following, mutually tolerant options are provided:

' OptionA = 0: use either delta = eps^(1/j) or

' OptionA = 1: use delta = vX0 * eps^(1/j), the default

' OptionB = 0: use the specified j-value, the default

' OptionB = 1: use j = 3 (2) 15

' OptionC = 0: do not compute pE, the default

' OptionC = 1: use a reference value for the first

'              derivative to compute pE

' Be aware that the input boxes may change when you select

' a different option.

' Near the end of the macro, you have another option, viz.

' to display pE as computed, or as limited to the range

' from 0 to 14.

Dim i As Integer, j As Integer, OptionA As Integer

Dim OptionB As Integer, OptionC As Integer

Dim C As Long, R As Long

Dim Delta(3 To 15) As Double, eps As Double

Dim F1 As Double, jEnd As Double, jStart As Double

Dim pE(3 To 15) As Double

Dim Deriv(3 To 15) As Double, vF0 As Double

Dim vF(-8 To 8) As Double, vX(-8 To 8) As Double

Dim vX0 As Double, vXX0 As Double

Dim rgDeriv As Range, rgF As Range, rgF1 As Range

Dim rgJ As Range, rgX As Range

Dim aF, aF1, aJ, aX, A0, a1

Dim B0, B1, B2, B3, C00, C01, C1, fF, fX

Dim TextA, TextB, TextC, TextD

' Select your options (if different from their

' current values) by changing the corresponding

' value:

' OptionA = 0: use delta = eps^(1/j)

' OptionA = 1: use delta = vX0 * eps^(1/j)

OptionA = 0

' Option B = 0: use the specified j-value

' Option B = 1: use j = 3 (2) 15

OptionB = 0

' OptionC = 0: pE not computed

' OptionC = 1: use a reference value for the first

'              derivative to compute pE

OptionC = 0

eps = 2 ^ (-52)

j = 3

' Read the value of j

If OptionB = 1 Then GoTo B1

Set rgJ = Application.InputBox(Prompt:= _

  "Input one of these odd integers:" & Chr(13) & _

  "3, 5, 7, 9, 11, 13, or 15; with" & Chr(13) & _

  "'1' you will select all of these." & _

  Chr(13) & Chr(13) & _

  "The value of j is located in ", Type:=8)

rgJ.Select

aJ = Selection.Value

j = Selection.Value

If j = 1 Then OptionB = 1: GoTo B1

' Determine the input size

C = Selection.Columns.Count

R = Selection.Rows.Count

If C <> 1 Or R <> 1 Then

  MsgBox "The value of j should occupy a single cell."

  End

End If

' Make sure that j is a valid choice

If j <> 1 Then

  If j < 2.9 Or j > 15.1 Or (j > 3.1 And j < 4.9) _

  Or (j > 5.1 And j < 6.9) Or (j > 7.1 And j < 8.9) _

  Or (j > 9.1 And j < 10.9) Or (j > 11.1 And j < 12.9) _

  Or (j > 13.1 And j < 14.9) Then

    MsgBox "Make j equal to one of the allowed choices."

    End

  End If

End If

If j >= 2.9 And j <= 3.1 Then j = 3

If j >= 4.9 And j <= 5.1 Then j = 5

If j >= 6.9 And j <= 7.1 Then j = 7

If j >= 8.9 And j <= 9.1 Then j = 9

If j >= 10.9 And j <= 11.1 Then j = 11

If j >= 12.9 And j <= 13.1 Then j = 13

If j >= 14.9 And j <= 15.1 Then j = 15

B1:

' Read the function F(x)

Set rgF = Application.InputBox(Prompt:= _

  "The function F(x) is located in ", Type:=8)

rgF.Select

aF = Selection.Address

fF = Selection.Formula

vF0 = Selection.Value

' Determine the input size

C = Selection.Columns.Count

R = Selection.Rows.Count

If C <> 1 Or R <> 1 Then

  MsgBox "The function F(x) should occupy a single cell."

  End

End If

' Read the variable x

Set rgX = Application.InputBox(Prompt:= _

  "The parameter x is located in ", Type:=8)

rgX.Select

aX = Selection.Address

fX = Selection.Formula

vX0 = Selection.Value

If Abs(vX0) < 1 Then OptionA = 0

If OptionA = 0 Then vXX0 = 1

If OptionA = 1 Then vXX0 = vX0

' Determine the input size

C = Selection.Columns.Count

R = Selection.Rows.Count

If C <> 1 Or R <> 1 Then

  MsgBox "The x-value should occupy a single cell."

  End

End If

' Optionally, read the value of the first derivative

If OptionC = 0 Then GoTo C00

Set rgF1 = Application.InputBox(Prompt:= _

  "The reference value for the first" & Chr(13) & _

  "derivative is located in ", Type:=8)

rgF1.Select

F1 = rgF1.Value

aF1 = rgF1.Address

' Determine the input size

C = Selection.Columns.Count

R = Selection.Rows.Count

If C <> 1 Or R <> 1 Then

  MsgBox "The value of the first derivative" & Chr(13) _

    & "should occupy a single cell."

  End

End If

C00:

' Read where to deposit the derivative

Set rgDeriv = Application.InputBox(Prompt:= _

  "The calculated value of the first" & Chr(13) & _

  "derivative should be placed in ", Type:=8)

rgDeriv.Select

' Determine the input size

C = Selection.Columns.Count

R = Selection.Rows.Count

If C <> 1 Or R <> 1 Then

  MsgBox "The derivative should be placed in a single cell."

  End

End If

' Sample the necessary values of the function F(x)

' by changing the spreadsheet values of x

If OptionB = 0 Then

  jStart = j

  jEnd = j

End If

If OptionB = 1 Then

  jStart = 3

  jEnd = 15

End If

For j = jStart To jEnd Step 2

Delta(j) = eps ^ (1 / j)

For i = -(j + 1) / 2 To (j + 1) / 2

  vX(i) = vX0 + i * vXX0 * Delta(j)

  rgX.Select

  Selection.Value = vX(i)

  rgF.Select

  vF(i) = Selection.Value

Next i

i = i - 1

' Estimate the jth derivative, vFJ(j)

If j = 3 Then Deriv(j) = (-vF(-1) + vF(1)) _

  / (2 * vXX0 * Delta(j))

If j = 5 Then Deriv(j) = (vF(-2) - vF(2) _

  + 8 * (vF(1) - vF(-1))) / (12 * vXX0 * Delta(j))

If j = 7 Then Deriv(j) = (vF(3) - vF(-3) _

  + 9 * (vF(-2) - vF(2)) + 45 * (vF(1) - vF(-1))) / _

  (60 * vXX0 * Delta(j))

If j = 9 Then Deriv(j) = (3 * (vF(-4) - vF(4)) _

  + 32 * (vF(3) - vF(-3)) + 168 * (vF(-2) - vF(2)) _

  + 672 * (vF(1) - vF(-1))) / (840 * vXX0 * Delta(j))

If j = 11 Then Deriv(j) = (2 * (vF(5) - vF(-5)) _

  + 25 * (vF(-4) - vF(4)) + 150 * (vF(3) - vF(-3)) _

  + 600 * (vF(-2) - vF(2)) + 2100 * (vF(1) - vF(-1))) _

  / (2520 * vXX0 * Delta(j))

If j = 13 Then Deriv(j) = (5 * (vF(-6) - vF(6)) _

  + 72 * (vF(5) - vF(-5)) + 495 * (vF(-4) - vF(4)) _

  + 2200 * (vF(3) - vF(-3)) + 7425 * (vF(-2) - vF(2)) _

  + 23760 * (vF(1) - vF(-1))) / (27720 * vXX0 * Delta(j))

If j = 15 Then Deriv(j) = (15 * (vF(7) - vF(-7)) _

  + 245 * (vF(-6) - vF(6)) + 1911 * (vF(5) - vF(-5)) _

  + 9555 * (vF(-4) - vF(4)) + 35035 * (vF(3) - vF(-3)) _

  + 105105 * (vF(-2) - vF(2)) + 315315 * (vF(1) - vF(-1))) _

  / (360360 * vXX0 * Delta(j))

Next j

j = j - 2

' Place supporting information in a cell comment

If OptionA = 0 Then TextA = ""

If OptionA = 1 Then TextA = " X0 *"

If OptionB = 0 Then TextB = j

If OptionB = 1 Then TextB = "3 to 15"

If OptionC = 0 Then TextC = "not "

If OptionC = 1 Then TextC = ""

If OptionC = 1 Then TextD = " using" & Chr(10) & _

  "    the value of F'(ref) in " & aF1

rgDeriv.Select

ActiveCell.Select

ActiveCell.ClearComments

ActiveCell.AddComment

ActiveCell.Comment.Visible = False

ActiveCell.Comment.Shape.Width = 150

ActiveCell.Comment.Shape.Height = 100

ActiveCell.Comment.Text Text:="Deriv v 7" & Chr(10) & _

  Chr(10) & "computes the first derivative F' of " & _

  Chr(10) & "    F" & fF & " in cell " & aF & Chr(10) & _

  "    at x =  " & vX0 & " in cell  " & aX & Chr(10) & _

  "delta = " & TextA & "epsilon^(1/j)" & Chr(10) & _

  "j =  " & TextB & ",    pE is " & TextC & _

  "computed" & TextD & Chr(10) & _

  "date: " & Date & ",  time: " & Time

' Display the result

rgDeriv.Select

If OptionB = 0 Then

  jStart = j

  jEnd = j

End If

If OptionB = 1 Then

  jStart = 3

  jEnd = 15

End If

For j = jStart To jEnd Step 2

Selection.Value = Deriv(j)

If OptionC = 0 Then GoTo C01

' Include pE

  If F1 = 0 Then pE(j) = -Log(Abs(Deriv(j))) _

    / Log(10)

  If F1 <> 0 And (Deriv(j) - F1) <> 0 Then _

    pE(j) = -Log(Abs((Deriv(j) - F1) / F1)) / Log(10)

  If F1 <> 0 And (Deriv(j) - F1) = 0 Then _

    pE(j) = "15"

' Optional output cleanup; if unwanted,

' comment-out the next two lines of code

  If pE(j) < 0 Then pE(j) = 0

  If pE(j) > 14 Then pE(j) = 14

' Output pE

  Selection.Offset(0, 1).Select

  Selection.Value = pE(j)

  Selection.Offset(0, -1).Select

C01:

  Selection.Offset(1, 0).Select

Next j

' Clean up

rgX.Select

Selection.Value = vX0

End Sub

'''''''''''''''''''''''''''''''''''''''''''''''''''''''''''

''''''''''''''*******************************''''''''''''''

''''''''''''''*                             *''''''''''''''

''''''''''''''*            Deriv1           *''''''''''''''

''''''''''''''*                             *''''''''''''''

''''''''''''''*******************************''''''''''''''

''''''''''''''''''''''''''''''''''''''''''' (c) R. de Levie

'''''''''''''''''''''''''''''''''''''''''''  v 12, May 2012

' PURPOSE

' This macro uses central differencing to estimate the

' first derivative dF(x)/dx of a function F(x) of a single

' variable x at x = x0. It is based on a detailed analysis

' of systematic and cancellation noise in central

' differencing. The theoretical justification for this

' algorithm is explained in sections 9.2.9 and 9.2.10 of my 

' 2012 book on Advanced Excel, 3rd ed., pp. 416-423.

' The macro assumes the provisional value x0 * 2^(-52/j)

' for the step size Delta used in such calculations, and

' with this value estimates the derivatives required to

' find a better value for Delta, with which it computes

' its final answer.

' This macro has so far been subjected only to preliminary

' testing, and is therefore labeled as a beta version. The

' preliminary testing suggests that the macro yields

' answers good to about 10 decimal places for j = 3, to

' about 11 decimals for j = 5, 12 for j = 7, and 13 for

' j = 9. There seems to be no advantage in going beyond

' j = 9, because cancellation noise then becomes dominant,

' except when averaging is used. (However, significant

' averaging noticeably slows down the macro, hardly worth

' the gain of about one more significant decimal, and

' therefore is not implemented here.) The macro can provide

' answers for j of up to 15, where j is an odd positive

' integer larger than 2.

' If this macro proves reliable under more rigorous

' testing, it will replace Deriv in the MacroBundle, and

' its algorithm will be incorporated into the Propagation

' and SolverAid macros.

'        PLEASE TRY IT OUT, and report any deficiencies,

'          suggestions or comments to me at my email,

'                    rdelevie@bowdoin.edu.

'                            THANKS!

' INPUT

' In successive input boxes, the user is asked to point to

' three individual spreadsheet cells, which should contain:

' (1) The desired value of j, the number of equidistant

'     data samples to be taken. The choice is restricted to

'     the seven odd integers 3, 5, 7, 9, 11, 13, and 15.

'     A "1" will generate the first four (or, with a minor

'     change in the macro, up to all seven) results.

' (2) The function to be differentiated.

' (3) The value of x at which the derivative of the

'     function should be evaluated. For j = 1 the location

'     given will specify the left-most cell of a row of

'     seven cells.

' Notes:

' The function in the cell provided in input#2 should be

' the actual expression as a function of the x-value

' listed in the location specified in input#3.

' It is easiest to arrange these three input cell in a

' single spreadsheet row, with the output cell(s) to their

' right.

' OUTPUT

' In answer to a fourth query, the user should indicate the

' spreadsheet location where the output should appear. When

' all seven results are requested, they will appear in a

' single row, starting in the selected cell with the result

' for j = 3, and moving successively to the right for

' increasingly higher j-values. In that case make sure that

' the then required seven output cells do not contain

' information that should not be overwritten.

' SUBROUTINES

' No subroutines are needed.

' APPLICABILITY

' Central differencing assumes that the function and the

' derivatives involved in the calculation are "smooth",

' i.e., that they contain no discontinuities or singula-

' rities. Moreover, the sampling used should not reach

' regions where the function is ill-defined, such as

' negative x-values for a square root, or non-positive

' numbers for a logarithm. In the latter cases, lateral

' differencing or other methods should be used instead.

' When applied to a function close to a discontinuity, 

' singularity, or boundary of applicability, it may be 

' advisable to use extended precision, perhaps with 

' forward or backward differencing, in order to stay away

' from such discontinuities, singularities, or other 

' boundaries and still achieve sufficient numerical 

' precision.  

' EXAMPLES:

' AE3 sections 9.29 and 9.2.10 describe the uses of this 

' macro great detail. It can be used as a general replace-

' ment for Deriv, and is subject to the same constraints. 

' SELECTING AN OPTIMAL j-VALUE

' When F(x) has no discontinuities or singularities,

' use j = 9 for best accuracy, often with pE >= 14.

' When the nearest discontinuity or singularity in F(x) is

' at F(y), select j such that abs(footprint) < abs(x-y).

' Use the following approximate values as your guide:

'   j = 3   abs(footprint) < 0.001

'   j = 5   abs(footprint) < 0.01

'   j = 7   abs(footprint) < 0.1

'   j = 9   abs(footprint) < 0.5

' Then run Deriv1, and check the actual footprint value in

' the cell comment. If this value has exceeded abs(x-y),

' lower your j-value and repeat. If this indicates that

' even j = 3 has too large a footprint for your problem,

' switch to lateral differencing.

' The macro fails (and alerts the user) when the necessary

' higher derivative is computed as equal to zero.

Sub Deriv1()

Dim i As Integer, j As Integer

Dim jEnd As Integer, jStart As Integer

Dim N(3 To 17) As Integer

Dim OptionA As Integer, OptionB As Integer, Z As Integer

Dim C As Long, R As Long

Dim bb(3 To 17) As Double, cc(3 To 17) As Double

Dim dd(3 To 17) As Double

Dim Delta(3 To 17) As Double, DeltaOpt(3 To 17) As Double

Dim Deriv(3 To 17) As Double, DerivJ(3 To 17) As Double

Dim DerivOpt(3 To 17) As Double, Ecanc(3 To 17) As Double

Dim Eest As Double, Eopt(3 To 17) As Double, eps As Double

Dim F(-9 To 9) As Double, Footprint(3 To 17) As Double

Dim F0 As Double, F1 As Double, jj As Double

Dim pDelta(3 To 17) As Double, pDeltaOpt(3 To 17) As Double

Dim pEcanc(3 To 17) As Double, pEest(3 To 17) As Double

Dim pEopt(3 To 17) As Double

Dim X(-9 To 9) As Double, X0 As Double, XX0 As Double

Dim rgDeriv As Range, rgF As Range, rgF1 As Range

Dim rgJ As Range, rgX As Range

Dim aF, aF1, aJ, aX, fF, fX

Dim IfFIsZeroAndXIsZero

Dim InputChoiceJ, InputFunctionF, InputParamaterX

Dim OutputLocation, StartDerivJ, TextA, TextB

' Internally, the macro sets and uses two mutually

' tolerant options that have the following functions:

' OptionA = 0: use delta = 2^(-52/j) when X0 close to 0

' OptionA = 1: use delta = X0 * 2^(-52/j)

' OptionB = 0: use the specified j-value when j > 2

' OptionB = 1: use j = 3 (2) 9 when j is specified as 1

'   (which can be changed to 3 (2) 15 if so desired)

eps = 2 ^ (-52)

OptionA = 1

OptionB = 0

j = 3

Z = 0

bb(3) = 6                       ' bb(j) = 1 / b(j)

bb(5) = 30

bb(7) = 140

bb(9) = 630

bb(11) = 2772

bb(13) = 12012

bb(15) = 51480

bb(17) = 218790

cc(3) = 4.5324665183684E-17

cc(5) = 6.09031888443656E-17

cc(7) = 6.93498646345351E-17

cc(9) = 7.48317772924943E-17

cc(11) = 7.87541807882321E-17

cc(13) = 8.17386474122853E-17

cc(15) = 8.4107643769096E-17

cc(17) = 8.60471943284807E-17

' Read the value of j

InputChoiceJ:

Set rgJ = Application.InputBox(Prompt:= _

  "Input the cell location where j is stored." & Chr(13) & _

  "The possible j-values are 3, 5, 7, 9, 11, 13," & Chr(13) & _

  "or 15, while 1 selects all seven j-values.", Type:=8)

rgJ.Select

aJ = Selection.Value

jj = Selection.Value

' Determine the input size

C = Selection.Columns.Count

R = Selection.Rows.Count

If C <> 1 Or R <> 1 Then

  MsgBox "The value of j should occupy a single cell."

  GoTo InputChoiceJ

End If

' Make sure that j is a valid choice

If jj >= 0.5 And jj < 1.5 Then OptionB = 1

If jj < 0.5 Or jj > 15.5 Or (jj > 1.1 And jj < 2.9) _

  Or (jj > 3.1 And jj < 4.9) Or (jj > 5.1 And jj < 6.9) _

  Or (jj > 7.1 And jj < 8.9) Or (jj > 9.1 And jj < 10.9) _

  Or (jj > 11.1 And jj < 12.9) Or (jj > 13.1 And jj < 14.9) _

  Then

    MsgBox "Make j equal to one of the allowed choices."

    GoTo InputChoiceJ

End If

If jj >= 2.9 And jj <= 3.1 Then j = 3

If jj >= 4.9 And jj <= 5.1 Then j = 5

If jj >= 6.9 And jj <= 7.1 Then j = 7

If jj >= 8.9 And jj <= 9.1 Then j = 9

If jj >= 10.9 And jj <= 11.1 Then j = 11

If jj >= 12.9 And jj <= 13.1 Then j = 13

If jj >= 14.9 And jj <= 15.1 Then j = 15

' Read the function F(x)

InputFunctionF:

Set rgF = Application.InputBox(Prompt:= _

  "The function F(x) is located in ", Type:=8)

rgF.Select

aF = Selection.Address

fF = Selection.Formula

F0 = Selection.Value

' Determine the input size

C = Selection.Columns.Count

R = Selection.Rows.Count

If C <> 1 Or R <> 1 Then

  MsgBox "The function F(x) should occupy a single cell."

  GoTo InputFunctionF

End If

' Read the variable x

InputParameterX:

Set rgX = Application.InputBox(Prompt:= _

  "The parameter x is located in ", Type:=8)

rgX.Select

aX = Selection.Address

fX = Selection.Formula

X0 = Selection.Value

If Abs(X0) < 1 Then OptionA = 0

If OptionA = 0 Then XX0 = 1

If OptionA = 1 Then XX0 = X0

' Determine the input size

C = Selection.Columns.Count

R = Selection.Rows.Count

If C <> 1 Or R <> 1 Then

  MsgBox "The x-value should occupy a single cell."

  GoTo InputParameterX

End If

' Read where to deposit the derivative

OutputLocation:

Set rgDeriv = Application.InputBox(Prompt:= _

  "The calculated value of the first" & Chr(13) & _

  "derivative should be placed in", Type:=8)

rgDeriv.Select

' Determine the input size

C = Selection.Columns.Count

R = Selection.Rows.Count

If OptionB = 0 And (C <> 1 Or R <> 1) Then

  MsgBox "The derivative should be placed in a single cell."

  GoTo OutputLocation

End If

If OptionB = 1 And (C <> 1 Or R <> 1) Then

  MsgBox "Please specify the location" & Chr(13) & _

         "of the left-most output cell."

  GoTo OutputLocation

End If

' Suppress screen updating

Application.ScreenUpdating = False

' Sample the necessary values of the function F(x)

' by changing the spreadsheet values of x

'On Error GoTo Handler

If OptionB = 0 Then

  jStart = j

  jEnd = j

End If

If OptionB = 1 Then

  jStart = 3

' HOW TO CHANGE THE RANGE OF j-VALUES SHOWN FOR j = 1:

' THE DEFAULT VALUE OF jEnd IS SET TO 9; IF YOU WANT TO

' DISPLAY MORE (OR FEWER) THAN THE FIRST FOUR j-VALUES,

' IN THE NEXT LINE CHANGE THE VALUE OF jEnd ACCORDINGLY,

' TO UP TO 15. NOTE: FOR jEnd ONLY USE ONE OF THE INTEGER

' VALUES 5, 7, 9, 11, 13, OR 15.

  jEnd = 9

End If

For j = jStart To jEnd Step 2

  Delta(j) = XX0 * eps ^ (1 / j)

  For i = -(j - 1) / 2 To (j - 1) / 2

    X(i) = X0 + i * Delta(j)

    rgX.Select

    Selection.Value = X(i)

    rgF.Select

    F(i) = Selection.Value

  Next i

  i = i - 1

' Estimate the first derivative, Deriv(j)

  If j = 3 Then Deriv(j) = (F(1) - F(-1)) _

    / (2 * Delta(j))

  If j = 5 Then Deriv(j) = (F(-2) - F(2) _

    + 8 * (F(1) - F(-1))) / (12 * Delta(j))

  If j = 7 Then Deriv(j) = (F(3) - F(-3) _

    + 9 * (F(-2) - F(2)) + 45 * (F(1) - F(-1))) _

    / (60 * Delta(j))

  If j = 9 Then Deriv(j) = (3 * (F(-4) - F(4)) _

    + 32 * (F(3) - F(-3)) + 168 * (F(-2) - F(2)) _

    + 672 * (F(1) - F(-1))) / (840 * Delta(j))

  If j = 11 Then Deriv(j) = (2 * (F(5) - F(-5)) _

    + 25 * (F(-4) - F(4)) + 150 * (F(3) - F(-3)) _

    + 600 * (F(-2) - F(2)) + 2100 * (F(1) - F(-1))) _

    / (2520 * Delta(j))

  If j = 13 Then Deriv(j) = (5 * (F(-6) - F(6)) _

    + 72 * (F(5) - F(-5)) + 495 * (F(-4) - F(4)) _

    + 2200 * (F(3) - F(-3)) + 7425 * (F(-2) - F(2)) _

    + 23760 * (F(1) - F(-1))) / (27720 * Delta(j))

  If j = 15 Then Deriv(j) = (15 * (F(7) - F(-7)) _

    + 245 * (F(-6) - F(6)) + 1911 * (F(5) - F(-5)) _

    + 9555 * (F(-4) - F(4)) + 35035 * (F(3) - F(-3)) _

    + 105105 * (F(-2) - F(2)) + 315315 * (F(1) - F(-1))) _

    / (360360 * Delta(j))

N(j) = 0

StartDerivJ:

' Estimate the Jth derivative, DerivJ(j)

  For i = -(j + 1) / 2 To (j + 1) / 2

    X(i) = X0 + i * Delta(j)

    rgX.Select

    Selection.Value = X(i)

    rgF.Select

    F(i) = Selection.Value

  Next i

  i = i - 1

  If j = 3 Then DerivJ(j) = ((F(2) - F(-2)) _

    + 2 * (F(-1) - F(1))) / (2 * (Delta(j)) ^ (j))

  If j = 5 Then DerivJ(j) = (F(3) - F(-3) _

    + 4 * (F(-2) - F(2)) + 5 * (F(1) - F(-1))) _

    / (2 * (Delta(j)) ^ (j))

  If j = 7 Then DerivJ(j) = (F(4) - F(-4) _

    + 6 * (F(-3) - F(3)) + 14 * (F(2) - F(-2) _

    + F(-1) - F(1))) / (2 * (Delta(j)) ^ (j))

  If j = 9 Then DerivJ(j) = (F(5) - F(-5) _

    + 8 * (F(-4) - F(4)) + 27 * (F(3) - F(-3)) _

    + 48 * (F(-2) - F(2)) + 42 * (F(1) - F(-1))) _

    / (2 * (Delta(j)) ^ (j))

  If j = 11 Then DerivJ(j) = (F(6) - F(-6) _

    + 10 * (F(-5) - F(5)) + 44 * (F(4) - F(-4)) _

    + 110 * (F(-3) - F(3)) + 165 * (F(2) - F(-2)) _

    + 132 * (F(-1) - F(1))) / (2 * (Delta(j)) ^ (j))

  If j = 13 Then DerivJ(j) = (F(7) - F(-7) _

    + 12 * (F(-6) - F(6)) + 65 * (F(5) - F(-5)) _

    + 208 * (F(-4) - F(4)) + 429 * (F(3) - F(-3)) _

    + 572 * (F(-2) - F(2)) + 429 * (F(1) - F(-1))) _

    / (2 * (Delta(j)) ^ (j))

  If j = 15 Then DerivJ(j) = (F(8) - F(-8) _

    + 14 * (F(-7) - F(7)) + 90 * (F(6) - F(-6)) _

    + 350 * (F(-5) - F(5)) + 910 * (F(4) - F(-4)) _

    + 1638 * (F(-3) - F(3)) + 2002 * (F(2) - F(-2)) _

    + 1430 * (F(-1) - F(1))) / (2 * (Delta(j)) ^ (j))

  If j = 17 Then DerivJ(j) = (F(9) - F(-9) _

    + 16 * (F(-8) - F(8)) + 119 * (F(7) - F(-7)) _

    + 544 * (F(-6) - F(6)) + 1700 * (F(5) - F(-5)) _

    + 3808 * (F(-4) - F(4)) + 6188 * (F(3) - F(-3)) _

    + 7072 * (F(-2) - F(2)) + 4862 * (F(1) - F(-1))) _

    / (2 * (Delta(j)) ^ (j))

  If Abs(DerivJ(j)) <= eps Then

    Delta(j) = 2 * Delta(j)

    N(j) = N(j) + 1

    If N(j) < 5 Then GoTo StartDerivJ

    If N(j) = 5 Then _

      MsgBox "The needed higher derivative is zero; " & Chr(10) _

         & "the algorithm failed at j = " & j: GoTo N

  End If

  Delta(j) = (Delta(j)) / (2 ^ N(j))

' Take care of a predictable problem

  If Abs(F0) <= eps And Abs(X0) <= eps Then _

    X0 = 0.000000000001: Z = 1

IfFIsZeroAndXIsZero:

  If Z = -1 Then X0 = -0.000000000001

  If Z <> 0 Then

    rgX.Select

    Selection.Value = X0

    rgF.Select

    F0 = Selection.Value

  End If

  If Abs(DerivJ(j)) < eps Then _

    MsgBox "The value of the Jth derivative " _

    & "is (too close to) zero.": GoTo E

' Compute DeltaOpt and pEOpt

  dd(j) = (cc(j) * bb(j) / (j - 1)) ^ (1 / j)

  DeltaOpt(j) = dd(j) * ((Abs(F0) + Abs(X0 * Deriv(j))) _

    / (Abs(DerivJ(j)))) ^ (1 / j)

  Eopt(j) = Abs((j * cc(j)) / ((j - 1) * dd(j)) _

    * ((Abs(F0) + Abs(X0 * Deriv(j))) / Deriv(j)) _

    * ((Abs(DerivJ(j))) / (Abs(F0) + Abs(X0 * Deriv(j)))) _

    ^ (1 / j))

  Ecanc(j) = cc(j) * (Abs(F0) + Abs(X0 * Deriv(j))) _

    / (Delta(j) * Abs(Deriv(j)))

  pDelta(j) = -Log(DeltaOpt(j)) / Log(10)

  pDeltaOpt(j) = -Log(Delta(j)) / Log(10)

  pEopt(j) = -Log(Abs(Eopt(j))) / Log(10)

  pEcanc(j) = -Log(Ecanc(j)) / Log(10)

  pEest(j) = -Log(Eopt(j) + 3 * Ecanc(j)) / Log(10)

  ' Re-sample the data with DeltaOpt

  For i = -(j - 1) / 2 To (j - 1) / 2

    X(i) = X0 + i * DeltaOpt(j)

    rgX.Select

    Selection.Value = X(i)

    rgF.Select

    F(i) = Selection.Value

  Next i

  i = i - 1

' Again compute the first derivative, Deriv(j),

' but now with DeltaOpt(j)

  If j = 3 Then DerivOpt(j) = (-F(-1) + F(1)) _

    / (2 * DeltaOpt(j))

  If j = 5 Then DerivOpt(j) = (F(-2) - F(2) _

    + 8 * (F(1) - F(-1))) / (12 * DeltaOpt(j))

  If j = 7 Then DerivOpt(j) = (F(3) - F(-3) _

    + 9 * (F(-2) - F(2)) + 45 * (F(1) - F(-1))) / _

    (60 * DeltaOpt(j))

  If j = 9 Then DerivOpt(j) = (3 * (F(-4) - F(4)) _

    + 32 * (F(3) - F(-3)) + 168 * (F(-2) - F(2)) _

    + 672 * (F(1) - F(-1))) / (840 * DeltaOpt(j))

  If j = 11 Then DerivOpt(j) = (2 * (F(5) - F(-5)) _

    + 25 * (F(-4) - F(4)) + 150 * (F(3) - F(-3)) _

    + 600 * (F(-2) - F(2)) + 2100 * (F(1) - F(-1))) _

    / (2520 * DeltaOpt(j))

  If j = 13 Then DerivOpt(j) = (5 * (F(-6) - F(6)) _

    + 72 * (F(5) - F(-5)) + 495 * (F(-4) - F(4)) _

    + 2200 * (F(3) - F(-3)) + 7425 * (F(-2) - F(2)) _

    + 23760 * (F(1) - F(-1))) / (27720 * DeltaOpt(j))

  If j = 15 Then DerivOpt(j) = (15 * (F(7) - F(-7)) _

    + 245 * (F(-6) - F(6)) + 1911 * (F(5) - F(-5)) _

    + 9555 * (F(-4) - F(4)) + 35035 * (F(3) - F(-3)) _

    + 105105 * (F(-2) - F(2)) + 315315 * (F(1) - F(-1))) _

    / (360360 * DeltaOpt(j))

  Dim DerivOptPlus(3 To 17) As Double

  Dim DerivOptMin(3 To 17) As Double

  Dim DeltaOptPlus(3 To 17) As Double

  Dim DeltaOptMin(3 To 17) As Double

  If Z = 1 Then

    DerivOptPlus(j) = DerivOpt(j)

    DeltaOptPlus(j) = DeltaOpt(j)

    Z = -1

    GoTo IfFIsZeroAndXIsZero

  End If

  If Z = -1 Then

    DerivOptMin(j) = DerivOpt(j)

    DeltaOptMin(j) = DeltaOpt(j)

    If DerivOptPlus(j) * DerivOptMin(j) > 0 Then

      DerivOpt(j) = (DerivOptPlus(j) + DerivOptMin(j)) / 2

      DeltaOpt(j) = DeltaOptPlus(j) + DeltaOptMin(j) / 2

    End If

  End If

  If Z = -1 Then X0 = 0

  rgX.Select

  Selection.Value = X0

  pDeltaOpt(j) = Round(pDeltaOpt(j), 4)

  pEopt(j) = Round(pEopt(j), 2)

  pEest(j) = Round(pEest(j) - 0.05, 1)

  If pEest(j) > 14 Then pEest(j) = "14"

  Footprint(j) = 0.5 * (j + 1) * (2 ^ N(j)) * Delta(j)

  If (j - 1) * DeltaOpt(j) > Footprint(j) Then _

    Footprint(j) = 0.5 * (j - 1) * DeltaOpt(j)

' Release screen updating

  Application.ScreenUpdating = False

' Place supporting information in a cell comment

  If OptionA = 0 Then TextA = ""

  If OptionA = 1 Then TextA = " X0 *"

' Display the result

  rgDeriv.Select

  If OptionB = 1 Then _

    Selection.Offset(0, (j - 3) / 2).Select

  ActiveCell.Select

  ActiveCell.ClearComments

  ActiveCell.AddComment

  ActiveCell.Comment.Visible = False

  ActiveCell.Comment.Shape.Width = 150

  ActiveCell.Comment.Shape.Height = 120

  ActiveCell.Comment.Text Text:="Deriv1 v 9 (beta)" & Chr(10) _

    & Chr(10) & "computes the first derivative F' of " & _

    Chr(10) & "    F" & fF & " in cell " & aF & Chr(10) & _

    "    at x=" & X0 & " in cell  " & aX & Chr(10) & _

    "j =  " & j & ",  pEest(" & j & ") > " & pEest(j) & _

    Chr(10) & "pDeltaOpt(" & j & ") = " & pDeltaOpt(j) & _

    Chr(10) & "Footprint(" & j & ") = " & X0 & _

    " plus or minus " & Footprint(j) & Chr(10) & _

    "date: " & Date & ",  time: " & Time

  Selection.Value = DerivOpt(j)

' Begin of optional outputs:

'  Selection.Offset(1, 0).Select

'  Selection.Value = pEopt(j)

'  Selection.Offset(1, 0).Select

'  Selection.Value = pEcanc(j)

'  Selection.Offset(1, 0).Select

'  Selection.Value = pEest(j)

'  Selection.Offset(-3, 0).Select

' End of optional outputs

N:

Next j

GoTo E

Handler:

If Err.Number = 13 Then

  MsgBox "The function cannot be evaluated at one of the" & _

    Chr(13) & "sampled x-values. Retry with a lower j-value."

  GoTo E

ElseIf Err.Number <> 13 Then _

  MsgBox "An error occurred, with error number " & _

    Err.Number & ":" & Chr(13) & "'" & _

    Err.Description & "'." & Chr(13) & Chr(13) & _

    "Press OK. Sorry, the macro will end."

  GoTo E

End If

E:

' Clean up

rgX.Select

Selection.Formula = fX

End Sub

'''''''''''''''''''''''''''''''''''''''''''''''''''''''''''

'''''''''''''''''''''''''''''''''''''''''''''''''''''''''''

'''''''''''''''''''*********************'''''''''''''''''''

'''''''''''''''''''*                   *'''''''''''''''''''

'''''''''''''''''''*     DerivScan     *'''''''''''''''''''

'''''''''''''''''''*                   *'''''''''''''''''''

'''''''''''''''''''*********************'''''''''''''''''''

''''''''''''''''''''''''''''''''''''''''''' (c) R. de Levie

'''''''''''''''''''''''''''''''''''''''''''  v 12, May 2012

Sub DerivScan()

' A simplified version of Deriv() for scanning the first

' derivative as a function of pDelta = -log(Delta) using

' central differencing, with Delta0 = X(0) * eps^1/j as

' its anchor. The range of Delta-values used in the macro

' is set in the macro code just just above the label C:

' An increment of i by 40 changes pDelta by one. For

' negative i-values (in the range of predominantly

' systematic errors) the range may need to be restricted

' at higher j-values.

' The input values are most conveniently arranged in a row

' of adjacent cells, with (from left to right) the values

' of j, F(x), and x. When so organized, the first derivative

' will appear below F(x), and pDelta below x, leaving space

' below j for a spreadsheet calculation of pE when a

' reference value for the first derivative is available,

' making the output convenient to plot.

' EXAMPLES

' The figures in AE3 sections 9.2.8 and 9.2.9 were computed

' with DerivScan

Dim i As Integer, j As Integer, k As Integer

Dim cF As Long, cX As Long, rF As Long, rX As Long

Dim Delta(3 To 15, -120 To 300) As Double

Dim eps(3 To 15) As Double

Dim vDeriv(-120 To 300) As Double, vF(-8 To 8) As Double

Dim vX(-8 To 8) As Double

Dim rgDeriv As Range, rgF As Range

Dim rgJ As Range, rgX As Range

Dim aF, aJ, aX, fF(0), fX(0)

' Read the value of j

Set rgJ = Application.InputBox(Prompt:= _

  "The j-value is located in ", Type:=8)

rgJ.Select

j = rgJ.Value

aJ = rgJ.Address

eps(j) = (1.11022302462532E-16) ^ (1 / j)

' Read the function F(x)

Set rgF = Application.InputBox(Prompt:= _

  "The function F(x) is located in ", Type:=8)

rgF.Select

aF = rgF.Address

fF(0) = rgF.Formula

vF(0) = rgF.Value

' Determine the input size

cF = Selection.Columns.Count

rF = Selection.Rows.Count

If cF <> 1 Or rF <> 1 Then

  MsgBox "The function F(x) should be" & _

  Chr(13) & "  located in a single cell."

  End

End If

' Read the variable x

Set rgX = Application.InputBox(Prompt:= _

  "The parameter x is located in ", Type:=8)

rgX.Select

aX = rgX.Address

fX(0) = rgX.Formula

vX(0) = rgX.Value

' Determine the input size

cX = Selection.Columns.Count

rX = Selection.Rows.Count

If cX <> 1 Or rX <> 1 Then

     MsgBox "The x-value should" & _

  Chr(13) & "occupy a single cell."

  End

End If

' Start the calculation

Application.ScreenUpdating = False

' HERE SELECT THE RANGE OF i-VALUES TO BE USED

For i = -40 To 300

C:

  Delta(j, i) = Abs(vX(0)) ^ (1 / j) * eps(j) * (0.01 ^ (i / 80))

  For k = -(j + 1) / 2 To (j + 1) / 2

    On Error GoTo ErrorHandlerC

    vX(k) = vX(0) + k * Delta(j, i)

    rgX.Select

    Selection.Value = vX(k)

    rgF.Select

    vF(k) = Selection.Value

  Next k

  GoTo D

ErrorHandlerC:

  j = j - 2

  MsgBox "Value of j is lowered to " & j

  GoTo C

D:

' Compute vDeriv

  If j = 3 Then vDeriv(i) = (-vF(-1) + vF(1)) _

    / (2 * Delta(j, i))

  If j = 5 Then vDeriv(i) = (vF(-2) - 8 * vF(-1) _

    + 8 * vF(1) - vF(2)) / (12 * Delta(j, i))

  If j = 7 Then vDeriv(i) = (vF(3) - vF(-3) _

    + 9 * (vF(-2) - vF(2)) + 45 * (vF(1) - vF(-1))) _

    / (60 * Delta(j, i))

  If j = 9 Then vDeriv(i) = (3 * (vF(-4) - vF(4)) + _

    32 * (vF(3) - vF(-3)) + 168 * (vF(-2) - vF(2)) _

    + 672 * (vF(1) - vF(-1))) / (840 * Delta(j, i))

  If j = 11 Then vDeriv(i) = (2 * (vF(5) - vF(-5)) + _

    25 * (vF(-4) - vF(4)) + 150 * (vF(3) - vF(-3)) _

    + 600 * (vF(-2) - vF(2)) + 2100 * (vF(1) - vF(-1))) _

    / (2520 * Delta(j, i))

  If j = 13 Then vDeriv(i) = (5 * (vF(-6) - vF(6)) _

    + 72 * (vF(5) - vF(-5)) + 495 * (vF(-4) - vF(4)) _

    + 2200 * (vF(3) - vF(-3)) + 7425 * (vF(-2) - vF(2)) _

    + 23760 * (vF(1) - vF(-1))) / (27720 * Delta(j, i))

  If j = 15 Then vDeriv(i) = (15 * (vF(7) - vF(-7)) _

    + 245 * (vF(-6) - vF(6)) + 1911 * (vF(5) - vF(-5)) _

    + 9555 * (vF(-4) - vF(4)) + 35035 * (vF(3) - vF(-3)) _

    + 105105 * (vF(-2) - vF(2)) + 315315 * (vF(1) - vF(-1))) _

    / (360360 * Delta(j, i))

Next i

' Clean up

rgX.Select

Selection.Formula = fX(0)

' Add a cell comment

Selection.Offset(4, -1).Select

ActiveCell.Select

ActiveCell.ClearComments

ActiveCell.AddComment

ActiveCell.Comment.Visible = False

ActiveCell.Comment.Shape.Width = 135

ActiveCell.Comment.Shape.Height = 60

ActiveCell.Comment.Text Text:="DerivScan  v 7" _

  & Chr(10) & "j=" & j & "  in cell  " & aJ & Chr(10) _

  & "F(x)" & fF(0) & "  in cell  " & aF & Chr(10) _

  & "x=" & vX(0) & "  in cell  " & aX _

  & Chr(10) & "Date: " & Date & ", Time: " & Time

' Write the result

'''Selection.Offset(4, -1).Select

'''Selection.Offset(5, 0).Select

For i = -40 To 300

  Selection = vDeriv(i)

  Selection.Offset(0, 1).Select

  Selection = -Log(Delta(j, i)) / Log(10)

  Selection.Offset(1, -1).Select

Next i

End Sub

'''''''''''''''''''''''''''''''''''''''''''''''''''''''''''

'''''''''''''*********************************'''''''''''''

'''''''''''''*                               *'''''''''''''

'''''''''''''*     Semi-Integration and      *'''''''''''''

'''''''''''''*     Semi-Differentiation      *'''''''''''''

'''''''''''''*                               *'''''''''''''

'''''''''''''*********************************'''''''''''''

''''''''''''''''''''''''''''''''''''''''''' (c) R. de Levie

'''''''''''''''''''''''''''''''''''''''''''  v 12, May 2012

' PURPOSE:

' This macro performs semi-integration and semi-differen-

' tiation, specialized (de)convolution methods applicable

' to problems of planar heat conduction, planar diffusion,

' and signal transmission along a cable with uniformly

' distributed resistance and capacitance, see K. B. Oldham,

' Anal. Chem. 41 (1969) 1121; 44 (1972) 196.

' The macro uses a simple semi-integration algorithm; for

' more sophisticated algorithms see K. B. Oldham & J. C.

' Myland, Fundamentals of electrochemical science, Academic

' Press 1994. The semi-differential is then found by taking

' first differences.

' In the context of linear sweep voltammetry or cyclic

' voltammetry (which are both methods used in electro-

' chemical analysis), semi-integration can convert a

' current-time-voltage curve into the corresponding

' steady-state current-voltage curve, which is often

' much easier to interpret quantitatively. For a discussion

' of such an application, see any modern electrochemical

' textbook, or consult my How to use Excel in analytical

' chemistry, Cambridge Univ. Press, 2001, pp. 257-263.

' SUBROUTINES:

' The drivers SemiIntegrate() and SemiDifferentiate()

' call the subroutine Semi

' INPUT:

' The input data must be equidistant, and should be

' stored in a single, continuous column.

' OUTPUT:

' The resulting transformed output data will be written

' in the column immediately to the right of the input

' data. That column is therefore best left free.

' PROCEDURE:

' Call the macro, and input the data by following the

' instructions with its input boxes.

Sub SemiIntegrate()

Dim iIntDif As Integer

iIntDif = 1

Call Semi(iIntDif)

End Sub

'''''''''''''''''''''''''''''''''''''''''''''''''''''''''''

Sub SemiDifferentiate()

Dim iIntDif As Integer

iIntDif = -1

Call Semi(iIntDif)

End Sub

'''''''''''''''''''''''''''''''''''''''''''''''''''''''''''

Sub Semi(iIntDif)

' The macro uses a simple semi-integration algorithm due

' to Oldham which is adequate to illustrate the approach.

' The semi-differential is then found by taking first

' differences.

' The input data should be stored in a single, continuous

' column. The resulting transformed output data will be

' written in the column immediately to the right of the

' input data. That column is therefore best left free.

' Parameter names used:

'     cnmax         number of columns in

'                   highlighted data block

'     rnmax         number of rows in

'                   highlighted data block

'     dataArray     array of all input data

'     rn            index specifying row number

'                   in input data array

Dim cnMax As Integer, j As Integer, k As Integer

Dim N As Integer, rnMax As Integer

Dim a As Double, s As Double, deltaTime As Double

Dim myRange As Range

Dim inputData As Variant, semiInt As Variant

Dim outputData As Variant, testValue As Variant

Dim Z As Variant

Dim myAddress, Ans

N = 0

' Input the time increment and use it

' to compute the parameter s

If iIntDif = 1 Then

  deltaTime = _

    InputBox("Enter the time increments, in seconds.", _

    "SemiIntegration InputBox 1: Time increments")

Else

  deltaTime = _

    InputBox("Enter the time increments, in seconds.", _

    "SemiDifferentiation InputBox 1: Time increments")

End If

s = Sqr(Abs(deltaTime))

' Enter the input data

If iIntDif = 1 Then

  Set myRange = Application.InputBox(Prompt:= _

  "The current values are located in:", Title:= _

  "SemiIntegration InputBox 2: Input data", Type:=8)

Else

  Set myRange = Application.InputBox(Prompt:= _

  "The current values are located in:", Title:= _

  "SemiDifferentiation InputBox 2: Input data", Type:=8)

End If

myRange.Select

If Selection.Columns.Count <> 1 Then

  End

Else

  rnMax = Selection.Rows.Count

  inputData = Selection.Value

  semiInt = Selection.Value

  outputData = Selection.Value

  myAddress = Selection.Address

End If

' Check that the output will not overwrite valuable data

Selection.Offset(0, 1).Select

testValue = Selection.Value

For j = 1 To rnMax

  Z = testValue(j, 1)

  If IsEmpty(Z) Then

    N = N

  Else

    N = N + 1

  End If

Next j

If N > 0 Then

  Ans = MsgBox("There are data in the space where " _

    & "the output will be" & Chr(13) & " written. " _

    & "Proceed anyway and overwrite those data?", _

  vbYesNo)

  If Ans = vbNo Then

    Selection.Offset(0, -1).Select

    End

  End If

End If

' Compute the semi-integration

ReDim P(1 To rnMax) As Double

P(1) = 1

For j = 2 To rnMax

  P(j) = P(j - 1) * (j - 1.5) / (j - 1)

Next j

For k = 1 To rnMax

  a = 0

  For j = 1 To k

    a = a + P(k - j + 1) * inputData(j, 1)

  Next j

  semiInt(k, 1) = a * s

Next k

If iIntDif = -1 Then

  For k = 2 To rnMax

    outputData(k, 1) = _

      (semiInt(k, 1) - semiInt(k - 1, 1)) / deltaTime

  Next k

  semiInt(1, 1) = ""

Else

  For k = 1 To rnMax

    outputData(k, 1) = semiInt(k, 1)

  Next k

End If

' Write the result

Selection.Value = outputData

End Sub

'''''''''''''''''''''''''''''''''''''''''''''''''''''''''''

'''''''''''''''''''*********************'''''''''''''''''''

'''''''''''''''''''*                   *'''''''''''''''''''

'''''''''''''''''''*      Mapper       *'''''''''''''''''''

'''''''''''''''''''*                   *'''''''''''''''''''

'''''''''''''''''''*********************'''''''''''''''''''

''''''''''''''''''''''''''''''''''''''''''' (c) R. de Levie

'''''''''''''''''''''''''''''''''''''''''''  v 11, May 2012

' PURPOSE:

'

' This macro reads a highlighted block of data, including

' axis labels at its top and left-hand side. This format

' was chosen in order to be compatible with that of Excel

' 2-D graphs and 3-D surface plots. The macro separates

' these labels, converts the remaining data into RGB colors

' with either a gray scale (for iColor = 10 through 12) or

' one of three sample color schemes provided (iColor = 20

' to 42), and uses these to make a graph. If you wish, look

' at the subroutines BitMap10 through BitMap42 to see how

' it is done, then add your own color scheme.

'

' This macro is most useful when there are many (100 or

' more) data points in both directions. If not, Excel's 3-D

' plot may well be the more satisfactory. Both of them

' require the input data to be equidistant in x as well as

' in y. When the plot contains relatively few data points,

' the simple, bilinear interpolation to generate pixel

' values can cause star-like artifacts, as if one looks at

' a street lamp through a finely woven cloth. This could be

' remedied by using a more sophisticated interpolation

' scheme.

'

' The resulting figure can be moved, enlarged or reduced,

' annotated, and modified as any other XY graph. Data

' points, curves, and text can be added, scales can be

' modified or deleted, etc.

'

' In order to facilitate displaying scales, the macro

' checks whether the top left cell in the highlighted

' array is empty, and the cells below it and to its right

' are filled. Note, however, that the scales displayed

' merely reflect the numbers provided in the top row and

' the left column, and are NOT functional, so that modi-

' fying them afterwards will not rescale or otherwise

' affect the map. Moreover, because this is a macro rather

' than a function, the map does not respond to subsequent

' changes in the input data until the macro is called again

' to update the input information.

'

' When the top left cell is not empty but contains a 1,

' the macro will not autoscale, but will instead ask the

' user for a minimum and maximum values for its scale, as

' may be useful for comparing different maps. In order to

' avoid artifacts from exceeding the minimum and/or maximum

' values, use the widest range that will accomodate all maps.

'

' SUBROUTINES:

'

' The drivers Mapper0 through Mapper333 call the main routine

' Mapper, which in turn calls on a number of subroutines:

' BitmapHeader, WriteAWord, WriteALong, WriteAPixel,

' WriteAByte, and Bitmap0 through Bitmap333. A single digit

' indicates a gradual scale, a double digit a stepwise scale,

' and a triple digit a scale that has its maximum color in

' the middle. If you want to add a different color scheme,

' just add one as Bitmap4 with a corresponding driver

' (Mapper4), using the existing code as example. If you use

' the MBToolbar or MBMenu, make sure to include the new

' driver(s) there.

'

' INPUT:

'

' The macro requires an input array together with the

' corresponding scales, which must be placed in the top

' row and in the left-hand column, leaving the top left-

' hand cell empty.

'

' OUTPUT:

'

' The resulting map can be resized, rescaled, moved, and

' otherwise manipulated as any graph. You can also insert

' text, overlay curves, add inserts, etc.

'

' PROCEDURE:

'

' Highlight the array to be mapped together with the

' corresponding scales, which must be placed in the

' top row and in the left-hand column, leaving the

' top left-hand cell empty. Then call the macro.

'

' WARNING:

'

' This macro goes OUTSIDE OF Excel in order to write

' a bitmap, which is then re-imported as 'background'

' for the graph. Depending on how the rest of the Office

' bundle is configured (or in its absence, when only

' Excel is loaded) the macro may not always run properly.

'

' CREDIT:

'

' This macro was written by William Craig of the Georgetown

' University Chemistry Department.

'

' EXAMPLE:

'

' Use of this macro is illustrated throughout EA3, starting 

' with section 1.5. Section 8.22 describes how to modify

' Mapper’s BitMap to select your own color scheme.

Sub Mapper0()

Dim iColor

iColor = 0

Call Mapper(0)

End Sub

'''''''''''''''''''''''''''''''''''''''''''''''''''''''''''

Sub Mapper00()

Dim iColor

iColor = 10

Call Mapper(10)

End Sub

'''''''''''''''''''''''''''''''''''''''''''''''''''''''''''

Sub Mapper000()

Dim iColor

iColor = 100

Call Mapper(100)

End Sub

'''''''''''''''''''''''''''''''''''''''''''''''''''''''''''

Sub Mapper1()

Dim iColor

iColor = 1

Call Mapper(1)

End Sub

'''''''''''''''''''''''''''''''''''''''''''''''''''''''''''

Sub Mapper11()

Dim iColor

iColor = 11

Call Mapper(11)

End Sub

'''''''''''''''''''''''''''''''''''''''''''''''''''''''''''

Sub Mapper111()

Dim iColor

iColor = 111

Call Mapper(111)

End Sub

'''''''''''''''''''''''''''''''''''''''''''''''''''''''''''

Sub Mapper2()

Dim iColor

iColor = 2

Call Mapper(2)

End Sub

'''''''''''''''''''''''''''''''''''''''''''''''''''''''''''

Sub Mapper22()

Dim iColor

iColor = 22

Call Mapper(22)

End Sub

'''''''''''''''''''''''''''''''''''''''''''''''''''''''''''

Sub Mapper222()

Dim iColor

iColor = 222

Call Mapper(222)

End Sub

'''''''''''''''''''''''''''''''''''''''''''''''''''''''''''

Sub Mapper3()

Dim iColors

iColor = 3

Call Mapper(3)

End Sub

'''''''''''''''''''''''''''''''''''''''''''''''''''''''''''

Sub Mapper33()

Dim iColor

iColor = 33

Call Mapper(33)

End Sub

'''''''''''''''''''''''''''''''''''''''''''''''''''''''''''

Sub Mapper333()

Dim iColor

iColor = 333

Call Mapper(333)

End Sub

'''''''''''''''''''''''''''''''''''''''''''''''''''''''''''

Sub Mapper4()

Dim iColors

iColor = 4

Call Mapper(4)

End Sub

'''''''''''''''''''''''''''''''''''''''''''''''''''''''''''

Sub Mapper(iColor As Integer)

' Nomenclature used:

'

' inputArray    the highlighted block containing the data

'                 plus (at its top and left-hand side)

'                 either axis labels or space for them.

' scaledArray   the input array scaled to fit the range

'                 from 0 to 255.

' valueArray    the interpolated array. It has as many

'                 terms as there are pixels in the plotting

'                 area. All terms are double precision, in

'                 the range from 0 to 255

' pixelArray    the value array expressed in terms of

'                 integers.

' rn            input data height.

' ph            pixel (instead or row) height, starting at

'                 the bottom with 0.

' cw            input data width.

' pw            pixel (instead of column) width, starting

'                 at the left with 0.

' cwMax-1, rhMax-1     number of input data, in terms of

'                 columns and rows respectively.

' hMax, wMax    number of pixels, in terms of height and

'                 width respectively.

' HH, WW        number of pixels per input data point, in

'                 terms of height and width respectively.

Dim cN As Integer, cw As Integer, cwMax As Integer

Dim ph As Integer, pw As Integer

Dim rh As Integer, rN As Integer, rhMax As Integer

Dim i As Integer, iH As Integer, iW As Integer

Dim j As Integer

Dim OutputFileNumber As Integer

Dim maxValue As Double, minValue As Double

Dim Product As Long

Dim XLabelMax As Double, XLabelMin As Double

Dim YLabelMax As Double, YLabelMin As Double

Dim a1 As Integer, a2 As Integer, B1 As Integer

Dim H As Integer, HH As Integer, hScale As Integer

Dim hMax As Integer, hMax0 As Integer  ' plot pixel height

Dim wMax As Integer, wMax0 As Integer  ' plot pixel width

Dim ns As Integer                  ' number of data series

Dim P As Integer, Q As Integer

Dim w As Integer, WW As Integer, wScale As Integer

Dim h1 As Double, h2 As Double

Dim w1 As Double, w2 As Double

Dim InputArray As Variant, scaledArray As Variant

Dim pixelArray As Variant, ValueArray As Variant

Dim OutputFileName As String, Sheetname As String

Dim myRange As Range, rgMax As Range, rgMin As Range

Dim hAnswer, ThisSheet

' Read the inputArray

Begin:

InputArray = Selection.Value

cwMax = Selection.Columns.Count

rhMax = Selection.Rows.Count

' If input area was not highlighted

If rhMax = 1 And cwMax = 1 Then

  hAnswer = MsgBox("You forgot to highlight" _

    & Chr(13) & "the block of input data." _

    & Chr(13) & "Do you want to do so now?" _

    , vbYesNo, "Mapper")

  If hAnswer = vbNo Then End

  If hAnswer = vbYes Then

    Set myRange = Application.InputBox(Prompt:= _

      "The input data are located in:", Type:=8)

    myRange.Select

  End If

  GoTo Begin

End If

Application.ScreenUpdating = False

If rhMax < 3 Then

  MsgBox "The block, including (space for)" & _

    Chr(13) & "a scale at the top row," & _

    Chr(13) & "should be at least 3 cells high."

End

End If

If cwMax < 3 Then

  MsgBox "The block, including (space for)" & _

    Chr(13) & "a scale at the left-most column," & _

    Chr(13) & "should be at least 3 cells wide."

  End

End If

' Make sure that the top left cell is either empty or

' contains a 1, and that its two nearest neighbors are

' not empty. The macro will autoscale when it finds an

' empty top left cell; a value of 1 in that cell will

' prompt two input boxes, for the minimum and maximum

' values respectively of the scale to be used.

'

' Note: the top row and the left-most column are used

' for creating the map scales, and will NOT be plotted.

' On the spreadsheet you might make them clearly

' different from the rest of the array, e.g., by using

' italics, boldface, and/or color.

P = 0

Q = 0

If InputArray(1, 1) = 1 Then P = 1: Q = 1

If IsEmpty(InputArray(1, 1)) Or IsNull(InputArray(1, 1)) _

  Then P = P + 1

If P = 0 Then

  MsgBox "Make sure that the top left cell" & _

  Chr(13) & "of the highlighted area is empty."

  End

End If

If IsEmpty(InputArray(2, 1)) Or IsNull(InputArray(2, 1)) _

  Then P = P + 1

If IsEmpty(InputArray(1, 2)) Or IsNull(InputArray(1, 2)) _

  Then P = P + 1

If P <> 1 Then

  MsgBox "   Make sure that the top row and the" & _

  Chr(13) & "left-hand column contain the plot scales," & _

  Chr(13) & "while leaving the left top cell empty."

  End

End If

' Find maximum and minimum values in XLabels and YLabels

YLabelMax = InputArray(2, 1)

YLabelMin = InputArray(2, 1)

For rN = 3 To rhMax

  If InputArray(rN, 1) > YLabelMax Then YLabelMax = _

    InputArray(rN, 1)

  If InputArray(rN, 1) < YLabelMin Then YLabelMin = _

    InputArray(rN, 1)

Next rN

XLabelMax = InputArray(1, 2)

XLabelMin = InputArray(1, 2)

For cN = 3 To cwMax

  If InputArray(1, cN) > XLabelMax Then XLabelMax = _

    InputArray(1, cN)

  If InputArray(1, cN) < XLabelMin Then XLabelMin = _

    InputArray(1, cN)

Next cN

' Find maximum and minimum values in inputArray

maxValue = InputArray(2, 2)

minValue = InputArray(2, 2)

For rN = 2 To rhMax

  For cN = 2 To cwMax

    If InputArray(rN, cN) > maxValue Then maxValue = _

      InputArray(rN, cN)

    If InputArray(rN, cN) < minValue Then minValue = _

      InputArray(rN, cN)

  Next cN

Next rN

' If Q = 1 ask for possibly different maximum and minimum values

Application.ScreenUpdating = True

If Q = 1 Then

  minValue = InputBox("The minimum value is: ", _

  "(Use this or a lower value)", minValue)

  maxValue = InputBox("The minimum value is: ", _

  "(Use this or a higher value)", maxValue)

End If

Application.ScreenUpdating = False

' Scale the inputArray on a scale from 0 to 255

ReDim scaledArray(0 To rhMax, 0 To cwMax)

For rN = 2 To rhMax

  For cN = 2 To cwMax

    scaledArray(rN, cN) = 255 * (InputArray(rN, cN) _

      - minValue) / (maxValue - minValue)

  Next cN

Next rN

' Draw graph

cw = Columns(1).Width

rh = Rows(1).Height

Sheetname = ActiveSheet.Name

Charts.Add

ActiveChart.ChartType = xlXYScatter

ActiveChart.Location Where:=xlLocationAsObject, Name:=Sheetname

ActiveChart.Axes(xlValue).Select

With ActiveChart.Axes(xlValue)

  .MinimumScaleIsAuto = True

  .MaximumScaleIsAuto = True

  .MajorUnitIsAuto = True

  .MinorUnitIsAuto = True

  .Crosses = xlAutomatic

  .MajorTickMark = xlCross

  .MinorTickMark = xlNone

  .TickLabelPosition = xlLow

End With

Selection.TickLabels.NumberFormat = "General"

ActiveChart.Axes(xlValue).MajorGridlines.Select

Selection.Delete

ActiveChart.Axes(xlCategory).Select

With ActiveChart.Axes(xlCategory)

  .MinimumScaleIsAuto = True

  .MaximumScaleIsAuto = True

  .MajorUnitIsAuto = True

  .MinorUnitIsAuto = True

  .Crosses = xlAutomatic

  .MajorTickMark = xlCross

  .MinorTickMark = xlNone

  .TickLabelPosition = xlLow

End With

With Selection.Border

   .Weight = xlHairline

   .LineStyle = xlAutomatic

End With

ActiveChart.PlotArea.Select

With Selection.Border

  .ColorIndex = 57

  .Weight = xlHairline

  .LineStyle = xlContinuous

End With

' Round plot area height and width to integer

' multiples of the data in the input array

wScale = 1

If wMax > cwMax - 1 Then wScale = Int(wMax0 / (cwMax - 2))

hScale = 1

If hMax > rhMax - 1 Then hScale = Int(hMax0 / (rhMax - 2))

wMax = (cwMax - 2) * wScale

hMax = (rhMax - 2) * hScale

HH = hMax / (rhMax - 2)

WW = wMax / (cwMax - 2)

' Calculate valueArray for cn = 2 to cwMax

ReDim ValueArray(0 To hMax, 0 To wMax) As Double

ReDim pixelArray(0 To hMax, 0 To wMax) As Integer

For cN = 2 To cwMax

  H = hMax

  iH = Int(H / HH)

  h1 = (iH + 1) - H / HH

  h2 = H / HH - iH

  ValueArray(hMax, (cN - 2) * WW) = _

    h1 * scaledArray(iH + 2, cN)

Next cN

For cN = 2 To cwMax

  For H = 0 To hMax - 1

    iH = Int(H / HH)

    h1 = (iH + 1) - H / HH

    h2 = H / HH - iH

    ValueArray(H, (cN - 2) * WW) = _

      h1 * scaledArray(iH + 2, cN) _

      + h2 * scaledArray(iH + 3, cN)

  Next H

Next cN

' Calculate all of valueArray

For H = 0 To hMax

  For w = 1 To wMax - 1

    iW = Int(w / WW)

    w1 = (iW + 1) - w / WW

    w2 = w / WW - iW

    ValueArray(H, w) = w1 * ValueArray(H, iW * WW) + _

       w2 * ValueArray(H, (iW + 1) * WW)

  Next w

Next H

' Convert pixel values to integers

For H = 0 To hMax

  For w = 0 To wMax

    pixelArray(H, w) = CInt(ValueArray(H, w))

  Next w

Next H

'Create, write, and close the bitmap file

OutputFileName = "ColorMap.BMP"

OutputFileNumber = 1

Open OutputFileName For Binary Access Write As _

  #OutputFileNumber

Call BitmapHeader(hMax, wMax)

If iColor = O Then Call BitMap0(hMax, wMax, pixelArray)

If iColor = 10 Then Call BitMap00(hMax, wMax, pixelArray)

If iColor = 100 Then Call BitMap000(hMax, wMax, pixelArray)

If iColor = 1 Then Call BitMap1(hMax, wMax, pixelArray)

If iColor = 11 Then Call BitMap11(hMax, wMax, pixelArray)

If iColor = 111 Then Call BitMap111(hMax, wMax, pixelArray)

If iColor = 2 Then Call BitMap2(hMax, wMax, pixelArray)

If iColor = 22 Then Call BitMap22(hMax, wMax, pixelArray)

If iColor = 222 Then Call BitMap222(hMax, wMax, pixelArray)

If iColor = 3 Then Call BitMap3(hMax, wMax, pixelArray)

If iColor = 33 Then Call BitMap33(hMax, wMax, pixelArray)

If iColor = 333 Then Call BitMap333(hMax, wMax, pixelArray)

If iColor = 4 Then Call BitMap4(hMax, wMax, pixelArray)

Close #OutputFileNumber

ThisSheet = ActiveSheet.Name

ActiveChart.PlotArea.Fill.UserPicture OutputFileName

' Maximize the plot area

ActiveChart.PlotArea.Select

Selection.Fill.Visible = True

' Set the axis labels

ActiveChart.Axes(xlCategory).Select

ActiveChart.Axes(xlCategory).MinimumScale = XLabelMin

ActiveChart.Axes(xlCategory).MaximumScale = XLabelMax

ActiveChart.Axes(xlValue).Select

ActiveChart.Axes(xlValue).MinimumScale = YLabelMin

ActiveChart.Axes(xlValue).MaximumScale = YLabelMax

' Remove all markers

ns = ActiveChart.SeriesCollection.Count

For i = 1 To ns

  ActiveChart.SeriesCollection(i).Select

  ActiveChart.SeriesCollection(i).MarkerStyle = xlNone

Next i

ActiveChart.ChartArea.Select

End Sub

'''''''''''''''''''''''''''''''''''''''''''''''''''''''''''

' This routine writes a minimal 54 byte bitmap file header.

Private Sub BitmapHeader(ph As Integer, pw As Integer)

' Values for pixel width pW, pixel height pH, ImageSize-

' InByte and TotalFileSize change with the image size.

'

Dim RowLengthInBytes  As Long

Dim ImageSizeInByte As Long

Dim TotalFileSize As Long

' Calculate three values needed write header. In

' particular, row length is the number of bytes

' rounded up to the next DWORD (4 byte) boundary.

RowLengthInBytes = (((pw + 1) * 3) \ 4) * 4

ImageSizeInByte = RowLengthInBytes * ph

TotalFileSize = 54 + (ph * RowLengthInBytes)

' Write the BitmapFileHeader

WriteAWord 19778        ' bfType   The two characters BM

WriteALong TotalFileSize        ' bfSize   Headers + image

WriteALong 0        ' bfReserved1&2  Two words of zero.

WriteALong 54       ' bfOffBits      Byte offset to image.

' Write the BitmapInfoHeader

WriteALong 40   ' biSize       BITMAPINFOHEADER size = 40.

WriteALong pw       ' biWidth           Image width.

WriteALong ph       ' biHeight          Image width.

WriteAWord 1        ' biPlanes          Always 1.

WriteAWord 24       ' biBitCount   Always 24 bits per pixel.

WriteALong 0        ' biCompression     No compression = 0

WriteALong ImageSizeInByte

                ' biSizeImage       Image size in bytes

WriteALong 3000     ' biXPelsPerMeter   A safe value

WriteALong 3000     ' biYPelsPerMeter   A safe value

WriteALong 0        ' biClrUsed         Always 0

WriteALong 0        ' biClrImportant    Always 0

End Sub

'''''''''''''''''''''''''''''''''''''''''''''''''''''''''''

Private Sub BitMap0(hMax As Integer, _

  wMax As Integer, pixelArray As Variant)

' For a continuous gray-scale, without color,

' from black to white

Dim H As Integer, w As Integer, RedVal As Integer

Dim GreenVal As Integer, BlueVal As Integer

For H = hMax To 0 Step -1

  For w = 0 To wMax - 1

    RedVal = pixelArray(H, w)

    GreenVal = pixelArray(H, w)

    BlueVal = pixelArray(H, w)

    WriteAPixel RedVal Mod 256, GreenVal Mod 256, _

      BlueVal Mod 256

  Next w

' Do not change the following, essential row padding

  w = wMax * 3

  Do While (w Mod 4) <> 0

    WriteAByte 0

    w = w + 1

  Loop

Next H

End Sub

'''''''''''''''''''''''''''''''''''''''''''''''''''''''''''

Private Sub BitMap00(hMax As Integer, _

  wMax As Integer, pixelArray As Variant)

' For quasi-contour diagram using 9 different

' shades of gray, from black to white

Dim H As Integer, w As Integer

For H = hMax To 0 Step -1

  For w = 0 To wMax - 1

    If pixelArray(H, w) < 16 Then

      RedVal = 0

      GreenVal = 0

      BlueVal = 0

    ElseIf pixelArray(H, w) < 48 Then

      RedVal = 32

      GreenVal = 32

      BlueVal = 32

    ElseIf pixelArray(H, w) < 80 Then

      RedVal = 64

      GreenVal = 64

      BlueVal = 64

    ElseIf pixelArray(H, w) < 112 Then

      RedVal = 96

      GreenVal = 96

      BlueVal = 96

    ElseIf pixelArray(H, w) < 144 Then

      RedVal = 128

      GreenVal = 128

      BlueVal = 128

    ElseIf pixelArray(H, w) < 176 Then

      RedVal = 160

      GreenVal = 160

      BlueVal = 160

    ElseIf pixelArray(H, w) < 208 Then

      RedVal = 192

      GreenVal = 192

      BlueVal = 192

    ElseIf pixelArray(H, w) < 240 Then

      RedVal = 224

      GreenVal = 224

      BlueVal = 224

    ElseIf pixelArray(H, w) >= 240 Then

      RedVal = 255

      GreenVal = 255

      BlueVal = 255

    End If

    WriteAPixel RedVal Mod 256, GreenVal Mod 256, _

      BlueVal Mod 256

  Next w

' Do not change the following, essential row padding

  w = wMax * 3

  Do While (w Mod 4) <> 0

    WriteAByte 0

    w = w + 1

  Loop

Next H

End Sub

'''''''''''''''''''''''''''''''''''''''''''''''''''''''''''

Private Sub BitMap000(hMax As Integer, _

  wMax As Integer, pixelArray As Variant)

' For quasi-contour diagram with 17 bands of 9

' shades of gray, from black to white to black

Dim H As Integer, w As Integer

For H = hMax To 0 Step -1

  For w = 0 To wMax - 1

    If pixelArray(H, w) < 8 Then

      RedVal = 0

      GreenVal = 0

      BlueVal = 0

    ElseIf pixelArray(H, w) < 24 Then

      RedVal = 32

      GreenVal = 32

      BlueVal = 32

    ElseIf pixelArray(H, w) < 40 Then

      RedVal = 64

      GreenVal = 64

      BlueVal = 64

    ElseIf pixelArray(H, w) < 56 Then

      RedVal = 96

      GreenVal = 96

      BlueVal = 96

    ElseIf pixelArray(H, w) < 72 Then

      RedVal = 128

      GreenVal = 128

      BlueVal = 128

    ElseIf pixelArray(H, w) < 88 Then

      RedVal = 160

      GreenVal = 160

      BlueVal = 160

    ElseIf pixelArray(H, w) < 104 Then

      RedVal = 192

      GreenVal = 192

      BlueVal = 192

    ElseIf pixelArray(H, w) < 120 Then

      RedVal = 224

      GreenVal = 224

      BlueVal = 224

    ElseIf pixelArray(H, w) < 136 Then

      RedVal = 255

      GreenVal = 255

      BlueVal = 255

    ElseIf pixelArray(H, w) < 152 Then

      RedVal = 224

      GreenVal = 224

      BlueVal = 224

    ElseIf pixelArray(H, w) < 168 Then

      RedVal = 192

      GreenVal = 192

      BlueVal = 192

    ElseIf pixelArray(H, w) < 184 Then

      RedVal = 160

      GreenVal = 160

      BlueVal = 160

    ElseIf pixelArray(H, w) < 200 Then

      RedVal = 128

      GreenVal = 128

      BlueVal = 128

    ElseIf pixelArray(H, w) < 216 Then

      RedVal = 96

      GreenVal = 96

      BlueVal = 96

    ElseIf pixelArray(H, w) < 232 Then

      RedVal = 64

      GreenVal = 64

      BlueVal = 64

    ElseIf pixelArray(H, w) < 248 Then

      RedVal = 32

      GreenVal = 32

      BlueVal = 32

    Else

      RedVal = 0

      GreenVal = 0

      BlueVal = 0

End If

    WriteAPixel RedVal Mod 256, GreenVal Mod 256, _

      BlueVal Mod 256

  Next w

' Do not change the following, essential row padding

  w = wMax * 3

  Do While (w Mod 4) <> 0

    WriteAByte 0

    w = w + 1

  Loop

Next H

End Sub

'''''''''''''''''''''''''''''''''''''''''''''''''''''''''''

Private Sub BitMap1(hMax As Integer, _

  wMax As Integer, pA As Variant)

' For a continuously changing, multicolor map;

' pixelArray(H, w) abbreviated to pA(H, w)

Dim H As Integer, w As Integer, RedVal As Long

Dim GreenVal As Long, BlueVal As Long

For H = hMax To 0 Step -1

  For w = 0 To wMax - 1

    If pA(H, w) < 16 Then

      RedVal = Int(60 + 2.42 * (pA(H, w) - 4) _

        + 0 * ((pA(H, w) - 4) ^ 2) _

        + 0.000523 * ((pA(H, w) - 4) ^ 3))

      GreenVal = Int(5 - 0.003323 * (pA(H, w) - 4) _

        + 0 * ((pA(H, w) - 4) ^ 2) _

        + 0.0000231 * ((pA(H, w) - 4) ^ 3))

      BlueVal = Int(90 + 2.097 * (pA(H, w) - 4) _

        + 0 * ((pA(H, w) - 4) ^ 2) _

        + 0.0028 * ((pA(H, w) - 4) ^ 3))

    ElseIf pA(H, w) < 32 Then

      RedVal = Int(90 + 2.65 * (pA(H, w) - 16) _

        + 0.0191 * ((pA(H, w) - 16) ^ 2) _

        - 0.000566 * ((pA(H, w) - 16) ^ 3))

      GreenVal = Int(5 + 0.00665 * (pA(H, w) - 16) _

        + 0.00831 * ((pA(H, w) - 16) ^ 2) _

        - 0.0000779 * ((pA(H, w) - 16) ^ 3))

      BlueVal = Int(120 + 3.306 * (pA(H, w) - 16) _

        + 0.101 * ((pA(H, w) - 16) ^ 2) _

        - 0.00945 * ((pA(H, w) - 16) ^ 3))

    ElseIf pA(H, w) < 48 Then

      RedVal = Int(135 + 2.83 * (pA(H, w) - 32) _

        - 0.0081 * ((pA(H, w) - 32) ^ 2) _

        + 0.000447 * ((pA(H, w) - 32) ^ 3))

      GreenVal = Int(5 - 0.0266 * (pA(H, w) - 32) _

        - 0.00291 * ((pA(H, w) - 32) ^ 2) _

        + 0.000286 * ((pA(H, w) - 32) ^ 3))

      BlueVal = Int(160 - 0.726 * (pA(H, w) - 32) _

        - 0.353 * ((pA(H, w) - 32) ^ 2) _

        + 0.0078 * ((pA(H, w) - 32) ^ 3))

    ElseIf pA(H, w) < 64 Then

      RedVal = Int(180 + 2.91 * (pA(H, w) - 48) _

        + 0.0134 * ((pA(H, w) - 48) ^ 2) _

        - 0.00122 * ((pA(H, w) - 48) ^ 3))

      GreenVal = Int(5 + 0.0997 * (pA(H, w) - 48) _

        + 0.0108 * ((pA(H, w) - 48) ^ 2) _

        + 0.00138 * ((pA(H, w) - 48) ^ 3))

      BlueVal = Int(90 - 6.028 * (pA(H, w) - 48) _

        + 0.0215 * ((pA(H, w) - 48) ^ 2) _

        + 0.003893 * ((pA(H, w) - 48) ^ 3))

    ElseIf pA(H, w) < 80 Then

      RedVal = Int(225 + 2.4 * (pA(H, w) - 64) _

        - 0.0453 * ((pA(H, w) - 64) ^ 2) _

        - 0.000442 * ((pA(H, w) - 64) ^ 3))

      GreenVal = Int(15 + 1.503 * (pA(H, w) - 64) _

        + 0.0769 * ((pA(H, w) - 64) ^ 2) _

        + 0.00129 * ((pA(H, w) - 64) ^ 3))

      BlueVal = Int(15 - 2.351 * (pA(H, w) - 64) _

        + 0.208 * ((pA(H, w) - 64) ^ 2) _

        - 0.00506 * ((pA(H, w) - 64) ^ 3))

    ElseIf pA(H, w) < 96 Then

      RedVal = Int(250 + 0.611 * (pA(H, w) - 80) _

        - 0.0665 * ((pA(H, w) - 80) ^ 2) _

        + 0.00177 * ((pA(H, w) - 80) ^ 3))

      GreenVal = Int(64 + 4.591 * (pA(H, w) - 80) _

        + 0.139 * ((pA(H, w) - 80) ^ 2) _

        - 0.00457 * ((pA(H, w) - 80) ^ 3))

      BlueVal = Int(10 + 0.4307 * (pA(H, w) - 80) _

        - 0.0345 * ((pA(H, w) - 80) ^ 2) _

        + 0.00169 * ((pA(H, w) - 80) ^ 3))

    ElseIf pA(H, w) < 112 Then

      RedVal = Int(250 - 0.158 * (pA(H, w) - 96) _

        + 0.0185 * ((pA(H, w) - 96) ^ 2) _

        - 0.000536 * ((pA(H, w) - 96) ^ 3))

      GreenVal = Int(160 + 5.879 * (pA(H, w) - 96) _

        - 0.0806 * ((pA(H, w) - 96) ^ 2) _

        - 0.000835 * ((pA(H, w) - 96) ^ 3))

      BlueVal = Int(15 + 0.628 * (pA(H, w) - 96) _

        + 0.0468 * ((pA(H, w) - 96) ^ 2) _

        + 0.00317 * ((pA(H, w) - 96) ^ 3))

    ElseIf pA(H, w) < 128 Then

      RedVal = Int(250 + 0.02063 * (pA(H, w) - 112) _

        - 0.00729 * ((pA(H, w) - 112) ^ 2) _

        - 0.00451 * ((pA(H, w) - 112) ^ 3))

      GreenVal = Int(230 + 2.657 * (pA(H, w) - 112) _

        - 0.121 * ((pA(H, w) - 112) ^ 2) _

        + 0.00205 * ((pA(H, w) - 112) ^ 3))

      BlueVal = Int(50 + 4.557 * (pA(H, w) - 112) _

        + 0.199 * ((pA(H, w) - 112) ^ 2) _

        - 0.0131 * ((pA(H, w) - 112) ^ 3))

    ElseIf pA(H, w) < 144 Then

      RedVal = Int(230 - 3.675 * (pA(H, w) - 128) _

        - 0.224 * ((pA(H, w) - 128) ^ 2) _

        + 0.00343 * ((pA(H, w) - 128) ^ 3))

      GreenVal = Int(250 + 0.3679 * (pA(H, w) - 128) _

        - 0.0223 * ((pA(H, w) - 128) ^ 2) _

        - 0.0000413 * ((pA(H, w) - 128) ^ 3))

      BlueVal = Int(120 + 0.8302 * (pA(H, w) - 128) _

        - 0.432 * ((pA(H, w) - 128) ^ 2) _

        + 0.00909 * ((pA(H, w) - 128) ^ 3))

    ElseIf pA(H, w) < 160 Then

      RedVal = Int(128 - 8.197 * (pA(H, w) - 144) _

        - 0.059 * ((pA(H, w) - 144) ^ 2) _

        + 0.00812 * ((pA(H, w) - 144) ^ 3))

      GreenVal = Int(250 - 0.3784 * (pA(H, w) - 144) _

        - 0.0243 * ((pA(H, w) - 144) ^ 2) _

        - 0.00189 * ((pA(H, w) - 144) ^ 3))

      BlueVal = Int(60 - 6.003 * (pA(H, w) - 144) _

        + 0.00463 * ((pA(H, w) - 144) ^ 2) _

        + 0.0134 * ((pA(H, w) - 144) ^ 3))

    ElseIf pA(H, w) < 176 Then

      RedVal = Int(15 - 3.849 * (pA(H, w) - 160) _

        + 0.331 * ((pA(H, w) - 160) ^ 2) _

        - 0.00808 * ((pA(H, w) - 160) ^ 3))

      GreenVal = Int(230 - 2.604 * (pA(H, w) - 160) _

        - 0.115 * ((pA(H, w) - 160) ^ 2) _

        + 0.000257 * ((pA(H, w) - 160) ^ 3))

      BlueVal = Int(20 + 4.432 * (pA(H, w) - 160) _

        + 0.648 * ((pA(H, w) - 160) ^ 2) _

        - 0.0212 * ((pA(H, w) - 160) ^ 3))

    ElseIf pA(H, w) < 192 Then

      RedVal = Int(5 + 0.529 * (pA(H, w) - 176) _

        - 0.0572 * ((pA(H, w) - 176) ^ 2) _

        + 0.00151 * ((pA(H, w) - 176) ^ 3))

      GreenVal = Int(160 - 6.08 * (pA(H, w) - 176) _

        - 0.102 * ((pA(H, w) - 176) ^ 2) _

        + 0.00574 * ((pA(H, w) - 176) ^ 3))

      BlueVal = Int(170 + 8.9 * (pA(H, w) - 176) _

        - 0.368 * ((pA(H, w) - 176) ^ 2) _

        + 0.00632 * ((pA(H, w) - 176) ^ 3))

    ElseIf pA(H, w) < 208 Then

      RedVal = Int(5 - 0.142 * (pA(H, w) - 192) _

        + 0.0153 * ((pA(H, w) - 192) ^ 2) _

        - 0.000399 * ((pA(H, w) - 192) ^ 3))

      GreenVal = Int(60 - 4.951 * (pA(H, w) - 192) _

        + 0.173 * ((pA(H, w) - 192) ^ 2) _

        - 0.00246 * ((pA(H, w) - 192) ^ 3))

      BlueVal = Int(244 + 1.969 * (pA(H, w) - 192) _

        - 0.0648 * ((pA(H, w) - 192) ^ 2) _

        - 0.00634 * ((pA(H, w) - 192) ^ 3))

    ElseIf pA(H, w) < 224 Then

      RedVal = Int(5 + 0.0403 * (pA(H, w) - 208) _

        - 0.00386 * ((pA(H, w) - 208) ^ 2) _

        + 0.0013 * ((pA(H, w) - 208) ^ 3))

      GreenVal = Int(15 - 1.303 * (pA(H, w) - 208) _

        + 0.055 * ((pA(H, w) - 208) ^ 2) _

        - 0.000791 * ((pA(H, w) - 208) ^ 3))

      BlueVal = Int(244 - 2.902 * (pA(H, w) - 208) _

        - 0.24 * ((pA(H, w) - 208) ^ 2) _

        + 0.00581 * ((pA(H, w) - 208) ^ 3))

    ElseIf pA(H, w) < 240 Then

      RedVal = Int(10 + 0.9186 * (pA(H, w) - 224) _

        + 0.0586 * ((pA(H, w) - 224) ^ 2) _

        - 0.00482 * ((pA(H, w) - 224) ^ 3))

      GreenVal = Int(5 - 0.1494 * (pA(H, w) - 224) _

        + 0.0171 * ((pA(H, w) - 224) ^ 2) _

        - 0.000484 * ((pA(H, w) - 224) ^ 3))

      BlueVal = Int(160 - 6.11 * (pA(H, w) - 224) _

        + 0.0392 * ((pA(H, w) - 224) ^ 2) _

        - 0.003 * ((pA(H, w) - 224) ^ 3))

    ElseIf pA(H, w) >= 240 Then

      RedVal = Int(20 - 0.902 * (pA(H, w) - 240) _

        - 0.173 * ((pA(H, w) - 240) ^ 2) _

        + 0.0139 * ((pA(H, w) - 240) ^ 3))

      GreenVal = Int(5 + 0.02554 * (pA(H, w) - 240) _

        - 0.00614 * ((pA(H, w) - 240) ^ 2) _

        + 0.000357 * ((pA(H, w) - 240) ^ 3))

      BlueVal = Int(60 - 7.157 * (pA(H, w) - 240) _

        - 0.105 * ((pA(H, w) - 240) ^ 2) _

        + 0.0298 * ((pA(H, w) - 240) ^ 3))

    End If

    WriteAPixel RedVal Mod 256, GreenVal Mod 256, _

      BlueVal Mod 256

  Next w

' Do not change the following, essential row padding

  w = wMax * 3

  Do While (w Mod 4) <> 0

    WriteAByte 0

    w = w + 1

  Loop

Next H

End Sub

'''''''''''''''''''''''''''''''''''''''''''''''''''''''''''

Private Sub BitMap11(hMax As Integer, _

  wMax As Integer, pixelArray As Variant)

' For a quasi-contour diagram using 9 different colors

Dim H As Integer, w As Integer

For H = hMax To 0 Step -1

  For w = 0 To wMax - 1

' Dark brown:

    If pixelArray(H, w) < 16 Then

      RedVal = 100

      GreenVal = 0

      BlueVal = 50

' Brown:

    ElseIf pixelArray(H, w) < 48 Then

      RedVal = 150

      GreenVal = 0

      BlueVal = 40

' Red:

    ElseIf pixelArray(H, w) < 80 Then

      RedVal = 255

      GreenVal = 0

      BlueVal = 0

' Orange:

    ElseIf pixelArray(H, w) < 112 Then

      RedVal = 255

      GreenVal = 128

      BlueVal = 0

' Yellow:

    ElseIf pixelArray(H, w) < 144 Then

      RedVal = 255

      GreenVal = 255

      BlueVal = 0

' Green:

    ElseIf pixelArray(H, w) < 176 Then

      RedVal = 0

      GreenVal = 255

      BlueVal = 0

' Light blue:

    ElseIf pixelArray(H, w) < 208 Then

      RedVal = 0

      GreenVal = 255

      BlueVal = 255

' Blue:

    ElseIf pixelArray(H, w) < 240 Then

      RedVal = 0

      GreenVal = 20

      BlueVal = 255

' Violet:

    Else

      RedVal = 20

      GreenVal = 0

      BlueVal = 150

    End If

    WriteAPixel RedVal Mod 256, GreenVal Mod 256, _

      BlueVal Mod 256

  Next w

' Do not change the following, essential row padding

  w = wMax * 3

  Do While (w Mod 4) <> 0

    WriteAByte 0

    w = w + 1

  Loop

Next H

End Sub

'''''''''''''''''''''''''''''''''''''''''''''''''''''''''''

Private Sub BitMap111(hMax As Integer, _

  wMax As Integer, pixelArray As Variant)

' For a quasi-contour diagram using 9 different colors

Dim H As Integer, w As Integer

For H = hMax To 0 Step -1

  For w = 0 To wMax - 1

' Near black

    If pixelArray(H, w) < 8 Then

      RedVal = 60

      GreenVal = 0

      BlueVal = 90

' Dark Violet:

    ElseIf pixelArray(H, w) < 24 Then

      RedVal = 90

      GreenVal = 0

      BlueVal = 120

' Violet:

    ElseIf pixelArray(H, w) < 40 Then

      RedVal = 120

      GreenVal = 0

      BlueVal = 160

' Dark red

    ElseIf pixelArray(H, w) < 56 Then

      RedVal = 180

      GreenVal = 0

      BlueVal = 90

' Red:

    ElseIf pixelArray(H, w) < 72 Then

      RedVal = 255

      GreenVal = 0

      BlueVal = 0

' Dark Orange:

    ElseIf pixelArray(H, w) < 88 Then

      RedVal = 255

      GreenVal = 64

      BlueVal = 0

' Orange:

    ElseIf pixelArray(H, w) < 104 Then

      RedVal = 255

      GreenVal = 128

      BlueVal = 0

' Yellow:

    ElseIf pixelArray(H, w) < 120 Then

      RedVal = 255

      GreenVal = 255

      BlueVal = 0

' Light yellow:

    ElseIf pixelArray(H, w) < 136 Then

      RedVal = 255

      GreenVal = 255

      BlueVal = 128

' Light green:

    ElseIf pixelArray(H, w) < 152 Then

      RedVal = 128

      GreenVal = 255

      BlueVal = 128

' Green:

    ElseIf pixelArray(H, w) < 168 Then

      RedVal = 0

      GreenVal = 255

      BlueVal = 0

' Blue-green:

    ElseIf pixelArray(H, w) < 184 Then

      RedVal = 0

      GreenVal = 160

      BlueVal = 160

' Light blue:

    ElseIf pixelArray(H, w) < 200 Then

      RedVal = 0

      GreenVal = 60

      BlueVal = 255

' Blue:

    ElseIf pixelArray(H, w) < 216 Then

      RedVal = 0

      GreenVal = 0

      BlueVal = 255

' Dark blue:

    ElseIf pixelArray(H, w) < 232 Then

      RedVal = 0

      GreenVal = 0

      BlueVal = 180

' Blue-black:

    ElseIf pixelArray(H, w) < 248 Then

      RedVal = 20

      GreenVal = 0

      BlueVal = 120

' Near black:

    Else

      RedVal = 15

      GreenVal = 0

      BlueVal = 60

    End If

    WriteAPixel RedVal Mod 256, GreenVal Mod 256, _

      BlueVal Mod 256

  Next w

' Do not change the following, essential row padding

  w = wMax * 3

  Do While (w Mod 4) <> 0

    WriteAByte 0

    w = w + 1

  Loop

Next H

End Sub

'''''''''''''''''''''''''''''''''''''''''''''''''''''''''''

Private Sub BitMap2(hMax As Integer, _

  wMax As Integer, pixelArray As Variant)

' For a map with continuous colors, from brown

' through red, orange and yellow to white

Dim H As Integer, w As Integer, RedVal As Integer

Dim GreenVal As Integer, BlueVal As Integer

For H = hMax To 0 Step -1

  For w = 0 To wMax - 1

' from black to red

     If pixelArray(H, w) < 64 Then

      RedVal = Int(4 * pixelArray(H, w))

      GreenVal = 0

      BlueVal = 0

 ' from red to orange

     ElseIf pixelArray(H, w) < 128 Then

      RedVal = 255

      GreenVal = Int(2 * (pixelArray(H, w) - 64))

      BlueVal = 0

 ' from orange to yellow

     ElseIf pixelArray(H, w) < 192 Then

      RedVal = 255

      GreenVal = Int(2 * (pixelArray(H, w) - 64))

      BlueVal = 0

 ' from yellow to white

     Else

      RedVal = 255

      GreenVal = 255

      BlueVal = Int(4 * (pixelArray(H, w) - 192))

    End If

    WriteAPixel RedVal Mod 256, GreenVal Mod 256, _

      BlueVal Mod 256

  Next w

' Do not change the following, essential row padding

  w = wMax * 3

  Do While (w Mod 4) <> 0

    WriteAByte 0

    w = w + 1

  Loop

Next H

End Sub

'''''''''''''''''''''''''''''''''''''''''''''''''''''''''''

Private Sub BitMap22(hMax As Integer, _

  wMax As Integer, pixelArray As Variant)

' For a quasi-contour map with colors from brown

' through red, orange and yellow to white

Dim H As Integer, w As Integer, RedVal As Integer

Dim GreenVal As Integer, BlueVal As Integer

For H = hMax To 0 Step -1

  For w = 0 To wMax - 1

' black:

    If pixelArray(H, w) < 16 Then

      RedVal = 0

      GreenVal = 0

      BlueVal = 0

' dark brown:

    ElseIf pixelArray(H, w) < 48 Then

      RedVal = 80

      GreenVal = 0

      BlueVal = 0

' brown:

    ElseIf pixelArray(H, w) < 80 Then

      RedVal = 160

      GreenVal = 0

      BlueVal = 0

' red:

    ElseIf pixelArray(H, w) < 112 Then

      RedVal = 255

      GreenVal = 0

      BlueVal = 0

' dark orange:

    ElseIf pixelArray(H, w) < 144 Then

      RedVal = 255

      GreenVal = 64

      BlueVal = 0

' dark orange:

    ElseIf pixelArray(H, w) < 176 Then

      RedVal = 255

      GreenVal = 96

      BlueVal = 0

' orange:

    ElseIf pixelArray(H, w) < 208 Then

      RedVal = 255

      GreenVal = 144

      BlueVal = 0

' Yellow:

    ElseIf pixelArray(H, w) < 240 Then

      RedVal = 255

      GreenVal = 255

      BlueVal = 0

' light yellow:

    Else

      RedVal = 255

      GreenVal = 255

      BlueVal = 128

    End If

    WriteAPixel RedVal Mod 256, GreenVal Mod 256, _

      BlueVal Mod 256

  Next w

' Do not change the following, essential row padding

  w = wMax * 3

  Do While (w Mod 4) <> 0

    WriteAByte 0

    w = w + 1

  Loop

Next H

End Sub

'''''''''''''''''''''''''''''''''''''''''''''''''''''''''''

Private Sub BitMap222(hMax As Integer, _

  wMax As Integer, pixelArray As Variant)

' For a quasi-contour map with colors from brown

' through red, orange and yellow to white

Dim H As Integer, w As Integer, RedVal As Integer

Dim GreenVal As Integer, BlueVal As Integer

For H = hMax To 0 Step -1

  For w = 0 To wMax - 1

' black:

    If pixelArray(H, w) < 8 Then

      RedVal = 0

      GreenVal = 0

      BlueVal = 0

' dark brown:

    ElseIf pixelArray(H, w) < 24 Then

      RedVal = 80

      GreenVal = 0

      BlueVal = 0

' brown:

    ElseIf pixelArray(H, w) < 40 Then

      RedVal = 120

      GreenVal = 0

      BlueVal = 0

' dark red:

    ElseIf pixelArray(H, w) < 56 Then

      RedVal = 192

      GreenVal = 0

      BlueVal = 0

' red:

    ElseIf pixelArray(H, w) < 72 Then

      RedVal = 255

      GreenVal = 0

      BlueVal = 0

' dark orange:

    ElseIf pixelArray(H, w) < 88 Then

      RedVal = 255

      GreenVal = 64

      BlueVal = 0

' orange:

    ElseIf pixelArray(H, w) < 104 Then

      RedVal = 255

      GreenVal = 128

      BlueVal = 0

' light orange:

    ElseIf pixelArray(H, w) < 120 Then

      RedVal = 255

      GreenVal = 192

      BlueVal = 0

' yellow:

    ElseIf pixelArray(H, w) < 136 Then

      RedVal = 255

      GreenVal = 255

      BlueVal = 0

' light orange:

    ElseIf pixelArray(H, w) < 152 Then

      RedVal = 255

      GreenVal = 192

      BlueVal = 0

' orange:

    ElseIf pixelArray(H, w) < 168 Then

      RedVal = 255

      GreenVal = 128

      BlueVal = 0

' dark orange:

    ElseIf pixelArray(H, w) < 184 Then

      RedVal = 255

      GreenVal = 64

      BlueVal = 0

' red:

    ElseIf pixelArray(H, w) < 200 Then

      RedVal = 255

      GreenVal = 0

      BlueVal = 0

' dark red:

    ElseIf pixelArray(H, w) < 216 Then

      RedVal = 192

      GreenVal = 0

      BlueVal = 0

' brown:

    ElseIf pixelArray(H, w) < 232 Then

      RedVal = 120

      GreenVal = 0

      BlueVal = 0

' dark brown:

    ElseIf pixelArray(H, w) < 248 Then

      RedVal = 80

      GreenVal = 0

      BlueVal = 0

' black:

    Else

      RedVal = 0

      GreenVal = 0

      BlueVal = 0

    End If

    WriteAPixel RedVal Mod 256, GreenVal Mod 256, _

      BlueVal Mod 256

  Next w

' Do not change the following, essential row padding

  w = wMax * 3

  Do While (w Mod 4) <> 0

    WriteAByte 0

    w = w + 1

  Loop

Next H

End Sub

'''''''''''''''''''''''''''''''''''''''''''''''''''''''''''

Private Sub BitMap3(hMax As Integer, _

  wMax As Integer, pixelArray As Variant)

' For a map with continuous colors,

' from black through blue to white

Dim H As Integer, w As Integer, RedVal As Integer

Dim GreenVal As Integer, BlueVal As Integer

For H = hMax To 0 Step -1

  For w = 0 To wMax - 1

' from black to dark blue

    If pixelArray(H, w) < 64 Then

      RedVal = 0

      GreenVal = 0

      BlueVal = Int(4 * pixelArray(H, w))

' from dark blue to light blue

    ElseIf pixelArray(H, w) < 192 Then

      RedVal = 0

      GreenVal = Int(2 * (pixelArray(H, w) - 64))

      BlueVal = 255

' from light blue to white

    Else

      RedVal = Int(4 * (pixelArray(H, w) - 192))

      GreenVal = 255

      BlueVal = 255

    End If

    WriteAPixel RedVal Mod 256, GreenVal Mod 256, _

      BlueVal Mod 256

  Next w

' Do not change the following, essential row padding

  w = wMax * 3

  Do While (w Mod 4) <> 0

    WriteAByte 0

    w = w + 1

  Loop

Next H

End Sub

'''''''''''''''''''''''''''''''''''''''''''''''''''''''''''

Private Sub BitMap33(hMax As Integer, _

  wMax As Integer, pixelArray As Variant)

' For a quasi-contour map with 9 colors,

' from black through blue to white

Dim H As Integer, w As Integer, RedVal As Integer

Dim GreenVal As Integer, BlueVal As Integer

For H = hMax To 0 Step -1

  For w = 0 To wMax - 1

' black:

    If pixelArray(H, w) < 16 Then

      RedVal = 0

      GreenVal = 0

      BlueVal = 0

' very dark blue:

    ElseIf pixelArray(H, w) < 48 Then

      RedVal = 0

      GreenVal = 0

      BlueVal = 80

' dark blue:

    ElseIf pixelArray(H, w) < 80 Then

      RedVal = 0

      GreenVal = 0

      BlueVal = 160

' blue

    ElseIf pixelArray(H, w) < 112 Then

      RedVal = 0

      GreenVal = 0

      BlueVal = 255

' gray-blue

    ElseIf pixelArray(H, w) < 144 Then

      RedVal = 0

      GreenVal = 64

      BlueVal = 255

' light gray-blue

    ElseIf pixelArray(H, w) < 176 Then

      RedVal = 0

      GreenVal = 154

      BlueVal = 255

' dark green-blue

    ElseIf pixelArray(H, w) < 208 Then

      RedVal = 64

      GreenVal = 196

      BlueVal = 255

' green-blue

    ElseIf pixelArray(H, w) < 240 Then

      RedVal = 64

      GreenVal = 255

      BlueVal = 255

' light green-blue

    Else

      RedVal = 128

      GreenVal = 255

      BlueVal = 255

    End If

    WriteAPixel RedVal Mod 256, GreenVal Mod 256, _

      BlueVal Mod 256

  Next w

' Do not change the following, essential row padding

  w = wMax * 3

  Do While (w Mod 4) <> 0

    WriteAByte 0

    w = w + 1

  Loop

Next H

End Sub

'''''''''''''''''''''''''''''''''''''''''''''''''''''''''''

Private Sub BitMap333(hMax As Integer, _

  wMax As Integer, pixelArray As Variant)

' For a quasi-contour map with 17 bands using 9 colors,

' from black through blue to light blue and back

Dim H As Integer, w As Integer, RedVal As Integer

Dim GreenVal As Integer, BlueVal As Integer

For H = hMax To 0 Step -1

  For w = 0 To wMax - 1

' black

    If pixelArray(H, w) < 8 Then

      RedVal = 0

      GreenVal = 0

      BlueVal = 0

' blue-black

    ElseIf pixelArray(H, w) < 24 Then

      RedVal = 0

      GreenVal = 0

      BlueVal = 160

' dark blue

    ElseIf pixelArray(H, w) < 40 Then

      RedVal = 0

      GreenVal = 0

      BlueVal = 255

' blue

    ElseIf pixelArray(H, w) < 56 Then

      RedVal = 0

      GreenVal = 48

      BlueVal = 255

' gray-blue

    ElseIf pixelArray(H, w) < 72 Then

      RedVal = 0

      GreenVal = 96

      BlueVal = 255

' light gray-blue

    ElseIf pixelArray(H, w) < 88 Then

      RedVal = 0

      GreenVal = 154

      BlueVal = 255

' dark green-blue

    ElseIf pixelArray(H, w) < 104 Then

      RedVal = 64

      GreenVal = 196

      BlueVal = 255

' green-blue

    ElseIf pixelArray(H, w) < 120 Then

      RedVal = 64

      GreenVal = 255

      BlueVal = 255

' light green-blue

    ElseIf pixelArray(H, w) < 136 Then

      RedVal = 128

      GreenVal = 255

      BlueVal = 255

' green-blue

    ElseIf pixelArray(H, w) < 152 Then

      RedVal = 64

      GreenVal = 255

      BlueVal = 255

 ' dark green-blue

    ElseIf pixelArray(H, w) < 168 Then

      RedVal = 64

      GreenVal = 196

      BlueVal = 255

' light gray-blue

    ElseIf pixelArray(H, w) < 184 Then

      RedVal = 0

      GreenVal = 154

      BlueVal = 255

' gray-blue

    ElseIf pixelArray(H, w) < 200 Then

      RedVal = 0

      GreenVal = 96

      BlueVal = 255

' blue

    ElseIf pixelArray(H, w) < 216 Then

      RedVal = 0

      GreenVal = 48

      BlueVal = 255

' dark blue

    ElseIf pixelArray(H, w) < 232 Then

      RedVal = 0

      GreenVal = 0

      BlueVal = 255

' dark blue

    ElseIf pixelArray(H, w) < 248 Then

      RedVal = 0

      GreenVal = 0

      BlueVal = 160

' black

    Else

      RedVal = 0

      GreenVal = 0

      BlueVal = 0

    End If

    WriteAPixel RedVal Mod 256, GreenVal Mod 256, _

      BlueVal Mod 256

  Next w

' Do not change the following, essential row padding

  w = wMax * 3

  Do While (w Mod 4) <> 0

    WriteAByte 0

    w = w + 1

  Loop

Next H

End Sub

'''''''''''''''''''''''''''''''''''''''''''''''''''''''''''

Private Sub BitMap4(hMax As Integer, _

  wMax As Integer, pixelArray As Variant)

' For a map with continuous colors,

' from black through blue to white

Dim H As Integer, w As Integer, RedVal As Integer

Dim GreenVal As Integer, BlueVal As Integer

For H = hMax To 0 Step -1

  For w = 0 To wMax - 1

' from black to dark green

    If pixelArray(H, w) < 64 Then

      RedVal = 0

      GreenVal = 128 + Int(2 * pixelArray(H, w))

      BlueVal = 128 - Int(2 * pixelArray(H, w))

' from dark green to yellow

    ElseIf pixelArray(H, w) < 192 Then

      RedVal = Int(2 * (pixelArray(H, w) - 64))

      GreenVal = 255

      BlueVal = 0

' from yellow to white

    Else

      RedVal = 255

      GreenVal = 255

      BlueVal = Int(3 * (pixelArray(H, w) - 192))

    End If

    WriteAPixel RedVal Mod 256, GreenVal Mod 256, _

      BlueVal Mod 256

  Next w

' Do not change the following, essential row padding

  w = wMax * 3

  Do While (w Mod 4) <> 0

    WriteAByte 0

    w = w + 1

  Loop

Next H

End Sub

'''''''''''''''''''''''''''''''''''''''''''''''''''''''''''

Private Sub BitMapRGB(hMax As Integer, _

  wMax As Integer, pixelArray As Variant)

' For a map with continuous colors,

' from black through blue to white

Dim H As Integer, w As Integer, RedVal As Integer

Dim GreenVal As Integer, BlueVal As Integer

For H = hMax To 0 Step -1

  For w = 0 To wMax - 1

    RedVal = Int(Red(pixelArray(H, w)))

    GreenVal = Int(Green(pixelArray(H, w)))

    BlueVal = Int(Blue(pixelArray(H, w)))

    WriteAPixel RedVal Mod 256, GreenVal Mod 256, _

      BlueVal Mod 256

  Next w

' Do not change the following, essential row padding

  w = wMax * 3

  Do While (w Mod 4) <> 0

    WriteAByte 0

    w = w + 1

  Loop

Next H

End Sub

'''''''''''''''''''''''''''''''''''''''''''''''''''''''''''

Private Sub WriteAWord(ByVal WordValue As Integer)

' Write a two byte short word to the binary data stream.

Dim OutputFileNumber As Integer

OutputFileNumber = 1

Put #OutputFileNumber, , WordValue

End Sub

'''''''''''''''''''''''''''''''''''''''''''''''''''''''''''

Private Sub WriteALong(ByVal LongValue As Long)

' Write a four-byte long word to the binary data stream.

Dim OutputFileNumber As Integer

OutputFileNumber = 1

Put #OutputFileNumber, , LongValue

End Sub

'''''''''''''''''''''''''''''''''''''''''''''''''''''''''''

Private Sub WriteAPixel(ByVal RedValue As Byte, _

  ByVal GreenValue As Byte, ByVal BlueValue As Byte)

' Write a three-byte pixel color value to

' the binary data stream. Note that there is

' no padding up to a word or long boundary.

Dim OutputFileNumber As Integer

OutputFileNumber = 1

Put #OutputFileNumber, , BlueValue

Put #OutputFileNumber, , GreenValue

Put #OutputFileNumber, , RedValue

End Sub

'''''''''''''''''''''''''''''''''''''''''''''''''''''''''''

Private Sub WriteAByte(ByVal ByteValue As Byte)

' Write a byte.

Dim OutputFileNumber As Integer

OutputFileNumber = 1

Put #OutputFileNumber, , ByteValue

End Sub

'''''''''''''''''''''''''''''''''''''''''''''''''''''''''''

'''''''''''''''''''''''''''''''''''''''''''''''''''''''''''

'''''''''''''''''************************''''''''''''''''''

'''''''''''''''''*                      *''''''''''''''''''

'''''''''''''''''*         ScanF        *''''''''''''''''''

'''''''''''''''''*                      *''''''''''''''''''

'''''''''''''''''************************''''''''''''''''''

''''''''''''''''''''''''''''''''''''''''''' (c) R. de Levie

'''''''''''''''''''''''''''''''''''''''''''  v 12, May 2012

' PURPOSE:

' This subroutine plots an array of values of a function F

' where F depends on two parameters, x and y, whose values

' are specified in numerical axes on the spreadsheet.

'

' For example, the function F can be the sum of squares of

' the residuals SSR of the comparison of an experimental

' data set with those of a numerical simulation, and the

' x- and y-values two coefficients determining SSR. In that

' case, the macro will compute values that can subsequently

' be displayed by Mapper as an error surface and, with

' ScanF1, be imported into SimonLuca Santoro's macro IsoL.

'

' The macro works by changing the spreadsheet parameters x

' and y by invasive sampling, reading the corresponding

' results as array elements. The parameter values must be

' specified in the X- and Y-axes of the array.

'

' The subroutine is called by two small driver macros that

' set the integer parameter L to either 0 or 1. When L is

' set to 0 the subroutine merely fills the xy array with

' the requested values of either log F or F. When L = 1 the

' macro also Lists these F-values in the input format for

' SimonLuca Santoro's macro IsoL for generating contour

' diagrams, at a location specified by the user. This is a

' long list, which will occupy (CMax-1)times (RMax-1) rows.

' Make sure there is enough place for it, and it doesn't

' overwrite useful results.

' PROCEDURE:

'

' Leave two empty cells, one below the other, at the top

' left corner of the space for the XY plot. The top cell

' will become the macro label, the bottom the anchor cell

' of the XY plot. To the right of the anchor cell, put the

' labels of the X-axis, and below the anchor cell, the

' labels for the Y-axis. These two sets of labels define

' the specific x- and y-values for which F will be evaluated

' and the result placed in the appropriate cell of the XY

' array.

'

' The macro reads the values of the X- and Y axes, substi-

' tutes them for x and y respectively, then fills the XY

' array with the resulting values of F.

'

' Highlight the array area, including its axes, and call

' the macro.

' For subsequent use of Mapper, either leave the anchor

' left cell empty (so that Mapper will autoscale), or place

' a 1 there (in which case Mapper will ask for minimum and

' maximum scale values). The latter option is useful when

' several maps should have the same color (or gray) scales.

' EXAMPLES:

' AE3 section 4.17 illustrates the use of ScanF

'''''''''''''''''''''''''''''''''''''''''''''''''''''''''''

Sub ScanF0()        ' for computing & outputting

                
' the data array for F(x,y)

Dim L As Integer

L = 0

Call ScanF(0)

End Sub

'''''''''''''''''''''''''''''''''''''''''''''''''''''''''''

Sub ScanF1()        ' for also computing & outputting

                
' an IsoList as input for IsoL.xla

Dim L As Integer

L = 1

Call ScanF(1)

End Sub

'''''''''''''''''''''''''''''''''''''''''''''''''''''''''''

Sub ScanF(L)

Dim C As Integer, CMax As Integer, R As Long

Dim RMax As Long, RR As Long, RRMax As Long

Dim fF As Variant, IsoList As Variant, vF As Variant

Dim fXY As Variant, vXY As Variant

Dim aA, aAA, aF, aTL, aX, aY, Begin, Ending

Dim fFInit, fXInit, fYInit, Percentage, PosColon

Dim rgF As Range, rgL As Range, rgTL As Range, rgX As Range

Dim rgXY As Range, rgXXYY As Range, rgY As Range

CMax = 0

RMax = 0

Begin:

CMax = Selection.Columns.Count

RMax = Selection.Rows.Count

Set rgXY = Range(Selection.Address)

RR = 0

' If area is not highlighted before the macro is called

If CMax < 2 And RMax < 2 Then

  hAnswer = MsgBox("You forgot to highlight" _

    & Chr(13) & "the block of input data." _

    & Chr(13) & "Do you want to do so now?" _

    , vbYesNo)

  If hAnswer = vbNo Then End

  If hAnswer = vbYes Then

    Set rgXY = Application.InputBox(Prompt:= _

      "The input data are located in:", Type:=8)

    rgXY.Select

  End If

  GoTo Begin

End If

RRMax = (CMax - 1) * (RMax - 1)

ReDim IsoList(1 To RRMax, 1 To 3) As Double

' Copy the values of the X- and Y-axes

ReDim vF(1 To RMax, 1 To CMax)

ReDim vXY(1 To RMax - 1, 1 To CMax - 1)

For C = 1 To CMax

  For R = 1 To RMax

    vF(R, C) = Selection(R, C).Value

  Next R

Next C

' Read the values of X and Y, and save their generating

' formulas for restoration when this macro is done

Set rgX = Application.InputBox(Prompt:="X", _

  Title:="ScanF" & L, Type:=8)

rgX.Select

aX = Selection.Address

fXInit = Selection.Formula

Set rgY = Application.InputBox(Prompt:="Y", _

  Title:="ScanF" & L, Type:=8)

rgY.Select

aY = Selection.Address

fYInit = Selection.Formula

' Read the address of the function F

' that should be placed in the XY plot

Set rgF = Application.InputBox(Prompt:="F", _

  Title:="ScanF" & L, Type:=8)

rgF.Select

aF = Selection.Address

fFInit = Selection.Formula

If L = 1 Then

  Set rgL = Application.InputBox(Prompt:= _

    "Place the top left-hand cell of the list here:", _

    Title:="ScanF" & L, Type:=8)

  rgL.Select

  aL = Selection.Address

End If

' Keep the user informed about

' the progress of the computation

Application.StatusBar = True

Application.StatusBar = "The computation is " & _

  "0% complete."

Application.ScreenUpdating = False

' Read the values of F to be displayed in the XY plot

rgXY.Select

aA = Selection.Address

Selection.Offset(1, 1).Select

Selection.Resize(RMax - 1, CMax - 1).Select

aAA = Selection.Address

Set rgXXYY = Range(aAA)

rgXXYY.Select

For C = 2 To CMax

  For R = 2 To RMax

    rgX.Select

    Selection.Value = vF(1, C)

    rgY.Select

    Selection.Value = vF(R, 1)

    rgF.Select

    vXY(R - 1, C - 1) = Selection.Value

' If a logarithmic scale is NOT desired (e.g., when vXY is

' negative, place an apostrophy in front of the next line:

    vXY(R - 1, C - 1) = Log(vXY(R - 1, C - 1)) / Log(10)

' to 'comment out' the preceding line.

  Next R

  Percentage = 5 * Int(20 * (C / CMax))

  Application.StatusBar = "The computation is " & _

    Percentage & "% complete."

Next C

Application.StatusBar = False

Application.ScreenUpdating = True

' Display the output results

rgXXYY.Select

Selection.Value = vXY

' Reset the parameter formulas

rgX.Select

Selection.Formula = fXInit

rgY.Select

Selection.Formula = fYInit

rgF.Select

Selection.Formula = fFInit

' Label the macro and add self-documentation

PosColon = InStr(1, aA, ":")

aTL = Left(aA, PosColon - 1)

Set rgTL = Range(aTL)

rgTL.Select

Selection.Offset(-1, 0).Select

With ActiveCell

  .Value = "ScanF" & L

  .Font.Bold = True

  .Font.Italic = False

  .Font.ColorIndex = 5

  .HorizontalAlignment = xlCenter

  .Interior.ColorIndex = 20

  .ClearComments

  .AddComment

  .Comment.Visible = False

  .Comment.Shape.Width = 200

  .Comment.Shape.Height = 80

  .Comment.Text Text:="ScanF" & L & " version 10" & _

    Chr(10) & Chr(10) & "Input: " & aA & Chr(10) & "X in " & _

    aX & " with initial formula/value " & fXInit & Chr(10) & _

    "Y in " & aY & " with initial formula/value " & _

    fYInit & Chr(10) & "F in " & aF & _

    " with initial formula " & fFInit & Chr(10) & _

    "date = " & Date & ",   time = " & Time

End With

If L = 0 Then GoTo Ending

' This part, used only when L = 1, puts the same F values in a list

' suitable for direct importation into XXX Santoro's IsoL macro.

' The list is generated starting in cell IA1 (as the default) or any

' other cell so designated by the user. The input box allowing the user

' such designation also lists the length of the list; its 

' width is fixed at three cells. The first column contains 

' x, the second y, and the third the function F(x,y).

'

' Highlight an array three columns wide, and length (RMax+1)

' times (CMax+1), then specify the input array I including

' its X and Y axes, and enter with Ctrl+Shift+Enter.

For C = 1 To CMax - 1

  For R = 1 To RMax - 1

    RR = RR + 1

    IsoList(RR, 1) = vF(1, C + 1)

    IsoList(RR, 2) = vF(R + 1, 1)

    IsoList(RR, 3) = vXY(R, C)

  Next R

Next C

' Label and output IsoList

rgL.Select

With ActiveCell

  .ClearComments

  .AddComment

  .Comment.Visible = False

  .Comment.Shape.Width = 160

  .Comment.Shape.Height = 40

  .Comment.Text Text:="IsoList with ScanF1 Version 10" & _

     Chr(10) & Chr(10) & "date = " & Date & ",   time = " & Time

End With

rgL.Resize(RR, 3).Select

Selection = IsoList

Ending:

End Sub

'''''''''''''''''''''''''''''''''''''''''''''''''''''''''''

''''''''''''''********************************'''''''''''''

''''''''''''''*                              *'''''''''''''

''''''''''''''*           RootFinder         *'''''''''''''

''''''''''''''^                             ^''''''''''''''

''''''''''''''********************************'''''''''''''

''''''''''''''''''''''''''''''''''''''''''' (c) R. de Levie

'''''''''''''''''''''''''''''''''''''''''''  v 12, May 2012

' PURPOSE:

' This macro finds the real root of a function F(x) of a

' single variable x by bisection inside a user-defined

' range. Note: because the macro looks for a sign change

' in F(x), it will also find singularities, such as with

' tan(x) for x = pi/2.

' The value of the resolution criterion FRes can be changed

' depending on the requirements of the problem. For simple

' functions, a high resolution (i.e., a small value of

' FRes, such as 1E-12) may be desired, while complicated

' expressions may require a lower resolution. For functions

' that reach values much larger than 1, a relative rather

' than an absolute resolution may be more appropriate.

' Similarly, the criterion for deciding between a root and

' a singularity may have to be changed for functions with

' extremely steep slopes near their roots.

' SUBROUTINES:

' This macro does not require any subroutines.

' INPUT:

' The locations of the function F(x) and the independent

' variable x must be in columns of equal length, and are

' entered into the macro via input boxes.

' The parameter x must change monotonically (i.e., either

' ascend or descend continuously) in its column, while

' the function F(x) should have one and only one zero

' crossing in the selected column or column fragment.

' The parameter x can be a value or a function of some

' other parameters.

' If the function on the spreadsheet exhibits more than one

' zero crossing, restrict the size of the block so that it

' will contain only one of them.

' OUTPUT:

' The root is displayed in a message box.

' PROCEDURE:

' Call the macro, then enter the column ranges for F(x) and

' x in the input boxes.

Sub RootFinder()

Dim i As Integer, N As Integer

Dim nc As Integer, nr As Integer

Dim FFirst As Double, FLast As Double, FNew As Double

Dim FRes As Double, Ratio As Double

Dim XFirst As Double, XLast As Double, XNew As Double

Dim FFormula As Variant, FValue As Variant

Dim XFormula As Variant, XValue As Variant

Dim FFormulaSave As Variant, XFormulaSave As Variant

Dim FRange As Range, XRange As Range

Dim FPrompt, XPrompt, FTitle, XTitle

' Input the function F(x) and its associated files

FPrompt = "The function F(x) is in column:"

FTitle = "RootFinder: Input F(x)"

Set FRange = Application.InputBox(FPrompt, FTitle, Type:=8)

FRange.Select

FFormula = Selection.Formula

FValue = Selection.Value

FFormulaSave = Selection.Formula

' Test the F(x) input

nc = Selection.Columns.Count

If nc <> 1 Then

  MsgBox "The function should be in a single column."

  End

End If

nr = Selection.Rows.Count

If nr < 3 Then

  MsgBox "The column should have at least three rows."

  End

End If

' Input the variable x and its associated files

XPrompt = "The variable x is in column:"

XTitle = "RootFinder: Input x"

Set XRange = Application.InputBox(XPrompt, XTitle, Type:=8)

XRange.Select

XFormula = Selection.Formula

XValue = Selection.Value

XFormulaSave = Selection.Formula

' Test the x input

nc = Selection.Columns.Count

If nc <> 1 Then

  MsgBox "The variable should be in a single column."

  End

End If

If nr <> Selection.Rows.Count Then

  MsgBox "The F(x) and x columns must have equal lengths."

  End

End If

' Check for one root

' Note: an odd number of roots will also pass this test

FFirst = FValue(1, 1)

FLast = FValue(nr, 1)

If FFirst * FLast >= 0 Then

  MsgBox "The array should encompass a single root."

  End

End If

' Set the resolution; FRes must be positive

FRes = 0.00000000000002

' Suppress screen updating

Application.ScreenUpdating = False

' Bisection

N = 0

Do

  XRange.Select

  XFirst = XValue(1, 1)

  XLast = XValue(nr, 1)

  XNew = (XFirst + XLast) / 2

  XValue(2, 1) = XNew

  Selection.Value = XValue

  FRange.Select

  FValue = Selection.Value

  FFirst = FValue(1, 1)

  FLast = FValue(nr, 1)

  FNew = FValue(2, 1)

  XRange.Select

  If FFirst * FNew = 0 Then GoTo a

  If FFirst * FNew > 0 Then

    For i = 1 To nr

      XValue(i, 1) = XNew + _

        (XLast - XNew) * (i - 1) / (nr - 1)

    Next i

  Ratio = Abs((FLast - FNew) / (XLast - XNew))

  End If

  If FFirst * FNew < 0 Then

    For i = 1 To nr

      XValue(i, 1) = XFirst + _

        (XNew - XFirst) * (i - 1) / (nr - 1)

    Next i

  Ratio = Abs((FNew - FFirst) / (XNew - XFirst))

  End If

  N = N + 1

  Selection.Value = XValue

Loop Until Abs(FLast - FFirst) < FRes Or N = 100

' Restore data input area and display the found root

a:

XRange.Select

Selection.Formula = XFormulaSave

FRange.Select

Selection.Formula = FFormulaSave

If Ratio > 1E+20 Then

  MsgBox "Singularity at x = " & XNew

Else

  MsgBox "Root at x = " & XNew

End If

End Sub

'''''''''''''''''''''''''''''''''''''''''''''''''''''''''''

'''''''''''''''''''''''''''''''''''''''''''''''''''''''''''

''''''''''***************************************''''''''''

''''''''''*                                     *''''''''''

''''''''''*     Movie Demonstration Macros      *''''''''''

''''''''''*                                     *''''''''''

''''''''''***************************************''''''''''

''''''''''''''''''''''''''''''''''''''''''' (c) R. de Levie

'''''''''''''''''''''''''''''''''''''''''''  v 12, May 2012

' The following macros merely demonstrate how to make

' simple Excel 'movies'. Their use is illustrated in

' section 1.6 of AE3.

'

''''''''''''''''''''''' MovieDemo 1 '''''''''''''''''''''''

'

' This demonstration macro requires that you enter the

' number 0 in cells A1 and B1, and the number 10 in A2 and

' B2. Then highlight the area A1:B2, and make a correspon-

' ding graph. Format both axes to have manual scales from 0

' to 10. Make sure that the graph displays B1:B2 vs. A1:A2

' as individual points ('markers'). Then call the macro,

' and enjoy the show.

Sub MovieDemo1()

Range("A1") = 0

Range("A2") = 0

For i = 1 To 400

  Range("A1") = 10 - 0.05 * Abs(i - 200)

  Range("A2") = 10 * Exp(-0.001 * (i - 300) ^ 2)

  Application.ScreenUpdating = True

Next i

Range("A1") = 0

Range("A2") = 0

End Sub

''''''''''''''''''''''' MovieDemo 2 '''''''''''''''''''''''

'

' This demonstration macro requires that you enter the num-

' ber 0 in cells H1 and I1, and the number 10 in H2 and I2.

' Then highlight the area H1:I2, and make a corresponding

' graph. Format both axes to have manual scales from 0 to

' 10. Make sure that the graph displays I1:I2 vs. H1:H2 as

' individual points ('markers') connected by a line. Then

' call the macro.

Sub MovieDemo2()

Range("H1") = 5: Range("I1") = 1

Range("H2") = 5: Range("I2") = 5

'Application.ScreenUpdating = True

For i = 1 To 1000

  Range("H1") = 5 - 5 * Sin(0.06 * i)

  Range("I1") = 5 - 5 * Cos(0.07 * i)

  Application.ScreenUpdating = True

Next i

Range("H1") = 5: Range("I1") = 1

Range("H2") = 5: Range("I2") = 5

End Sub

''''''''''''''''''''''' MovieDemo 3 '''''''''''''''''''''''

' This demonstration macro requires that you enter the

' numbers 0 through 10 in cells A15 through A25, and the

' number 5 in cells B15:B25. Again use a graph with manual

' scales from 0 to 10, and display the data in B15:B25 vs.

' A15:A25 as markers plus (smoothed) connecting lines.

Sub MovieDemo3()

Range("A15") = 0: Range("B15") = 5

Range("A16") = 1: Range("B16") = 5

Range("A17") = 2: Range("B17") = 5

Range("A18") = 3: Range("B18") = 5

Range("A19") = 4: Range("B19") = 5

Range("A20") = 5: Range("B20") = 5

Range("A21") = 6: Range("B21") = 5

Range("A22") = 7: Range("B22") = 5

Range("A23") = 8: Range("B23") = 5

Range("A24") = 9: Range("B24") = 5

Range("A25") = 10: Range("B25") = 5

Application.ScreenUpdating = True

For i = 1 To 100

  Application.ScreenUpdating = False

  Range("B15") = 5 + 0.04 * i * Sin(0.01 * i)

  Range("B16") = 5 + 0.04 * i * Sin(0.01 * i - 100)

  Range("B17") = 5 + 0.04 * i * Sin(0.01 * i - 200)

  Range("B18") = 5 + 0.04 * i * Sin(0.01 * i - 300)

  Range("B19") = 5 + 0.04 * i * Sin(0.01 * i - 400)

  Range("B20") = 5 + 0.04 * i * Sin(0.01 * i - 500)

  Range("B21") = 5 + 0.04 * i * Sin(0.01 * i - 600)

  Range("B22") = 5 + 0.04 * i * Sin(0.01 * i - 700)

  Range("B23") = 5 + 0.04 * i * Sin(0.01 * i - 800)

  Range("B24") = 5 + 0.04 * i * Sin(0.01 * i - 900)

  Range("B25") = 5 + 0.04 * i * Sin(0.01 * i - 1000)

  Application.ScreenUpdating = True

Next i

For i = 101 To 900

  Application.ScreenUpdating = False

  Range("B15") = 5 + 4 * Sin(0.01 * i)

  Range("B16") = 5 + 4 * Sin(0.01 * i - 100)

  Range("B17") = 5 + 4 * Sin(0.01 * i - 200)

  Range("B18") = 5 + 4 * Sin(0.01 * i - 300)

  Range("B19") = 5 + 4 * Sin(0.01 * i - 400)

  Range("B20") = 5 + 4 * Sin(0.01 * i - 500)

  Range("B21") = 5 + 4 * Sin(0.01 * i - 600)

  Range("B22") = 5 + 4 * Sin(0.01 * i - 700)

  Range("B23") = 5 + 4 * Sin(0.01 * i - 800)

  Range("B24") = 5 + 4 * Sin(0.01 * i - 900)

  Range("B25") = 5 + 4 * Sin(0.01 * i - 1000)

  Application.ScreenUpdating = True

Next i

For i = 901 To 1000

  Application.ScreenUpdating = False

  Range("B15") = 5 + 0.04 * (1000 - i) * Sin(0.01 * i)

  Range("B16") = _

    5 + 0.04 * (1000 - i) * Sin(0.01 * i - 100)

  Range("B17") = _

    5 + 0.04 * (1000 - i) * Sin(0.01 * i - 200)

  Range("B18") = _

    5 + 0.04 * (1000 - i) * Sin(0.01 * i - 300)

  Range("B19") = _

    5 + 0.04 * (1000 - i) * Sin(0.01 * i - 400)

  Range("B20") = _

    5 + 0.04 * (1000 - i) * Sin(0.01 * i - 500)

  Range("B21") = _

    5 + 0.04 * (1000 - i) * Sin(0.01 * i - 600)

  Range("B22") = _

    5 + 0.04 * (1000 - i) * Sin(0.01 * i - 700)

  Range("B23") = _

    5 + 0.04 * (1000 - i) * Sin(0.01 * i - 800)

  Range("B24") = _

    5 + 0.04 * (1000 - i) * Sin(0.01 * i - 900)

  Range("B25") = _

    5 + 0.04 * (1000 - i) * Sin(0.01 * i - 1000)

  Application.ScreenUpdating = True

Next i

Range("A15") = 0: Range("B15") = 5

Range("A16") = 1: Range("B16") = 5

Range("A17") = 2: Range("B17") = 5

Range("A18") = 3: Range("B18") = 5

Range("A19") = 4: Range("B19") = 5

Range("A20") = 5: Range("B20") = 5

Range("A21") = 6: Range("B21") = 5

Range("A22") = 7: Range("B22") = 5

Range("A23") = 8: Range("B23") = 5

Range("A24") = 9: Range("B24") = 5

Range("A25") = 10: Range("B25") = 5

End Sub

''''''''''''''''''''''' MovieDemo 4 '''''''''''''''''''''''

'

' This demonstration macro requires that you enter the

' numbers 0, 0, 5, 5, 5, 5, 10 in cells A29:A36, and the

' number 0,1,7,5,5,5,5,0 in cells B29:B36. Again use a

' graph with manual scales from 0 to 10, and display the

' data as individually distinct points, each of a different

' type and/or color. For example, point A32,B32 might be a

' hollow black circle of size 18, point A33,B33 a hollow

' red circle of size 14, point A34,B34 a hollow black cir-

' cle of size 9, and point A35,B35 be a solid red circle of

' size 6. Also make the other data markers distinct.

Sub MovieDemo4()

Range("A1") = 0: Range("B1") = 0

Range("A2") = 0: Range("B2") = 1

Range("A3") = 5: Range("B3") = 9

Range("A4") = 5: Range("B4") = 5

Range("A5") = 5: Range("B5") = 5

Range("A6") = 5: Range("B6") = 5

Range("A7") = 5: Range("B7") = 5

Range("A8") = 10: Range("B8") = 0

Application.ScreenUpdating = True

Application.Volatile True

For i = 1 To 1000

  Application.ScreenUpdating = False

  Range("A1") = 5 - 5 * Sin(0.01 * (i - 500))

  Range("B1") = 5 - 5 * Sin(0.01 * (i - 500))

  Range("A2") = Range("A2") + 0.01

  Range("B2") = 5 - 4 * Cos(0.03 * i)

  Range("A3") = 5 + 4 * Sin(0.2 * i)

  Range("B3") = 5 - 4 * Cos(0.2 * i)

  If CDbl(i) / 160 - CInt(i / 160) = 0 Then

    Range("A4") = 5 + (5 * Rnd() - 2.5)

    Range("B4") = 5 + (5 * Rnd() - 2.5)

  End If

  If CDbl(i) / 170 - CInt(i / 170) = 0 Then

    Range("A5") = Range("A4")

    Range("B5") = Range("B4")

  End If

  If CDbl(i) / 180 - CInt(i / 180) = 0 Then

    Range("A6") = Range("A4")

    Range("B6") = Range("B4")

  End If

  If CDbl(i) / 190 - CInt(i / 190) = 0 Then

    Range("A7") = Range("A4")

    Range("B7") = Range("B4")

  End If

  If CDbl(i) / 10 - CInt(i / 10) = 0 Then

    Range("A8") = 5 + 5 * Sin(0.01 * (i - 500))

  End If

  Range("B8") = Range("B8") + 0.01

  Application.ScreenUpdating = True

Next i

Range("A1") = 0: Range("B1") = 0

Range("A2") = 0: Range("B2") = 1

Range("A3") = 5: Range("B3") = 9

Range("A4") = 5: Range("B4") = 5

Range("A5") = 5: Range("B5") = 5

Range("A6") = 5: Range("B6") = 5

Range("A7") = 5: Range("B7") = 5

Range("A8") = 10: Range("B8") = 0

End Sub

''''''''''''''''''''''' MovieDemo5 '''''''''''''''''''''''

'

' This demonstration macro requires that you enter the

' numbers 4, 4, 4, 4, 6, 6, 6, and 6 in cells H15:H22,

' and the number 5 in cells I15:I22. Again use a graph

' with manual scales from 0 to 10, and display the points

' H15, I15, and H19:I19 as small solid circles, preferably

' of different color, and the others as larger, hollow

' (concentric) rings around them.

Sub MovieDemo5()

Range("H15") = 4: Range("I15") = 5

Range("H16") = 4: Range("I16") = 5

Range("H17") = 4: Range("I17") = 5

Range("H18") = 4: Range("I18") = 5

Range("H19") = 6: Range("I19") = 5

Range("H20") = 6: Range("I20") = 5

Range("H21") = 6: Range("I21") = 5

Range("H22") = 6: Range("I22") = 5

For i = 1 To 252

  Range("H15") = 4 + 0.2 * Sin(0.1 * i)

  Range("H16") = 4 + 0.3 * Sin(0.1 * i)

  Range("H17") = 4 + 0.4 * Sin(0.1 * i)

  Range("H18") = 4 + 0.5 * Sin(0.1 * i)

  Range("H19") = 6 - 0.2 * Sin(0.1 * i)

  Range("H20") = 6 - 0.3 * Sin(0.1 * i)

  Range("H21") = 6 - 0.4 * Sin(0.1 * i)

  Range("H22") = 6 - 0.5 * Sin(0.1 * i)

  Application.ScreenUpdating = True

Next i

For i = 253 To 502

  Range("H15") = _

    5 + Cos(0.025 * i) * (-1 + 0.2 * Sin(0.1 * i))

  Range("I15") = _

    5 + Sin(0.025 * i) * (-1 + 0.2 * Sin(0.1 * i))

  Range("H16") = _

    5 + Cos(0.025 * i) * (-1 + 0.3 * Sin(0.1 * i))

  Range("I16") = _

    5 + Sin(0.025 * i) * (-1 + 0.3 * Sin(0.1 * i))

  Range("H17") = _

    5 + Cos(0.025 * i) * (-1 + 0.4 * Sin(0.1 * i))

  Range("I17") = _

    5 + Sin(0.025 * i) * (-1 + 0.4 * Sin(0.1 * i))

  Range("H18") = _

    5 + Cos(0.025 * i) * (-1 + 0.5 * Sin(0.1 * i))

  Range("I18") = _

    5 + Sin(0.025 * i) * (-1 + 0.5 * Sin(0.1 * i))

  Range("H19") = _

    5 + Cos(0.025 * i) * (1 - 0.2 * Sin(0.1 * i))

  Range("I19") = _

    5 + Sin(0.025 * i) * (1 - 0.2 * Sin(0.1 * i))

  Range("H20") = _

    5 + Cos(0.025 * i) * (1 - 0.3 * Sin(0.1 * i))

  Range("I20") = _

    5 + Sin(0.025 * i) * (1 - 0.3 * Sin(0.1 * i))

  Range("H21") = _

    5 + Cos(0.025 * i) * (1 - 0.4 * Sin(0.1 * i))

  Range("I21") = _

    5 + Sin(0.025 * i) * (1 - 0.4 * Sin(0.1 * i))

  Range("H22") = _

    5 + Cos(0.025 * i) * (1 - 0.5 * Sin(0.1 * i))

  Range("I22") = _

    5 + Sin(0.025 * i) * (1 - 0.5 * Sin(0.1 * i))

  Application.ScreenUpdating = True

Next i

For i = 503 To 630

  Range("H15") = 4 + 0.2 * Sin(0.1 * i)

  Range("H16") = 4 + 0.3 * Sin(0.1 * i)

  Range("H17") = 4 + 0.4 * Sin(0.1 * i)

  Range("H18") = 4 + 0.5 * Sin(0.1 * i)

  Range("H19") = 6 - 0.2 * Sin(0.1 * i)

  Range("H20") = 6 - 0.3 * Sin(0.1 * i)

  Range("H21") = 6 - 0.4 * Sin(0.1 * i)

  Range("H22") = 6 - 0.5 * Sin(0.1 * i)

  Application.ScreenUpdating = True

Next i

Range("H15") = 4: Range("I15") = 5

Range("H16") = 4: Range("I16") = 5

Range("H17") = 4: Range("I17") = 5

Range("H18") = 4: Range("I18") = 5

Range("H19") = 6: Range("I19") = 5

Range("H20") = 6: Range("I20") = 5

Range("H21") = 6: Range("I21") = 5

Range("H22") = 6: Range("I22") = 5

Application.ScreenUpdating = True

End Sub

'''''''''''''''''''''''''''''''''''''''''''''''''''''''''''

'''''''''''''''''''''''''''''''''''''''''''''''''''''''''''

'''''''''''*************************************'''''''''''

'''''''''''*                                   *'''''''''''

'''''''''''*      Insert MacroBundle Toolbar   *'''''''''''

'''''''''''*                                   *'''''''''''

'''''''''''*************************************'''''''''''

''''''''''''''''''''''''''''''''''''''''''' (c) R. de Levie

'''''''''''''''''''''''''''''''''''''''''''  v 12, May 2012

' PURPOSE:

' This subroutine places an extra toolbar in the Excel

' spreadsheet in order to facilitate access to the custom

' macros of this MacroBundle. The MBToolbar shows seven

' categories, many with subcategories, and provides access

' to all macros of the MacroBundle. To remove this toolbar,

' select RemoveMBToolbar in the last category, Miscellany.

' You can of course modify this toolbar in order to delete

' macros you do not need and/or add your own custom macros.

' All macros, whether listed in this toolbar or not, can

' always be called with Tools > Macro > Macros or AltÈF8.

' SUBROUTINES:

' All custom macros of the current MacroBundle

' can be called with this toolbar.

' LIMITATIONS:

' This subroutine works in Excel 97 through Excel 2003; it

' will not work in Excel 95 or in earlier versions, which

' handled toolbars differently. It will work with Excel

' 2007, but there must be placed in its Developer ribbon.

' If that ribbon does not already exist, create it as

' follows. Either click on the Office Button in the top

' left-hand corner of the Excel 2007 spreadsheet, select

' Excel Options, and place a tickmark with Show Developer

' tab, or simply press Alt followed by the letters fid.

''''''''''''''''''''''''''''''''''''''''''''''''''''''''''''

Sub InsertMBToolbar()

Dim TBar As CommandBar

Dim Button1 As CommandBarButton

Dim Button2 As CommandBarPopup

Dim Button21 As CommandBarPopup

Dim Button210 As CommandBarButton

Dim Button211 As CommandBarButton

Dim Button22 As CommandBarPopup

Dim Button221 As CommandBarButton

Dim Button222 As CommandBarButton

Dim Button23 As CommandBarPopup

Dim Button230 As CommandBarButton

Dim Button231 As CommandBarButton

Dim Button24 As CommandBarButton

Dim Button25 As CommandBarPopup

Dim Button250 As CommandBarButton

Dim Button251 As CommandBarButton

Dim Button26 As CommandBarPopup

Dim Button260 As CommandBarButton

Dim Button261 As CommandBarButton

Dim Button27 As CommandBarButton

Dim Button3 As CommandBarPopup

Dim Button31 As CommandBarButton

Dim Button32 As CommandBarPopup

Dim Button321 As CommandBarButton

Dim Button322 As CommandBarButton

Dim Button323 As CommandBarButton

Dim Button33 As CommandBarButton

Dim Button4 As CommandBarPopup

Dim Button41 As CommandBarPopup

Dim Button411 As CommandBarButton

Dim Button412 As CommandBarButton

Dim Button42 As CommandBarButton

Dim Button43 As CommandBarPopup

Dim Button430 As CommandBarButton

Dim Button431 As CommandBarButton

Dim Button5 As CommandBarPopup

Dim Button51 As CommandBarPopup

Dim Button511 As CommandBarButton

Dim Button512 As CommandBarButton

Dim Button52 As CommandBarPopup

Dim Button521 As CommandBarButton

Dim Button522 As CommandBarButton

Dim Button523 As CommandBarButton

Dim Button524 As CommandBarButton

Dim Button6 As CommandBarPopup

Dim Button61 As CommandBarPopup

Dim Button611 As CommandBarButton

Dim Button612 As CommandBarButton

Dim Button62 As CommandBarPopup

Dim Button621 As CommandBarButton

Dim Button622 As CommandBarButton

Dim Button63 As CommandBarButton

Dim Button6301 As CommandBarButton

Dim Button64 As CommandBarButton

Dim Button65 As CommandBarPopup

Dim Button651 As CommandBarButton

Dim Button652 As CommandBarButton

Dim Button7 As CommandBarPopup

Dim Button71 As CommandBarPopup

Dim Button710 As CommandBarButton

Dim Button711 As CommandBarButton

Dim Button712 As CommandBarButton

Dim Button713 As CommandBarButton

Dim Button714 As CommandBarButton

Dim Button72 As CommandBarPopup

Dim Button720 As CommandBarButton

Dim Button721 As CommandBarButton

Dim Button722 As CommandBarButton

Dim Button723 As CommandBarButton

Dim Button73 As CommandBarPopup

Dim Button730 As CommandBarButton

Dim Button731 As CommandBarButton

Dim Button732 As CommandBarButton

Dim Button733 As CommandBarButton

Dim Button8 As CommandBarPopup

Dim Button81 As CommandBarPopup

Dim Button810 As CommandBarButton

Dim Button811 As CommandBarButton

Dim Button82 As CommandBarButton

Dim Button83 As CommandBarPopup

Dim Button830 As CommandBarButton

Dim Button831 As CommandBarButton

Dim Button832 As CommandBarButton

Dim Button833 As CommandBarButton

Dim Button834 As CommandBarButton

Dim Button84 As CommandBarPopup

Dim Button841 As CommandBarButton

Dim Button842 As CommandBarButton

' In order to prevent conflicts, delete any earlier

' of the MacroBundle toolbar, if present

On Error Resume Next

CommandBars("MacroBundle").Delete

' Create a commandbar

Set TBar = CommandBars.Add

With TBar

  .Name = "MacroBundle"

  .Position = msoBarTop

  .Visible = True

End With

Set Button1 = CommandBars("MacroBundle") _

  .Controls.Add(Type:=msoControlButton)

With Button1

  .Caption = "Propagation"

  .Style = msoButtonCaption

  .OnAction = "Propagation"

End With

' Create menu for Linear LS

Set Button2 = CommandBars("MacroBundle") _

  .Controls.Add(Type:=msoControlPopup)

Button2.Caption = "linear LS"

Button2.BeginGroup = True

' Create submenu for LS

Set Button21 = Button2.Controls.Add(Type:=msoControlPopup)

Button21.Caption = " LS"

Button21.TooltipText = "Highlight array" & Chr(13) & _

  "before pressing" & Chr(13) & "LSO or LSI"

' Create buttons for LSO and LSI

Set Button210 = Button21.Controls _

  .Add(Type:=msoControlButton)

With Button210

  .Caption = "LSO"

  .Style = msoButtonCaption

  .OnAction = "LSO"

End With

Set Button211 = Button21.Controls _

  .Add(Type:=msoControlButton)

With Button211

  .Caption = "LSI"

  .Style = msoButtonCaption

  .OnAction = "LSI"

End With

' Create a submenu for ELS

Set Button22 = Button2.Controls.Add(Type:=msoControlPopup)

Button22.Caption = "ELS"

Button22.TooltipText = "Highlight array" & Chr(13) & _

  "before pressing" & Chr(13) & "ELSauto or ELSfixed"

' Create buttons for ELSauto and ELSfixed

Set Button221 = Button22.Controls _

  .Add(Type:=msoControlButton)

Button221.Caption = "ELSauto"

Button221.OnAction = "ELSauto"

Set Button222 = Button22.Controls _

  .Add(Type:=msoControlButton)

Button222.Caption = "ELSfixed"

Button222.OnAction = "ELSfixed"

' Create a submenu for WLS

Set Button23 = Button2.Controls.Add(Type:=msoControlPopup)

Button23.Caption = "WLS"

Button23.TooltipText = "Highlight array" & Chr(13) & _

  "before pressing" & Chr(13) & "WLSO or WLSI"

' Create buttons for WLSO and WLSI

Set Button230 = Button23.Controls _

  .Add(Type:=msoControlButton)

Button230.Caption = "WLSO"

Button230.OnAction = "WLSO"

Set Button231 = Button23.Controls _

  .Add(Type:=msoControlButton)

Button231.Caption = "WLSI"

Button231.OnAction = "WLSI"

' Create a button for ScaleBySf

Set Button24 = Button2.Controls.Add(Type:=msoControlButton)

With Button24

  .Caption = "GradeBySf"

  .Style = msoButtonCaption

  .OnAction = "GradeBySf "

End With

' Create a submenu for LSPoly

Set Button25 = Button2.Controls.Add(Type:=msoControlPopup)

Button25.Caption = "LSPoly"

Button25.TooltipText = "Highlight array" & Chr(13) & _

  "before pressing" & Chr(13) & "LSPoly0 or LSPoly1"

' Create buttons for LSPolyO and LSPolyI

Set Button250 = Button25.Controls _

  .Add(Type:=msoControlButton)

With Button250

  .Caption = "LSPolyO"

  .Style = msoButtonCaption

  .OnAction = "LSPolyO"

End With

Set Button251 = Button25.Controls _

  .Add(Type:=msoControlButton)

With Button251

  .Caption = "LSPolyI"

  .Style = msoButtonCaption

  .OnAction = "LSPolyI"

End With

' Create a submenu for LSMulti

Set Button26 = Button2.Controls.Add(Type:=msoControlPopup)

Button26.Caption = "LSMulti"

Button26.TooltipText = "Highlight array" & Chr(13) & _

  "before pressing" & Chr(13) & "LSMulti0 or LSMulti1"

' Create buttons for LSMultiO and LSMultiI

Set Button260 = Button26.Controls _

  .Add(Type:=msoControlButton)

With Button260

  .Caption = "LSMultiO"

  .Style = msoButtonCaption

  .OnAction = "LSMultiO"

End With

Set Button261 = Button26.Controls _

  .Add(Type:=msoControlButton)

With Button261

  .Caption = "LSMultiI"

  .Style = msoButtonCaption

  .OnAction = "LSMultiI"

End With

' Create a button for Permute

Set Button27 = Button2.Controls.Add(Type:=msoControlButton)

With Button27

  .Caption = "LSPermute"

  .Style = msoButtonCaption

  .OnAction = "LSPermute"

End With

' Create a menu for Nonlinear LS

Set Button3 = CommandBars("MacroBundle") _

  .Controls.Add(Type:=msoControlPopup)

Button3.Caption = "nonlinear LS"

Button3.BeginGroup = True

' Create a button for SolverAid

Set Button31 = Button3.Controls.Add(Type:=msoControlButton)

With Button31

  .Caption = "SolverAid"

  .Style = msoButtonCaption

  .OnAction = "SolverAid"

End With

' Create a button for ColumnSolver

Set Button32 = Button3.Controls.Add(Type:=msoControlPopup)

Button32.Caption = "ColumnSolver"

' Create submenus for ColSolverMax, ColSolverMin,

' and ColSolverValue respectively

Set Button321 = Button32.Controls _

  .Add(Type:=msoControlButton)

Button321.Caption = "ColumnSolverMax"

Button321.OnAction = "ColumnSolverMax"

Set Button322 = Button32.Controls _

  .Add(Type:=msoControlButton)

Button322.Caption = "ColumnSolverMin"

Button322.OnAction = "ColumnSolverMin"

Set Button323 = Button32.Controls _

  .Add(Type:=msoControlButton)

Button323.Caption = "ColumnSolverValue"

Button323.OnAction = "ColumnSolverValue"

' Create a button for SolverScan

Set Button33 = Button3.Controls.Add(Type:=msoControlButton)

With Button33

  .Caption = "SolverScan"

  .Style = msoButtonCaption

  .OnAction = "SolverScan"

End With

' Create a button for Transforms

Set Button4 = CommandBars("MacroBundle").Controls _

  .Add(Type:=msoControlPopup)

Button4.Caption = "transforms"

Button4.BeginGroup = True

' Create a button for Fourier

Set Button41 = Button4.Controls.Add(Type:=msoControlPopup)

Button41.Caption = "Fourier"

Button41.TooltipText = "Highlight array" & Chr(13) & _

    "before pressing" & Chr(13) & "ForwardFT or" & _

    Chr(13) & "InverseFT"

' Create submenus for ForwardFT and InverseFT

Set Button411 = Button41.Controls _

  .Add(Type:=msoControlButton)

With Button411

  .Caption = "ForwardFT"

  .Style = msoButtonCaption

  .OnAction = "ForwardFT"

End With

Set Button412 = Button41.Controls. _

  Add(Type:=msoControlButton)

With Button412

  .Caption = "InverseFT"

  .Style = msoButtonCaption

  .OnAction = "InverseFT"

End With

' Create a button for Gabor

Set Button42 = Button4.Controls.Add(Type:=msoControlButton)

With Button42

  .Caption = "Gabor"

  .Style = msoButtonCaption

  .OnAction = "Gabor"

End With

' Create a button for Ortho

Set Button43 = Button4.Controls.Add(Type:=msoControlPopup)

Button43.Caption = "Ortho"

' Create submenus for OrthoO and OrthoI

Set Button430 = Button43.Controls _

  .Add(Type:=msoControlButton)

With Button430

  .Caption = "OrthoO"

  .Style = msoButtonCaption

  .OnAction = "OrthoO"

End With

Set Button431 = Button43.Controls _

  .Add(Type:=msoControlButton)

With Button431

  .Caption = "OrthoI"

  .Style = msoButtonCaption

  .OnAction = "OrthoI"

End With

' Create a button for (de)convolution

Set Button5 = CommandBars("MacroBundle") _

  .Controls.Add(Type:=msoControlPopup)

Button5.Caption = "(de)convolution"

Button5.BeginGroup = True

' Create a button for convolution

Set Button51 = Button5.Controls.Add(Type:=msoControlPopup)

Button51.Caption = "convolution"

' Create submenus for Convolve and ConvolveFT

Set Button511 = Button51.Controls _

  .Add(Type:=msoControlButton)

With Button511

  .Caption = "Convolve"

  .Style = msoButtonCaption

  .OnAction = "Convolve"

End With

Set Button512 = Button51.Controls _

  .Add(Type:=msoControlButton)

With Button512

  .Caption = "ConvolveFT"

  .Style = msoButtonCaption

  .OnAction = "ConvolveFT"

End With

' Create a button for Deconvolution

Set Button52 = Button5.Controls.Add(Type:=msoControlPopup)

Button52.Caption = "deconvolution"

' Create submenus for Deconvolve, DeconvolveFT,

' DeconvolveIt0, and DeconvolveIt1

Set Button521 = Button52.Controls _

  .Add(Type:=msoControlButton)

With Button521

  .Caption = "Deconvolve"

  .Style = msoButtonCaption

  .OnAction = "Deconvolve"

End With

Set Button522 = Button52.Controls _

  .Add(Type:=msoControlButton)

With Button522

  .Caption = "DeconvolveFT"

  .Style = msoButtonCaption

  .OnAction = "DeconvolveFT"

End With

Set Button523 = Button52.Controls _

  .Add(Type:=msoControlButton)

With Button523

  .Caption = "DeconvolveIt0"

  .Style = msoButtonCaption

  .OnAction = "DeconvolveIt0"

End With

Set Button524 = Button52.Controls _

  .Add(Type:=msoControlButton)

With Button524

  .Caption = "DeconvolveIt1"

  .Style = msoButtonCaption

  .OnAction = "DeconvolveIt1"

End With

' Create a button for calculus

Set Button6 = CommandBars("MacroBundle") _

  .Controls.Add(Type:=msoControlPopup)

Button6.Caption = "calculus"

Button6.BeginGroup = True

' Create a button for Romberg

Set Button61 = Button6.Controls.Add(Type:=msoControlPopup)

Button61.Caption = "Romberg"

' Create submenus for RombergAuto and RombergSpecify

Set Button611 = Button61.Controls _

  .Add(Type:=msoControlButton)

With Button611

  .Caption = "RombergAuto"

  .Style = msoButtonCaption

  .OnAction = "RombergAuto"

End With

Set Button612 = Button61.Controls _

  .Add(Type:=msoControlButton)

With Button612

  .Caption = "RombergSpecify"

  .Style = msoButtonCaption

  .OnAction = "RombergSpecify"

End With

' Create a button for Trapez

Set Button62 = Button6.Controls.Add(Type:=msoControlPopup)

Button62.Caption = "Trapez"

' Create submenus for TrapezAuto and TrapezSpecify

Set Button621 = Button62.Controls _

  .Add(Type:=msoControlButton)

With Button621

  .Caption = "TrapezAuto"

  .Style = msoButtonCaption

  .OnAction = "TrapezAuto"

End With

Set Button622 = Button62.Controls _

  .Add(Type:=msoControlButton)

With Button622

  .Caption = "TrapezSpecify"

  .Style = msoButtonCaption

  .OnAction = "TrapezSpecify"

End With

' Create a button for Deriv

Set Button63 = Button6.Controls.Add(Type:=msoControlButton)

With Button63

  .Caption = "Deriv"

  .Style = msoButtonCaption

  .OnAction = "Deriv"

End With

' Create a button for Deriv1

Set Button6301 = Button6.Controls.Add(Type:=msoControlButton)

With Button6301

  .Caption = "Deriv1"

  .Style = msoButtonCaption

  .OnAction = "Deriv1"

End With

' Create a button for DerivScan

Set Button64 = Button6.Controls.Add(Type:=msoControlButton)

With Button64

  .Caption = "DerivScan"

  .Style = msoButtonCaption

  .OnAction = "DerivScan"

End With

' Create a button for SemiIntegration

' and SemiDifferentiation

Set Button65 = Button6.Controls.Add(Type:=msoControlPopup)

Button65.Caption = "SemiDiff/Int"

' Create buttons for SemiDifferentiate and SemiIntegrate

Set Button651 = Button65.Controls _

  .Add(Type:=msoControlButton)

With Button651

  .Caption = "SemiIntegrate"

  .Style = msoButtonCaption

  .OnAction = "SemiIntegrate"

End With

Set Button652 = Button65.Controls _

  .Add(Type:=msoControlButton)

With Button652

  .Caption = "SemiDifferentiate"

  .Style = msoButtonCaption

  .OnAction = "SemiDifferentiate"

End With

' Create a button for mapper

Set Button7 = CommandBars("MacroBundle").Controls _

  .Add(Type:=msoControlPopup)

Button7.Caption = "mapper"

Button7.BeginGroup = True

' Create a button for Gradual Mapper

Set Button71 = Button7.Controls _

  .Add(Type:=msoControlPopup)

Button71.TooltipText = "gradual scale"

Button71.Caption = "Gradual Mapper"

' Create submenus for Mapper 0 through 3

Set Button710 = Button71.Controls _

  .Add(Type:=msoControlButton)

With Button710

  .Caption = "gray scale"

  .Style = msoButtonCaption

  .OnAction = "Mapper0"

End With

Set Button711 = Button71.Controls _

  .Add(Type:=msoControlButton)

With Button711

  .Caption = "spectrum"

  .Style = msoButtonCaption

  .OnAction = "Mapper1"

End With

Set Button712 = Button71.Controls _

  .Add(Type:=msoControlButton)

With Button712

  .Caption = "red scale"

  .Style = msoButtonCaption

  .OnAction = "Mapper2"

End With

Set Button713 = Button71.Controls _

  .Add(Type:=msoControlButton)

With Button713

  .Caption = "blue scale"

  .Style = msoButtonCaption

  .OnAction = "Mapper3"

End With

' Create a button for 9-band mapper

Set Button72 = Button7.Controls _

  .Add(Type:=msoControlPopup)

Button72.TooltipText = "9-band scale"

Button72.Caption = "9-Band Mapper"

' Create submenus for Mapper00 through Mapper33

Set Button720 = Button72.Controls _

  .Add(Type:=msoControlButton)

With Button720

  .Caption = "gray scale"

  .Style = msoButtonCaption

  .OnAction = "Mapper00"

End With

Set Button721 = Button72.Controls _

  .Add(Type:=msoControlButton)

With Button721

  .Caption = "spectrum"

  .Style = msoButtonCaption

  .OnAction = "Mapper11"

End With

Set Button722 = Button72.Controls _

  .Add(Type:=msoControlButton)

With Button722

  .Caption = "red scale"

  .Style = msoButtonCaption

  .OnAction = "Mapper22"

End With

Set Button723 = Button72.Controls _

  .Add(Type:=msoControlButton)

With Button723

  .Caption = "blue scale"

  .Style = msoButtonCaption

  .OnAction = "Mapper33"

End With

' Create a button for 17-band mapper

Set Button73 = Button7.Controls _

  .Add(Type:=msoControlPopup)

Button73.TooltipText = "17-band scale"

Button73.Caption = "17-Band Mapper"

' Create submenus for Mapper000 through Mapper333

Set Button730 = Button73.Controls _

  .Add(Type:=msoControlButton)

With Button730

  .Caption = "gray scale"

  .Style = msoButtonCaption

  .OnAction = "Mapper000"

End With

Set Button731 = Button73.Controls _

  .Add(Type:=msoControlButton)

With Button731

  .Caption = "spectrum"

  .Style = msoButtonCaption

  .OnAction = "Mapper111"

End With

Set Button732 = Button73.Controls _

  .Add(Type:=msoControlButton)

With Button732

  .Caption = "red scale"

  .Style = msoButtonCaption

  .OnAction = "Mapper222"

End With

Set Button733 = Button73.Controls _

  .Add(Type:=msoControlButton)

With Button733

  .Caption = "blue scale"

  .Style = msoButtonCaption

  .OnAction = "Mapper333"

End With

' Create a button for miscellany

Set Button8 = CommandBars("MacroBundle").Controls _

  .Add(Type:=msoControlPopup)

Button8.Caption = "miscellany"

Button8.BeginGroup = True

' Create a button for ScanF

Set Button81 = Button8.Controls.Add(Type:=msoControlPopup)

Button81.TooltipText = "Select ScanF0 for F(x,y) or" _

  & Chr(13) & "ScanF1 for added IsoList "

Button81.Caption = "ScanF"

' Create submenus for ScanF0 and ScanF1

Set Button810 = Button81.Controls _

  .Add(Type:=msoControlButton)

With Button810

  .Caption = "ScanF0"

  .Style = msoButtonCaption

  .OnAction = "ScanF0"

End With

Set Button811 = Button81.Controls _

  .Add(Type:=msoControlButton)

With Button811

  .Caption = "ScanF1"

  .Style = msoButtonCaption

  .OnAction = "ScanF1"

End With

' Create a button for RootFinder

Set Button82 = Button8.Controls _

  .Add(Type:=msoControlButton)

With Button82

  .Caption = "RootFinder"

  .Style = msoButtonCaption

  .OnAction = "RootFinder"

End With

' Create a button for MovieDemos

Set Button83 = Button8.Controls.Add(Type:=msoControlPopup)

Button83.Caption = "MovieDemos"

' Create submenus for MovieDemo1 through MovieDemo5

Set Button830 = Button83.Controls _

  .Add(Type:=msoControlButton)

With Button830

  .Caption = "MovieDemo1"

  .Style = msoButtonCaption

  .OnAction = "MovieDemo1"

End With

Set Button831 = Button83.Controls _

  .Add(Type:=msoControlButton)

With Button831

  .Caption = "MovieDemo2"

  .Style = msoButtonCaption

  .OnAction = "MovieDemo2"

End With

Set Button832 = Button83.Controls _

  .Add(Type:=msoControlButton)

With Button832

  .Caption = "MovieDemo3"

  .Style = msoButtonCaption

  .OnAction = "MovieDemo3"

End With

Set Button833 = Button83.Controls _

  .Add(Type:=msoControlButton)

With Button833

  .Caption = "MovieDemo4"

  .Style = msoButtonCaption

  .OnAction = "MovieDemo4"

End With

Set Button834 = Button83.Controls _

  .Add(Type:=msoControlButton)

With Button834

  .Caption = "MovieDemo5"

  .Style = msoButtonCaption

  .OnAction = "MovieDemo5"

End With

' Create a button for MBToolbar

Set Button84 = Button8.Controls.Add(Type:=msoControlPopup)

Button84.Caption = "MBToolbar"

' Create submenus for InsertMBToolbar and RemoveMBToolbar

Set Button841 = Button84.Controls _

  .Add(Type:=msoControlButton)

With Button841

  .Caption = "InsertMBToolbar"

  .Style = msoButtonCaption

  .OnAction = "InsertMBToolbar"

End With

Set Button842 = Button84.Controls _

  .Add(Type:=msoControlButton)

With Button842

  .Caption = "RemoveMBToolbar"

  .Style = msoButtonCaption

  .OnAction = "RemoveMBToolbar"

End With

End Sub

'''''''''''''''''''''''''''''''''''''''''''''''''''''''''''

'''''''''''''''''''''''''''''''''''''''''''''''''''''''''''

'''''''''''*************************************'''''''''''

'''''''''''*                                   *'''''''''''

'''''''''''*     Remove MacroBundle Toolbar    *'''''''''''

'''''''''''*                                   *'''''''''''

'''''''''''*************************************'''''''''''

''''''''''''''''''''''''''''''''''''''''''' (c) R. de Levie

'''''''''''''''''''''''''''''''''''''''''''  v 12, May 2012

Sub RemoveMBToolbar()

On Error Resume Next

CommandBars("MacroBundle").Delete

End Sub

'''''''''''''''''''''''''''''''''''''''''''''''''''''''''''

'''''''''''''''''''''''''''''''''''''''''''''''''''''''''''

'''''''''''*************************************'''''''''''

'''''''''''*                                   *'''''''''''

'''''''''''*       Lagrange Interpolation      *'''''''''''

'''''''''''*                                   *'''''''''''

'''''''''''*************************************'''''''''''

''''''''''''''''''''''''''''''''''''''''''' (c) R. de Levie

'''''''''''''''''''''''''''''''''''''''''''  v 12, May 2012

' EXAMPLE:

'

' Use of this stand-alone function is illustrated in AE3 

' section 1.10.

Function Lagrange(XArray, YArray, X, m)

' m denotes the order of the polynomial used,

' and must be an integer between 1 and 14

Dim Row As Integer, i As Integer, j As Integer

Dim Term As Double, Y As Double

Row = Application.Match(X, XArray, 1)

If Row < (m + 1) / 2 Then Row = (m + 1) / 2

If Row > XArray.Count - (m + 1) / 2 Then _

  Row = XArray.Count - (m + 1) / 2

For i = Row - (m - 1) / 2 To Row + (m + 1) / 2

  Term = 1

  For j = Row - (m - 1) / 2 To Row + (m + 1) / 2

    If i <> j Then Term = Term * (X - XArray(j)) / _

      (XArray(i) - XArray(j))

  Next j

  Y = Y + Term * YArray(i)

Next i

Lagrange = Y

End Function
